
Le basi
Questo tutorial presuppone che il lettore sappia come usare MASM. Se non avete dimistichezza con 
MASM, scaricatevi win32asm.exe e studiate i documenti all'interno del package prima di procedere con 
questo tutorial. Bene. Ora siete pronti. Andiamo! 

Preliminari
I programmi per Win32 sono eseguiti in "protected mode" che e' disponibile sin dai tempi degli 80286. Ma
gli 80286 ormai sono storia. Percio' dovremo relazionarci all' 80386 e i suoi discendenti. Windows esegue 
ogni singolo programma Win32 in uno spazio virtuale separato e unico. Cio' significa che ogni programma
Win32 avra' i suoi propri 4 GB di address space. Ogni programma e' solo nel suo address space. Cio' e' in 
contrasto con la situazione presente in Win16. Tutti i programmi Win16 possono *vedersi* gli uni con gli 
altri. Questo non accade in Win32. Tale caratteristica aiuta a ridurre la probabilita' che un programma 
scriva sul codice/dati di un altro. Il memory model (modello di memoria, NdT) e' parimenti drasticamente
diverso dai vecchi giorni del mondo a 16-bit. In Win32, non e' piu' necessario preoccuparsi del modello di 
memoria o dei segmenti! C'e' solo UN modello di memoria: il Flat memory model. Non esistono piu' 
segmenti da 64K. La memoria e' un largo e continuo spazio di 4 GB. Questo significa anche che non 
bisognera' piu' giocare con i segment registers. Potrete usare qualsiasi segment register per indirizzare 
qualsiasi punto nello spazio di memoria. Questo e' un GRANDE aiuto ai programmatori, ed e' cio' che 
rende la programmazione in assembly per Win32 semplice quanto quella in C. 

Contenuto
Ecco lo scheletro del programma. Se non comprendete alcune parti del codice, niente panico. Spieghero' 
ciascuna di esse in seguito. 

.386 

.MODEL Flat, STDCALL 

.DATA 
    <I vostri dati inizializzati> 
    ...... 
.DATA? 
   <I vostri dati NON inizializzati>
   ...... 
.CONST 
   <Le vostre costanti>

   ...... 
.CODE 
   <etichetta>: 
    <Il vostro codice> 
   ..... 
    end <etichetta>

Ecco tutto! Analizziamo questo scheletro. 

.386 Questa e' una direttiva per l'assembler, a cui diciamo di usare il set di istruzioni 80386. Potreste 
anche usare .486, .586 ma la scelta piu' sicura e' quella di utilizzare sempre .386 . .MODEL FLAT, 
STDCALL .MODEL e' una direttiva per l'assembler che specifica il modello di memoria del nostro 
programma. Sotto Win32, esiste un solo modello di memoria, il modello FLAT. STDCALL comunica a 
MASM la convenzione nel passare i parametri. Questa convenzione specifica l'ordine con cui i parametri 
verranno passati, da sinistra-verso-destra o da destra-verso-sinistra, oltre a chi bilanciera' lo stack frame
dopo la chiamata di una call (procedura, NdT). In Win16, ci sono due tipi di convenzioni di chiamata, C e 
PASCAL La convenzione di chiamata C passa i parametri da destra a sinistra, cioe', il parametro 
all'estrema destra e' PUSHato per primo. Il caller e' responsabile del bilanciamento dello stack frame 
dopo la call. Ad esempio, volendo chiamare una funzione denominata foo(int primo_param, int 
secondo_param, int terzo_param) con la convenzione C, il codice assomiglierebbe a questo: 

push  [terzo_param]             // ; Pusha il terzo parametro
push  [secondo_param]           // ; Seguito dal secondo
push  [primo_param]             // ; E dal primo
call    foo 
add    sp, 12                   // ; Il caller bilancia lo stack frame

La convenzione PASCAL e' l'inverso di quella per il C. Essa passa i parametri da sinistra a destra, e il 
callee e' responsabile per il bilanciamento della stack dopo la call. Win16 adotta la convenzione PASCAL 
poiche' produce codici piu' piccoli. la convenzione C e' utile quando non si conosce il numero dei 
parametri che verranno passati alla funzione, come nel caso di wsprintf(). Nel caso di wsprintf(), la 
funzione non ha modo di determinare aprioristicamente il numero di parametri che verrano pushati sulla 
stack, percio' non puo' bilanciare lo stack frame. STDCALL e' un ibrido tra le convenzioni C e PASCAL. 
Essa passa i parametri da destra a sinistra ma il callee e' responsabile per il bilanciamento della stack 
dopo la call. La piattaforma Win32 usa esclusivamente STDCALL. Eccetto in un caso: wsprintf(). Dovete 
usare la convenzione C con wsprintf(). 

.DATA 

.DATA? 

.CONST 

.CODE

Tutte e quattro le direttive sono cio' che viene definito SEZIONE. Non avete segmenti in Win32, 
ricordate ? Ma potete dividere il vostro intero address space in sezioni logiche. L'inizio di una sezione 
definisce la fine della sezione precedente. Ci sono due gruppi di sezioni: dati e codice. Le sezioni dei dati 
sono divise in 3 categorie: 

 .DATA Questa sezione contiene i dati inizializzati del vostro programma. 

 .DATA? Questa sezione contiene i dati NON inizializzati del vostro programma. A volte capita di 
voler impegnare una parte di memoria (variabli, NdT) senza inizializzarla. Questa sezione serve 
a tale scopo. 

 .CONST Questa sezione contiene le costanti usate dal vostro programma. Le costanti in questa 
sezione non potranno mai essere modificate dal vostro programma. Sono semplicemente 
costanti. 

Non e' necessario utilizzare tutte e tre le sezioni nel vostro programma. Dichiarate solo la/e sezione/i che 
volete usare. C'e' solo una sezione per il codice: .CODE. Qui e' dove le vostre istruzioni risiedono. 
<etichetta>: end <etichetta> dove <etichetta> e' una qualsiasi etichetta arbitraria usata per specificare 



l'estensione del vostro programma. Entrambe le etichette devono essere identiche. Tutto il vostro codice 
deve risiedere tra <etichetta>: e end <etichetta> 

Note finali
Questi tutorials erano presenti nel sito di RingZero. Li rimettiamo a disposizione a chiunque voglia poterli 
leggere nella loro traduzione in italiano che fu curata da NeuralNoise

. 

Introduzione
In questo tutorial, creeremo un programma per Windows completamente funzionante che mostra un box 
con il messaggio "Win32 assembly is great!". Scaricate il file con l'esempio qui. 

Preliminari
Windows prepara una grossa quantita' di risorse per i programmi Windows. Al centro di cio' c'e' la 
Windows API (Application Programming Interface, Interfaccia per la Programmazione di Applicazioni, 
NdT). La Windows API e' un'immensa collezione di utilissime funzioni contenute in Windows stesso, 
pronte ad essere usate da qualsiasi programma per Windows. Queste funzioni risiedono in DLL (dynamic-
linked libraries, librerie collegate dinamicamente al programma, NdT) come kernel32.dll, user32.dll e 
gdi32.dll. Kernel32.dll contiene funzioni API relative alla memoria e alla gestione dei processi. User32.dll 
controlla gli aspetti dell'interfaccia utente del vostro programma.Gdi32.dll e' responsabile per le 
operazioni grafiche. Oltre alle "principali tre", ci sono altre DLL che il vostro programma puo' usare, 
ammesso che voi possediate abbastanza informazioni riguardo alla funzione API desiderata. I programmi 
per Windows si linkano (collegano, NdT) dinamicamente a queste DLL, in altre parole il codice per le 
funzioni API non e' incluso nell'eseguibile del programma per Windows. Per comunicare al vostro 
programma dove trovare le funzioni API desiderate al momento dell'esecuzione, dovrete accludere tale 
informazione nel file eseguibile. L'informazione risiede nelle import libraries (librerie importate, NdT). 
Dovrete linkare il vostro programma con le corrette import libraries o esso non sara' capace di localizzare 
le funzioni API. Esistono due tipi di funzioni API: Uno per ANSI e uno per Unicode. Il nome delle funzioni 
API per ANSI e' postfissato con "A", ad esempio MessageBoxA. Quelle per Unicode sono postfissate con 
"W" (per Wide Char, credo). Windows 95 supporta nativamente ANSI e l'Unicode Windows NT. Ma la 
maggior parte delle volte, utilizzerete un file di include che puo' determinare e selezionare le funzioni API 
appropriate per la vostra piattaforma. Semplicemente riferitevi alla funzione API senza il postfisso. 

Contenuto
Presentero' semplicemente lo scheletro del programma qui sotto. Lo riempiremo successivamente. 

.386 

.model flat, stdcall 

.data 

.code 
    Main: 
    end Main

http://quequero.org/Tutorial_2:_La_MessageBox#Disclaimer
http://quequero.org/Tutorial_2:_La_MessageBox#Note_finali
http://quequero.org/Tutorial_2:_La_MessageBox#Contenuto
http://quequero.org/Tutorial_2:_La_MessageBox#Preliminari
http://quequero.org/Tutorial_2:_La_MessageBox#Introduzione
http://quequero.org/Tutorial_2:_La_MessageBox#Tutorial_2:_La_MessageBox
javascript:toggleToc()


Ogni programma per Windows deve chiamare una funzione API, ExitProcess, quando vuole uscire a 
Windows. In quest'ottica, ExitProcess e' equivalente a int 21h, ah=4Ch in DOS. 

Ecco il prototipo per la funzione ExitProcess da winbase.h: void WINAPI ExitProcess(UINT uExitCode); 

void significa che la funzione non restituisce nessun valore al caller. 
WINAPI e' un alias della convenzione di chiamata STDCALL. 
UINT e un tipo di dati, "unsigned integer", che e' un valore a 32-bits sotto Win32 (e' un valore a 16-bits 
sotto Win16) 
uExitCode e' il codice a 32-bits di ritorno a Windows. 
Questo valore non e' usato da Windows al momento. 

Per chiamare ExitProcess da un programma in assembly, dovrete prima dichiarare il function prototype 
(prototipo di funzione, NdT) per ExitProcess. 

.386 

.model flat, stdcall 
ExitProcess     PROTO      ,:DWORD 
.data 
.code 
    Main: 
     INVOKE    ExitProcess, 0 
    end Main

Ecco tutto. Il vostro primo programma funzionante per Win32. Salvatelo come msgbox.asm. 

Presupponendo che ml.exe e' nella vostra path, assemblate msgbox.asm con: 
ml /c /coff /Cp msgbox.asm 
/c dice a MASM di assemblare soltanto. Non invoca Link. 
/coff dice a MASM di creare un file .obj in formato COFF. 
/Cp dice a MASM di conservare le caratteristiche di formattazione (maiuscole/minuscole) degli 
identificatori (variabili, NdT) dell'utente. 

Quindi procedete con link: link /SUBSYSTEM:WINDOWS /LIBPATH:c:\masm611\lib msgbox.obj 
kernel32.lib 
/SUBSYSTEM:WINDOWS dice a Link che tipo di eseguibile e' questo programma. 
/LIBPATH:<path delle import libraries> dice a Link dove sono le import libraries. Sul mio PC, sono sotto 
c:\masm611\lib 

Adesso avete ottenuto msgbox.exe. Andate avanti, fatelo partire. Scoprirete che non fa niente. Beh, non 
ci abbiamo ancora inserito niente di interessante. Ma e' senza ombra di dubbio un programma per 
Windows. E osservate le sue dimensioni! Sul mio PC, il file e' lungo 1,536 bytes. 
La linea: 
ExitProcess PROTO ,:DWORD 
e' un prototipo di funzione. 
Voi dichiarate il nome della funzione seguito dalla parola chiave "PROTO", una virgola, e la lista del tipo di
dati dei parametri. MASM usa il prototipo di funzione per controllare il numero e il tipo di parametri della 
funzione. Il miglior posto per i prototipi di funzione e' un file di include. Potete creare un file di include 
pieno di prototipi di funzioni e strutture di dati frequentemente usati e includerlo all'inizio del vostro 
programma asm. 

Chiamate le funzioni API usando la parola chiave INVOKE: 
INVOKE ExitProcess, 0 
INVOKE e' in pratica una specie di call specializzata. Essa controlla il numero e il tipo di parametri e li 
pusha sulla stack seguendo la convenzione di chiamata predefinita (in questo caso, stdcall). Usando 
INVOKE invece del normale CALL, potete prevenire gli errori della stack derivanti da un passaggio di 
parametri incorretto. Molto utile. La sintassi e': 
INVOKE espressione [,argomenti] 
dove espressione e' un'etichetta o il nome di una funzione. 



Successivamente, metteremo su una message box. La dichiarazione per questa funzione e': 
int WINAPI MessageBoxA(HWND hwnd, LPCSTR lpText, LPCSTR lpCaption, UINT uType); 
dove hwnd e' l'handle della parent window (finestra-genitrice, NdT :) 
lpText e' un puntatore al testo che volete mostrare nella client area (l'area a disposizione della message 
box, NdT) 
lpCaption e' un puntatore al titolo della message box 
uType specifica l'icona e il numero e tipo dei bottoni della message box 

Sotto la piattaforma Win32, HWND, LPCSTR, e UINT sono tutti valori della dimensione di 32 bits. 
Modifichiamo msgbox.asm per includere la message box. 

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\kernel32.inc
includelib \masm32\lib\kernel32.lib
include \masm32\include\user32.inc
includelib \masm32\lib\user32.lib

.data
MsgBoxCaption    db "Iczelion Tutorial No.2",0
MsgBoxText       db "Win32 Assembly is Great!",0

.code
start:
invoke MessageBox, NULL, addr MsgBoxText, addr MsgBoxCaption, MB_OK
invoke ExitProcess, NULL
end start

Assemblatelo e fatelo partire. (Dovrete includere user32.lib nel parametro di Link, poiche' le informazioni 
per linkare MessageBoxA risiedono in user32.lib) Vedrete una message box che mostra il testo "Win32 
Assembly is Great!". 
Diamo un'altra occhiata al codice. 
Definiamo due stringhe terminate con zero (zero-terminated) nella sezione .data . Ricordate che tutte le 
stringhe in Windows devono essere terminate con zero. 
Definiamo due costanti nella sezione .const . Utilizziamo le costanti per rendere piu' chiaro il codice. 
Osservate i parametri della funzione MessageBoxA. Il primo parametro e' NULL. 
Cio' significa che non c'e' nessuna finestra che possiede questa message box. 
L'operatore "ADDR" e' usato per passare l'indirizzo dell'etichetta alla funzione. 

Introduzione
In questo tutorial, costruiremo un programma per Windows che mostra una finestra pienamente 
funzionante sul desktop. 

Preliminari
I programmi per Windows si basano molto sulle funzioni API per la loro GUI (Graphic User Interface, 
Interfaccia Grafica Utente, NdT). Questo approccio va a vantaggio sia degli utenti che dei programmatori.
Per gli utenti, il vantaggio e' che non devono imparare come navigare nella GUI di ogni nuovo 
programma, poiche' le GUI dei programmi Window sono simili. Per i programmatori, il vantaggio e' che i 
codici per la GUI stanno gia' li', testati e pronti all'uso. Il rovescio della medaglia per i programmatori e' la
maggiore complessita' implicata. Per poter creare o manipolare ogni oggetto della GUI come finestre, 
menu o icone, i programmatori devono seguire una strada obbligata e molto stretta (come avreste 
tradotto voi "strict recipe" ?!?, NdT ;). Ma cio' puo' essere superato grazie alla programmazione modulare
o al paradigma OOP. Vi mostrero' qui di seguito i passi necessari per poter creare una finestra sul 
desktop: 
1.Trovare l'instance handle del vostro programma (necessario) 
2.Ricevere la riga di comando (non necessario a meno che il vostro programma non riceva una riga di 
comando) 
3.Registrare la classe della finestra (window class) (necessario, a meno che non vengano usati tipi di 
finestre predefinite, come ad esempio la MessageBox) 
4.Creare la finestra (necessario) 
5.Mostrare la finestra sul desktop (necessario a meno che non si voglia mostrare la finestra 
immediatamente) 
6.Rinfrescare (refresh) la client area della finestra 
7.Entrare in un loop infinito, che controlla l'arrivo di messaggi da Windows 
8.Se arrivano messaggi, vengono elaborati da una funzione specializzata che e' responsabile della finestra

9.Uscire dal programma se l'utente chiude la finestra 
Come potete vedere, la struttura di un programma per Windows e' abbastanza complessa se messa a 
confronto con quella di un programma per DOS. Ma il mondo di Windows e' drasticamente diverso dal 
mondo del DOS. I programmi per Windows devono essere capaci di coesistere pacificamente. Devono 
seguire delle regole ferree. Voi, in quanto programmatori, dovrete parimenti diventare piu' severi nello 
stile e nelle abitudini di programmazione. 

Contenuto
Qui di seguito c'e' il codice del nostro semplice programma di creazione della finestra. Prima di inoltrarci 
nei dettagli insanguinati della programmazione in assembly per Win32, vi delineero' alcuni punti 
importanti che vi semplificheranno la programmazione. 

 Dovreste inserire tutte le costanti, strutture e prototipi di funzioni di Windows in un file di include 
e includerlo all'inizio del vostro file .asm. Cio' vi fara' risparmiare moltissimo tempo e vi aiutera' 
ad evitare errori di battitura. La maggior parte delle volte, potrete usare il file di include di 
alcuni esempi di asm per Win32. Io ho usato windows.inc dall'esempio "Small Is Beautiful" di 
Steve Gibson (scaricatevi quest'esempio anche se windows.inc e' contenuto nel package 
dell'esempio di questo tutorial...Small Is Beautiful e' da vedere per crederci, NdT :) e ci ho fatto 
alcune aggiunte personali. 

 Usate la direttiva IncludeLib per specificare le import libraries usate dal vostro programma. Ad 
esempio, se il vostro programma chiama MessageBoxA, dovete inserire la riga: 

IncludeLib user32.lib all'inizio del vostro file .asm. Questa direttiva dice a MASM che il vostro 
programma fara' uso di funzioni presenti in quella import library. Se il vostro programma chiama funzioni 
in piu' di una libreria, semplicemente aggiungete un IncludeLib per ogni libreria usata. Usando la direttiva
IncludeLib, non dovete preoccuparvi delle import libraries al momento del link. Potete comunque usare lo 
switch /LIBPATH del linker per comunicare a Link la path delle librerie. 

 Quando dichiarate i prototipi per le funzioni API, le strutture o le costanti nel vostro file di include,
cercate di utilizzare i nomi originali usati nei file include di Windows, inclusa la formattazione 
(maiuscole/minuscole). Cio' vi evitera' un bel po' di mal di testa quando cercherete qualcosa 
nella Win32 API reference. 

 Usate makefile per automatizzare il processo di assemblaggio. Cio' vi fara' evitare di scrivere 
molto. 

include windows.inc   // ; .386 e .model sono gia' dichiarati in windows.inc
includelib user32.lib // ; chiamate a funzioni incluse in user32.lib e kernel32.lib 
includelib kernel32.lib 
.DATA // ; dati inizializzati
    ClassName db "SimpleWinClass",0 // ; il nome della classe della nostra finestra 
    AppName  db "Our First Window",0 // ; il nome della nostra finestra
.DATA? // ; dati NON inizializzati
hInstance HINSTANCE ? // ; Instance handle del nostro programma
CommandLine LPSTR ? 
.CODE // ; Qui comincia il nostro codice
 start: 
     invoke GetModuleHandle, NULL // ; trova l'instance handle del nostro programma. 
                                  // ; Sotto Win32, hmodule==hinstance 
     mov    hInstance,eax 
     invoke GetCommandLine // ; rileva la riga di comando. Non dovete chiamare questa funzione
SE
                           // ; il vostro programma non elabora la riga di comando
     mov    CommandLine,eax 
     invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT    // ; chiama la funzione 
principale 
     invoke ExitProcess,eax // ; esci dal nostro programma. Il codice di uscita e' conservato 
in 
                            // ; eax dopo WinMain.
WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD 
    LOCAL wc:WNDCLASSEX // ; crea variabili locali sulla stack
    LOCAL msg:MSG 
    LOCAL hwnd:HWND 
    mov   wc.cbSize,SIZEOF WNDCLASSEX // ; riempi con valori i membri di wc
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance 
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 



    mov   wc.hIcon,eax 
    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc // ; registra la classe della nostra finestra
    invoke CreateWindowEx,NULL,\ 
                ADDR ClassName,\ 
                ADDR AppName,\ 
                WS_OVERLAPPEDWINDOW,\ 
                CW_USEDEFAULT,\ 
                CW_USEDEFAULT,\ 
                CW_USEDEFAULT,\ 
                CW_USEDEFAULT,\ 
                NULL,\ 
                NULL,\ 
                hInst,\ 
                NULL 
    mov   hwnd,eax 
    invoke ShowWindow, hwnd,CmdShow //  ; mostra la nostra finestra sul desktop 
    invoke UpdateWindow, hwnd   //  ; rinfresca la client area
    .WHILE TRUE              //  ; Entra nel loop dei messaggi
                invoke GetMessage, ADDR msg,NULL,0,0 
                .BREAK .IF (!eax) 
                invoke TranslateMessage, ADDR msg 
                invoke DispatchMessage, ADDR msg 
   .ENDW 
    mov     eax,msg.wParam   //  ; metti il codice di uscita in eax
    ret 
WinMain endp 
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
    mov   eax,uMsg  // ; metti il messaggio di windows in eax per efficienza
    .IF eax==WM_DESTROY // ; se l'utente chiude la nostra finestra
        invoke PostQuitMessage,NULL // ; Esci dalla nostra applicazione
        xor eax,eax 
    .ELSE 
       invoke DefWindowProc,hWnd,uMsg,wParam,lParam // ; Elaborazione di 
                                                    // ; default (predefinita, NdT) 
                                                    // ; dei messaggi
    .ENDIF 
 ret 
WndProc endp 
end start

Potreste essere sconvolti dal fatto che un semplice programma per Windows richiede tutto questo codice. 
Ma la maggior parte di queste istruzioni sono semplicemente delle istruzioni di *template* che potrete 
copiare da un codice sorgente ad un altro. O, se preferite, potete assemblare alcuni di queste istruzioni in
una libreria da usare come prologo ed epilogo. Potete scrivere solo le istruzioni nella funzione WinMain. 
Infatti, questo e' quello che fanno i compilatori C. Vi fanno scrivere le istruzioni di WinMain senza farvi 
preoccupare delle faccende di casa. L'unico rovescio della medaglia e' che dovrete avere una funzione 
denominata WinMain o altrimenti i compilatori C non saranno capaci di combinare il vostro codice con il 
prologo e l'epilogo. Non avete tale restrizione in assembly. Potete usare qualsiasi nome di funzione invece
di WinMain o anche nessuna funzione. Preparatevi. Questo sara' un lungo, lungo tutorial. Analizziamo 
questo programma fino all'osso!
include windows.inc 
includelib user32.lib 
includelib kernel32.lib 
Dobbiamo includere windows.inc all'inizio del codice. Esso contiene prototipi di funzioni API importanti, 
strutture e costanti che vengono utilizzate dal nostro programma. Il file include, windows.inc, e' un 
semplice file di testo. Potete aprirlo con un qualsiasi editor di testo. Le prime due linee sono .386 e le 
direttive .model, per cui non dovete specificare queste due linee all'inizio del vostro codice. 
Successivamente, ci sono diverse macros che il suo autore (Steve Gibson) usa frequentemente. Il resto 
del file contiene strutture importanti, costanti e prototipi di funzioni. Notate che windows.inc non 
contiente tutte le strutture, costanti e prototipi di funzioni di Windows. Esso contiene solo le piu' usate. 
Potete aggiungere nuove cose se snon sono nel file. Il nostro programma chiama funzioni API che 
risiedono in user32.dll (CreateWindowEx, RegisterWindowClassEx, per esempio) e kernel32.dll 
(ExitProcess), percio' dovremo linkare il nostro programma con quelle due import libraries. La prossima 
domanda e': come faccio a sapere quali import libraries devono essere linkate al mio programma? La 
risposta: dovete sapere dove risiedono le funzioni API chiamate dal vostro programma. Per esempio, se 
chiamate una funzione API in gdi32.dll, dovrete linkare con gdi32.lib. Questo e' l'approccio di MASM. Il 



sistema di importazione di librerie di TASM e' molto piu' semplice: semplicemente, bisogna linkare ad UN 
solo file: import32.lib. 

.DATA 
    ClassName db "SimpleWinClass",0 
    AppName  db "Our First Window",0 
.DATA? 
hInstance HINSTANCE ? 
CommandLine LPSTR ?

Successivamente incontriamo le sezioni "DATA".
In .DATA, dichiariamo due stringhe terminate con zero (ASCIIZ strings): ClassName che e' il nome della 
classe della nostra finestra e AppName che e' il nome della nostra finestra. Notate che le due variabili 
sono inizializzate. In .DATA?, vengono dichiarate tre variabili: hInstance (instance handle del nostro 
programma), CommandLine (riga di comando del nostro programma), e CommandShow (stato della 
nostra finestra alla prima comparsa). I tipi di dati poco familiari, HINSTANCE e LPSTR, sono in realta' 
nuovi nomi per DWORD. Puoi esaminarli in windows.inc. Notate che nessuna variabile in .DATA? e' 
inizializzata, a significare che non devono contenere un valore specifico all'avvio, ma ne vogliamo 
riservare lo spazio per un uso futuro. 

.CODE 
 start: 
     invoke GetModuleHandle, NULL 
     mov    hInstance,eax 
     invoke GetCommandLine 
     mov    CommandLine,eax 
     invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT 
     invoke ExitProcess,eax 
     ..... 
end start

.CODE contiene tutte le vostre istruzioni. Il vostro codice deve risiedere tutto tra <etichetta iniziale>: e 
end <etichetta iniziale>. Il nome dell'etichetta non e' importante. Potete chiamarla come volete, sempre 
se non viola la convenzione dei nomi di MASM. La nostra prima istruzione e' la chiamata a 
GetModuleHandle per trovare l'instance handle del nostro programma. Sotto Win32, l'instance handle e 
l'handle del modulo sono una sola cosa. Potreste pensare all'instance handle come all' ID del nostro 
programma. Esso e' usato come parametro per molte funzioni API che il nostro programma deve 
chiamare, percio' e' generalmente una buona idea trovarlo all'inizio del nostro programma. Al ritorno da 
una funzione Win32, il valore di ritorno (return value), se esiste, puo' essere trovato in eax. Tutti gli altri 
valori sono ritornati attraverso variabili passate nella lista dei parametri di funzione che avete definito per
la call. Quando chiamate una funzione Win32, essa preservera' sempre i segment registers e i registri 
ebx, edi, esi e ebp. Contrariamente, ecx e edx sono considerati registri di scratch (cioe' un po' come il 
general purpose, NdT), e non potete mai prevedere come saranno al ritorno da una funzione Win32. La 
conclusione e' questa: quando chiamate una funzione API, aspettatevi che il valore di ritorno sia in eax. 
Se una delle vostre funzioni verranno chiamate da Windows, dovete ugualmente giocare seguendo la 
regola: salvate e ripristinate i segment registers, ebx, edi, esi e ebp al ritorno da una funzione o il vostro 
programma crashera' in breve tempo. La chiamata a GetCommandLine e' inutile se il vostro programma 
non elabora una riga di comando. In questo esempio, io vi mostro come utilizzarla in caso ne abbiate 
bisogno nei vostri programmi. Successivamente troviamo la call a WinMain(). Qui riceve quattro 
parametri: l'instance handle del nostro programma, l'instance handle della precedente istanza del nostro 
programma, la riga di comando e lo stato della finestra alla prima comparsa. Sotto Win32, NON ESISTE 
istanza precedente. Ogni programma e' solo nel suo address space, percio' il valore di hPrevInst e' 
sempre 0. Questo parametro e' stato lasciato dai tempi di Win16. Nota: Non dovete dichiarare il nome 
della funzione come WinMain. Infatti, avete liberta' completa riguardo a questo. Non siete per niente 
obbligati ad usare funzioni equivalenti a WinMain. Potete copiare le istruzioni contenute in WinMain subito
dopo GetCommandLine e il vostro programma funzionera' ancora perfettamente. Al ritorno da WinMain, 
eax contiene l'exit code (codice di uscita, NdT). Noi passiamo l'exit code come parametro ad ExitProcess 
che chiude la nostra applicazione. WinMain proc 
Inst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD La riga qui sopra e' la 
dichiarazione di funzione di WinMain. Notate le coppie Parametro:TIPO che seguono la direttiva PROC. 
Essi sono parametri che WinMain riceve dal caller. Potete riferirvi a questi parametri per nome piuttosto 
che per manipolazione di stack. In piu', MASM generera' le istruzioni di prologo ed epilogo per la 
funzione. Pertanto non ci dobbiamo preoccupare dello stack frame all'ingresso ed all'uscita dalle funzioni. 



   LOCAL wc:WNDCLASSEX 

   LOCAL msg:MSG 

   LOCAL hwnd:HWND 

La direttiva LOCAL riserva memoria dalla stack per variabili locali usate nella funzione. La direttiva LOCAL
e' immediatamente seguita da <nome della variabile locale>:<tipo di variabile>. Percio' LOCAL 
wc:WNDCLASSEX dice a MASM di riservare dalla stack una quantita' di memoria della dimensione della 
struttura WNDCLASSEX per la variabile denominata wc. Possiamo riferirci a wc nel nostro codice senza 
nessuna difficolta' implicata nella manipolazione della stack. E' un vero dono di Dio, a mio parere. Il 
rovescio della medaglia e' che le variabili locali non possono essere usate al di fuori della funzione in cui 
sono state create, e vengono automaticamente distrutte quando la funzione ritorna al caller. Un altro 
svantaggio e' che non potete inizializzare variabili locali perche' esse consistono in memoria di stack 
riservata dinamicamente all'inizio della funzione. Dovete assegnarvi manualmente il valore desiderato 
dopo la direttiva LOCAL. 

mov   wc.cbSize,SIZEOF WNDCLASSEX 
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 
    mov   wc.hIcon,eax 
    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc  ; registra la classe della nostra finestra

Le spaventose righe qui sopra sono in realta' molto semplici nel concetto. E' solo che ci vogliono molte 
linee di codice per compierlo. Il concetto che giace dietro tutte queste linee e' una window class (classe 
della finestra, NdT). Una window class non e' nient'altro che il "progetto" di una finestra. Essa definisce 
molte importanti caratteristiche di una finestra, come la sua icona, il suo cursore, la funzione 
responsabile per essa, il suo colore ecc. Voi create una finestra da una window class. Questo e' una 
specie di concetto "object oriented" (orientato agli oggetti, NdT). Se volete creare piu' di una finestra con 
le stesse caratteristiche, e' ragionevole conservare tutte queste caratteristiche in un solo posto e riferirsi 
ad esse quando se ne ha bisogno. Questo schema risparmiera' molta memoria, evitando la duplicazione 
delle informazioni. Ricordate, Windows e' stato disegnato in passato, quando i costi dei chip di memoria 
erano proibitivi e la maggior parte dei computer aveva 1 MB di memoria. Windows deve essere molto 
efficiente nell'uso di scarse risorse di memoria. Il punto e': se definite la vostra finestra personale, dovete
riempire le caratteristiche desiderate per la vostra finestra in strutture WNDCLASS o WNDCLASSEX e 
chiamare RegisterClass o RegisterClassEx prima di essere in grado di creare la vostra finestra. Dovete 
registrare la window class una sola volta per ogni tipo di finestra da cui volete crearne un'altra. Windows 
ha molte window classes predefinite, come "button" e "edit box". Per queste finestre (o controls - 
controlli, NdT), non dovete registrare una window class, semplicemente chiamerete CreateWindowEx con 
il class name predefinito. L'unico membro importante in WNDCLASSEX e' lpfnWndProc. lpfn significa "long
pointer to function". Sotto Win32, non ci sono puntatori "near" o "far", bensi' solo "pointer", a causa del 
nuovo modello di memoria FLAT. Ma questa, ancora una volta, e' una reminescenza dai tempi di Win16. 
Ogni window class deve essere associata con una funzione chiamata window procedure. La window 
procedure e' responsabile per la gestione dei messaggi di tutte le finestre create dalla window class 
associata. Windows mandera' messaggi alla window procedure per notificarle eventi importanti inerenti le
finestre di cui e' responsabile, come input da tastiera o da mouse. Sta poi alla window procedure 
rispondere adeguatamente ad ogni messaggio che riceve. Passerete la maggior parte del tempo a 
scrivere codice che gestisce gli eventi in una window procedure. Descrivero' qui di seguito ogni elemento 
di WNDCLASSEX:



typedef struct tagWNDCLASSEX { 
    UINT      cbSize; 
    UINT      style; 
    WNDPROC   lpfnWndProc; 
    int       cbClsExtra; 
    int       cbWndExtra; 
    HINSTANCE hInstance; 
    HICON     hIcon; 
    HCURSOR   hCursor; 
    HBRUSH    hbrBackground; 
    LPCSTR    lpszMenuName; 
    LPCSTR    lpszClassName; 
    HICON     hIconSm; 
} WNDCLASSEX;

cbSize: La dimensione della struttura WNDCLASSEX in bytes. Possiamo usare l'operatore SIZEOF per 
ottenere il valore.
style: Lo stile delle finestre create da questa class. Potete combinare diversi stili usando l'operatore "or".
lpfnWndProc: L'indirizzo della window procedure responsabile per le finestre create da questa class. 
cbClsExtra: Specifica il numero di extra bytes da riservare in seguito alla struttura window-class. Il 
sistema operativo inzializza questi bytes con zero.
cbWndExtra: Specifica il numero di extra bytes to allocate in seguito alla window instance. Il sistema 
operativo inzializza questi bytes con zero. Se un'applicazione usa la struttura WNDCLASS per registrare 
un dialog box creato usando la direttiva CLASS nel file .res (resource file), la stessa deve modificare 
questo membro in DLGWINDOWEXTRA.
hInstance: L'instance handle del modulo.
hIcon: Handle dell'icona. Lo si ottiene dalla call a LoadIcon.
hCursor: Handle del cursore. Lo si ottiene dalla call a LoadCursor.
hbrBackground: Colore di sfondo delle finestre create da questa class.
lpszMenuName: Menu handle predefinito per le finestre create da questa class.
lpszClassName: Il nome di questa window class. 
hIconSm: Handle di una piccola icona che viene associata con la window class. Se questo membro e' 
NULL, il sistema cerca l'icon resource specificata dal membro hIcon per una icona di dimensione adeguata
ad essere usata come icona piccola.

invoke CreateWindowEx, NULL,\ 
    ADDR ClassName,\ 
    ADDR AppName,\ 
    WS_OVERLAPPEDWINDOW,\ 
    CW_USEDEFAULT,\ 
    CW_USEDEFAULT,\ 
    CW_USEDEFAULT,\ 
    CW_USEDEFAULT,\ 
    NULL,\ 
    NULL,\ 
    hInst,\ 
    NULL

Dopo aver registrato la window class, possiamo chiamare CreateWindowEx per creare la nostra finestra in
base alla window class scelta. Notate che ci sono 12 parametri per questa funzione. Il prototipo di 
funzione in C per CreateWindowEx e' il seguente:

HWND 
WINAPI 
CreateWindowExA( 
    DWORD dwExStyle, 
    LPCSTR lpClassName, 
    LPCSTR lpWindowName, 
    DWORD dwStyle, 
    int X, 
    int Y, 
    int nWidth, 
    int nHeight, 
    HWND hWndParent , 
    HMENU hMenu, 
    HINSTANCE hInstance, 
    LPVOID lpParam);



Diamo un'occhiata alla descrizione dettagliata per ciascun parametro:
dwExStyle: Extra window styles (stili extra per la finestra, NdT). Questo e' il nuovo parametro che e' 
stato aggiunto alla vecchia CreateWindow. Potete inserire nuovi stili per Windows 95 ed NT qui. Potete 
specificare il vostro window style ordinario in dwStyle ma se volete qualche stile speciale come topmost 
window, dovete specificarli qui. Potete usare NULL se non volete extra window styles. 
lpClassName: (Necessario). Indirizzo della stringa ASCIIZ contenente il nome della window class che 
volete usare come template per questa finestra. La class puo' essere la vostra propria classe registrata o 
una window class predefinita. Come detto prima, ogni finestra che create deve essere basata su una 
window class.
lpWindowName: Indirizzo della stringa ASCIIZ contenente il nome della finestra. Sara' mostrata sulla 
barra di titolo della finestra. Se questo parametro e' NULL, la barra di titolo della finestra sara' vuota.
dwStyle: Stili della finestra. Dovete specificare l'apparenza della finestra qui. Passare NULL va bene ma la
finestra non avra' un box con menu di sistema, nessun bottone riduci a icona/ingrandisci, e nessun 
bottone chiudi finestra. La finestra non servirebbe poi a tanto. Dovreste premere Alt+F4 per chiuderla. Il 
window style piu' comune e' WS_OVERLAPPEDWINDOW. Un window style e' solo una bit flag. In tal modo
puoi combinare molti stili con l'operatore "or" per ottenere il desiderato look della finestra. Lo stile 
WS_OVERLAPPEDWINDOW e' in realta' una combinazione degli stili piu' comuni, messi insieme con 
questo metodo.
X,Y: Le coordinate dell'angolo superiore sinistro della finestra. Normalmente questo valore dovrebbe 
essere CW_USEDEFAULT, che significa che sara' Windows a decidere per voi dove posizionare la finestra 
sul desktop.
nWidth, nHeight: La larghezza e l'altezza della finestra in pixels. Potete anche usare CW_USEDEFAULT 
per lasciare a Windows il compito di scegliere una larghezza ed un'altezza appropriate per voi.
hWndParent: L'handle della parent window della finestra (se esiste). Questo parametro dice a Windows se
questa finestra e' una child window (subordinata) di qualche altra finestra, e, se lo e', quale finestra e' la 
parent. Notate che questa non e' la stessa relazione parent-child degli MDI (multiple document interface, 
interfaccia multi-documento, NdT). Le child windows non sono legate alla client area della parent window.
Questa relazione e' esclusivamente per un uso interno a Windows. Se la parent window e' distrutta, tutte 
le child windows verranno distrutte automaticamente. E' davvero cosi' semplice. Poiche' nel nostro 
esempio c'e' solo una finestra, noi definiremo questo parametro come NULL.
hMenu: L'handle del menu della finestra. NULL se deve essere usato il menu della window class. Ritornate
un attimo al membro lpszMenuName della struttura WNDCLASSEX. lpszMenuName specifica il menu 
predefinito per la window class. Ogni finestra creata per questa window class avra' lo stesso menu per 
default, a meno che voi non specifichiate un *overriding* menu per una specifica finestra tramite il suo 
parametro hMenu. hMenu e' in realta' un parametro con due scopi. Nel caso in cui la finestra che volete 
creare e' di tipo predefinito (ad esempio un control), tale control non puo' possedere un menu. hMenu 
verra' usato come l'ID di quel control. Windows puo' distinguere se hMenu e' un menu handle o una 
control ID mediante il parametro lpClassName. Se e' il nome di una window class predefinita, hMenu e' 
una control ID. Altrimenti, e' un handle del menu della finestra.
hInstance: L'instance handle del program module che crea la finestra.
lpParam: Puntatore opzionale ad una struttura di dati passata alla finestra. E' usato dalle finestre MDI per
passare i dati CLIENTCREATESTRUCT. Normalmente, questo valore e' NULL, a significare che nessun dato
e' passato tramite CreateWindow(). La finestra puo' ritrovare il valore di questo parametro tramite la 
chiamata alla funzione GetWindowLong. 

mov   hwnd,eax 
    invoke ShowWindow, hwnd,CmdShow 
    invoke UpdateWindow, hwnd

Al ritorno con successo da CreateWindowEx, l'handle e' presente in eax. Dobbiamo conservare questo 
valore per un uso futuro. La finestra che abbiamo appena creato non e' mostrata automaticamente. 
Dovete chiamare ShowWindow con l'handle della finestra e il desiderato *display state* della finestra per 
far si che essa venga mostrata su schermo. Successivamente potete chiamare UpdateWindow per 
ordinare alla vostra finestra di ridisegnare la sua client area. Questa funzione e' utile quando volete 
aggiornare il contenuto della client area. Potete comunque omettere questa call.

.WHILE TRUE 
      invoke GetMessage, ADDR msg,NULL,0,0 
      .BREAK .IF (!eax) 
         invoke TranslateMessage, ADDR msg 
         invoke DispatchMessage, ADDR msg 
   .ENDW

A questo punto, la nostra finestra e' presente sullo schermo. Ma non puo' ricevere input dal mondo. 
Percio' dobbiamo *informarla* di eventi rilevanti. Facciamo cio' con un message loop. C'e' un unico 



message loop per ogni modulo. Questo message loop controlla continuamente se ci sono messaggi da 
Windows con la call a GetMessage. GetMessage passa un puntatore ad una struttura MSG a Windows. 
Questa struttura MSG sara' riempita con informazioni inerenti il messaggio che Windows vuole inviare ad 
una finestra nel modulo. La funzione GetMessage non ritornera' finche' non c'e' un messaggio per una 
finestra nel modulo. Durante quel lasso di tempo, Windows puo' dare il controllo ad altri programmi.can 
give control to other programs. Questo e' il principio che delinea lo schema di multitasking cooperativo 
della piattaforma Win16. GetMessage restituisce FALSE se viene ricevuto il messaggio WM_QUIT, che nel 
message loop, terminera' il loop ed uscira' dal programma. TranslateMessage e' una funzione che prende 
il semplice input dalla tastiera e genera un nuovo messaggio (WM_CHAR) che viene aggiunto alla coda 
dei messaggi. Il messaggio con WM_CHAR contiene il valore ASCII del tasto premuto, con il quale e' 
sicuramente piu' semplice interagire che con lo scan code del tasto premuto sulla tastiera. Potete 
omettere questa call se il vostro programma non elabora pressioni di tasti. DispatchMessage manda i dati
del messaggio alla window procedure responsabile per la specifica finestra a cui e' destinato il messaggio.

mov     eax,msg.wParam 
   ret 
WinMain endp

Se il message loop termina, l'exit code e' presente nel membro wParam della struttura MSG. Puoi 
conservare questo exit code in eax per restituirlo a Windows. Al momento, Windows non fa uso del valore
di ritorno, ma e' meglio essere sicuri e giocare secondo le regole.
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
Questa e' la procedura della nostra finestra, la nostra window procedure. Non siete obbligati a chiamarla 
WndProc. Il primo parametro, hWnd, e' l'handle della finestra a cui il messaggio e' destinato. uMsg e' il 
messaggio. Notate che uMsg non e' una struttura MSG. E' solo un numero, davvero. Window definisce 
centinaia di messaggi, la maggior parte dei quali non interesseranno mai il vostro programma. Windows 
mandera' un appropriato messaggio ad una finestra nel caso in cui qualcosa di rilevante accada a quella 
finestra. La window procedure riceve il messaggio e reagisce di conseguenza. wParam ed lParam sono 
solo parametri extra usati da alcuni message. Alcuni messaggi non mandano nessun dato o valore 
assieme al messaggio stesso. Questi dati sono passati alla window procedure tramite lParam e wParam. 

mov   eax,uMsg 
   .IF eax==WM_DESTROY 
      invoke PostQuitMessage,NULL 
      xor eax,eax 
   .ELSE 
      invoke DefWindowProc,hWnd,uMsg,wParam,lParam 
   .ENDIF 
ret 
WndProc endp

Ecco la parte fondamentale. Qui e' dove risiede la gran parte dell'intelligenza del vostro programma. Il 
codice che risponde ad ogni messaggio di Windows risiede nella window procedure. Il vostro codice deve 
controllare i messaggi di Windows per vedere se vi e' interessato o no. Se lo e', fate qualsiasi cosa 
vogliate in risposta a quel messaggio e quindi ritornate con zero in eax. Se non lo e', voi DOVETE passare
TUTTI i parametri per l'elaborazione di default del messaggio, eseguita da DefWindowProc. Questa 
DefWindowProc e' una funzione API che elabora i messaggi a cui il vostro programma non e' interessato. 
L'unico messaggio a cui DOVETE rispondere e' WM_DESTROY. Questo messaggio e' inviato alla vostra 
window procedure ogni qual volta la vostra finestra viene chiusa. Nel momento in cui la vostra window 
procedure riceve questo messaggio, la vostra finestra viene rimossa dallo schermo. Questa e' solo una 
notifica del fatto che la vostra finestra e' ormai distrutta, e voi dovreste prepararvi a ritornare a Windows.
In risposta a questo, potete sbrigare le "faccende di casa" prima di ritornare a Windows. Non avete scelta
se non uscire quando si arriva a questo punto. Se volete avere la possibilita' di fermare l'utente dal 
chiudere la vostra finestra, dovete elaborare il messaggio WM_CLOSE. Ma torniamo a WM_DESTROY. 
Dopo aver sbrigato le faccende di casa, dovete chiamare PostQuitMessage che rimandera' il messaggio 
WM_QUIT indietro al vostro module. WM_QUIT fara' si che GetMessage ritorni con il valore zero in eax, 
che, a sua volta, terminera' il message loop e uscira' a Windows. Potete mandare il messaggio 
WM_DESTROY alla vostra stessa window procedure chiamando la funzione DestroyWindow. 

Tutorial 4: Disegnare con il Testo

Introduzione
In questo tutorial, impareremo come "disegnare" del testo nella client area di una finestra. Impareremo 
anche cos'e' un device context (contesto di periferica, NdT). 

Preliminari
Il testo in Windows e' un tipo di oggetto GUI. Ogni carattere e' composto da numerosi pixels che sono 
ammassai insieme in un pattern ben definito. Ecco perche' e' chiamato "painting" (disegnare, NdT) invece
di "writing" (scrivere, NdT). Normalmente, disegnerete il testo nella vostra propria client area (in realta', 
potete disegnare anche al di fuori della client area ma quella e' un'altra storia). Mettere testo su schermo 
in Windows e' drasticamente diverso dal DOS. In DOS, potete considerare lo schermo di dimensioni 
80x25. Ma, in Windows, lo schermo e' condiviso da diversi programmi. Bisogna seguire alcune regole ben
precise per far si che dei programmi scrivano sulla parte dello schermo di altri programmi e cosi' via. 
Windows assicura che cio' non accada limitando la painting area (area in cui e' possibile 
scrivere/disegnare, NdT) di ogni finestra alla PROPRIA CLIENT AREA unicamente. Le dimensioni della 
client area di una finestra non sono costanti. L'utente puo' cambiare le dimensioni in qualsiasi momento. 
Pertanto, dovete determinare le dimensioni della client area dinamicamente, durante l'esecuzione. Prima 
di poter disegnare qualcosa nella client area, dovete chiedere il permesso a Windows. Esatto, non avete il
controllo assoluto dello schermo come ai tempi del DOS. Dovete chiedere a Windows il permesso di 
disegnare nella vostra client area. Windows determinera' le dimensioni della client area, le font, i colori e 
altri attributi GDI, e mandera' un handle per il contesto di periferica (device context) in ritorno al vostro 
programma. Potete quindi utilizzare il device context come un passaporto per disegnare nella vostra 
client area. Cos'e' un device context? E' semplicemente un struttura di dati mantenuta internamente da 
Windows. Un device context e' associato con una determinata periferica (device, NdT), come ad esempio 
una stampante o lo schermo. Per lo schermo, solitamente un device context e' associato con una 
particolare finestra presente su di esso. Alcuni dei valori nel device context sono attributi grafici come 
colore, tipo di font, ecc. Questi sono valori predefiniti che potete cambiare a piacimento. Essi esistono per
aiutare a ridurre il fastidio di dover specificare tali attributi in ogni chiamata a funzioni GDI. Quando un 
programma deve disegnare, esso deve ottenere l'handle di un device context. Normalmente, cio' si puo' 
fare in due modi. • chiamare BeginPaint in risposta al messaggio WM_PAINT. • chiamare GetDC in 
risposta ad altri messaggi. Una cosa che dovete ricordare, una volta completate le operazioni relative 
all'handle del device context, e' che dovete rilasciarlo durante l'elaborazione di un singolo messaggio. Non
commettete l'errore di trovare l'handle in risposta ad un messaggio e poi di rilasciarlo in riposta ad un 
altro. Windows manda i messaggi WM_PAINT ad una finestra per notificare che e' ora di ridisegnare la 
propria client area. Windows non salva il contenuto della client area di una finestra. Invece, quando si 
verifica una situazione che garantisce il ridipingimento di una client area (come ad esempio quando una 
finestra che era stata coperta da un'altra viene rimessa in primo piano), Windows aggiunge il messaggio 
WM_PAINT nella coda dei messaggi di quella finestra. E' poi responsabilita' di quella finestra il 
ridipingimento della propria client area. Voi dovete inserire tutte le informazioni inerenti il ridipingimento 

http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Disclaimer
http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Note_finali
http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Contenuto
http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Preliminari
http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Introduzione
http://quequero.org/Tutorial_4:_Disegnare_con_il_Testo#Tutorial_4:_Disegnare_con_il_Testo
javascript:toggleToc()


della vostra client area nella sezione relativa a WM_PAINT della vostra window procedure, cosicche' la la 
window procedure puo' ridisegnare la client area quando sopraggiunge il messaggio WM_PAINT. Un altro 
concetto con cui dovrete avere a che fare e' l'invalid rectangle (rettangolo invalido, NdT). Windows 
definisce un rettangolo invalido come la piu' piccola area rettangolare nella client area che abbia bisogno 
di essere ridisegnata. Quando Windows rileva un rettangolo invalido nella client area di una finestra, 
manda il messaggio WM_PAINT a quella finestra. In risposta al messaggio WM_PAINT, la finestra puo' 
ritrovare una struttura paintstruct che contiene, tra l'altro, le coordinate del rettangolo invalido. Voi 
chiamate BeginPaint in risposta al messaggio WM_PAINT per validare il rettangolo invalido. Se non 
elaborate il messaggio WM_PAINT, quanto meno dovete chiamare DefWindowProc o ValidateRect per 
validare il rettangolo invalido, altrimenti Windows continuera' all'infinito a mandarvi il messaggio 
WM_PAINT. Ecco i passi che dovete seguire in risposta ad un messaggio WM_PAINT: • Ottenere l'handle 
del device context con BeginPaint. • Disegnare la client area. • Rilasciare l'handle del device context con 
EndPaint Notate che non dovete esplicitamente validare il rettangolo invalido. Cio' e' fatto 
automaticamente dalla chiamata a BeginPaint. Nelle istruzioni tra BeginPaint ed EndPaint, potete 
chiamare qualsiasi funzione GDI per disegnare la vostra client area. Praticamente tutte richiedono un 
handle di un device context come uno dei parametri. 

Contenuto
Scriveremo un programma che mostra la stringa testuale "Win32 assembly is great and easy!" al centro 
della client area. 

include windows.inc 
includelib user32.lib 
includelib kernel32.lib 
.DATA 
ClassName db "SimpleWinClass",0 
AppName  db "Our First Window",0 
OurText  db "Win32 assembly is great and easy!",0 
.DATA? 
hInstance HINSTANCE ? 
CommandLine LPSTR ? 
.CODE 
 start: 
 invoke GetModuleHandle, NULL 
 mov    hInstance,eax 
 invoke GetCommandLine 
 invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT 
 invoke ExitProcess,eax 
WinMain proc hinst:HINSTANCE, hPrevInst:HINSTANCE, CmdLine:LPSTR, CmdShow:SDWORD 
    LOCAL wc:WNDCLASSEX 
    LOCAL msg:MSG 
    LOCAL hwnd:HWND 
    mov   wc.cbSize,SIZEOF WNDCLASSEX 
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance 
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 
    mov   wc.hIcon,eax 
    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc 
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ 
           hInst,NULL 
    mov   hwnd,eax 
    invoke ShowWindow, hwnd,SW_SHOWNORMAL 
    invoke UpdateWindow, hwnd 
        .WHILE TRUE 
                invoke GetMessage, ADDR msg,NULL,0,0 



                .BREAK .IF (!eax) 
                invoke TranslateMessage, ADDR msg 
                invoke DispatchMessage, ADDR msg 
        .ENDW 
        mov     eax,msg.wParam 
        ret 
WinMain endp 
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
    LOCAL hdc:HDC 
    LOCAL ps:PAINTSTRUCT 
    LOCAL rect:RECT 
    mov   eax,uMsg 
    .IF eax==WM_DESTROY 
        invoke PostQuitMessage,NULL 
    .ELSEIF eax==WM_PAINT 
        invoke BeginPaint,hWnd, ADDR ps 
        mov    hdc,eax 
        invoke GetClientRect,hWnd, ADDR rect 
        invoke DrawText, hdc,ADDR OurText,-1, ADDR rect, \ 
                DT_SINGLELINE or DT_CENTER or DT_VCENTER 
        invoke EndPaint,hWnd, ADDR ps 
    .ELSE 
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam 
        ret 
    .ENDIF 
    xor   eax, eax 
    ret 
WndProc endp 
end start

La maggior parte del codice e' uguale a quello nell'esempio del tutorial 3. Spieghero' soltanto i 
cambiamenti importanti. 

   LOCAL hdc:HDC  

   LOCAL ps:PAINTSTRUCT  

   LOCAL rect:RECT  

 

Queste sono variabili locali che vengono usate dalle funzioni GDI nella nostra sezione WM_PAINT. hdc e' 
usata per conservare l'handle del device context restituito dalla chiamata a BeginPaint. ps e' una 
struttura PAINTSTRUCT. Normalmente non utilizzate i valori in ps. E' passata alla funzione BeginPaint e 
Windows la riempie con i valori appropriati. Successivamente, passate ps alla funzione EndPaint quando 
avete finito di ridisegnare la client area. rect e' una struttura RECT definita come segue: 

RECT Struct 
    left    LONG ? 
    top     LONG ? 
    right   LONG ? 
    bottom  LONG ? 
RECT ends

Left e top sono le coordinate dell'angolo superiore sinistro di un rettangolo. Right e bottom sono le 
coordinate dell'angolo inferiore destro. Una cosa da ricordare: L'origine delle assi x-y e' nell'angolo 
superiore sinistro della client area. Pertanto il punto y=10 e' AL DI SOTTO del punto y=0. 

invoke BeginPaint,hWnd, ADDR ps 
   mov    hdc,eax 
   invoke GetClientRect,hWnd, ADDR rect 
   invoke DrawText, hdc,ADDR OurText,-1, ADDR rect, \ 
          DT_SINGLELINE or DT_CENTER or DT_VCENTER 
   invoke EndPaint,hWnd, ADDR ps



In risposta al messaggio WM_PAINT, voi chiamate BeginPaint con l'handle della finestra in cui volete 
disegnare ed una struttura PAINTSTRUCT non inizializzata come parametri. Dopo che la call e' stata 
eseguita con successo, eax contiene l'handle del device context. Successivamente, chiamate 
GetClientRect per trovare la dimensione della client area. La dimensione viene restituita nella variabile 
rect che voi passate a DrawText come uno dei suoi parametri. La sintassi di DrawText e': 

int WINAPI DrawText(HDC hdc, 
    LPCSTR lpString, 
    int nCount, 
    LPRECT lpRect, 
    UNIT uFormat);

DrawText e' una funzione API di output ad alto livello. Essa gestisce alcuni dettagli come il word wrap (a 
capo automatico, NdT), testo centrato, ecc. percio' vi potete concentrare sulla stringa che volete 
disegnare. Il suo fratello a basso livello, TextOut, sara' esaminato nel prossimo tutorial. DrawText 
formatta una stringa di testo per far si che essa rientri nei limiti di un rettangolo. Essa usa la font, il 
colore e lo sfondo (nel device context) correntemente impostati per disegnare il testo. Le linee sono 
automaticamente interrotte e riprese al nuovo rigo per rientrare nei limiti del rettangolo. La funzione 
restituisce l'altezza del testo di output in unita' relative alla periferica, nel nostro caso pixels. Diamo 
un'occhiata ai suoi parametri: 
hdc handle del device context 
lpString Un puntatore alla stringa che volete disegnare nel rettangolo. La stringa deve essere null-
terminated (terminata con zero, NdT) o altrimenti dovreste specificare la sua lunghezza nel parametro 
successivo, nCount. 
nCount Il numero di caratteri da scrivere. Se la stringa e' null-terminated, nCount deve essere -1. 
Altrimenti nCount deve contenere il numero dei caratteri nella stringa che voi volete disegnare. 
lpRect Un puntatore ad un rettangolo (una struttura di tipo RECT) all'interno del quale volete disegnare la
stringa. Notate che questo rettangolo e' anche un clipping rectangle, che significa che non potete 
disegnare la stringa al di fuori di questo rettangolo. 
uFormat Il valore che specifica in che modo la stringa viene mostrata nel rettangolo. Noi utilizziamo tre 
valori combinati dall'operatore "or": 

 DT_SINGLELINE specifica una singola linea di testo. 

 DT_CENTER centra il testo orizzontalmente. 

 DT_VCENTER centra il testo verticalmente. Deve essere usato con DT_SINGLELINE. 

Quando avete finito di disegnare la client area, dovete chiamare la funzione EndPaint per rilasciare 
l'handle del device context. Ecco tutto. Possiamo sintetizzare i punti salienti in questo modo: 

 Chiamate la coppia BeginPaint-EndPaint in risposta al messaggio WM_PAINT. 

 Fate cio' che volete con la client area tra le chiamate a BeginPaint e ad EndPaint. 

 Se volete ridisegnare la vostra client area in risposta ad altri messaggi, avete due scelte: 

 Usare la coppia GetDC-ReleaseDC e eseguire le vostre operazioni di disegno tra queste chiamate. 

 Chiamare InvalidateRect o UpdateWindow per invalidare l'intera client area, forzando Windows ad
inserire il messaggio WM_PAINT nella coda dei messaggi della vostra finestra, ed eseguire le 
vostre operazioni di disegno nella sezione relativa a WM_PAINT. 

Introduzione
Sperimenteremo di piu' con gli attributi del testo, ad esempio il tipo di font (= tipo di carattere, NdT) e il 
colore. 

Preliminari
Il sistema dei colori di Windows si basa su valori RGB, R=red (Rosso, NdT), G=Green (Verde, NdT), 
B=Blue (Indovinate, NdT :). Se volete specificare un colore in Windows, dovete dichiarare il colore da voi 
desiderato in funzione di questi tre colori principali. Ogni valore di colore ha un raggio che va da 0 a 255 
(un valore di un byte). Ad esempio, se volete un colore rosso puro, dovete usare 255,0,0. O se volete un 
colore bianco puro, dovete usare 255,255,255. Potete osservare dagli esempi che ottenere il colore di cui 
avete bisogno e' molto difficile con questo sistema, dal momento in cui dovete sapere bene come 
miscelare e unire i colori. Per il colore del testo e per lo sfondo, userete SetTextColor e SetBkColor, ed 
entrambi necessitano dell'handle di un device context e un valore RGB a 32-bits. La struttura del valore 
RGB a 32-bits e' definita cosi': 

RGB_value struct 
    unused   db 0 
    blue     db ? 
    green    db ? 
    red      db ? 
RGB_value ends

Notate che il primo byte non e' utilizzato e dovrebbe essere zero. L'ordine dei tre bytes rimanenti e' 
invertito, cioe' e' blue, green, red. Comunque, non useremo questa struttura poiche' e' complicata da 
inizializzare e da usare. Creeremo una macro, invece. La macro ricevera' i tre parametri: i valori red, 
green e blue. Produrra' quindi il valore RGB a 32-bits desiderato e lo conservera' in eax. La macro e' 
definita come segue: 

RGB macro red,green,blue 
    xor  eax,eax 
    mov  ah,blue 
    shl  eax,8 
    mov  ah,green 
    mov  al,red 
endm

Potete inserire questa macro nel file di include per usarla in futuro. 
Potete "creare" una font chiamando CreateFont o CreateFontIndirect. La differenza tra le due e' che 
CreateFontIndirect riceve solo un parametro: un puntatore ad una struttura font logica, LOGFONT. 
CreateFontIndirect e' la piu' flessibile tra le due specialmente se i vostri programmi hanno bisogno di 
cambiare font frequentemente. Comunque, nel nostro esempio, "creeremo" solo una font per 
dimostrazione, e possiamo farlo tranquillamente con CreateFont. Dopo la chiamata a CreateFont, essa 
restituira' l'handle di una font che dovete selezionare nel device context. Successivamente, ogni funzione 
API testuale utilizzera' la font che abbiamo selezionato nell'ambito del device context. 

Contenuto

Segue il nostro codice: 

include windows.inc 
includelib user32.lib 
includelib kernel32.lib 
includelib gdi32.lib 
RGB macro red,green,blue 
        xor eax,eax 
        mov ah,blue 
        shl eax,8 
        mov ah,green 
        mov al,red 
endm 
.data 
ClassName db "SimpleWinClass",0 
AppName  db "Our First Window",0 
TestString  db "Win32 assembly is great and easy!",0 
FontName db "script",0 
.data? 
hInstance HINSTANCE ? 
CommandLine LPSTR ? 
.code 
 start: 
invoke GetModuleHandle, NULL 
mov    hInstance,eax 
invoke GetCommandLine 
invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT 
invoke ExitProcess,eax 
WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD 
    LOCAL wc:WNDCLASSEX 
    LOCAL msg:MSG 
    LOCAL hwnd:HWND 
    mov   wc.cbSize,SIZEOF WNDCLASSEX 
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance 
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 
    mov   wc.hIcon,eax 
    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc 
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ 
           hInst,NULL 
    mov   hwnd,eax 
    invoke ShowWindow, hwnd,SW_SHOWNORMAL 
    invoke UpdateWindow, hwnd 
        .WHILE TRUE 
                invoke GetMessage, ADDR msg,NULL,0,0 
                .BREAK .IF (!eax) 
                invoke TranslateMessage, ADDR msg 
                invoke DispatchMessage, ADDR msg 
        .ENDW 
        mov     eax,msg.wParam 
        ret 
WinMain endp 
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
    LOCAL hdc:HDC 
    LOCAL ps:PAINTSTRUCT 
    LOCAL hfont:HFONT 
    mov   eax,uMsg 
    .IF eax==WM_DESTROY 
        invoke PostQuitMessage,NULL 
    .ELSEIF eax==WM_PAINT 
        invoke BeginPaint,hWnd, ADDR ps 



        mov    hdc,eax 
        invoke CreateFont,24,16,0,0,400,0,0,0,OEM_CHARSET,\ 
                                       OUT_DEFAULT_PRECIS,CLIP_DEFAULT_PRECIS,\ 
                                       DEFAULT_QUALITY,DEFAULT_PITCH or FF_SCRIPT,\ 
                                       ADDR FontName 
        invoke SelectObject, hdc, eax 
        mov    hfont,eax 
        RGB    200,200,50 
        invoke SetTextColor,hdc,eax 
        RGB    0,0,255 
        invoke SetBkColor,hdc,eax 
        invoke TextOut,hdc,0,0,ADDR TestString,SIZEOF TestString 
        invoke SelectObject,hdc, hfont 
        invoke EndPaint,hWnd, ADDR ps 
    .ELSE 
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam 
        ret 
    .ENDIF 
    xor    eax,eax 
    ret 
WndProc endp 
end start

Iniziamo la nostra analisi : ) 

invoke CreateFont,24,16,0,0,400,0,0,0,OEM_CHARSET,\ 
       OUT_DEFAULT_PRECIS,CLIP_DEFAULT_PRECIS,\ 
       DEFAULT_QUALITY,DEFAULT_PITCH or FF_SCRIPT,\ 
       ADDR FontName

CreateFont crea una font logica che si avvicini il piu' possibile ai parametri inseriti e ai dati sulle fonts 
disponibili. Questa funzione ha piu' parametri di ogni altra funzione di Windows. Essa restituisce l'handle 
della font logica che deve essere usata con la funzione SelectObject. Esamineremo i suoi parametri in 
dettaglio. 

HFONT CreateFont(int nHeight, 
                 int nWidth, 
                 int nEscapement, 
                 int nOrientation, 
                 int nWeight, 
                 BYTE cItalic, 
                 BYTE cUnderline, 
                 BYTE cStrikeOut, 
                 BYTE cCharSet, 
                 BYTE cOutputPrecision, 
                 BYTE cClipPrecision, 
                 BYTE cQuality, 
                 BYTE cPitchAndFamily, 
                 LPSTR lpFacename);

nHeight L'altezza dei caratteri desiderata. 0 significa dimensione predefinita. 
nWidth La larghezza dei caratteri desiderata. Normalmente questo valore dovrebbe essere 0, che 
permette a Windows di far combaciare la larghezza con l'altezza. Comunque, in questo esempio, la 
larghezza predefinita rende i caratteri quasi illeggibili, percio' ho usato la larghezza 16. 
nEscapement Specifica l'orientamento del carattere successivo relativo al precedente in decimi di grado. 
Normalmente, e' 0. Usate 900 per far si che i caratteri vengano scritti dal basso verso l'alto rispetto al 
primo, 1800 per farli scrivere da destra a sinistra, o 2700 per scrivere ciascun carattere dall'alto verso il 
basso. 
nOrientation Specifica quanto il carattere deve essere ruotato al momento della scrittura in decimi di 
grado. Usate 900 per far giacere i caratteri sulla schiena, 1800 per scrittura invertita, ecc. 
nWeight Regola lo spessore di ogni carattere. Windows definisce le seguenti dimensioni: 

FW_DONTCARE    equ 0 
FW_THIN        equ 100 
FW_EXTRALIGHT  equ 200 
FW_ULTRALIGHT  equ 200 
FW_LIGHT       equ 300 
FW_NORMAL      equ 400 
FW_REGULAR     equ 400 



FW_MEDIUM      equ 500 
FW_SEMIBOLD    equ 600 
FW_DEMIBOLD    equ 600 
FW_BOLD        equ 700 
FW_EXTRABOLD   equ 800 
FW_ULTRABOLD   equ 800 
FW_HEAVY       equ 900 
FW_BLACK       equ 900

cItalic 0 per normale, qualsiasi altro valore per caratteri in corsivo. 
cUnderline 0 per normale, qualsiasi altro valore per caratteri sottolineati. 
cStrikeOut 0 per normale, qualsiasi altro valore per caratteri attraversati da una linea nel centro. 
cCharSet Il set di caratteri per la font. Normalmente dovrebbe essere OEM_CHARSET che permette a 
Windows di selezionare la font che sta funzionando in dipendenza del sistema. 
cOutputPrecision Specifica con quale precisione la font selezionata deve essere vicina alle caratteristiche 
che noi vogliamo. Normalmente dovrebbe essere OUT_DEFAULT_PRECIS che definisce il comportamento 
di font mapping predefinito. 
cClipPrecision Specifica la clipping precision (precisione di taglio, NdT). La precisione di taglio definisce 
come tagliare i caratteri che stanno parzialmente al di fuori della clipping region. Non dovreste avere 
problemi usando CLIP_DEFAULT_PRECIS che definisce il comportamento di clipping predefinito. 
cQuality Specifica la qualita' dell'output. La qualita' dell'output definisce il grado di precisione con cui GDI 
deve cercare di far combaciare gli attributi della font logica con quelli di una reale font fisica. Ci sono tre 
scelte: DEFAULT_QUALITY, PROOF_QUALITY e DRAFT_QUALITY. 
cPitchAndFamily Specifica il pitch (tono, NdT) e la famiglia della font. Dovete combinare i valori pitch e 
family con l'operatore "or". 
lpFacename Un puntatore ad una stringa null-terminated che specifica il nome attribuito alla font. 
La descrizione precedente non e' assolutamente completa di tutti i dettagli. Dovreste fare riferimento alla 
vostra Win32 API reference per maggiori dettagli. 

invoke SelectObject, hdc, eax 
        mov    hfont,eax

Dopo aver ottenuto l'handle della font logica, dobbiamo usarlo per selezionare la font all'interno del 
device context chiamando SelectObject. SelectObject inserisce i nuovi oggetti GDI come pens (penne, 
NdT), brushs (spazzole, NdT) e fonts nel device context che deve essere usato dalle funzioni GDI. Essa 
restituisce l'handle dell'oggetto rimpiazzato che dovremmo conservare per future chiamate a 
SelectObject. Dopo la chiamata a SelectObject, qualsiasi output testuale usera' la font che abbiamo 
appena selezionato nel device context. 

RGB    200,200,50 
        invoke SetTextColor,hdc,eax 
        RGB    0,0,255 
        invoke SetBkColor,hdc,eax

Usate la macro RGB per creare un valore RGB a 32-bits da essere usato con SetColorText e SetBkColor. 

invoke TextOut,hdc,0,0,ADDR TestString,SIZEOF TestString

Chiamate la funzione TextOut per disegnare il testo nella client area. Il testo avra' la font e il colore che 
abbiamo specificato in precedenza. La sinstassi di TextOut e' questa: 

BOOL TextOut( 
    HDC  hdc,          // handle del device context 
    int  nXStart,      // coordinata x della posizione di partenza
    int  nYStart,      // coordinata y della posizione di partenza
    LPCTSTR  lpString, // indirizzo della stringa
    int  cbString      // numero di caratteri nella stringa
   ); 

invoke SelectObject,hdc, hfont

Quando abbiamo finito con la font, dovremmo ripristinare la vecchia font nel device context. Dovreste 
sempre ripristinare l'oggetto che avete rimpiazzato nel device context. 

Introduzione
Impareremo come fa Windows a ricevere input dalla tastiera. 

Preliminiari
Dal momento in cui c'e' solo una tastiera in ogni PC, tutti i programmi Windows in esecuzione devono 
condividerla. Windows e' responsabile per l'invio dell'input di pressione del tasto alla finestra che detiene 
l'input focus (quando un cursore lampeggia in un box, si dice che quel box ha l' "input focus", NdT). 
Nonostante possono coesistere diverse finestre sullo schermo, solo una di esse ha l'input focus. La 
finestra che ha l'input focus e' l'unica che puo' ricevere l'input dai tasti. Potete distinguere la finestra che 
ha l'input focus dalle altre osservando la sua barra del titolo, che e' evidenziata. In realta', ci sono due 
tipi principali di messaggi da tastiera. Potete considerare una tastiera come un gruppo di tasti. Per 
esempio, se premete il tasto "a", Windows manda un messaggio WM_KEYDOWN alla finestra che ha 
l'input focus, notificando che un tasto e' stato premuto. Quando sollevate il dito dal tasto, Windows 
manda un messaggio WM_KEYUP. In questo caso, trattate un tasto come se fosse un bottone. Un altro 
modo di considerare la tastiera e' vederla come un periferica grazie alla quale si ricevono dei caratteri 
come input. Quando premete il tasto "a", Windows manda un messaggio WM_CHAR alla finestra che ha 
l'input focus, dicendole che l'utente le sta inviando un carattere "a". Infatti, Windows manda i messaggi 
WM_KEYDOWN, WM_CHAR, e WM_KEYUP alla finestra che ha l'input focus. La window procedure puo' 
decidere se elaborare tutti e tre i messaggi o solo quelli a cui e' interessata. La maggior parte delle volte, 
potrete ignorare WM_KEYDOWN e WM_KEYUP dal momento che la funzione TranslateMessage nel 
message loop traduce i messaggi WM_KEYDOWN e WM_KEYUP in un messaggio WM_CHAR. Noi ci 
concentreremo su WM_CHAR in questo tutorial. 

Contenuto
include windows.inc 
includelib user32.lib 
includelib kernel32.lib 
includelib gdi32.lib 
.data 
ClassName db "SimpleWinClass",0 
AppName  db "Our First Window",0 
char WPARAM 20h                         // il carattere che il programma riceve dalla tastiera
.data? 
hInstance HINSTANCE ? 
CommandLine LPSTR ? 
.code 
start: 
    invoke GetModuleHandle, NULL 

    mov    hInstance,eax 
    invoke GetCommandLine 
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT 
    invoke ExitProcess,eax 
    WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD 
    LOCAL wc:WNDCLASSEX 
    LOCAL msg:MSG 
    LOCAL hwnd:HWND 
    mov   wc.cbSize,SIZEOF WNDCLASSEX 
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance 
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 
    mov   wc.hIcon,eax 
    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc 
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ 
           hInst,NULL 
    mov   hwnd,eax 
    invoke ShowWindow, hwnd,SW_SHOWNORMAL 
    invoke UpdateWindow, hwnd 
    .WHILE TRUE 
                invoke GetMessage, ADDR msg,NULL,0,0 
                .BREAK .IF (!eax) 
                invoke TranslateMessage, ADDR msg 
                invoke DispatchMessage, ADDR msg 
        .ENDW 
    mov     eax,msg.wParam 
    ret 
WinMain endp 
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
    LOCAL hdc:HDC 
    LOCAL ps:PAINTSTRUCT 
    mov   eax,uMsg 
    .IF eax==WM_DESTROY 
        invoke PostQuitMessage,NULL 
    .ELSEIF eax==WM_CHAR 
        push wParam 
        pop  char 
        invoke InvalidateRect, hWnd,NULL,TRUE 
    .ELSEIF eax==WM_PAINT 
        invoke BeginPaint,hWnd, ADDR ps 
        mov    hdc,eax 
        invoke TextOut,hdc,0,0,ADDR char,1 
        invoke EndPaint,hWnd, ADDR ps 
    .ELSE 
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam 
        ret 
    .ENDIF 
    xor    eax,eax 
    ret 
WndProc endp 
end start

Analizziamo il codice
char WPARAM 20h     //; il carattere che il programma riceve dalla tastiera

Questa e' la variabile che conserva il carattere ricevuto dalla tastiera. Dal momento che il carattere e' 
inviato al WPARAM della window procedure, noi definiamo la variabile di tipo WPARAM per semplicita'. Il 



valore iniziale e' 20h (spazio) poiche' quando la nostra finestra "rinfresca" la sua client area la prima 
volta, non c'e' nessun input di caratteri. Percio' noi mostreremo uno spazio. 

.ELSEIF eax==WM_CHAR 
   push wParam 
   pop  char 
   invoke InvalidateRect, hWnd,NULL,TRUE

Questa parte e' aggiunta alla window procedure per gestire il messaggio WM_CHAR. Essa semplicemente 
mette il carattere nella variabile "char" e quindi chiama InvalidateRect. InvalidateRect rende invalido uno 
specifico rettangolo della client area, la qual cosa obbliga Windows a mandare il messaggio WM_PAINT 
alla window procedure. La sua sintassi e' questa: 

BOOL InvalidateRect( 
    HWND  hWnd,                    // handle della finestra la cui regione deve essere 
aggiornata
    CONST RECT  * lpRect,     // indirizzo delle coordinate del rettangolo
    BOOL  bErase                       // flag "cancella lo sfondo"
   );

lpRect e' un puntatore al rettangolo nella client area che vogliamo dichiarare invalido. Se questo 
parametro e' NULL (=0, NdT), l'intera client area sara' segnata come invalida. bErase e' una flag che 
comunica a Windows se si deve cancellare lo sfondo. Se questa flag e' TRUE (=1, NdT), Windows 
cancellera' lo sfondo del rettangolo invalido quando viene chiamata la funzione BeginPaint. Percio' la 
strategia che abbiamo usato qui e' questa: conserviamo tutte le informazioni necessarie su come 
disegnare la client area e generiamo il messaggio WM_PAINT per disegnare la client area. Ovviamente, le
istruzioni nella sezione WM_PAINT devono sapere cosa ci si aspetta da loro. Questo sembrerebbe una 
specie di sistema "alternativo" nel fare le cose, una sorta di scappatoia, ma e' cosi' che lavora Windows. 
In realta' possiamo disegnare nella client area mentre elaboriamo il messaggio WM_CHAR chiamando la 
coppia di funzioni GetDC e ReleaseDC. Non c'e' problema. Ma il divertimento comincia quando la nostra 
finestra ha bisogno di ridisegnare la propria client area. Poiche' le istruzioni che disegnano il carattere 
risiedono nella sezione relativa a WM_CHAR, la window procedure non sara' in grado di ridisegnare il 
nostro carattere nella client area. Percio' la conclusione e': mettete tutti i dati e le istruzioni necessarie 
per disegnare sotto WM_PAINT. Potete mandare il messaggio WM_PAINT da qualsiasi posto nel vostro 
codice in qualsiasi momento vi venga voglia di ridisegnare la vostra client area. 

invoke TextOut,hdc,0,0,ADDR char,1

Quando viene chiamata la funzione InvalidateRect, essa restituisce un messaggio WM_PAINT alla window 
procedure. Di conseguenza vengono eseguite le istruzioni nella sezione WM_PAINT. Questa chiama, come
al solito, BeginPaint per trovare l'handle del device context e quindi chiama TextOut che disegna il nostro 
carattere nella client area alle coordinate x=0, y=0. Quando fate partire il programma e premete un tasto
qualsiasi, vedrete il carattere riprodursi nell'angolo superiore sinistro della client window. E quando la 
finestra viene ridotta ad icona e quindi ingrandita di nuovo, il carattere sta ancora li' poiche' tutte le 
istruzioni e i dati essenziali per ridisegnarlo risiedono nella sezione WM_PAINT. 

Introduzione
Impareremo come ricevere input dal mouse e rispondervi nella nostra window procedure. Il programma 
esempio attendera' i clicks del tasto sinistro del mouse e mostrera' una stringa di testo nel punto esatto 
della client area in cui l'utente ha clickato. 

Preliminari
Come con l'input da tastiera, Windows rileva e notifica le attivita' del mouse che sono rilevanti per 
ciascuna finestra. Tali attivita' includono clicks del tasto sinistro e destro, movimenti del puntatore nella 
finestra, doppi clicks. Diversamente dall'input da tastiera che e' diretto alla finestra che detiene l'input 
focus, i messaggi del mouse sono inviati a qualsiasi finestra su cui si trova il puntatore del mouse, sia 
essa attiva o non. In piu', ci sono messaggi del mouse relativi anche alla non-client area. Ma la maggior 
parte delle volte, possiamo tranquillamente ignorarli. Possiamo focalizzare la nostra attenzione su quelli 
relativi alla client area. Per ciascun tasto, sinistro e destro, del mouse, ci sono due messaggi associati: 
WM_LBUTTONDOWN, WM_RBUTTONDOWN e WM_LBUTTONUP, WM_RBUTTONUP. Per un mouse a tre 
tasti, ci sono in piu' WM_MBUTTONDOWN e WM_MBUTTONUP. Quando il puntatore del mouse si muove 
all'interno della client area, Windows manda messaggi WM_MOUSEMOVE alla finestra relativa. Una 
finestra puo' ricevere messaggi di doppio click, WM_LBUTTONDBCLK o WM_RBUTTONDBCLK, se e solo se
la sual window class e' stata istruita a riceverli, tramite l'inclusione della flag CS_DBLCLKS nel class style,
altrimenti la finestra ricevera' soltanto una serie di "mouse button up" e "mouse button down". Per tutti 
questi messaggi, il valore di lParam equivale alla posizione del mouse. La low word e' la coordinata x, e la
high word e' la coordinata y, relative all'angolo superiore sinistro della client area della finestra. wParam 
indica lo stato dei tasti del mouse e dei tasti Shift e Ctrl. 

Contenuto
include windows.inc 
includelib user32.lib 
includelib kernel32.lib 
includelib gdi32.lib 
.data 
ClassName db "SimpleWinClass",0 
AppName  db "Our First Window",0 
MouseClick db 0         ; 0=non si e' ricevuto ancora nessun click
.data? 
hInstance HINSTANCE ? 
CommandLine LPSTR ? 
hitpoint POINT <> 
.code 
start: 
    invoke GetModuleHandle, NULL 
    mov    hInstance,eax 
    invoke GetCommandLine 
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT 
    invoke ExitProcess,eax 
WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD 
    LOCAL wc:WNDCLASSEX 
    LOCAL msg:MSG 
    LOCAL hwnd:HWND 
    mov   wc.cbSize,SIZEOF WNDCLASSEX 
    mov   wc.style, CS_HREDRAW or CS_VREDRAW 
    mov   wc.lpfnWndProc, OFFSET WndProc 
    mov   wc.cbClsExtra,NULL 
    mov   wc.cbWndExtra,NULL 
    push  hInstance 
    pop   wc.hInstance 
    mov   wc.hbrBackground,COLOR_WINDOW+1 
    mov   wc.lpszMenuName,NULL 
    mov   wc.lpszClassName,OFFSET ClassName 
    invoke LoadIcon,NULL,IDI_APPLICATION 
    mov   wc.hIcon,eax 



    mov   wc.hIconSm,0 
    invoke LoadCursor,NULL,IDC_ARROW 
    mov   wc.hCursor,eax 
    invoke RegisterClassEx, addr wc 

    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ 
           hInst,NULL 

    mov   hwnd,eax 
    invoke ShowWindow, hwnd,SW_SHOWNORMAL 
    invoke UpdateWindow, hwnd 
    .WHILE TRUE 
                invoke GetMessage, ADDR msg,NULL,0,0 
                .BREAK .IF (!eax) 
                invoke DispatchMessage, ADDR msg 
    .ENDW 
    mov     eax,msg.wParam 
    ret 
WinMain endp 

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM 
    LOCAL hdc:HDC 
    LOCAL ps:PAINTSTRUCT 
    mov   eax,uMsg 
    .IF eax==WM_DESTROY 
        invoke PostQuitMessage,NULL 
    .ELSEIF eax==WM_LBUTTONDOWN 
        mov eax,lParam 
        shl eax,16 
        shr eax,16 
        mov hitpoint.x,eax 
        mov eax,lParam 
        shr eax,16 
        mov hitpoint.y,eax 
        mov MouseClick,TRUE 
        invoke InvalidateRect,hWnd,NULL,TRUE 
    .ELSEIF eax==WM_PAINT 
        invoke BeginPaint,hWnd, ADDR ps 
        mov    hdc,eax 
        .IF MouseClick 
            invoke lstrlen,ADDR AppName 
            invoke TextOut,hdc,hitpoint.x,hitpoint.y,ADDR AppName,eax 
        .ENDIF 
        invoke EndPaint,hWnd, ADDR ps 
    .ELSE 
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam 
        ret 
    .ENDIF 
    xor    eax,eax 
    ret 
WndProc endp 
end start

E' ora di analizzare il programma. 

.ELSEIF eax==WM_LBUTTONDOWN 
    mov eax,lParam 
    and eax,0FFFFh ; comment
    mov hitpoint.x,eax 
    mov eax,lParam 
    shr eax,16 
    mov hitpoint.y,eax 
    mov MouseClick,TRUE 
    invoke InvalidateRect,hWnd,NULL,TRUE

La window procedure attende un click dal tasto sinistro del mouse. Quando riceve il messaggio 
WM_LBUTTONDOWN, lParam contiene le coordinate del puntatore del mouse nella client area. Le 
coordinate vengono salvate in una variabile di tipo POINT che e' definita cosi': 



POINT STRUCT 
    x   dd ? 
    y   dd ? 
POINT ENDS

e cambia la flag, MouseClick, in TRUE, a significare che c'e' stato almeno un click del tasto sinistro del 
mouse nella client area. 

mov eax,lParam 
        and eax,0FFFFh 
        mov hitpoint.x,eax

Poiche' la coordinata x e' la low word di lParam e i membri della struttura POINT sono a 32-bits come 
dimensioni, noi dobbiamo azzerare la high word di eax prima di conservarla in hitpoint.x. 

shr eax,16 
        mov hitpoint.y,eax

E poiche' la coordinata y e' la high word di lParam, dobbiamo metterla nella low word di eax prima di 
conservarla in hitpoint.y. Facciamo cio' spostando (shifting) eax di 16 bit a destra (con shr). Dopo aver 
salvato la posizione del mouse, settiamo la flag, MouseClick, a TRUE per far si che le istruzioni di 
"scrittura" nella sezione WM_PAINT sappiano che c'e' stato almeno un click nella client area, e di 
conseguenza disegnino la stringa alla posizione del mouse. Successivamente chiamiamo la funzione 
InvalidateRect per costringere la finestra a ridisegnare la sua intera client area. 

.IF MouseClick 
            invoke lstrlen,ADDR AppName 
            invoke TextOut,hdc,hitpoint.x,hitpoint.y,ADDR AppName,eax 
        .ENDIF

Le istruzioni di scrittura nella sezione WM_PAINT devono controllare se MouseClick e' TRUE, dal momento
che quando la finestra e' stata creata, ha ricevuto ugualmente un messaggio WM_PAINT, ma in quel 
momento non c'era stato nessun click e quindi non doveva disegnare la stringa nella client area. Pertanto
noi inizializziamo MouseClick come FALSE e cambiamo il suo valore in TRUE quando l'utente clicka 
effettivamente il tasto sinistro del mouse. Se c'e' stato almeno un click del mouse, il codice disegna la 
stringa nella client area alla posizione del mouse. Notate che chiamiamo lstrlen per ottenere la lunghezza 
della stringa da mostrare, e utilizziamo tale lunghezza come ultimo parametro della funzione TextOut. 

Introduzione
In questo tutorial, impareremo come incorporare un menu nella nostra finestra. 

Preliminari
Il Menu e' uno dei componenti piu' importanti nella vostra finestra. Il menu presenta una lista dei servizi 
che il programma offre all'utente. L'utente non deve necessariamente leggere il manuale incluso con il 
programma per poterlo usare, puo' sperimentare con i menu per avere una visione generale delle 
capacita' di un determinato programma, e pertanto cominciare a sfruttarlo immediatamente. Dal 
momento che un menu e' un mezzo per mettere l'utente a suo agio e fargli cominciare subito ad 
utilizzare il programma, dovreste seguire gli standard. In poche parole, i primi due menu items 
dovrebbero essere File ed Edit e l'ultimo dovrebbe essere Help. Potete inserire i vostri menu tra Edit ed 
Help. Se un menu item chiama una dialog box, dovreste aggiungere dei puntini sospensivi (...) alla 
stringa con il nome del menu. I menu sono una specie di resource (risorsa, NdT). Ci sono diversi tipi di 
resources, come il dialog box, la string table, le icone, i bitmap, i menu ecc. Le resources sono definite in 
un file separato chiamato "resource file", che normalmente ha un'estensione .rc. Voi quindi combinate le 
resources con il codice al momento del link. Il prodotto finale e' un file eseguibile che contiene sia le 
istruzioni che le resources. Potete scrivere scripts per le resources usando qualsiasi editor di testo. Essi 
sono composti di frasi che descrivono l'aspetto e altri attributi delle resources usate in un determinato 
programma. Nonostante possiate utilizzare un editor di testo per scrivere i resource scripts, cio' e' 
abbastanza difficile. Una scelta migliore e' quella di utilizzare un editor di risorse (resource editor), che vi 
permette di creare visualmente qualsiasi resource con semplicita'. I resource editors sono spesso inclusi 
nei package dei compilatori, come nel Visual C++, Borland C++, ecc. Essi scrivono il resource file per voi.
Descrivete una menu resource in questo modo: 

MyMenu  MENU 
{ 
   [qui va la lista dei menu] 
}

Chi programma in C potrebbe riconoscere che il procedimento e' simile alla dichiarazione di una structure.
MyMenu e' il menu name, seguito dalla keyword MENU e la lista dei menu tra parentesi graffe. In 
alternativa, potete usare BEGIN e END invece delle parentesi graffe, se volete. Questa sintassi e' piu' 
consona a chi programma in Pascal. La lista dei menu puo' contenere sia statements MENUITEM che 
statements POPUP. Lo statement MENUITEM definisce una barra di menu che non chiama un menu di 
popup quando viene selezionato. La sintassi e' la seguente: 

MENUITEM "&testo", ID [,opzioni]

Si comincia con la keyword MENUITEM seguita dal testo che volete usare come nome del menu. Notate 
l'amperstand (il simbolo "&", NdT). Esso fa si che il carattere successivo sia sottolineato. Dopo la stringa 
di testo, c'e' la ID del menu item. La ID e' un numero che verra' usato per identificare il menu item nel 
messaggio inviato alla window procedure quando il menu item viene selezionato. Pertanto, le ID di menu 
devono essere uniche tra di loro. Le opzioni sono opzionali ( :-), NdT). Le opzioni disponibili sono queste: 

 GRAYED Il menu item e' disattivato, e non genera un messaggio WM_COMMAND. Il testo e' grigio
(per intenderci, quello dei menu non selezionabili, NdT). 

 INACTIVE Il menu item e' disattivato, e non genera un messaggio WM_COMMAND. Il testo viene 
mostrato normalmente. 

 MENUBREAK Questo item e gli items successivi appaiono su una linea nuova nel menu. 

 HELP Questo item e gli items successivi sono giustificati a destra. 

Potete usare una delle opzioni di cui sopra o combinarle con l'operatore "or". Ricordatevi pero' che 
INACTIVE e GRAYED non possono essere combinati. Lo statement POPUP ha la seguente sintassi: 

POPUP "&testo" [,opzioni] 
{ 
  [lista dei menu] 
}



Lo statement POPUP definisce una barra di menu che, quando selezionata, apre una lista di menu items 
in una piccola finestra di popup. La lista dei menu puo' essere di tipo MENUTIEM o POPUP. C'e' un tipo 
speciale di statement MENUITEM, e cioe' MENUITEM SEPARATOR, che disegna una linea orizzontale nella 
finestra di popup. Il passo successivo, una volta terminato il menu resource script, e' di riferirvisi dal 
vostro programma. Potete farlo in due posti diversi del programma. 

 Nel membro lpszMenuName della struttura WNDCLASSEX. Ad esempio, se avete un menu 
chiamato "FirstMenu", potete assegnare il menu alla vostra finestra in questo modo: 

.DATA 
MenuName  db "FirstMenu",0
. . .
. . .
.CODE 
. . .
mov   wc.lpszMenuName, OFFSET MenuName 
. . .

 Nel parametro "menu handle" di CreateWindowEx, in questo modo: 
.DATA 
MenuName  db "FirstMenu",0 
hMenu HMENU ?
. . .  
. . . 
.CODE 
. . . 
invoke LoadMenu, hInst, OFFSET MenuName 
mov   hMenu, eax 
invoke CreateWindowEx,NULL,OFFSET ClsName,\ 
            OFFSET Caption, WS_OVERLAPPEDWINDOW,\ 
            CW_USEDEFAULT,CW_USEDEFAULT,\ 
            CW_USEDEFAULT,CW_USEDEFAULT,\ 
            NULL,\ 
            hMenu,\ 
            hInst,\ 
            NULL\ 
. . .

A questo punto potreste chiedervi, qual'e' la differenza tra questi due metodi ?
Quando riferite un menu nella struttura WNDCLASSEX, il menu diventa il menu predefinito, di "default", 
per la window class. Ogni finestra di quella class avra' lo stesso menu. Se volete che ogni finestra creata 
dalla stessa class abbia menu diversi, dovete scegliere il secondo metodo. In questo caso, qualsiasi 
finestra a cui e' passato un menu handle nella funzione CreateWindowEx avra' un menu che "sovrascrive"
il menu di default definito nella struttura WNDCLASSEX. Successivamente esamineremo in che modo un 
menu notifica alla window procedure che l'utente seleziona un menu item. Quando l'utente seleziona un 
menu item, la window procedure ricevera' un messaggio WM_COMMAND. La low word di wParam 
contiene la menu ID del menu item selezionato. A questo punto abbiamo abbastanza informazioni per 
creare ed usare un menu. Facciamolo. 

Contenuto
Il primo esempio mostra come creare ed usare un menu specificando il nome del menu nella window 
class. 

include windows.inc <br>
includelib user32.lib <br>
includelib kernel32.lib <br>
includelib gdi32.lib <br>
.data <br>
ClassName db "SimpleWinClass",0 <br>
AppName  db "Our First Window",0 <br>
MenuName db "FirstMenu",0                //; Il nome del nostro menu nel resource file. <br>
Test_string db "You selected Test menu item",0 <br>
Hello_string db "Hello, my friend",0 <br>
Goodbye_string db "See you again, bye",0 <br>
.data? <br>
hInstance HINSTANCE ? <br>
CommandLine LPSTR ? <br>



.const <br>
IDM_TEST equ 1                    //; le Menu IDs <br>
IDM_HELLO equ 2 <br>
IDM_GOODBYE equ 3 <br>
IDM_EXIT equ 4 <br>
.code <br>
start: <br>
    invoke GetModuleHandle, NULL <br>
    mov    hInstance,eax <br>
    invoke GetCommandLine <br>
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT <br>
    invoke ExitProcess,eax <br>
WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD <br>
    LOCAL wc:WNDCLASSEX <br>
    LOCAL msg:MSG <br>
    LOCAL hwnd:HWND <br>
    mov   wc.cbSize,SIZEOF WNDCLASSEX <br>
    mov   wc.style, CS_HREDRAW or CS_VREDRAW <br>
    mov   wc.lpfnWndProc, OFFSET WndProc <br>
    mov   wc.cbClsExtra,NULL <br>
    mov   wc.cbWndExtra,NULL <br>
    push  hInstance <br>
    pop   wc.hInstance <br>
    mov   wc.hbrBackground,COLOR_WINDOW+1 <br>
    mov   wc.lpszMenuName,OFFSET MenuName        //; Mettete il nostro nome qui<br>
    mov   wc.lpszClassName,OFFSET ClassName <br>
    invoke LoadIcon,NULL,IDI_APPLICATION <br>
    mov   wc.hIcon,eax <br>
    mov   wc.hIconSm,0 <br>
    invoke LoadCursor,NULL,IDC_ARROW <br>
    mov   wc.hCursor,eax <br>
    invoke RegisterClassEx, addr wc <br>
    
    invoke CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ <br>
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ <br>
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ <br>
           hInst,NULL <br>
    
    mov   hwnd,eax <br>
    invoke ShowWindow, hwnd,SW_SHOWNORMAL <br>
    invoke UpdateWindow, hwnd <br>
    .WHILE TRUE <br>
                invoke GetMessage, ADDR msg,NULL,0,0 <br>
                .BREAK .IF (!eax) <br>
                invoke DispatchMessage, ADDR msg <br>
    .ENDW <br>
    mov     eax,msg.wParam <br>
    ret <br>
WinMain endp <br>
WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM <br>
    mov   eax,uMsg <br>
    .IF eax==WM_DESTROY <br>
        invoke PostQuitMessage,NULL <br>
    .ELSEIF eax==WM_COMMAND <br>
        mov eax,wParam <br>
        .IF ax==IDM_TEST <br>
            invoke MessageBox,NULL,ADDR Test_string,OFFSET AppName,MB_OK <br>
        .ELSEIF ax==IDM_HELLO <br>
            invoke MessageBox, NULL,ADDR Hello_string, OFFSET AppName,MB_OK <br>
        .ELSEIF ax==IDM_GOODBYE <br>
            invoke MessageBox,NULL,ADDR Goodbye_string, OFFSET AppName, MB_OK <br>
        .ELSE <br>
            invoke DestroyWindow,hWnd <br>
        .ENDIF <br>
    .ELSE <br>
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam <br>
        ret <br>
    .ENDIF <br>
    xor    eax,eax <br>
    ret <br>
WndProc endp <br>
end start <br>



//***********************
Menu.rc
//***********************
&#35;define IDM_TEST 1 
&#35;define IDM_HELLO 2 
&#35;define IDM_GOODBYE 3 
&#35;define IDM_EXIT 4 
FirstMenu MENU 
{ 
 POPUP "&PopUp" 
        { 
         MENUITEM "&Say Hello",IDM_HELLO 
         MENUITEM "Say &GoodBye", IDM_GOODBYE 
         MENUITEM SEPARATOR 
         MENUITEM "E&xit",IDM_EXIT 
        } 
 MENUITEM "&Test", IDM_TEST 
}

Analizziamo prima il resource file. 

&#35;define IDM_TEST 1                ///* uguale a IDM_TEST equ 1*/ 
&#35;define IDM_HELLO 2 
&#35;define IDM_GOODBYE 3 
&#35;define IDM_EXIT 4

Queste righe definiscono le menu IDs usate dal menu script. Potete assegnare qualsiasi valore alle ID 
sempre che tale valore sia unico nel menu. 

FirstMenu MENU

Dichiara il vostro menu con la keyword MENU. 

POPUP "&PopUp" 
        { 
         MENUITEM "&Say Hello",IDM_HELLO 
         MENUITEM "Say &GoodBye", IDM_GOODBYE 
         MENUITEM SEPARATOR 
         MENUITEM "E&xit",IDM_EXIT 
        }

Definiscono un menu a popup con quattro menu items, di cui il terzo e' un menu separator. 

MENUITEM "&Test", IDM_TEST

Definisce una barra di menu nel menu principale.
Successivamente esamineremo il codice: 

MenuName db "FirstMenu",0                //; Il nome del nostro menu nel resource file. 
Test_string db "You selected Test menu item",0 
Hello_string db "Hello, my friend",0 
Goodbye_string db "See you again, bye",0

MenuName e' il nome del menu nel resource file. Notate che potete definire piu' di un menu nel resource 
file, pertanto dovete specificare quale menu usare. Le restanti tre righe definiscono le stringe testuali che
devono essere mostrate in message boxes, che vengono chiamate quando l'appropriato menu item viene 
selezionato dall'utente. 

IDM_TEST equ 1                    //; Le Menu IDs 
IDM_HELLO equ 2 
IDM_GOODBYE equ 3 
IDM_EXIT equ 4



Definiscono le menu IDs per poterle utilizzare nella window procedure. Questi valori DEVONO essere 
identici a quelli definiti nel resource file. 

.ELSEIF eax==WM_COMMAND 
        mov eax,wParam 
        .IF ax==IDM_TEST 
            invoke MessageBox,NULL,ADDR Test_string,OFFSET AppName,MB_OK 
        .ELSEIF ax==IDM_HELLO 
            invoke MessageBox, NULL,ADDR Hello_string, OFFSET AppName,MB_OK 
        .ELSEIF ax==IDM_GOODBYE 
            invoke MessageBox,NULL,ADDR Goodbye_string, OFFSET AppName, MB_OK 
        .ELSE 
            invoke DestroyWindow,hWnd 
        .ENDIF

Nella window procedure, elaboriamo i messaggi WM_COMMAND. Quando l'utente seleziona un menu 
item, la menu ID di quel menu item e' inviata alla window procedure nella low word di wParam assieme al
messaggio WM_COMMAND. Pertanto quando salviamo il valore di wParam in eax, compariamo il valore in
ax alle menu IDs che abbiamo definito in precedenza e ci comportiamo di conseguenza. Nei primi tre casi,
quando l'utente sceglie i menu items Test, Say Hello, e Say GoodBye, noi mostreremo semplicemente 
una stringa di testo in una message box. Se l'utente seleziona il menu item Exit, chiamiamo 
DestroyWindow con l'handle della nostra finestra come parametro, il che chiudera' la nostra finestra. 
Come potete vedere, specificare un nome di menu in una window class e' un'operazione abbastanza 
semplice e diretta. Comunque potete anche utilizzare un metodo alternativo per caricare un menu nella 
vostra finestra. Non vi mostrero' l'intero codice qui. Il resource file e' identico con entrambi i metodi. Ci 
sono solo alcuni piccoli cambiamenti nel file sorgente che vi mostrero' qui sotto. 

.data? 
hInstance HINSTANCE ? 
CommandLine LPSTR ? 
hMenu HMENU ?                    //; handle del nostro menu

Definisce una variabile di tipo HMENU per conservare l'handle del nostro menu. 

invoke LoadMenu, hInst, OFFSET MenuName 
        mov    hMenu,eax 
        INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ 
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,hMenu,\ 
           hInst,NULL

Prima di chiamare CreateWindowEx, chiamiamo LoadMenu con l'instance handle e un puntatore al nome 
del nostro menu. LoadMenu restituisce l'handle del nostro menu nel resource file che poi passiamo a 
CreateWindowEx. 

Tutorial 9: I Controlli di Child Window

Introduzione
In questo tutorial, esploreremo i controlli di Child Window, che sono importanti periferiche di input ed 
output dei nostri programmi. 

Tools
Esempio 

Preliminari
Windows fornisce diverse window classes predefinite che possiamo usare istantaneamente nei nostri 
programmi. La maggior parte delle volte le usiamo come componenti di una dialog box, percio' vengono 
chiamate solitamente "controlli di child window". I controlli di child window elaborano i loro propri 
messaggi dal mouse e dalla tastiera e notificano la parent window (finestra a cui appartengono, NdT) 
quando il loro stato e' cambiato. Essi tolgono un gran peso dalle spalle dei programmatori, percio' 
dovreste usarli quanto piu' possibile. In questo tutorial, li inseriro' in una finestra normale semplicemente 
per dimostrarvi come potete crearli ed usarli, ma solitamente dovreste utilizzarli in una dialog box. 
Esempi di window classes predefinite sono il button, la listbox, la checkbox, il radio button, l'edit ecc. Per 
poter usare un controllo di child window, dovete crearlo con CreateWindow o CreateWindowEx. Notate 
che non dovrete registrare la window class, poiche' essa e' registrata per voi da Windows. Il parametro 
class name DEVE essere il class name predefinito. Ad esempio, se volete creare un button, dovete 
specificare "button" come class name in CreateWindowEx. Gli altri parametri che dovete riempire sono 
l'handle della parent window e la control ID. La control ID deve essere unica tra i controlli. La control ID 
e' l'ID di quel controllo. La usate per distinguere i controlli. Una volta creato il controllo, questo mandera' 
dei messaggi per notificare alla parent window quando il suo stato e' cambiato. Normalmente, potete 
creare le child windows nella sezione relativa al messaggio WM_CREATE della parent window. La child 
window manda messaggi WM_COMMAND alla parent window con la sua control ID nella low word di 
wParam, il codice di notifica nella high word di wParam, e il suo window handle in lParam. Ogni controllo 
di child window ha un diverso codice di notifica (notification codes) rispetto agli altri, riferitevi alla vostra 
Win32 API reference per maggiori informazioni. Anche la parent window puo' mandare comandi alle child 
windows, chiamando la funzione SendMessage. La funzione SendMessage manda il messaggio specificato 
con i relativi valori in wParam e lParam alla finestra specificata dal window handle. E' una funzione 
estremamente utile poiche' puo' mandare messaggi a qualsiasi finestra, posto che voi conosciate il suo 
window handle. Percio', una volta create le child windows, la parent window deve elaborare i messaggi 
WM_COMMAND per essere in grado di ricevere codici di notifica dalle child windows. 

Creeremo una finestra che contiene un edit control ed un pushbutton. Quando clickate sul button, una 
message box apparira', mostrandovi il testo che avevate scritto nell'edit box. C'e' anche un menu con 4 
menu items: 

1. Say Hello -- Scrive una stringa di testo nell'edit box 
2. Clear Edit Box -- Cancella il contenuto dell'edit box 
3. Get Text -- Mostra una message box con il testo nell'edit box 
4. Exit -- Chiude il programma. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.data

ClassName db "SimpleWinClass",0
AppName  db "Our First Window",0
MenuName db "FirstMenu",0
ButtonClassName db "button",0
ButtonText db "My First Button",0
EditClassName db "edit",0
TestString db "Wow! I'm in an edit box now",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hwndButton HWND ?
hwndEdit HWND ?
buffer db 512 dup(?),0                    ; buffer per salvare il testo catturato dall'edit 
box

.const
ButtonID equ 1                                ; La control ID del button control
EditID equ 2                                     ; La control ID dell'edit control
IDM_HELLO equ 1
IDM_CLEAR equ 2
IDM_GETTEXT equ 3
IDM_EXIT equ 4

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_BTNFACE+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName, \
                        ADDR AppName, WS_OVERLAPPEDWINDOW,\
                        CW_USEDEFAULT, CW_USEDEFAULT,\
                        300,200,NULL,NULL, hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    .WHILE TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg ; Traduce l'input "grezzo" da tastiera in input di 
un carattere
        invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_DESTROY
        invoke PostQuitMessage,NULL



    .ELSEIF eax==WM_CREATE
        invoke CreateWindowEx,WS_EX_CLIENTEDGE, ADDR EditClassName,NULL,\
                        WS_CHILD or WS_VISIBLE or WS_BORDER or ES_LEFT or\
                        ES_AUTOHSCROLL,\
                        50,35,200,25,hWnd,8,hInstance,NULL
        mov  hwndEdit,eax
        invoke SetFocus, hwndEdit
        invoke CreateWindowEx,NULL, ADDR ButtonClassName,ADDR ButtonText,\
                        WS_CHILD or WS_VISIBLE or ES_DEFPUSHBUTTON,\
                        75,70,140,25,hWnd,ButtonID,hInstance,NULL
        mov  hwndButton,eax
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .IF lParam==0
            .IF ax==IDM_HELLO
                invoke SetWindowText,hwndEdit,ADDR TestString
            .ELSEIF ax==IDM_CLEAR
                invoke SetWindowText,hwndEdit,NULL
            .ELSEIF  ax==IDM_GETTEXT
                invoke GetWindowText,hwndEdit,ADDR buffer,512
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK
            .ELSE
                invoke DestroyWindow,hWnd
            .ENDIF
        .ELSE
            .IF ax==ButtonID
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_GETTEXT,0
                .ENDIF
            .ENDIF
        .ENDIF
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .ENDIF
     xor    eax,eax
    ret
WndProc endp
end start

Analisi
Analizziamo il programma. 

.ELSEIF eax==WM_CREATE
        invoke CreateWindowEx,WS_EX_CLIENTEDGE, \
                        ADDR EditClassName,NULL,\
                        WS_CHILD or WS_VISIBLE or WS_BORDER or ES_LEFT\
                        or ES_AUTOHSCROLL,\
                        50,35,200,25,hWnd,EditID,hInstance,NULL
        mov  hwndEdit,eax
        invoke SetFocus, hwndEdit
        invoke CreateWindowEx,NULL, ADDR ButtonClassName,\
                        ADDR ButtonText,\
                        WS_CHILD or WS_VISIBLE or ES_DEFPUSHBUTTON,\
                        75,70,140,25,hWnd,ButtonID,hInstance,NULL
        mov  hwndButton,eax

Creiamo i controlli durante l'elaborazione del messaggio WM_CREATE. Chiamiamo CreateWindowEx con 
un window style extra, WS_EX_CLIENTEDGE, che fa sembrare la client area infossata. Il nome di ogni 
controllo e' un nome predefinito, "edit" per l'edit control, "button" per il button control. Successivamente 
specifichiamo gli stili della child window. Ciascun controllo ha stili extra in aggiunta ai normali stili della 
finestra. Ad esempio, gli stili relativi ai buttons hanno come prefisso "BS_", che significa "button style", 
mentre gli stili relativi all'edit hanno come prefisso "ES_" che significa "edit style". Dovete cercare questi 
stili nella Win32 API reference. Notate che potete mettere una control ID al posto del menu handle. 
Questo non causa nessun problema poiche' un controllo di una child window non puo' avere un menu. 



Dopo aver i creato controlli, conserviamo gli handle di ciascuno di essi in delle variabili per un uso futuro. 
SetFocus viene chiamata per dare l'input focus all'edit box cosicche' l'utente puo' scrivere il testo dentro 
di esso immediatamente. Adesso arriva la parte davvero interessante. Ogni controllo di child window 
notifica gli eventi alla propria parent window con WM_COMMAND. 

.ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .IF lParam==0

Tenete presente che anche i menu mandano messaggi WM_COMMAND per notificare la finestra circa il 
loro stato. Come si puo' distinguere un messaggo WM_COMMAND originato da un menu e uno originato 
da un controllo? Qui sotto c'e' la risposta. 

Low word di wParam High word di wParam lParam 

Menu Menu ID 0 0 

Control Control ID Codice di notifica Handle della Child Window 

Potete osservare che e' necessario controllare lParam. Se e' zero, il messaggo WM_COMMAND elaborato 
proviene da un menu. Potete utilizzare wParam per distinguere tra un menu ed un controllo dal momento
che la menu ID e la control ID potrebbero essere uguali ed il codice di notifica potrebbe essere zero. 

.IF ax==IDM_HELLO
                invoke SetWindowText,hwndEdit,ADDR TestString
            .ELSEIF ax==IDM_CLEAR
                invoke SetWindowText,hwndEdit,NULL
            .ELSEIF  ax==IDM_GETTEXT
                invoke GetWindowText,hwndEdit,ADDR buffer,512
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK

Potete scrivere una stringa di testo in un edit box chiamando SetWindowText. Potete cancellare il 
contenuto di un edit box chiamando SetWindowText con NULL. SetWindowText e' una funzione API utile 
per piu' scopi. Potete usare SetWindowText anche per cambiare il titolo di una finestra o la scritta su un 
bottone. 

Per prendere il testo da un edit box, userete GetWindowText. 

.IF ax==ButtonID
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_GETTEXT,0
                .ENDIF
            .ENDIF

Queste istruzioni hanno a che fare con la condizione in cui ci si trova quando l'utente preme il bottone. 
Innanzitutto, la routine controlla la low word di wParam per vedere se la control ID coincide con quella 
del button. Se si, controlla la high word di wParam per vedere se c'e' il codice di notifica BN_CLICKED che
viene inviato quando un button e' stato premuto. La parte interessante viene dopo che ci si e' assicurati 
che il codice di notifica e' BN_CLICKED. Vogliamo prendere il testo scritto nell'edit box e mostrarlo in una 
message box. Potremmo duplicare il codice nella sezione IDM_GETTEXT di cui sopra ma questo sarebbe 
piuttosto scomodo. Se in qualche modo potessimo mandare un messaggio WM_COMMAND con la low 
word di wParam contenente il valore IDM_GETTEXT alla nostra window procedure, potremmo evitare la 
duplicazione del codice e semplificheremmo il nostro programma. La funzione SendMessage e' la 
soluzione. Questa funzione manda qualsiasi messaggio a qualsiasi finestra con qualsiasi wParam e lParam
a nostra scelta. Percio', invece di duplicare le istruzioni, chiamiamo SendMessage con l'handle della 
parent window, WM_COMMAND, IDM_GETTEXT, e 0. Questo procedimento ottiene lo stesso effetto di 
quando si seleziona il menu item "Get Text" dal menu. La window procedure non rileva alcuna differenza 
tra i due. Dovreste usare questa tecnica quanto piu' spesso possibile per rendere il vostro codice meglio 
strutturato. Infine, non dimenticatevi della funzione TranslateMessage nel message loop. Visto che dovete
digitare del testo nell'edit box, il vostro programma deve tradurre l'input "grezzo" (raw) dalla tastiera in 
testo leggibile. Se omettete questa funzione, non potrete scrivere niente nel vostro edit box. 

Tutorial 10: Il Dialog Box come Finestra Principale



Contents
[hide]

 1 Tutorial 10: Il Dialog Box come Finestra Principale 
o 1.1 Introduzione

o 1.2 Tools

o 1.3 Preliminari

o 1.4 Essay 1

o 1.5 Analisi 1

o 1.6 Essay 2

o 1.7 Analisi 2

o 1.8 Disclaimer

Introduzione
Adesso arriva la parte davvero interessante riguardo alla GUI, il dialog box. In questo tutorial (e nel 
prossimo), impareremo come usare un dialog box come finestra principale. 

Tools
Esempio 1 
Esempio 2 

Preliminari
Se analizzate ed usate piu' volte gli esempi del tutorial precedente, dopo un po' vi accorgerete che non 
potete passare l'input focus da un controllo di child window ad un altro con il tasto Tab. L'unico modo di 
farlo e' clickare sul controllo in cui volete l'input focus. La situazione e' piuttosto fastidiosa. Un'altra cosa 
che potreste notare e' che ho cambiato il colore dello sfondo della parent window da bianco normale, 
come nell'esempio precedente, in grigio. Con questo cambiamento, il colore dei controlli di child window 
possono mescolarsi in modo perfetto con il colore della client area della parent window. C'e' un modo per 
aggirare il problema ma non e' semplice. Dovete subclassare tutti i controlli di child window nella vostra 
parent window. Il motivo per cui esiste questo problema e' che i controlli di child window sono stati 
originariamente ideati per funzionare con un dialog box, non una finestra normale. Il colore predefinito 
dei controlli di child window (tipo il push button) e' grigio perche' la client area di un dialog box e' 
normalmente grigia, e pertanto i colori si mescolano senza nessun lavoro aggiuntivo da parte del 
programmatore. Prima di inoltrarci nei dettagli, dovremmo capire COS'E' un dialog box. Un dialog box e' 
semplicemente una finestra normale progettata per funzionare con i controlli di child window. Windows 
fornisce inoltre un "gestore di dialog box" interno che e' responsabile per la maggior parte delle 
operazioni da tastiera, come cambiare l'input focus se Tab viene premuto, premere un push button 
predefinito se viene premuto il tasto Enter, ecc. cosicche' i programmatori possono dedicarsi a compiti 
piu' importanti. I dialog boxes sono fondamentalmente usati come periferiche di input e output. Come 
tali, possono essere considerati "scatole nere" di input o di output, a significare che un programmatore 
non deve sapere come un dialog box funziona internamente per poterlo utilizzare, deve semplicemente 
sapere come interagire con esso. Questo e' uno dei principi dell'object oriented programming (OOP, 
programmazione orientata agli oggetti, NdT), chiamato information hiding (occultamento delle 
informazioni). Se la scatola nera e' *perfettamente* progettata, l'utente puo' farne uso senza avere la 

minima idea di come essa operi. Il problema e' che la scatola nera deve essere perfetta, e questa e' una 
cosa difficile da ottenere nel mondo reale. Anche la Win32 API e' progettata come una scatola nera. Beh, 
sembra che stiamo deviando dal nostro discorso. Ritorniamo al nostro argomento. I dialog boxes sono 
progettati per ridurre il carico di lavoro di un programmatore. Normalmente se mettete i controlli di child 
window in una finestra normale, dovete subclassarli e scrivere le risposte alle operazioni da tastiera da 
soli. Ma se li mettete in un dialog box, questo gestira' le operazioni da tastiera per voi. Dovrete in tal caso
solo sapere come rilevare l'input dal dialog box o come mandarvi comandi. Un dialog box e' definito come
una risorsa, proprio come un menu. Voi scrivete un "template" del dialog box descrivendo le 
caratteristiche dello stesso e i suoi controlli, e quindi compilate il resource script con un editor di risorse. 
Notate che tutte le risorse vanno messe assieme nello stesso resource script file. Potete usare qualsiasi 
editor di testo per scrivere il template di un dialog box ma ve lo sconsiglio. Fareste meglio ad usare un 
editor di risorse e fare il lavoro in modo "visivo", poiche' organizzare i controlli di child window in un 
dialog box e' difficile da fare manualmente. Esistono molti eccellenti editors di risorse. Molti dei piu' 
importanti compilatori includono il proprio editor di risorse nel package. Potete usarli per creare uno 
script di risorse per il vostro programma, e quindi rimuovere le righe inutili, come quelle inerenti il MFC. 
Ci sono due tipi principali di dialog box: modal e modeless. Un modeless dialog box vi permette di 
cambiare l'input focus alle altre finestre. Un esempio e' il dialog "Find" di MS Word. Esistono due sottotipi 
di modal dialog box: application modal e system modal. Un application modal dialog box non vi permette 
di spostare l'input focus su di un'altra finestra della stessa applicazione, ma vi permette di spostarlo su 
finestre di ALTRE applicazioni. Un system modal dialog box non vi permette di spostare l'input focus su 
nessun'altra finestra se non gli rispondete, prima. Un modeless dialog box viene creato tramite la 
funzione API CreateDialogParam. Un modal dialog box viene creato chiamando DialogBoxParam. L'unica 
distinzione tra un application modal dialog box e un system modal dialog box e' lo stile DS_SYSMODAL. 
Se includete lo stile DS_SYSMODAL nel template del dialog box, quel dialog box sara' system modal. 
Potete rilevare dati da un controllo di child window control o da un dialog box chiamando la funzione 
SendDlgItemMessage. La sua sintassi e' questa: 

LONG SendDlgItemMessage(

        HWND  hwndDlg, // handle del dialog box
        int  idControl, // id del controllo
        UINT  uMsg, // messaggio da mandare
        WPARAM  wParam, // primo parametro del messaggio
        LPARAM  lParam  // secondo parametro del messaggio
       );

Questa funzione API e' l'unica che dovete conoscere per essere in grado di rilevare e inviare messaggi e 
dati A e DA un controllo di child window. Ad esempio, se volete rilevare il testo da un edit control, potete 
farlo cosi': 

call SendDlgItemMessage, hDlg, ID_EDITBOX, WM_GETTEXT, 256, ADDR buffer_per_testo

Per sapere quale messaggio inviare, dovreste consultare la vostra Win32 API reference. Comunque, 
Windows fornisce anche molte funzioni API controllo-specifiche per prendere e mandare dati 
velocemente, ad esempio CheckRadioButton, CheckDlgButton ecc. Queste funzioni controllo-specifiche 
sono state create per la comodita' dei programmatori, di modo che essi non si trovino a dover cercare il 
significato di wParam ed lParam per ogni messaggio. Normalmente, dovreste usare funzioni API controllo-
specifiche quando disponibili, perche' rendono la gestione del codice piu' semplice. Ricorrete a 
SendDlgItemMessage solo se non disponete di una funzione API controllo-specifica. Un dialog box manda 
i suoi messaggi ad una funzione di callback specializzata, chiamata dialog box procedure che ha il 
seguente formato: 

BOOL DlgProc  
          
            (HWND hDlg,         // handle della finestra del dialog
            UINT iMsg,          // Messaggio
            WPARAM wParam ,
            LPARAM lParam);

Come potete vedere, una dialog box procedure e' molto simile ad una window procedure, eccetto per il 
tipo di valore di ritorno, che e' BOOL anziche' LRESULT. Il gestore di dialog boxes interno di Windows E' 
la vera window procedure per i dialog boxes. Esso chiama la nostra dialog box procedure con i molti 
messaggi che riceve. Percio' la regola generale e' questa: se la dialog box procedure elabora un 
messaggio, DEVE restituire il valore TRUE in eax e se non lo elabora, deve restituire FALSE in eax. 
Notate che una dialog box procedure non passa i messaggi che non elabora alla call a DefWindowProc, 



poiche' non e' una vera window procedure. Potete fare due usi distinti di un dialog box. Potete usarlo 
come finestra principale della vostra applicazione o usarlo come periferica di input. Esamineremo il primo 
approccio in questo tutorial. La frase "usare un dialog box come finestra principale" puo' essere 
interpretata in due modi diversi. 

1. Potete usare il template del dialog box come un template di class che registrate con la funzione 
RegisterClassEx. In questo caso, il dialog box si comporta come una finestra "normale": riceve 
messaggi tramite la window procedure riferita dal membro lpfnWndProc della window class, non
tramite una dialog box procedure. Il vantaggio di questo approccio sta nel fatto che non dovete 
creare i controlli di child window da soli, poiche' Windows li crea per voi quando viene creato il 
dialog box. Inoltre Windows gestisce la keyboard logic (Tab, Enter ecc.) per voi. In piu' potete 
specificare cursore ed icona della vostra finestra nella struttura della window class. 

2. Il vostro programma crea semplicemente il dialog box senza creare nessuna parent window. 
Questo approccio rende un message loop inutile poiche' i messaggi sono inviati direttamente 
alla dialog box procedure. Non dovete nemmeno preoccuparvi di registrare una window class! 
Solitamente la prima linea nel codice e' la call che crea il dialog box che verra' usato come 
finestra principale. Questo approccio e' generalmente il migliore ed il piu' semplice per creare 
una finestra principale con molti controlli di child window. Poiche' il gestore interno di dialog box
e' funzionante, non dovrete preoccuparvi della keyboard logic. 

Questo tutorial sara' lungo. Presentero' qui il primo approccio seguito dal secondo. 

Essay 1

Dialog.asm

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.data
ClassName db "DLGCLASS",0
MenuName db "MyMenu",0
DlgName db "MyDialog",0
AppName db "Our First Dialog Box",0
TestString db "Wow! I'm in an edit box now",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
buffer db 512 dup(?),0

.const
IDC_EDIT        equ 3000
IDC_BUTTON      equ 3001
IDC_EXIT        equ 3002
IDM_GETTEXT     equ 32000
IDM_CLEAR       equ 32001
IDM_EXIT        equ 32002

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hDlg:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX



    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,DLGWINDOWEXTRA
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_BTNFACE+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateDialogParam,hInstance,ADDR DlgName,NULL,NULL,NULL
    mov   hDlg,eax
    invoke ShowWindow, hDlg,SW_SHOWNORMAL
    invoke UpdateWindow, hDlg
   invoke GetDlgItem,hDlg,IDC_EDIT
   invoke SetFocus,eax
   .WHILE TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke IsDialogMessage, hDlg, ADDR msg
        .IF eax ==FALSE
            invoke TranslateMessage, ADDR msg
            invoke DispatchMessage, ADDR msg
        .ENDIF
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .IF lParam==0
            .IF ax==IDM_GETTEXT
                invoke GetDlgItemText,hWnd,IDC_EDIT,ADDR buffer,512
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK
            .ELSEIF ax==IDM_CLEAR
                invoke SetDlgItemText,hWnd,IDC_EDIT,NULL
            .ELSE
                invoke DestroyWindow,hWnd
            .ENDIF
        .ELSE
            .IF ax==IDC_BUTTON
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SetDlgItemText,hWnd,IDC_EDIT,ADDR TestString
                .ENDIF
            .ELSEIF ax==IDC_EXIT
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0
                .ENDIF
            .ENDIF
        .ENDIF
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
 .ENDIF
        xor    eax,eax
 ret
WndProc endp
end start

Dialog.rc



#include "c:\masm\include\winuser.h"

#define IDC_EDIT                                       3000
#define IDC_BUTTON                                3001
#define IDC_EXIT                                       3002

#define IDM_GETTEXT                             32000
#define IDM_CLEAR                                  32001
#define IDM_EXIT                                      32003
 

MyDialog DIALOG 10, 10, 205, 60
STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK
CAPTION "Our First Dialog Box"
CLASS "DLGCLASS"
BEGIN
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13, WS_GROUP
END
 

MyMenu  MENU
BEGIN
    POPUP "Test Controls"
    BEGIN
        MENUITEM "Get Text", IDM_GETTEXT
        MENUITEM "Clear Text", IDM_CLEAR
        MENUITEM "", , 0x0800 /*MFT_SEPARATOR*/
        MENUITEM "E&xit", IDM_EXIT
    END
END

Analisi 1
Analizziamo questo primo esempio. 

Questo esempio mostra come registrare il template di un dialog come window class e creare una 
"finestra" da quella class. Il risultato e' molto simile alla finestra nel tutorial n.9 e cioe': non potete 
passare l'input focus ad altri controlli con il tasto tab. Ma semplifica il vostro programma poiche' non 
dovete creare i controlli di child window da soli. Analizziamo prima il template del dialog. 

MyDialog DIALOG 10, 10, 205, 60

Dichiara il nome di un dialog, in questo caso "MyDialog", seguito dalla keyword "DIALOG". I quattro 
numeri successivi sono: x, y , larghezza, e altezza del dialog box in unita' di misura del dialog box (non 
uguali ai pixels). 

STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK

Dichiara gli stili del dialog box. 

CAPTION "Our First Dialog Box"

Questo e' il testo che apparira' nella barra del titolo del dialog box. 

CLASS "DLGCLASS"

Questa linea e' importantissima. E' questa keyword CLASS che ci permette di usare il template del dialog
box come una window class. Dopo la keyword c'e' il nome della "window class". 



BEGIN
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13
END

La parte sovrastante definisce i controlli di child window nel dialog box. Essi sono definiti tra le keywords 
BEGIN e END. Generalmente la sintassi e' questa: 

tipi-di-controlli "testo" ,controlID, x, y, larghezza,altezza,[,stili]

tipi-di-controlli sono costanti del compilatore di risorse pertanto dovete consultare il manuale. Adesso 
analizziamo il codice assembly. La parte interessante sta nella struttura della window class: 

mov   wc.cbWndExtra,DLGWINDOWEXTRA
      mov   wc.lpszClassName,OFFSET ClassName

Normalmente, questo membro e' settato a NULL, ma se vogliamo registrare il template di un dialog box 
come window class, dobbiamo settare questo membro con il valore DLGWINDOWEXTRA. Notate che il 
nome della class deve essere identico a quello che segue la keyword CLASS nel template del dialog box. 
I membri rimanenti sono inizializzati nel solito modo. Una volta riempita la struttura della window class, 
registratela con RegisterClassEx. Vi sembra familiare ? Questa e' la stessa routine che dovete seguire per 
registrare una normale window class. 

invoke CreateDialogParam,hInstance,ADDR DlgName ,NULL,NULL,NULL

Una volta registrata la "window class", creeremo il nostro dialog box. In questo esempio, io creo un 
modeless dialog box con la funzione CreateDialogParam. Questa funzione supporta 5 parametri ma voi 
dovrete riempire solo i primi due: l'instance handle e il puntatore al nome del template del dialog box. 
Notate che il secondo parametro non e' un puntatore al nome della class. A questo punto, Windows crea 
il dialog box e i suoi controlli di child window. La vostra window procedure ricevera', come al solito, il 
messaggo WM_CREATE. 

invoke GetDlgItem,hDlg,IDC_EDIT
      invoke SetFocus,eax

Una volta creato il dialog box, volete spostare l'input focus sull'edit control. Se scrivete questa parte di 
codice nella sezione WM_CREATE, la call a GetDlgItem fallira' poiche' in quel momento i controlli di child 
window non sono stati ancora creati. L'unico modo in cui potete farlo e' inserendola dopo che il dialog box
e tutte i suoi controlli di child window sono stati creati. Percio' ho messo queste due righe dopo la 
chiamata ad UpdateWindow. La funzione GetDlgItem ottiene l'ID del controllo e restituisce l'handle della 
finestra associata al controllo. E' cosi' che potete trovare l'handle di una finestra se conoscete una sua 
control ID. 

invoke IsDialogMessage, hDlg, ADDR msg
        .IF eax ==FALSE
            invoke TranslateMessage, ADDR msg
            invoke DispatchMessage, ADDR msg
        .ENDIF

Il programma entra nel message loop e prima di chiamare TranslateMessage e DispatchMessage, 
chiamiamo la funzione IsDialogMessage per far si che il gestore di dialog box gestisca la keyboard logic 
del nostro dialog box al posto nostro. Se questa funzione restituisce TRUE, significa che il messaggio e' 
destinato al dialog box ed e' elaborato dal gestore di dialog box. Notate un'altra differenza dal tutorial 
precedente. Quando la window procedure vuole rilevare del testo dall'edit control, chiama la funzione 
GetDlgItemText invece di GetWindowText. GetDlgItemText accetta l'ID di un controllo al posto dell'handle
di una finestra. Cio' rende la chiamata piu' semplice nel caso usiate un dialog box. 



Adesso passiamo al secondo approccio su come usare un dialog box come finestra principale. Nel 
prossimo esempio, creero' un application modal dialog box. Non troverete un message loop o una window
procedure perche' non sono necessari! 

Essay 2

Dialog.asm

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.data
DlgName db "MyDialog",0
AppName db "Our Second Dialog Box",0
TestString db "Wow! I'm in an edit box now",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
buffer db 512 dup(?),0

.const
IDC_EDIT            equ 3000
IDC_BUTTON     equ 3001
IDC_EXIT            equ 3002
IDM_GETTEXT  equ 32000
IDM_CLEAR       equ 32001
IDM_EXIT           equ 32002
 

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    mov  eax, OFFSET DlgProc
    invoke DialogBoxParam, hInst, ADDR DlgName,NULL, eax,NULL
    ret
WinMain endp

DlgProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_INITDIALOG
        invoke GetDlgItem, hWnd,IDC_EDIT
        invoke SetFocus,eax
    .ELSEIF eax==WM_CLOSE
        invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .IF lParam==0
            .IF ax==IDM_GETTEXT
                invoke GetDlgItemText,hWnd,IDC_EDIT,ADDR buffer,512
                invoke MessageBox,NULL,ADDR buffer,ADDR AppName,MB_OK
            .ELSEIF ax==IDM_CLEAR
                invoke SetDlgItemText,hWnd,IDC_EDIT,NULL
            .ELSEIF ax==IDM_EXIT
                invoke EndDialog, hWnd,NULL
            .ENDIF



        .ELSE
            .IF ax==IDC_BUTTON
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SetDlgItemText,hWnd,IDC_EDIT,ADDR TestString
                .ENDIF
            .ELSEIF ax==IDC_EXIT
                shr eax,16
                .IF ax==BN_CLICKED
                    invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0
                .ENDIF
            .ENDIF
        .ENDIF
    .ELSE
        mov eax,FALSE
        ret
    .ENDIF
    mov eax,TRUE
    ret
DlgProc endp
end start

Dialog.rc

#include "c:\masm\include\winuser.h"

#define IDC_EDIT                                       3000
#define IDC_BUTTON                                3001
#define IDC_EXIT                                       3002

#define IDR_MENU1                                  3003

#define IDM_GETTEXT                              32000
#define IDM_CLEAR                                   32001
#define IDM_EXIT                                       32003
 

MyDialog DIALOG 10, 10, 205, 60
STYLE 0x0004 | DS_CENTER | WS_CAPTION | WS_MINIMIZEBOX |
WS_SYSMENU | WS_VISIBLE | WS_OVERLAPPED | DS_MODALFRAME | DS_3DLOOK
CAPTION "Our Second Dialog Box"
MENU IDR_MENU1
BEGIN
    EDITTEXT         IDC_EDIT,   15,17,111,13, ES_AUTOHSCROLL | ES_LEFT
    DEFPUSHBUTTON   "Say Hello", IDC_BUTTON,    141,10,52,13
    PUSHBUTTON      "E&xit", IDC_EXIT,  141,26,52,13
END
 

IDR_MENU1  MENU
BEGIN
    POPUP "Test Controls"
    BEGIN
        MENUITEM "Get Text", IDM_GETTEXT
        MENUITEM "Clear Text", IDM_CLEAR
        MENUITEM "", , 0x0800 /*MFT_SEPARATOR*/
        MENUITEM "E&xit", IDM_EXIT
    END
END

Analisi 2
Segue l'analisi: 

mov  eax, OFFSET DlgProc
    invoke DialogBoxParam, hInst, ADDR DlgName,NULL, eax,NULL
    ret



Nella funzione WinMain, ci sono solo queste tre linee di codice. La prima linea salva l'offset della dialog 
box procedure in eax. La linea successiva e' la chiamata alla funzione DialogBoxParam che supporta 5 
parametri: l'instance handle, il nome del template del dialog box, l'handle della parent window, l'indirizzo 
della dialog box procedure, e i dati specifici del dialog. DialogBoxParam crea un modal dialog box. Non si 
ritornera' fino a che il dialog box non viene distrutto. 

.IF eax==WM_INITDIALOG
        invoke GetDlgItem, hWnd,IDC_EDIT
        invoke SetFocus,eax
    .ELSEIF eax==WM_CLOSE
        invoke SendMessage,hWnd,WM_COMMAND,IDM_EXIT,0

La dialog box procedure sembra una window procedure tranne per il fatto che non riceve il messaggio 
WM_CREATE. Il primo messaggio che riceve e' WM_INITDIALOG. Normalmente potete inserire qui il 
codice relativo all'inizializzazione. Notate che dovete restituire il valore TRUE in eax se elaborate il 
messaggio. Il gestore interno di dialog box non manda il messaggio WM_DESTROY alla nostra dialog box 
procedure. Percio' se vogliamo reagire quando l'utente preme il tasto di chiusura del nostro dialog box, 
dobbiamo elaborare il messaggio WM_CLOSE. Nel nostro esempio, mandiamo un messaggio 
WM_COMMAND con il valore IDM_EXIT in wParam. Cio' ha lo stesso effetto di quando l'utente sceglie il 
menu item Exit. L'elaborazione dei messaggi WM_COMMAND rimane invariata. Quando volete distruggere
il dialog box, l'unico modo e' chiamare la funzione EndDialog. Non provate con DestroyWindow! Adesso 
esaminiamo il resource file. Il cambiamento degno di nota e' che invece di usare una stringa testuale 
come nome di menu usiamo un valore, IDR_MENU1. Cio' e' necessario se volete incollare un menu ad un 
dialog box creato con DialogBoxParam. Notate che nel template del dialog box, dovete aggiungere la 
keyword MENU seguita dalla resource ID del menu. Una differenza tra i due esempi in questo tutorial che
potete immediatamente osservare e' la mancanza di un'icona nell'ultimo esempio. Comunque, potete 
settare un'icona mandando il messaggio WM_SETICON al dialog box durante WM_INITDIALOG. 

Tutorial 11: Altre Informazioni Riguardanti il Dialog Box

Introduzione
In questo tutorial impareremo altre cose riguardo al dialog box. Specificamente, ci inoltreremo nel modo 
di usare i dialog boxes come periferiche di input-output. Se avete letto il tutorial precedente, questo sara'
una passeggiata, poiche' e' necessaria solo una piccola modifica per poter usare i dialog boxes come 

aggiunte alla nostra finestra principale. Inoltre, in questo tutorial, impareremo come usare i dialog boxes 
comuni. 

Tools
Esempio 

Preliminari
C'e' molto poco da dire su come usare i dialog boxes come periferiche di input-output per un programma.
Il vostro programma crea la finestra principale come al solito, e quando volete mostrare il dialog box, 
semplicemente chiama CreateDialogParam o DialogBoxParam. Con la chiamata a DialogBoxParam, non 
dovete fare piu' niente, ma semplicemente elaborare i messaggi nella dialog box procedure. Con 
CreateDialogParam, dovete inserire la chiamata a IsDialogMessage nel message loop per far si che il 
gestore di dialog box gestisca automaticamente la navigazione da tastiera nel vostro dialog box. Poiche' i 
due casi sono triviali, non inseriro' il codice qui. Potete scaricare gli esempi e dare un'occhiata voi stessi, 
qui e qui (sono i 2 esempi del tutorial precedente NdT). 

Passiamo ai dialog boxes comuni. Windows contiene alcuni dialog boxes predefiniti che potete usare nelle
vostre applicazioni. Questi dialog boxes esistono per fornire una interfaccia utente standardizzata. Essi 
sono i dialog boxes file, print (stampa, NdT), color, font, e search (trova, NdT). Dovreste usarli quanto 
piu' possibile. I dialog boxes sono contenuti in commdlg.dll. Per poterli usare, dovete linkare anche 
commdlg.lib. Creerete questi dialog boxes chiamando funzioni appropriate nella common dialog library. 
Per il dialog "Apri File" la funzione e' GetOpenFileName, per il dialog "salva con nome" e' 
GetSaveFileName, per il dialog "stampa" e' PrintDlg e cosi' via. Ognuna di queste funzioni supporta un 
puntatore ad una struttura come parametro. Dovreste studiarle nella Win32 API reference. In questo 
tutorial, dimostrero' come creare ed usare un dialog "Apri File". Qui di seguito c'e' il prototipo di funzione 
di GetOpenFileName: 

BOOL GetOpenFileName(

      LPOPENFILENAME  lpofn //indirizzo della struttura con i dati di inizializzazione
       );

Potete vedere che necessita di un solo parametro, un puntatore ad una struttura OPENFILENAME. Il 
valore di ritorno TRUE indica che l'utente ha scelto un file da aprire, altrimenti e' FALSE. Esamineremo 
ora la struttura OPENFILENAME. 

OPENFILENAME  STRUCT

         lStructSize DWORD  ?
         hwndOwner HWND  ?
         hInstance HINSTANCE ?
         lpstrFilter LPCSTR  ?
         lpstrCustomFilter LPSTR  ?
         nMaxCustFilter DWORD  ?
         nFilterIndex DWORD  ?
         lpstrFile LPSTR  ?
         nMaxFile DWORD  ?
         lpstrFileTitle LPSTR  ?
         nMaxFileTitle DWORD  ?
         lpstrInitialDir LPCSTR  ?
         lpstrTitle LPCSTR  ?
         Flags  DWORD  ?
         nFileOffset WORD  ?
         nFileExtension WORD  ?
         lpstrDefExt LPCSTR  ?
         lCustData LPARAM  ?
         lpfnHook DWORD  ?

         lpTemplateName LPCSTR  ?

    OPENFILENAME  ENDS

Vediamo qual'e' il significato dei membri usati frequentemente. 

lStructSize La dimensione della struttura OPENFILENAME, in bytes 

hwndOwner Il window handle del dialog box "Apri File" 

hInstance L'instance handle dell'applicazione che crea il dialog box "Apri File" 

lpstrFilter 

Le stringhe-filtro in formato di coppie di stringhe null-terminated. La prima stringa in 
ciascuna coppia e' la descrizione. La seconda stringa e' il pattern per il filtro. Ad esempio:
FilterString   db "Tutti i file (*.*)",0, "*.*",0
               db "File di testo (*.txt)",0,"*.txt",0,0
Notate che solo il pattern nella seconda stringa e' effettivamente usato da Windows per 
filtrare i files. Notate inoltre che dovete mettere uno zero in piu' alla fine delle stringhe-
filtro per denotarne la conclusione. 

nFilterIndex 

Specifca quale coppia di stringhe-filtro sara' utilizzata inzialmente, quando il dialog box 
"Apri File" viene mostrato per la prima volta. L'indice e' in base 1, vale a dire che la 
prima coppia e' 1, la seconda coppia e' 2 e cosi via. Percio', dato l'esempio precedente, 
se dichiariamo nFilterIndex come 2, il secondo pattern, "*.txt", sara' usato. 

lpstrFile 

Puntatore al buffer che contiene il nome del file che viene usato per inizializzare l'edit 
control "Nome File" nel dialog box. Il buffer dovrebbe essere lungo almeno 260 bytes. 
Dopo che l'utente seleziona un file da aprire, il nome di tale file con la path completa 
viene salvato in questo buffer. Potete estrarne le informazioni successivamente. 

nMaxFile La grandezza del buffer lpstrFile. 

lpstrTitle Puntatore al titolo del dialog box "Apri File". 

Flags Determina gli stili e le caratteristiche del dialog box. 

nFileOffset 
Dopo che l'utente ha selezionato un file da aprire, questo membro contiene l'indice al 
primo carattere del nome effettivo del file. Ad esempio, se il nome completo con la path 
intera e' "c:\windows\system\lz32.dll", questo membro conterra' il valore 18. 

nFileExtension Dopo che l'utente ha selezionato un file da aprire, questo membro contiene l'indice al 
primo carattere dell'estensione del file. 

Essay
Il programma che segue mostra un dialog box "Apri File" quando l'utente seleziona File-> Open dal 
menu. Quando l'utente seleziona un file nel dialog box, il programma mostra una message box con il 
nome intero, il nome del file e l'estensione del file selezionato. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib
includelib comdlg32.lib

.const
IDM_OPEN equ 1
IDM_EXIT equ 2
MAXSIZE equ 260
OUTPUTSIZE equ 512

.data
ClassName db "SimpleWinClass",0
AppName  db "Our Main Window",0
MenuName db "FirstMenu",0
ofn   OPENFILENAME <>
FilterString db "All Files",0,"*.*",0
             db "Text Files",0,"*.txt",0,0



buffer db MAXSIZE dup(0)
OurTitle db "-=Our First Open File Dialog Box=-: Choose the file to open",0
FullPathName db "The Full Filename with Path is: ",0
FullName  db "The Filename is: ",0
ExtensionName db "The Extension is: ",0
OutputString db OUTPUTSIZE dup(0)
CrLf db 0Dh,0Ah,0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
           CW_USEDEFAULT,300,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    .WHILE TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .if ax==IDM_OPEN
            mov ofn.lStructSize,SIZEOF ofn
            push hWnd
            pop  ofn.hwndOwner
            push hInstance
            pop  ofn.hInstance
            mov  ofn.lpstrFilter, OFFSET FilterString
            mov  ofn.lpstrFile, OFFSET buffer
            mov  ofn.nMaxFile,MAXSIZE
            mov  ofn.Flags, OFN_FILEMUSTEXIST or \
                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\



                OFN_EXPLORER or OFN_HIDEREADONLY
            mov  ofn.lpstrTitle, OFFSET OurTitle
            invoke GetOpenFileName, ADDR ofn
            .if eax==TRUE
                invoke lstrcat,offset OutputString,OFFSET FullPathName
                invoke lstrcat,offset OutputString,ofn.lpstrFile
                invoke lstrcat,offset OutputString,offset CrLf
                invoke lstrcat,offset OutputString,offset FullName
                mov  eax,ofn.lpstrFile
                push ebx
                xor  ebx,ebx
                mov  bx,ofn.nFileOffset
                add  eax,ebx
                pop  ebx
                invoke lstrcat,offset OutputString,eax
                invoke lstrcat,offset OutputString,offset CrLf
                invoke lstrcat,offset OutputString,offset ExtensionName
                mov  eax,ofn.lpstrFile
                push ebx
                xor ebx,ebx
                mov  bx,ofn.nFileExtension
                add eax,ebx
                pop ebx
                invoke lstrcat,offset OutputString,eax
                invoke MessageBox,hWnd,OFFSET OutputString,ADDR AppName,MB_OK
                invoke FillMemory,offset OutputString,OUTPUTSIZE,0
            .endif
        .else
            invoke DestroyWindow, hWnd
        .endif
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .ENDIF
    xor    eax,eax
    ret
WndProc endp
 end start

Analisi
mov ofn.lStructSize,SIZEOF ofn
            push hWnd
            pop  ofn.hwndOwner
            push hInstance
            pop  ofn.hInstance

Riempiamo i membri necessari delle strutture ofn. 

mov  ofn.lpstrFilter, OFFSET FilterString

Questa Stringa-Filtro e' il filtro per il nome del file che dichiariamo come segue: 

FilterString db "All Files",0,"*.*",0
                 db "Text Files",0,"*.txt",0,0

Notate che TUTTE E QUATTRO le stringhe sono zero-terminated. La prima stringa e' la descrizione della 
stringa successiva. I patterns reali possono anche essere stringhe numeriche, in questo caso sono "*.*" e
"*.txt". In realta' possiamo specificare qualsiasi pattern qui. DOBBIAMO mettere uno zero in piu' alla 
fine dell'ultima stringa-pattern per denotare la fine della stringa-filtro. Non dimenticatevi questo dettaglio 
o il vostro dialog box si comportera' in modo strano. 

mov  ofn.lpstrFile, OFFSET buffer
mov  ofn.nMaxFile,MAXSIZE



Qui specifichiamo il buffer in cui il dialog box salvera' il nome del file che l'utente seleziona. Notate che 
dovete specificarne la grandezza nel membro nMaxFile. Possiamo poi estrarre il nome del file da questo 
buffer. 

mov  ofn.Flags, OFN_FILEMUSTEXIST or \
                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\
                OFN_EXPLORER or OFN_HIDEREADONLY

Flags specifica le caratteristiche del dialog box. Le flags OFN_FILEMUSTEXIST e OFN_PATHMUSTEXIST 
impongono che il nome del file la path che l'utente digita nell'edit control "Nome File" DEVONO esistere. 
La flag OFN_LONGNAMES indica al dialog box di mostrare i nomi dei file lunghi (in stile Windows per 
intenderci, NdT). La flag OFN_EXPLORER specifica che l'apparenza del dialog box deve assomigliare a 
quella di Explorer (=Gestione Risorse, o Esplora Risorse in italiano, NdT). La flag OFN_HIDEREADONLY 
nasconde la checkbox sola-lettura dal dialog box. Ci sono molte altre flags utilizzabili. Consultate la 
vostra Win32 API reference. 

mov  ofn.lpstrTitle, OFFSET OurTitle

Specifica il titolo del dialog box. 

invoke GetOpenFileName, ADDR ofn

Chiama la funzione GetOpenFileName, passando il puntatore alla struttura ofn come parametro. A questo 
punto, il dialog box "Apri File" viene mostrato sullo schermo. La funzione non ritornera' a meno che 
l'utente non selezioni un file da aprire, prema il tasto cancel o chiuda il dialog box. Essa restituira' il 
valore TRUE in eax se l'utente selziona un file da aprire. Altrimenti, restituira' FALSE. 

.if eax==TRUE
                invoke lstrcat,offset OutputString,OFFSET FullPathName
                invoke lstrcat,offset OutputString,ofn.lpstrFile
                invoke lstrcat,offset OutputString,offset CrLf
                invoke lstrcat,offset OutputString,offset FullName

Se l'utente ha selezionato un file da aprire, prepariamo un stringa che verra' mostrata in una message 
box. Allochiamo un blocco di memoria nella variabile OutputString e quindi usiamo una funzione API, 
lstrcat, per concatenare le stringhe insieme. Per mettere le stringhe su piu' linee, dobbiamo separare ogni
linea con una coppia carriage return-line feed. 

mov  eax,ofn.lpstrFile
 push ebx
 xor  ebx,ebx
 mov  bx,ofn.nFileOffset
 add  eax,ebx
 pop  ebx
 invoke lstrcat,offset OutputString,eax

Queste righe richiedono qualche spiegazione. nFileOffset contiene l'indice ad ofn.lpstrFile. Ma non potete 
addizionarli direttamente perche' nFileOffset e' una variabile di dimensioni WORD e lpstrFile e' invece una
DWORD. Percio' bisogna mettere il valore di nFileOffset nella low word di ebx e aggiungerlo al valore di 
lpstrFile. 

invoke MessageBox,hWnd,OFFSET OutputString,ADDR AppName,MB_OK

Mostriamo la stringa in una message box. 

invoke FillMemory,offset OutputString,OUTPUTSIZE,0

Dobbiamo *svuotare* la OutputString prima di potervi inserire un'altra stringa. Percio' usiamo la funzione
FillMemory per eseguire il lavoro. 

Introduzione
In questo tutorial impareremo le basi della gestione della memoria e delle operazioni di I/O con i files. In 
piu', utilizzeremo dei dialog boxes comuni come periferiche di input-output. 

Preliminari
La gestione della memoria sotto Win32 dal punto di vista delle applicazioni e' abbastanza semplice e 
diretta. Ogni processo possiede un memory address space privato di 4 GB. Il modello di memoria 
utilizzato e' chiamato flat memory model. In questo modello, tutti i segment registers (o selettori) 
puntano allo stesso indirizzo iniziale e l'offset e' a 32-bits cosicche' un'applicazione puo' accedere alla 
memoria in qualsiasi punto del proprio address space senza aver bisogno di cambiare il valore dei 
selettori. Cio' semplifica la gestione della memoria in modo eccezionale. Non ci sono piu' "near" o "far" 
pointers. Sotto Win16, ci sono due principali categorie di funzioni API: Global e Local. Le funzioni API di 
tipo Global hanno a che fare con memoria allocata in altri segmenti, pertanto sono funzioni di memoria di
tipo "far". Le funzioni API di tipo Local, invece, hanno a che fare con con la memoria locale del processo, 
e quindi sono funzioni di tipo "near". Sotto Win32, questi due tipi sono identici. Che chiamate GlobalAlloc 
o LocalAlloc, otterrete sempre lo stesso risultato. I passi per l'allocazione e l'utilizzo della memoria sono i 
seguenti: 

1. Allocate un blocco di memoria chiamando GlobalAlloc. Questa funzione restituisce l'handle del 
blocco di memoria richiesto. 

2. "Bloccate" la memoria chiamando GlobalLock. Questa funziona accetta l'handle del blocco di 
memoria e restituisce un puntatore al blocco di memoria. 

3. Potete usare il puntatore per leggere o scrivere la memoria. 
4. "Sbloccate" il blocco di memoria chiamando GlobalUnlock . Questa funzione invalida il puntatore

al blocco di memoria. 
5. Liberate il blocco di memoria chiamando GlobalFree. Questa funzione accetta l'handle del blocco 

di memoria. 

Potete anche sostituire "Local" a "Global", ad esempio LocalAlloc, LocalLock, ecc. 

L' I/O con i files sotto Win32 e' molto simile a quello sotto DOS. I passaggi necessari sono uguali. Dovete 
solo sostituire gli interrupts con chiamate API e il gioco e' fatto. I passi necessari sono i segueti: 

1. Aprite o create un file chiamando la funzione CreateFile. Questa funzione e' molto versatile: 
oltre ai files, puo' aprire porte di comunicazione, pipes, drives o console. In caso di successo, 
essa restituisce l'handle del file o della periferica. Potete quindi usare questo handle per 
eseguire operazioni sul file o sulla periferica. Muovete il file pointer alla locazione desiderata 
chiamando SetFilePointer. 

2. Eseguite l'operazione di lettura o scrittura chiamando ReadFile o WriteFile. Queste funzioni 
trasferiscono dati da un blocco di memoria a un file o viceversa. Percio' dovete allocare un 
blocco di memoria largo abbastanza da contenere i dati. 

3. Chiudete il file chiamando FreeHandle. Questa funzione accetta l'handle del file. 

Essay
Il programma che segue mostra un dialog box "apri file". Quindi, lascia scegliere all'utente un file di testo
da aprire e mostra il contenuto di quel file in un edit control della sua client area. L'utente puo' modificare
il testo nell'edit control a suo piacimento, e puo' scegliere di salvare il contenuto in un file. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib
includelib comdlg32.lib

.const
IDM_OPEN equ 1
IDM_SAVE equ 2
IDM_EXIT equ 3
MAXSIZE equ 260
MEMSIZE equ 65535



EditID equ 1                            ; ID dell'edit control

.data
ClassName db "Win32ASMEditClass",0
AppName  db "Win32 ASM Edit",0
EditClass db "edit",0
MenuName db "FirstMenu",0
ofn   OPENFILENAME <>
FilterString db "All Files",0,"*.*",0
             db "Text Files",0,"*.txt",0,0
buffer db MAXSIZE dup(0)

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hwndEdit HWND ?                                ; Handle dell'edit control
hFile HANDLE ?                                   ; handle del File
hMemory HANDLE ?                           ; handle del blocco di memoria allocato
pMemory DWORD ?                            ; puntatore al blocco di memoria allocato
SizeReadWrite DWORD ?                   ; numero di bytes effettivamente letti o scritti

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
           CW_USEDEFAULT,300,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    .WHILE TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc uses ebx hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    LOCAL rec:RECT
    mov   eax,uMsg
    .IF eax==WM_CREATE
        invoke CreateWindowEx,NULL,ADDR EditClass,NULL,\
                   WS_VISIBLE or WS_CHILD or ES_LEFT or ES_MULTILINE or\
                   ES_AUTOHSCROLL or ES_AUTOVSCROLL,0,\



                   0,0,0,hWnd,EditID,\
                   hInstance,NULL
        mov hwndEdit,eax
        invoke SetFocus,hwndEdit
;===============================================================================
;        Inizializza i membri della struttura OPENFILENAME
;===============================================================================
        mov ofn.lStructSize,SIZEOF ofn
        push hWnd
        pop  ofn.hwndOwner
        push hInstance
        pop  ofn.hInstance
        mov  ofn.lpstrFilter, OFFSET FilterString
        mov  ofn.lpstrFile, OFFSET buffer
        mov  ofn.nMaxFile,MAXSIZE
    .ELSEIF eax==WM_SIZE
        invoke GetClientRect,hWnd,ADDR rec
        invoke MoveWindow,hwndEdit,0,0,rec.right,rec.bottom,TRUE
    .ELSEIF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .if lParam==0
            .if ax==IDM_OPEN
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY
                invoke GetOpenFileName, ADDR ofn
                .if eax==TRUE
                    invoke CreateFile,ADDR buffer,\
                                GENERIC_READ or GENERIC_WRITE ,\
                                FILE_SHARE_READ or FILE_SHARE_WRITE,\
                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\
                                NULL
                    mov hFile,eax
                    invoke GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE
                    mov  hMemory,eax
                    invoke GlobalLock,hMemory
                    mov  pMemory,eax
                    invoke ReadFile,hFile,pMemory,MEMSIZE-1,ADDR SizeReadWrite,NULL
                    invoke SendMessage,hwndEdit,WM_SETTEXT,NULL,pMemory
                    invoke CloseHandle,hFile
                    invoke GlobalUnlock,pMemory
                    invoke GlobalFree,hMemory
                .endif
                invoke SetFocus,hwndEdit
            .elseif eax==IDM_SAVE
                mov ofn.Flags,OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY
                invoke GetSaveFileName, ADDR ofn
                    .if eax==TRUE
                        invoke CreateFile,ADDR buffer,\
                                                GENERIC_READ or GENERIC_WRITE ,\
                                                FILE_SHARE_READ or FILE_SHARE_WRITE,\
                                                NULL,CREATE_NEW,FILE_ATTRIBUTE_ARCHIVE,\
                                                NULL
                        mov hFile,eax
                        invoke GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE
                        mov  hMemory,eax
                        invoke GlobalLock,hMemory
                        mov  pMemory,eax
                        invoke SendMessage,hwndEdit,WM_GETTEXT,MEMSIZE-1,pMemory
                        invoke lstrlen,pMemory
                        invoke WriteFile,hFile,pMemory,eax,ADDR SizeReadWrite,NULL
                        invoke CloseHandle,hFile
                        invoke GlobalUnlock,pMemory
                        invoke GlobalFree,hMemory
                    .endif
                    invoke SetFocus,hwndEdit
                .else
                    invoke DestroyWindow, hWnd
                .endif
            .endif
        .ELSE



            invoke DefWindowProc,hWnd,uMsg,wParam,lParam
            ret
.ENDIF
xor    eax,eax
ret
WndProc endp
end start

Analisi
invoke CreateWindowEx,NULL,ADDR EditClass,NULL,\
                   WS_VISIBLE or WS_CHILD or ES_LEFT or ES_MULTILINE or\
                   ES_AUTOHSCROLL or ES_AUTOVSCROLL,0,\
                   0,0,0,hWnd,EditID,\
                   hInstance,NULL
        mov hwndEdit,eax

Nella sezione WM_CREATE, creiamo un edit control. Notate che i parametri che specificano x, y, 
larghezza, altezza del controllo sono tutti zero perche' ridimensioneremo il controllo successivamente, per
coprire l'intera client area della parent window. Notate che in questo caso, non dobbiamo chiamare 
ShowWindow per far comparire l'edit control sullo schermo, perche' includiamo lo stile WS_VISIBLE. 
Potete usare questo trucco anche nella parent window. 

;==========================================================
;        Inizializza i membri della struttura OPENFILENAME
;==========================================================

        mov ofn.lStructSize,SIZEOF ofn
        push hWnd
        pop  ofn.hwndOwner
        push hInstance
        pop  ofn.hInstance
        mov  ofn.lpstrFilter, OFFSET FilterString
        mov  ofn.lpstrFile, OFFSET buffer
        mov  ofn.nMaxFile,MAXSIZE

Dopo aver creato l'edit control, inzializziamo i membri di ofn. Poiche' vogliamo riutilizzare ofn anche nel 
dialog box "save as", riempiamo solo i membri *comuni* che vengono usati sia da GetOpenFileName che 
da GetSaveFileName. La sezione WM_CREATE e' un ottimo posto per inizializzare dati che lo necessitano 
una sola volta. 

.ELSEIF eax==WM_SIZE
        invoke GetClientRect,hWnd,ADDR rec
        invoke MoveWindow,hwndEdit,0,0,rec.right,rec.bottom,TRUE

Riceviamo i messaggi WM_SIZE quando la dimensione della client area della nostra finestra principale 
cambia. Ne riceviamo uno anche quando la finestra viene creata per la prima volta. Per poter essere in 
grado di ricevere questo messaggio, gli stili della window class devono includere CS_VREDRAW e 
CS_HREDRAW. Utilizziamo questa opportunita' per ridimensionare il nostro edit control alla stessa 
grandezza della client area della parent window. Innanzitutto dobbiamo conoscere la larghezza e l'altezza
correnti della client area della parent window. Otteniamo questa informazione chiamando la funzione 
GetClientRect, che necessita di un puntatore ad una struttura RECT, che e' definita come segue: 

RECT struct
        left         LONG ?
        top         LONG ?
        right       LONG ?
        bottom   LONG ?
    RECT ends

Dopo la chiamata a GetClientRect, la struttura RECT e' riempita con le dimensioni correnti della client 
area. Quindi noi utilizziamo le informazioni per ridimensionare l'edit control chiamando la funzione 
MoveWindow che, oltre a cambiare la posizione di una finestra, ne puo' anche modificare le dimensioni. 



.if ax==IDM_OPEN
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY
                invoke GetOpenFileName, ADDR ofn

Quando l'utente seleziona il menu item File/Open, riempiamo il membro Flags della struttura ofn, e 
chiamiamo la funzione GetOpenFileName per mostrare il dialog box "apri file". 

.if eax==TRUE
                    invoke CreateFile,ADDR buffer,\
                                GENERIC_READ or GENERIC_WRITE ,\
                                FILE_SHARE_READ or FILE_SHARE_WRITE,\
                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\
                                NULL
                    mov hFile,eax

Dopo che l'utente ha scelto il file da aprire, chiamiamo CreateFile per aprirlo. Specifichiamo che la 
funzione dovrebbe tentare di aprire il file per la lettura e per la scrittura. Una volta aperto il file, la 
funzione restituisce l'handle del file aperto, che noi conserviamo in una variabile globale per uso futuro. 
Questa funzione ha la seguente sintassi: 

HANDLE CreateFile(
    LPCTSTR  lpFileName,                 // indirizzo del nome del file
    DWORD  dwDesiredAccess,             // modo di accesso (lettura-scrittura)
    DWORD  dwShareMode,                // share mode (modo di condivisione, NdT)
    LPSECURITY_ATTRIBUTES  lpSecurityAttributes, // indirizzo del descrittore di sicurezza
    DWORD  dwCreationDistribution,  // come creare
    DWORD  dwFlagsAndAttributes,   // attributi del file
    HANDLE  hTemplateFile      // handle del file con gli attributi da copiare
   );

dwDesiredAccess specifica quali operazioni volete eseguire sul file. 

 0 Apre il file per rilevarne gli attributi. 

 GENERIC_READ Apre il file per la lettura. 

 GENERIC_WRITE Apre il file per la scrittura. 

dwShareMode specifica quali operazioni volete permettere agli altri processi di compiere sul file che sta 
per essere aperto. 

 0 Non dividere il file con altri processi. 

 FILE_SHARE_READ Permette agli altri processi di leggere dati dal file che sta per essere aperto.

 FILE_SHARE_WRITE Permette agli altri processi di scrivere dati sul file che sta per essere 
aperto. 

lpSecurityAttributes non ha signficato sotto Windows 95. dwCreationDistribution specifica l'azione 
che CreateFile eseguira' quando il file specificato in lpFileName esiste o quando non esiste. 

 CREATE_NEW Crea un nuovo file. La funzione fallisce se il file specificato esiste gia'. 

 CREATE_ALWAYS Crea un nuovo file. La funzione sovrascrive il file se e' gia' esistente. 

 OPEN_EXISTING Apre il file. La funzione fallisce se il file specificato non esiste. 

 OPEN_ALWAYS Apre il file, se esiste. Se il file non esiste la funzione lo crea come se 
dwCreationDistribution fosse stato CREATE_NEW. 

 TRUNCATE_EXISTING Apre il file. Una volta aperto, il file e' troncato di modo che la sua 
dimensione e' di zero bytes. Il processo di chiamata deve aprire il file con accesso almeno di 
GENERIC_WRITE. La funzione fallisce se il file non esiste. 

dwFlagsAndAttributes specifica gli attributi del file. 



 FILE_ATTRIBUTE_ARCHIVE Il file e' un file d'archivio. Le applicazioni usano questo attributo 
per marcare i files per il backup o la rimozione. 

 FILE_ATTRIBUTE_COMPRESSED Il file o la directory e' compresso. Per un file, cio' significa 
che tutti i dati nel file sono compressi. Per una directory, cio' significa che la compressione e' 
predefinita per i nuovi files creati e le subdirectories. 

 FILE_ATTRIBUTE_NORMAL Il file non ha attributi settati. Questo attributo e' valido solo se 
utilizzato da solo. 

 FILE_ATTRIBUTE_HIDDEN Il file e' nascosto. Non va incluso in una normale lista dei files nella 
directory. 

 FILE_ATTRIBUTE_READONLY Il file e' di sola lettura. Le applicazioni possono leggere il file ma 
non scrivervi o cancellarlo. 

 FILE_ATTRIBUTE_SYSTEM Il file e' una parte del sistema operativo (o e' usato esclusivamente 
da esso). 

invoke GlobalAlloc,GMEM_MOVEABLE or GMEM_ZEROINIT,MEMSIZE
                    mov  hMemory,eax
                    invoke GlobalLock,hMemory
                    mov  pMemory,eax

Quando il file viene aperto, allochiamo un blocco di memoria per usarlo con le funzioni ReadFile e 
WriteFile. Specifichiamo la flag GMEM_MOVEABLE per far si che Windows sposti il blocco per rendere 
l'intera memoria piu' solida. La flag GMEM_ZEROINIT dice a GlobalAlloc di riempire il blocco di memoria 
appena allocato con degli zeri. Quando GlobalAlloc ritorna con successo, eax contiene l'handle del blocco 
di memoria allocato. Passiamo questo handle alla funzione GlobalLock che restituisce un puntatore al 
blocco di memoria. 

invoke ReadFile,hFile,pMemory,MEMSIZE-1,ADDR SizeReadWrite,NULL
                    invoke SendMessage,hwndEdit,WM_SETTEXT,NULL,pMemory

Quando il blocco di memoria e' pronto all'uso, chiamiamo la funzione ReadFile per leggere dati dal file. 
Quando un file viene aperto o creato per la prima volta, il file pointer e' all'offset 0. Percio', in questo 
caso, cominciamo a leggere dal primo byte del file in avanti. Il primo parametro di ReadFile e' l'handle del
file da leggere, il secondo e' il puntatore al blocco di memoria che deve contenere i dati, successivamente
c'e' il numero di bytes da leggere dal file, e il quarto parametro e' l'indirizzo della variabile di dimensione 
DWORD che verra' riempita con il numero di bytes effettivamente letti dal file. Dopo aver riempito il 
blocco di memoria con i dati, mettiamo gli stessi nell'edit control mandando il messaggio WM_SETTEXT 
all'edit control con lParam contenente il puntatore al blocco di memoria. Dopo questa chiamata, l'edit 
control mostra i dati nella sua client area. 

invoke CloseHandle,hFile
                    invoke GlobalUnlock,pMemory
                    invoke GlobalFree,hMemory
                .endif

Arrivati a questo punto, non abbiamo piu' bisogno di lasciare il file aperto, poiche' il nostro scopo e' di 
scrivere i dati modificati dall'edit control in un altro file, non nell'originale. Pertanto chiudiamo il file 
chiamando CloseHandle con l'handle del file come parametro. Successivamente sblocchiamo il blocco di 
memoria e lo liberiamo. In realta', non dovete liberare la memoria per forza a questo punto, potreste 
riutilizzare il blocco durante la successiva operazione di salvataggio. Ma, a scopo dimostrativo, ho deciso 
di liberare la memoria qui. 

invoke SetFocus,hwndEdit

Quando il dialog box "apri file" viene mostrato sullo schermo, l'input focus passa a questi. Percio', quando
il dialog box viene chiuso, dobbiamo spostare l'input focus di nuovo nell'edit control. Questo termina 
l'operazione di lettura sul file. A questo punto, l'utente puo' modificare il contenuto dell'edit control. E 
quando vuole salvare i dati su di un altro file, deve selezionare il menuitem File/Save, che mostra un 
dialog box "Salva con nome". La creazione del dialog box "Salva con nome" non e' molto diversa da 
quella per il dialog "Apri file". Infatti, esse differiscono solo nel nome delle funzioni, GetOpenFileName e 
GetSaveFileName. Potete riutilizzare la maggior parte dei membri della struttura ofn eccetto il membro 
Flags. 



mov ofn.Flags,OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY

Nel nostro caso, vogliamo creare un nuovo file, percio' OFN_FILEMUSTEXIST e OFN_PATHMUSTEXIST 
devono essere evitati altrimenti il dialog box non ci permettera' di creare un file che non esiste gia'. Il 
parametro dwCreationDistribution della funzione CreateFile deve essere cambiato in CREATE_NEW 
poiche' vogliamo creare un nuovo file. Il codice rimanente e' identico a quello della sezione "apri file" 
tranne che per le istruzioni seguenti: 

invoke SendMessage,hwndEdit,WM_GETTEXT,MEMSIZE-1,pMemory
                        invoke lstrlen,pMemory
                        invoke WriteFile,hFile,pMemory,eax,ADDR SizeReadWrite,NULL

Mandiamo il messaggio WM_GETTEXT all'edit control per copiare i dati da esso al blocco di memoria che 
forniamo. Quando i dati sono nel blocco di memoria, determiniamo la lunghezza della stringa copiata in 
memoria e scriviamo i dati nel nuovo file. 

Introduzione
Vi mostrero' cosa sono i memory mapped files e come usarli per i vostri scopi. Usare un memory mapped
file e' abbastanza semplice, come vedrete in questo tutorial. 

Tools
Esempio 

Preliminari
Se esaminate l'esempio del tutorial precedente con molta attenzione, vi accorgerete che ha una 
limitazione abbastanza grave: che si fa se il file che volete leggere e' piu' grande del blocco di memoria 
allocato? O che facciamo se la stringa che vogliamo cercare e' tagliata a meta' alla fine del blocco di 
memoria? La risposta tradizionale alla prima domanda e' che si devono leggere i dati dal file 

ripetutamente, fino a che non si raggiunge la fine del file. La risposta alla seconda domanda e' che 
dovreste prepararvi alla particolare situazione della fine del blocco di memoria. Quest viene chiamato 
"problema di bonduary value". Presenta mal di testa per i programmatori e causa innumerevoli bugs. 
Sarebbe bello se potessimo allocare un blocco di memoria molto grande, abbastanza da contenere tutto il
file, ma il nostro programma sarebbe riempito come un uovo dalle risorse. La salvezza ci arriva con il file 
mapping. Usando il file mapping, potete pensare all'intero file come gia' caricato in memoria, e potete 
pertanto usare un puntatore di memoria per leggere o scrivere dati sul file. E' davvero cosi' semplice. Non
c'e' piu' bisogno di usare le funzioni API riguardanti la memoria e, separatamente, le funzioni API 
riguardanti l' I/O con i files, perche' sono un tutt'uno, con il file mapping. Ai memory mapped files 
possono accedere tutti i processi del sistema, non solo il processo che li crea. Il file mapping e' inoltre 
usato come mezzo per condividere dati tra i processi. Usando il file mapping in questo modo, non c'e' 
nessun file fisicamente coinvolto. E' piu' come un blocco di memoria riservato che tutti i processi possono 
*vedere*. Ma condividere i dati tra i processi e' un argomento delicato, e non va trattato 
superficialmente. Dovete implementare la process e la thread synchronization o altrimenti le vostre 
applicazioni crasheranno in breve tempo. Non affronteremo l'argomento del file mapping come un mezzo 
per creare una regione di memoria condivisa in questo tutorial. Ci concentreremo invece sull'uso del file 
mapping come modo per "mappare" un file in memoria. Infatti, il PE loader usa il file mapping per 
caricare gli eseguibili in memoria. E' molto conveniente perche' solo le porzioni necessarie possono essere
lette selettivamente dal file sul disco. Sotto Win32, dovreste usare il file mapping quanto piu' possibile. Ci
sono pero' anche alcune limitazioni nel file mapping. Una volta creato un memory mapped file, la sua 
grandezza non puo' essere variata durante quella sessione. Percio' il file mapping e' eccezionale per i files
di sola lettura o per operazioni con i files che non modificano la grandezza del file. Cio' non significa che 
non potete usare il file mapping se volete aumentare la grandezza del file. Potete stimare la nuova 
grandezza e creare il memory mapped file basandovi su di essa, e il file aumentera' a quella grandezza. 
E' semplicemente un piccolo inconveniente, ecco tutto. Ok, abbastanza spiegazioni. Tuffiamoci 
nell'implementazione del file mapping. Per usare il file mapping, questi tre passi devono essere seguiti: 

1. chiamate CreateFile, esclusivamente per aprire il file che volete mappare. Dovete specificare il 
parametro dwShareMode come 0 per evitare che gli altri processi scrivano sul file durante la 
sessione. 

2. chiamate CreateFileMapping con il file handle restituito dal CreateFile come uno dei parametri. 
Questa funzione crea un file mapping object dal file aperto da CreateFile. 

3. chiamate MapViewOfFile per mappare una regione del file selezionata (o l'intero file) in 
memoria. Questa funzione restituisce un puntatore al primo byte della regione del mapped file. 

4. Usate il puntatore per leggere o scrivere sul file. 
5. chiamate UnmapViewOfFile per "demappare" il file. 
6. chiamate CloseHandle con l'handle del mapped file come parametro per chiudere il mapped file. 
7. chiamate CloseHandle di nuovo, ma stavolta con il file handle restituito da CreateFile per 

chiudere fisicamente il file. 

Essay
Il seguente programma vi fa aprire un file con un dialog box "apri file". Esso apre il file usando il file 
mapping. Se riesce ad aprirlo con successo, il titolo della finestra viene cambiato con il nome del file 
aperto. Potete salvare il file con un altro nome selezionando il menuitem File/Save. Il programma copiera'
l'intero contenuto del file aperto nel nuovo file. Notate che non dovete chiamare GlobalAlloc per allocare 
un blocco di memoria in questo programma. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib
includelib comdlg32.lib

.const
IDM_OPEN equ 1
IDM_SAVE equ 2
IDM_EXIT equ 3
MAXSIZE equ 260

.data



ClassName db "Win32ASMFileMappingClass",0
AppName  db "Win32 ASM File Mapping Example",0
MenuName db "FirstMenu",0
ofn   OPENFILENAME <>
FilterString db "All Files",0,"*.*",0
             db "Text Files",0,"*.txt",0,0
buffer db MAXSIZE dup(0)
hMapFile HANDLE 0     ; Handle del memory mapped file, deve essere
                     ;inizializzato con 0 perche' lo vogliamo usare anche
                     ;come flag nella sezione WM_DESTROY.

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hFileRead HANDLE ?                     ; Handle del file sorgente
hFileWrite HANDLE ?                    ; Handle del file di output
hMenu HANDLE ?
pMemory DWORD ?                        ; puntatore ai dati del file sorgente
SizeWritten DWORD ?           ; numero di bytes effettivamente scritti da WriteFile
FileSize DWORD ?              ; low DWORD della file size (dimensione file) del file sorgente
FileSizeHighWord DWORD ?      ; high DWORD della file size del file sorgente

.code
start:
        invoke GetModuleHandle, NULL
        mov    hInstance,eax
        invoke GetCommandLine
        invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
 invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,\
                ADDR AppName, WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
               CW_USEDEFAULT,300,200,NULL,NULL,\
    hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    .WHILE TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc uses ebx hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_CREATE
        invoke GetMenu,hWnd                        ; Ottieni l'handle del menu
        mov  hMenu,eax
        mov ofn.lStructSize,SIZEOF ofn
        push hWnd



        pop  ofn.hwndOwner
        push hInstance
        pop  ofn.hInstance
        mov  ofn.lpstrFilter, OFFSET FilterString
        mov  ofn.lpstrFile, OFFSET buffer
        mov  ofn.nMaxFile,MAXSIZE
    .ELSEIF eax==WM_DESTROY
        .if hMapFile!=0
            call CloseMapFile
        .endif
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .if lParam==0
            .if ax==IDM_OPEN
                mov  ofn.Flags, OFN_FILEMUSTEXIST or \
                                OFN_PATHMUSTEXIST or OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY
                                invoke GetOpenFileName, ADDR ofn
                .if eax==TRUE
                    invoke CreateFile,ADDR buffer,\
                                                GENERIC_READ ,\
                                                0,\
                                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\
                                                NULL
                    mov hFileRead,eax
                    invoke CreateFileMapping,hFileRead,NULL,PAGE_READONLY,0,0,NULL
                    mov     hMapFile,eax
                    push    ebx
                    mov     eax,OFFSET buffer
                    movzx  bx,ofn.nFileOffset
                    add      eax,ebx
                    pop      ebx
                    invoke SetWindowText,hWnd,eax
                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_GRAYED
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_ENABLED
                .endif
            .elseif eax==IDM_SAVE
                mov ofn.Flags,OFN_LONGNAMES or\
                                OFN_EXPLORER or OFN_HIDEREADONLY
                invoke GetSaveFileName, ADDR ofn
                .if eax==TRUE
                    invoke CreateFile,ADDR buffer,\
                                                GENERIC_READ or GENERIC_WRITE ,\
                                                FILE_SHARE_READ or FILE_SHARE_WRITE,\
                                                NULL,CREATE_NEW,FILE_ATTRIBUTE_ARCHIVE,\
                                                NULL
                    mov hFileWrite,eax
                    invoke MapViewOfFile,hMapFile,FILE_MAP_READ,0,0,0
                    mov pMemory,eax
                    invoke GetFileSize,hFileRead,ADDR FileSizeHighWord
                    invoke WriteFile,hFileWrite,pMemory,eax,ADDR SizeWritten,NULL
                    invoke UnmapViewOfFile,pMemory
                    call   CloseMapFile
                    invoke CloseHandle,hFileWrite
                    invoke SetWindowText,hWnd,ADDR AppName
                    invoke EnableMenuItem,hMenu,IDM_OPEN,MF_ENABLED
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_GRAYED
                .endif
            .else
                invoke DestroyWindow, hWnd
            .endif
        .endif
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .ENDIF
    xor    eax,eax
    ret
WndProc endp

CloseMapFile PROC
        invoke CloseHandle,hMapFile
        mov    hMapFile,0



        invoke CloseHandle,hFileRead
        ret
CloseMapFile endp

end start

Analisi
invoke CreateFile,ADDR buffer,\
                                                GENERIC_READ ,\
                                                0,\
                                                NULL,OPEN_EXISTING,FILE_ATTRIBUTE_ARCHIVE,\
                                                NULL

Quando l'utente seleziona un file nel dialog box "apri file", noi chiamiamo CreateFile per aprirlo. Notate 
che specifichiamo GENERIC_READ per aprire questo file con accesso di sola lettura e dwShareMode e' 
zero perche' non vogliamo che altri processi modifichino il file durante la nostra operazione. 

invoke CreateFileMapping,hFileRead,NULL,PAGE_READONLY,0,0,NULL

Quindi chiamiamo CreateFileMapping per creare un memory mapped file dal file aperto. 
CreateFileMapping ha la seguente sintassi: 

HANDLE CreateFileMapping(
    HANDLE  hFile,                             ;// handle del file da mappare
    LPSECURITY_ATTRIBUTES  lpFileMappingAttributes, ;// attributi di sicurezza opzionali
    DWORD  flProtect,                        ;// protezione per il mapping object
    DWORD  dwMaximumSizeHigh,  ;// high-order 32 bits della grandezza dell'object
    DWORD  dwMaximumSizeLow,   ;// low-order 32 bits della grandezza dell'object
    LPCTSTR  lpName                        ;// nome del file-mapping object
   );

Dovreste prima sapere che CreateFileMapping non deve mappare per forza l'intero file in memoria. Potete
usare la funzione anche per mappare solo una parte del file fisico in memoria. Voi specificate la 
grandezza del memory mapped file nei parametri dwMaximumSizeHigh e dwMaximumSizeLow. Se 
specificate una grandezza che e' piu' grande della grandezza effettiva del file, il file fisico verra' espanso 
alla nuova grandezza. Se volete che il memory mapped file sia di dimensioni identiche al file fisico, 
mettete zero in entrambi i parametri. Potete usare NULL nel parametro lpFileMappingAttributes per dire a
Windows di creare un memory mapped file con gli attributi di sicurezza predefiniti. flProtect definisce la 
protezione desiderata per il memory mapped file. Nel nostro esempio, noi usiamo PAGE_READONLY per 
permettere solo operazioni di lettura sul memory mapped file. Notate che questo attributo non deve 
contraddire l'attributo usato in CreateFile o altrimenti CreateFileMapping fallira'. lpName punta al nome 
del memory mapped file. Se volete condividere questo file con altri processi, dovete dargli un nome. Ma, 
nel nostro esempio, il nostro processo e' l'unico che usa questo file quindi ignoreremo questo parametro. 

push    ebx
                    mov     eax,OFFSET buffer
                    movzx  bx,ofn.nFileOffset
                    add      eax,ebx
                    pop      ebx
                    invoke SetWindowText,hWnd,eax

Se CreateFileMapping ha successo, cambiamo il titolo della finestra con il nome del file aperto. Il nome 
del file con l'intera path viene conservato in buffer; noi vogliamo mostrare solo il nome del file nella barra
e quindi dobbiamo aggiungere il valore del membro nFileOffset della struttura OPENFILENAME all'indirizzo
di buffer. 

invoke EnableMenuItem,hMenu,IDM_OPEN,MF_GRAYED
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_ENABLED

Come precauzione, non vogliamo che l'utente apra piu' files alla volta, percio' disattiviamo (o meglio 
ingrigiamo) il menu item "Open" e abilitiamo il menu item "Save". EnableMenuItem e' usata per cambiare



gli attributi di un menu item. Dopo di cio', aspettiamo che l'utente selezioni il menu item File/Save per 
chiudere il nostro programma. Se l'utente sceglie di chiudere il programma, dobbiamo chiudere il 
memory mapped file e il file fisico con il seguente codice: 

.ELSEIF eax==WM_DESTROY
        .if hMapFile!=0
            call CloseMapFile
        .endif
        invoke PostQuitMessage,NULL

In questo spezzone, quando la window procedure riceve il messaggio WM_DESTROY, verifica prima se il 
valore di hMapFile e' zero o no. Se non e' zero, chiama la funzione CloseMapFile che contiene il codice 
seguente: 

CloseMapFile PROC
        invoke CloseHandle,hMapFile
        mov    hMapFile,0
        invoke CloseHandle,hFileRead
        ret
CloseMapFile endp

CloseMapFile chiude il memory mapped file e il file fisico cosi' che non ci sara' nessuna perdita di risorse 
quando il nostro programma esce a Windows. Se l'utente sceglie di salvare i dati in un altro file, il 
programma gli presenta un dialog box "Salva con nome". Dopo che l'utente ha scritto il nome del nuovo 
file, il file viene creato con la funzione CreateFile. 

invoke MapViewOfFile,hMapFile,FILE_MAP_READ,0,0,0
                    mov pMemory,eax

Immediatamente dopo che il file di output e' stato creato, chiamiamo MapViewOfFile per mappare in 
memoria la porzione desiderata del memory mapped file. Questa funzione ha la seguente sintassi: 

LPVOID MapViewOfFile(
    HANDLE  hFileMappingObject,          ;// file-mapping object da mappare nell'address space
    DWORD  dwDesiredAccess,               ;// modo di accesso
    DWORD  dwFileOffsetHigh,                ;// high-order 32 bits del file offset
    DWORD  dwFileOffsetLow,                ;// low-order 32 bits del file offset
    DWORD  dwNumberOfBytesToMap  ;//numero di bytes da mappare
   );

dwDesiredAccess specifica che operazione vogliamo eseguire sul file. Nel nostro esempio, vogliamo solo
leggere i dati, percio' usiamo FILE_MAP_READ. dwFileOffsetHigh e dwFileOffsetLow specificano il file 
offset iniziale della porzione di file che volete mappare in memoria. Nel nostro caso, vogliamo leggere 
tutto il file percio' cominciamo a mappare dall'offset 0 in avanti. dwNumberOfBytesToMap specifica il 
numero di bytes da mappare in memoria. Se volete mappare tutto il file (specificato da 
CreateFileMapping), passate 0 a MapViewOfFile. Dopo aver chiamato MapViewOfFile, la porzione 
desiderata e' caricata in memoria. Vi sara' dato il puntatore al blocco di memoria che contiene i dati dal 
file. 

invoke GetFileSize,hFileRead,ADDR FileSizeHighWord

Trova la lunghezza del file, che e' restituita in eax. Se il file e' piu' grande di 4 GB, la high DWORD della 
lunghezza del file e' conservata in FileSizeHighWord. 

invoke WriteFile,hFileWrite,pMemory,eax,ADDR SizeWritten,NULL

Scrive i dati che sono mappati in memoria nel file di output. 

invoke UnmapViewOfFile,pMemory

Quando abbiamo finito con il file sorgente, lo "demappiamo" dalla memoria... 

call   CloseMapFile
                    invoke CloseHandle,hFileWrite



...e chiudiamo tutti i files. 

invoke SetWindowText,hWnd,ADDR AppName

Ripristina il titolo originale della finestra. 

invoke EnableMenuItem,hMenu,IDM_OPEN,MF_ENABLED
                    invoke EnableMenuItem,hMenu,IDM_SAVE,MF_GRAYED

Abilita il menu item "Open" e disabilita "Save As". 

Introduzione
Impareremo cos'è un processo, come crearlo e come terminarlo. 

Tools

Preliminari
Cos'e' un processo? Cito questa definizione dalla Win32 API reference: 

"Un processo e' un'applicazione in esecuzione che consiste di un address space virtuale privato, codice, 
dati, e altre risorse del sistema operativo, come files, pipes, e synchronization objects che sono visibili 
per il processo." 

Come potete capire dalla definizione di cui sopra, un processo "possiede" diversi oggetti: l'address space, 
i/il modulo/i in esecuzione, e qualsiasi cosa che i moduli in esecuzione aprano o creino. Come minimo, un
processo deve essere costituito da un modulo in esecuzione, un address space privato e un thread. Ogni 
processo deve avere almeno un thread. Cos'e' un thread? Un thread e' in realta' una "coda di 
esecuzione". Quando Windows crea per la prima volta un processo, crea solo UN thread per processo. 
Questo thread solitamente comincia ad essere eseguito dalla prima istruzione nel modulo. Se il processo 
successivamente ha bisogno di piu' threads, esso puo' esplicitamente crearli. Quando Windows riceve un 
comando per creare un processo, crea il memory address space privato per quel processo e quindi mappa
il file eseguibile in tale spazio. Dopo di cio', crea il thread primario per il processo. Sotto Win32, potete 
anche creare processi per i vostri programmi chiamando la funzione CreateProcess. CreateProcess ha la 
seguente sintassi: 

BOOL CreateProcess(
    LPCTSTR  lpApplicationName,         // puntatore al nome del modulo eseguibile
    LPTSTR  lpCommandLine,                 // puntatore alla stringa della riga di comando
    LPSECURITY_ATTRIBUTES  lpProcessAttributes, // puntatore agli attributi di sicurezza del 
processo
    LPSECURITY_ATTRIBUTES  lpThreadAttributes, // puntatore agli attributi di sicurezza del 
thread
    BOOL  bInheritHandles,                     // flag di "eredita' degli handle" (handle 
inherit flag)
    DWORD  dwCreationFlags,                // flags di creazione
    LPVOID  lpEnvironment,                 // puntatore al nuovo blocco di environment 
(ambiente, NdT)
    LPCTSTR  lpCurrentDirectory,        // puntatore al nome della directory corrente
    LPSTARTUPINFO  lpStartupInfo,    // puntatore a STARTUPINFO
    LPPROCESS_INFORMATION  lpProcessInformation  // puntatore a PROCESS_INFORMATION 
   );

Non vi preoccupate per il numero dei parametri. Ne possiamo ignorare la maggior parte. 

lpApplicationName --> Il nome del file eseguibile (con o senza la path) che volete eseguire. Se questo 
parametro e' NULL, dovete fornire il nome del file eseguibile nel parametro lpCommandLine. 

lpCommandLine --> Gli argomenti della riga di comando del programma che volete eseguire. Notate 
che se lpApplicationName e' NULL, questo parametro deve contenere anche il nome del file eseguibile. 
Come questo : "notepad.exe readme.txt" 

lpProcessAttributes e lpthreadAttributes --> Specificano gli attributi di sicurezza per i processi e per 
il thread primario. Se sono NULLs, verranno usati gli attributi di sicurezza predefiniti. 

bInheritHandles --> Una flag che specifica se volete che il nuovo processo erediti tutti gli handle aperti 
dal vostro processo. 

dwCreationFlags --> Diverse flags che determinano il comportamento del processo che volete creare; 
ad esempio, volete che il processo venga creato ma immediatamente sospeso, in modo tale da 
permettervi di esaminarlo o modificarlo prima che parta? Potete anche specificare la classe di priorita' 
del/i thread(s) nel nuovo processo. Questa classe di priorita' e' usata per determinare la priorita' di 
comparsa dei threads all'interno del processo. Normalmente usiamo le flags CREATE_NEW_CONSOLE e 
NORMAL_PRIORITY_CLASS. 

lpEnvironment --> Un puntatore all'environment block che contiene diverse stringhe di environment per
il nuovo processo. Se questo parametro e' NULL, il nuovo processo eredita l'environment block dal parent
process (processo precedente che gli ha dato inizio, NdT). 

lpCurrentDirectory --> Un puntatore alla stringa che specfica il drive e la directory correnti per il child 
process (il processo nuovo, quello appena creato, NdT). NULL se volete che il child process erediti dal 
parent process. 

lpStartupInfo --> Punta ad una struttura STARTUPINFO che specifica come dovra' apparire la finestra 
principale del nuovo processo. La struttura STARTUPINFO contiene molti membri che specificano 
l'apparenza della finestra principale del child process. Se non volete niente di speciale, potete riempire la 
struttura STARTUPINFO con i valori dal parent process chiamando la funzione GetStartupInfo. 



lpProcessInformation --> Punta ad una struttura PROCESS_INFORMATION che riceve informazioni di 
identificazione circa il nuovo processo. La struttura PROCESS_INFORMATION ha i seguenti membri: 

PROCESS_INFORMATION STRUCT
        hProcess          HANDLE ?             ; handle del child process
        hThread            HANDLE ?             ; handle del thread primario del child process
        dwProcessId     DWORD ?             ; ID del child process
        dwThreadId      DWORD ?             ; ID del thread primario del child process
    PROCESS_INFORMATION ENDS

L'handle del processo e la ID del processo sono due cose diverse. Una process ID e' un identificatore 
unico per il processo nel sistema. Un process handle e' un valore restituito da Windows per essere 
utilizzato con altre funzioni API legate ai processi. Un process handle non puo' essere usato per 
identificare un processo poiche' non e' unico. 

Dopo la chiamata a CreateProcess, un nuovo processo viene creato e la chiamata a CreateProcess ritorna 
immediatamente. Potete controllare se il nuovo processo e' ancora attivo chiamando la funzione 
GetExitCodeProcess che ha la seguente sintassi: 

BOOL GetExitCodeProcess(
    HANDLE  hProcess,         // handle del processo
    LPDWORD  lpExitCode  // indirizzo che riceve lo stato di chiusura
   );

Se questa chiamata ha successo, lpExitCode contiene il termination status (stato di chiusura, NdT) del 
processo in questione. Se il valore in lpExitCode e' uguale a STILL_ACTIVE, allora il processo e' ancora 
in esecuzione. 

Potete terminare a piacimento un processo chiamando la funzione TerminateProcess che ha la seguente 
sintassi: 

BOOL TerminateProcess(
    HANDLE  hProcess,     // handle del processo
    UINT  uExitCode           // codice di uscita per il processo
   );

Potete specificare il codice di uscita desiderato per il processo, qualsiasi valore vogliate. TerminateProcess
non e' un modo pulito di terminare un processo poiche' nessuna dll relativa al processo sara' informata 
che questo e' stato terminato. 

Essay
L'esempio che segue creera' un nuovo processo quando l'utente seleziona il menu item "create process". 
Il programma cercera' di eseguire "msgbox.exe". Se l'utente vuole terminare il nuovo processo, puo' 
selezionare il menu item "terminate process". Il programma controllera' prima se il nuovo processo e' gia'
stato distrutto, e se no, chiamera' la funzione TerminateProcess per distruggerlo. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.const
IDM_CREATE_PROCESS equ 1
IDM_TERMINATE equ 2
IDM_EXIT equ 3

.data
ClassName db "Win32ASMProcessClass",0
AppName  db "Win32 ASM Process Example",0
MenuName db "FirstMenu",0
processInfo PROCESS_INFORMATION <>



programname db "msgbox.exe",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hMenu HANDLE ?
ExitCode DWORD ?              ; contiene il codice di uscita del processo, dalla chiamata    

                              ; GetExitCodeProcess.

.code
start:
        invoke GetModuleHandle, NULL
        mov    hInstance,eax
        invoke GetCommandLine
        invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
        invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
           CW_USEDEFAULT,300,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    invoke GetMenu,hwnd
    mov  hMenu,eax
    .WHILE TRUE
                invoke GetMessage, ADDR msg,NULL,0,0
                .BREAK .IF (!eax)
                invoke TranslateMessage, ADDR msg
                invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    LOCAL startInfo:STARTUPINFO
    mov   eax,uMsg
    .IF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_INITMENUPOPUP
        invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode
        .if eax==TRUE
            .if ExitCode==STILL_ACTIVE
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_GRAYED
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_ENABLED
            .else
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED
            .endif
        .else
            invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED



            invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED
        .endif
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .if lParam==0
            .if ax==IDM_CREATE_PROCESS
                invoke GetStartupInfo,ADDR startInfo
                invoke CreateProcess,ADDR programname,NULL,NULL,NULL,FALSE,\
                                        NORMAL_PRIORITY_CLASS,\
                                        NULL,NULL,ADDR startInfo,ADDR processInfo
            .elseif ax==IDM_TERMINATE
                invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode
                .if ExitCode==STILL_ACTIVE
                    invoke TerminateProcess,processInfo.hProcess,0
                .endif
            .else
                invoke DestroyWindow,hWnd
            .endif
        .endif
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .ENDIF
    xor    eax,eax
    ret
WndProc endp
end start

Analisi
Il programma crea la finestra principale e rileva il menu handle per uso futuro. Quindi aspetta che 
l'utente selezioni un comando dal menu. Quando l'utente seleziona il menu item "Process" dal menu 
principale, noi elaboriamo il messaggio WM_INITMENUPOPUP per modificare i menu items dentro il menu 
di popup prima che venga mostrato. 

.ELSEIF eax==WM_INITMENUPOPUP
        invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode
        .if eax==TRUE
            .if ExitCode==STILL_ACTIVE
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_GRAYED
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_ENABLED
            .else
                invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED
                invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED
            .endif
        .else
            invoke EnableMenuItem,hMenu,IDM_CREATE_PROCESS,MF_ENABLED
            invoke EnableMenuItem,hMenu,IDM_TERMINATE,MF_GRAYED
        .endif

Perche' vogliamo elaborare questo messaggio ? Perche' vogliamo preparare i menu items del menu di 
popup prima che l'utente li possa vedere. Nel nostro esempio, se il nuovo processo non e' ancora 
cominciato, vogliamo abilitare il menu item "start process" e disabilitare "terminate process". Facciamo il 
contrario se il nuovo processo e' gia' attivo. Prima controlliamo se il nuovo processo e' ancora in 
esecuzione chiamando la funzione GetExitCodeProcess con l'handle del processo che e' stato riempito 
dalla funzione CreateProcess. Se GetExitCodeProcess restituisce FALSE, significa che il processo non e' 
partito ancora, e quindi disabilitiamo il menu item "terminate process". Se GetExitCodeProcess restituisce
TRUE, sappiamo che un nuovo processo e' stato avviato, ma dobbiamo sapere se e' ancora in esecuzione.
Percio' compariamo il valore in ExitCode con il valore STILL_ACTIVE, e se sono uguali, il processo e' 
ancora in esecuzione: dobbiamo disabilitare pertanto il menu item "start process" poiche' non vogliamo 
far cominciare piu' processi contemporaneamente. 

.if ax==IDM_CREATE_PROCESS
                invoke GetStartupInfo,ADDR startInfo
                invoke CreateProcess,ADDR programname,NULL,NULL,NULL,FALSE,\



                                        NORMAL_PRIORITY_CLASS,\
                                        NULL,NULL,ADDR startInfo,ADDR processInfo

Quando l'utente seleziona il menu item "start process", chiamiamo la funzione GetStartupInfo per 
riempire la struttura startupinfo che passeremo alla funzione CreateProcess. Dopo di cio' chiamiamo la 
funzione CreateProcess per avviare il nuovo processo. 

.elseif ax==IDM_TERMINATE
                invoke GetExitCodeProcess,processInfo.hProcess,ADDR ExitCode
                .if ExitCode==STILL_ACTIVE
                    invoke TerminateProcess,processInfo.hProcess,0
            .endif

Quando l'utente seleziona il menu item "terminate process", controlliamo se il nuovo processo e' ancora 
attivo chiamando la funzione GetExitCodeProcess. Se e' ancora attivo, chiamiamo la funzione 
TerminateProcess per "ucciderlo". 

Introduzione
In questo tutorial impareremo come creare un programma multithread. Studieremo inoltre i metodi di 
comunicazione tra i threads. 

Tools
Esempio 

Preliminari
Nel tutorial precedente, avete imparato che un processo contiene almeno un thread: il thread primario 
(primary thread). Un thread e' una catena di esecuzione. Potete anche creare threads addizionali nel 
vostro programma. Potete considerare il multithreading come un multitasking all'interno di un solo 
programma. In termini di implementazione, un thread e' una funzione che viene eseguita in 
concomitanza col programma principale. Potete eseguire diverse istanze della stessa funzione o potete 
eseguire diverse funzioni simultaneamente, a seconda delle vostre necessita'. Il multithreading e' proprio 

http://www.quequero.org/uicwiki/images/Tut15.zip
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Disclaimer
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Analisi
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Essay
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Preliminari
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Tools
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Introduzione
http://quequero.org/Tutorial_15:_Programmazione_in_Multithreading#Tutorial_15:_Programmazione_in_Multithreading
javascript:toggleToc()


di Win32, non esistono equivalenti in Win16. I threads vengono eseguiti nello stesso processo, cosicche' 
possono accedere a qualsiasi risorsa del processo, come variabili globali, handles, ecc. Comunque, ogni 
thread ha la sua propria stack cosicche' le variabili locali in ciascun thread sono private. Ogni thread 
possiede inoltre il suo set privato di registri, di modo che quando Windows passa ad altri threads, il 
thread puo' "ricordarsi" il suo ultimo stato e puo' "riprendere" la funzione che stava espletando quando 
riottiene il controllo. Cio' viene gestito internamente da Windows. Possiamo dividere i threads in due 
caterorie: 

1. User interface thread: Questo tipo di thread crea la sua propria finestra cosicche' puo' ricevere 
messaggi di windows. Puo' rispondere all'utente tramite la propria finestra, e da qui proviene il 
nome. Questo tipo di thread e' soggetto alla regola Win16 Mutex che consente un solo user 
interface thread in user e gdi kernel a 16-bit. Mentre uno user interface thread sta eseguendo 
codice dello user o del gdi kernel a 16-bit, nessun altro UI thread puo' servirsi dello user o del 
gdi kernel a 16-bit. Notate che questa Win16 Mutex e' valida per Windows 95 poiche' in 
profondita', le funzioni API di Windows 95 si spezzettano in codice a 16-bit. Windows NT non ha 
Win16 Mutex percio' gli user interface threads sotto NT funzionano molto piu' liberamente che 
sotto Windows 95. 

2. Worker thread: Questo tipo di thread non crea una finestra, e percio' non puo' ricevere 
messaggi di windows. Esiste fondamentalmente per eseguire il lavoro assegnato in background, 
e da qui il nome "worker thread". 

Io consiglio la seguente strategia nell'utilizzo delle capacita' di multithreading di Win32: Lasciate che il 
thread primario esegua i compiti legati all'interfaccia utente e lasciate agli altri threads il compito di 
sbrigare i lavori pesanti in background. In tal modo, il thread primario e' come un Governatore, e gli altri 
threads sono lo staff del Governatore. Il Governatore assegna lavori allo staff, mentre egli mantiene il 
contatto con il pubblico. Lo staff esegue ubbidientemente il lavoro e poi ritorna dal Governatore. Se il 
Governatore avesse dovuto eseguire ciascun compito da solo, non sarebbe stato in grado di prestare 
molta attenzione al pubblico o alla stampa. Cio' e' quello che avviene ad una finestra impegnata ad 
eseguire un lungo lavoro nel suo thread primario: non risponde ai comandi dell'utente finche' il lavoro 
non e' completo. Un tale programma puo' beneficiare della creazione di un thread addizionale che e' 
responsabile dell'esecuzione del lungo lavoro, e che permette al thread primario di rispondere ai comandi 
dell'utente. Possiamo creare un thread chiamando la funzione CreateThread che ha la seguente sintassi: 

HANDLE CreateThread(
    LPSECURITY_ATTRIBUTES  lpThreadAttributes,  // indirizzo degli attributi di sicurezza del 
thread
    DWORD  dwStackSize,                        // dimensione inziale della stack del thread, 
in bytes
    LPTHREAD_START_ROUTINE  lpStartAddress,  // indirizzo della thread function
    LPVOID  lpParameter,                                                // argomento per il 
nuovo thread
    DWORD  dwCreationFlags,         // flags di creazione
    LPDWORD  lpThreadId            // indirizzo dell'identificatore di thread (thread ID) 
restituito
   );

La funzione CreateThread assomiglia moltissimo a CreateProcess. 

lpThreadAttributes --> Potete usare NULL se volete che il thread abbia il descrittore di sicurezza 
predefinito. 

dwStackSize --> Specifica la dimensione della stack del thread. Se volete che il thread abbia la stessa 
grandezza di stack del thread primario, usate NULL in questo parametro. 

lpStartAddress --> Indirizzo della thread function. E' la funzione che eseguira' il lavoro del thread. 
Questa funzione DEVE ricevere uno e soltanto un parametro a 32-bit e restituire un valore a 32-bit. 

lpParameter --> Il parametro che volete passare alla thread function. 

dwCreationFlags --> Uguali a quelle nella funzione CreateProcess. 

lpThreadId --> La funzione CreateThread riempira', a questo indirizzo, la thread ID del thread appena 
creato. 



Se la chiamata a CreateThread ha successo, restituisce l'handle del thread appena creato. Altrimenti, 
restituisce NULL. La thread function viene eseguita appena la chiamata a CreateThread ha successo, a 
meno che voi non specifichiate la flag CREATE_SUSPENDED in dwCreationFlags. In tal caso, il thread e' 
sospeso finche' non viene chiamata la funzione ResumeThread. Quando la thread function ritorna con 
l'istruzione ret, Windows chiama implicitamente la funzione ExitThread per la thread function. Potete 
chiamare anche voi la funzione ExitThread all'interno della vostra thread function, ma farlo non ha molto 
senso. Potete rilevare l'exit code chiamando la funzione GetExitCodeThread. Se volete terminare un 
thread da un altro thread, potete chiamare la funzione TerminateThread. Ma dovreste usare questa 
funzione in circostanze estreme, poiche' essa termina immediatamente il thread, senza dargli alcuna 
possibilita' di eliminare i suoi "resti". 

Ora spostiamoci sui metodi di comunicazione tra i threads. Ce ne sono tre: 

 Usando variabili globali; 

 Messaggi di Windows; 

 Eventi. 

I threads condividono le risorse del processo, incluse le variabili globali, percio' essi possono utilizzarle 
per comunicare l'un l'altro. Comunque questo metodo deve essere utilizzato con molta attenzione. 
Bisogna prendere in considerazione la sincronizzazione dei threads. Ad esempio, se due threads usano la 
stessa struttura di 10 membri, che succede se Windows improvvisamente toglie il controllo ad uno dei 
threads proprio mentre stava aggiornando la struttura ? L'altro thread sara' lasciato con dei dati 
inconsistenti in una struttura! Non commettete errori, i programmi che usano il multithreading sono piu' 
difficili da "debuggare" e da mantenere. Questo tipo di bug sembra capitare casualmente, il che rende il 
problema molto difficile da intercettare. Potete usare inoltre i messaggi di Windows per comunicare tra i 
threads. Se sono tutti user interface threads, non c'e' problema: questo metodo puo' essere usato come 
un mezzo di comunicazione biunivoco. Tutto cio' che dovete fare e' definire uno o piu' messaggi 
personalizzati che hanno significato per i threads. Definite un messaggio personalizzato utilizzando il 
messaggio WM_USER come valore base, ad esempio: 

WM_MYCUSTOMMSG equ WM_USER+100h

Windows non usera' nessun valore da WM_USER a salire per i propri messaggi, quindi potete usare il 
valore WM_USER e piu' per i valori dei vostri messaggi predefiniti. Se uno dei threads e' uno user 
interface thread e l'altro e' un worker, non potete usare questo metodo come sistema di comunicazione 
biunivoco, dal momento che un worker thread non ha la sua finestra e quindi nemmeno una coda di 
messaggi. Potete invece usare il seguente schema: 

User interface Thread ------> variabile/i globale/i ----> Worker thread
Worker Thread  ------> messaggio/i Windows personalizzati ----> User interface Thread

Infatti, useremo questo sistema nel nostro esempio. L'ultimo metodo di comunicazione e' l'event object. 
Potete considerare l'event object come una specie di flag. Se l'event object e' in stato "non-segnalato", il 
thread e' inattivo o "addormentato", e, in questo stato, il thread non ricevera' CPU time slice. Quando 
l'event object e' in stato "segnalato", Windows "sveglia" il thread e questo comincia ad eseguire il 
compito a lui assegnato. 

Essay
Dovreste scaricare il file zippato con l'esempio e far partire thread1.exe. Clickate sul menu item "Savage 
Calculation". Cio' dira' al programma di eseguire "add eax,eax " per 600,000,000 di volte. Notate che 
durante questo tempo, non potete fare niente con la finestra principale: non potete muoverla, non potete
attivarne i menu ecc. Quando il calcolo e' completato, una message box appare. Dopo di cio' la finestra 
accetta i vostri comandi normalmente. Per evitare questo tipo di inconveniente per l'utente, possiamo 
muovere la routine di "calcolo" in un worker thread separato e lasciare che il thread primario continui con
il suo compito di interfaccia. Potete vedere che anche se la finestra principale risponde piu' lentamente 
del solito, comunque risponde! 



include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.const
IDM_CREATE_THREAD equ 1
IDM_EXIT equ 2
WM_FINISH equ WM_USER+100h

.data
ClassName db "Win32ASMThreadClass",0
AppName  db "Win32 ASM MultiThreading Example",0
MenuName db "FirstMenu",0
SuccessString db "The calculation is completed!",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hwnd HANDLE ?
ThreadID DWORD ?

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
           CW_USEDEFAULT,300,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    .WHILE TRUE
            invoke GetMessage, ADDR msg,NULL,0,0
            .BREAK .IF (!eax)
            invoke TranslateMessage, ADDR msg
            invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam
        .if lParam==0
            .if ax==IDM_CREATE_THREAD



                mov  eax,OFFSET ThreadProc
                invoke CreateThread,NULL,NULL,eax,\
                                        NULL,NORMAL_PRIORITY_CLASS,\
                                        ADDR ThreadID
                invoke CloseHandle,eax
            .else
                invoke DestroyWindow,hWnd
            .endif
        .endif
    .ELSEIF eax==WM_FINISH
        invoke MessageBox,NULL,ADDR SuccessString,ADDR AppName,MB_OK
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .ENDIF
    xor    eax,eax
    ret
WndProc endp

ThreadProc PROC USES ecx Param:DWORD
        mov  ecx,600000000
Loop1:
        add  eax,eax
        dec  ecx
        jz   Get_out
        jmp  Loop1
Get_out:
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL
        ret
ThreadProc ENDP

end start

Analisi
Il programma principale si presenta all'utente con una normale finestra con un menu. Se l'utente sceglie 
il menu item "Create Thread", il programma crea un thread cosi': 

.if ax==IDM_CREATE_THREAD
                mov  eax,OFFSET ThreadProc
                invoke CreateThread,NULL,NULL,eax,\
                                        NULL,NORMAL_PRIORITY_CLASS,\
                                        ADDR ThreadID
                invoke CloseHandle,eax

La funzione sovrastante crea un thread che eseguira' una procedura chiamata ThreadProc assieme al 
thread primario. Dopo che la chiamata ha avuto successo, CreateThread ritorna immediatamente e 
ThreadProc comincia ad essere eseguita. Poiche' non usiamo handle di thread, dovremmo chiuderlo o ci 
saranno perdite di memoria. 

ThreadProc PROC USES ecx Param:DWORD
        mov  ecx,600000000
Loop1:
        add  eax,eax
        dec  ecx
        jz   Get_out
        jmp  Loop1
Get_out:
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL
        ret
ThreadProc ENDP

Come potete vedere, ThreadProc esegue una savage calculation che richiede un bel po' di tempo, e 
quando ha finito manda un messaggio WM_FINISH alla finestra principale. WM_FINISH e' il nostro 
messaggio personalizzato, definito come segue: 



WM_FINISH equ WM_USER+100h

Non dovete per forza aggiungere 100h a WM_USER ma cosi' state piu' sicuri. Il messaggio WM_FINISH e'
significativo solo all'interno del nostro programma. Quando la finestra principale riceve il messaggio 
WM_FINISH, risponde mostrando una message box che ci informa che il calcolo e' stato completato. 
Potete creare piu' threads in successione usando "Create Thread" diverse volte. In questo esempio, la 
comunicazione e' univoca perche' solo il thread puo' notificare qualcosa alla finestra principale. Se volete 
che il thread primario sia in grado di mandare comandi al worker thread, potete farlo cosi': 

 aggiungete un menu item che dice qualcosa come "Kill Thread" al menu. 

 aggiungete un variabile globale che viene usata come command flag. TRUE=Ferma il thread, 
FALSE=Continua il thread 

 Modificate ThreadProc e inserite nel loop un check del valore della command flag. 

Quando l'utente sceglie il menu item "Kill Thread", il programma principale settera' il valore TRUE nella 
command flag. Quando ThreadProc rileva che il valore della command flag e' TRUE, esce dal loop e 
ritorna, terminando il thread. 

Introduzione
Impareremo cos'è un event object e come usarlo in un programma in multithreading. 

Tools
Esempio 

Preliminari
Nel tutorial precedente, ho dimostrato come i thread comunicano tra di loro con un messaggio di windows
personalizzato. Non ho parlato di altri due metodi: le variabili globali e gli event objects. Li utilizzeremo 
entrambi in questo tutorial. Un event object e' come un interruttore: ha solo due stati, ON o OFF. Quando
un event object e' accesso (ON), e' in stato "segnalato". Quando e' spento (OFF), e' instato "non-
segnalato". Voi create un event object e inserite uno spezzone di codice nei thread interessati per 
controllare lo stato dell'event object. Se l'event object e' in stato "non-segnalato", i threads che lo 
aspettano saranno "addormentati". Quando i threads sono in stato di attesa, consumano poco CPU time. 
Creerete un event object chiamando la funzione CreateEvent che ha la seguente sintassi: 

HANDLE CreateEvent(
    LPSECURITY_ATTRIBUTES  lpEventAttributes, // indirizzo degli attributi di sicurezza
    BOOL  bManualReset,                  // flag per l'evento di reset manuale
    BOOL  bInitialState,                 // flag per lo stato iniziale
    LPCTSTR  lpName                      // indirizzo del nome dell'event object
   );

lpEventAttribute --> Se specificate il valore NULL, l'event object e' creato con il descrittore di sicurezza 
predefinito. 

bManualReset --> Se volete che Windows resetti automaticamente l'event object in stato "non-
segnalato" dopo la chiamata a WaitForSingleObject, dovete specificare FALSE in questo parametro. 
Altrimenti, dovete resettare manualmente l'event object con la chiamata a ResetEvent. 

bInitialState --> Se volete che l'event object venga creato in stato "segnalato", specificate TRUE come 
parametro o altrimenti l'event object sara' creato in stato "non-segnalato". 

lpName --> Puntatore ad una stringa ASCIIZ che contiene il nome dell'event object. Questo nome e' 
usato quando volete chiamare OpenEvent. 

Se la chiamata ha successo, restituisce l'handle del nuovo event object, altrimenti restituisce NULL. 
Potete modificare lo stato di un event object con due funzioni API: SetEvent e ResetEvent. La funzione 
SetEvent mette l'event object in stato "segnalato". ResetEvent fa il contrario. Quando un event object 
viene creato, dovete inserire la chiamata a WaitForSingleObject nel thread che deve controllare lo stato 
dell'event object. WaitForSingleObject ha la seguente sintassi: 

DWORD WaitForSingleObject(
    HANDLE  hObject,         // handle dell'object da aspettare
    DWORD  dwTimeout      // intervallo di time-out in millisecondi
   );

hObject --> L'handle di uno dei synchronization objects. L'event object e' un tipo di synchronization 
object. 

dwTimeout --> Specifica il tempo in millisecondi che questa funzione aspettera' per vedere se l'object e'
in stato "segnalato". Se il tempo specificato e' passato e l'event object e' ancora in stato "non-segnalato",
WaitForSingleObject ritorna al caller. Se volete aspettare l'object per un tempo indefinito, dovete 
specificare il valore INFINITE in questo parametro. 

Essay
L'esempio che segue mostra una finestra che aspetta che l'utente selezioni un comando dal menu. Se 
l'utente sceglie "run thread", il thread comincia la savage calculation. Quando ha finito, viene mostrata 
una message box che informa l'utente che il lavoro e' stato completato. Mentre il thread e' in esecuzione, 
l'utente puo' scegliere "stop thread" per fermarlo. 

include windows.inc
includelib user32.lib
includelib kernel32.lib
includelib gdi32.lib

.const
IDM_START_THREAD equ 1
IDM_STOP_THREAD equ 2
IDM_EXIT equ 3
WM_FINISH equ WM_USER+100h

.data
ClassName db "Win32ASMEventClass",0
AppName  db "Win32 ASM Event Example",0
MenuName db "FirstMenu",0



SuccessString db "The calculation is completed!",0
StopString db "The thread is stopped",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
hwnd HANDLE ?
hMenu HANDLE ?
ThreadID DWORD ?
ExitCode DWORD ?
hEventStart HANDLE ?
EventStop BOOL FALSE

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke GetCommandLine
    invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_WINDOW+1
    mov   wc.lpszMenuName,OFFSET MenuName
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,\
            ADDR  AppName,\
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\
           CW_USEDEFAULT,300,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    invoke ShowWindow, hwnd,SW_SHOWNORMAL
    invoke UpdateWindow, hwnd
    invoke GetMenu,hwnd
    mov  hMenu,eax
    .WHILE TRUE
            invoke GetMessage, ADDR msg,NULL,0,0
            .BREAK .IF (!eax)
            invoke TranslateMessage, ADDR msg
            invoke DispatchMessage, ADDR msg
    .ENDW
    mov     eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .IF eax==WM_CREATE
        invoke CreateEvent,NULL,FALSE,FALSE,NULL
        mov  hEventStart,eax
        mov  eax,OFFSET ThreadProc
        invoke CreateThread,NULL,NULL,eax,\
                             NULL,NORMAL_PRIORITY_CLASS,\
                             ADDR ThreadID
        invoke CloseHandle,eax
    .ELSEIF eax==WM_DESTROY
        invoke PostQuitMessage,NULL
    .ELSEIF eax==WM_COMMAND
        mov eax,wParam



        .if lParam==0
            .if ax==IDM_START_THREAD
                invoke SetEvent,hEventStart
                invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_GRAYED
                invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_ENABLED
            .elseif ax==IDM_STOP_THREAD
                mov  EventStop,TRUE
                invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_ENABLED
                invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_GRAYED
            .else
                invoke DestroyWindow,hWnd
            .endif
        .endif
    .ELSEIF eax==WM_FINISH
        invoke MessageBox,NULL,ADDR SuccessString,ADDR AppName,MB_OK
    .ELSE
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
.ENDIF
    xor    eax,eax
    ret
WndProc endp

ThreadProc PROC USES ecx Param:DWORD
        invoke WaitForSingleObject,hEventStart,INFINITE
        mov  ecx,600000000
        .WHILE ecx!=0
                .if EventStop!=TRUE
                        add  eax,eax
                        dec  ecx
                .else
                        invoke MessageBox,hwnd,ADDR StopString,ADDR AppName,MB_OK
                        mov  EventStop,FALSE
                        jmp ThreadProc
                .endif
        .ENDW
        invoke PostMessage,hwnd,WM_FINISH,NULL,NULL
        invoke EnableMenuItem,hMenu,IDM_START_THREAD,MF_ENABLED
        invoke EnableMenuItem,hMenu,IDM_STOP_THREAD,MF_GRAYED
        jmp   ThreadProc
        ret
ThreadProc ENDP
end start

Analisi
In quest'esempio, dimostro un altra tecnica di thread. 

.IF eax==WM_CREATE
        invoke CreateEvent,NULL,FALSE,FALSE,NULL
        mov  hEventStart,eax
        mov  eax,OFFSET ThreadProc
        invoke CreateThread,NULL,NULL,eax,\
                             NULL,NORMAL_PRIORITY_CLASS,\
                             ADDR ThreadID
        invoke CloseHandle,eax

Potete vedere che creo l'event object e il thread durante l'elaborazione del messaggio WM_CREATE. Creo 
l'event object in stato "non-segnalato" con reset automatico. Una volta creato l'event object, creo il 
thread. Comunque il thread non parte immediatamente perche' attende che l'event object sia in stato 
"segnalato" con il codice seguente: 

ThreadProc PROC USES ecx Param:DWORD
        invoke WaitForSingleObject,hEventStart,INFINITE
        mov  ecx,600000000



La prima linea della thread procedure e' la chiamata a WaitForSingleObject. Essa aspetta indefinitamente 
per lo stato "segnalato" dell'event object prima di ritornare. Cio' significa che nonostante noi creiamo il 
thread, lo mettiamo in stato di inattivita'. Quando l'utente sceglie il comando "run thread" dal menu, 
mettiamo l'event object in stato "segnalato" cosi': 

.if ax==IDM_START_THREAD
                invoke SetEvent,hEventStart

La chiamata a SetEvent mette l'event object in stato "segnalato", che a sua volta fa si che la chiamata a 
WaitForSingleObject nella thread procedure ritorni e quindi che il thread cominci ad essere eseguito. 
Quando l'utente seleziona il comando "stop thread", noi settiamo il valore della variabile globale 
"EventStop" a TRUE. 

.if EventStop==FALSE
                        add  eax,eax
                        dec  ecx
                .else
                        invoke MessageBox,hwnd,ADDR StopString,ADDR AppName,MB_OK
                        mov  EventStop,FALSE
                        jmp ThreadProc
                .endif

Cio' blocca il thread e salta di nuovo alla chiamata a WaitForSingleObject. Notate che non dobbiamo 
resettare manualmente l'event object in stato "non-segnalato", poiche' abbiamo specificato il parametro 
bManualReset della chiamata a CreateEvent come FALSE. 

Introduzione
In questo tutorial, impareremo cosa sono le DLLs, e come crearle. 

Tools
Esempio 

Preliminari
Se programmate per un periodo abbastanza lungo, vi renderete conto che i programmi che scrivete 
hanno alcune routines di codice in comune. E' un enorme spreco di tempo riscriverle ogni volta che 
cominciate un nuovo programma. Ai tempi del vecchio DOS, i programmatori conservavano le routines 
usate piu' comunemente in una o piu' librerie. Quando volevano usare le funzioni, non dovevano far altro 
che linkare le librerie all'object file e il linker estraeva le funzioni dalle librerie e le inseriva nel file 
eseguibile finale. Questo processo e' chiamato "static linking". Le C runtime libraries sono un buon 
esempio. Il rovescio della medaglia di questo metodo e' che dovete avere funzioni identiche in ogni 
programma che le chiama. Lo spazio sul disco e' sprecato nel conservare diverse copie uguali delle 
funzioni. Ma, per i programmi DOS, questo metodo e' abbastanza accettabile poiche' c'e' solitamente un 
solo programma attivo in memoria. Percio' non c'e' nessuno spreco di preziosa memoria.Sotto Windows, 
la situazione diventa molto piu' critica perche' potete avere diverse copie del programma in esecuzione 
simultanea. La memoria sara' mangiata molto velocemente se il vostro programma e' abbastanza grande.
Windows ha una soluzione per questo tipo di problema: le dynamic link libraries. Una dynamic link library
e' una specie di "raccolta" di funzioni comuni. Windows non carichera' diverse copie di una DLL in 

memoria, percio' anche se ci sono diverse copie del vostro programma in esecuzione simultanea, in 
memoria ci sara' sempre solo una copia della DLL che il programma usa. Il programma si linka ad una 
DLL durante l'esecuzione (runtime) a differenza delle vecchie static libraries. Ecco perche' le librerie 
vengono chiamate dynamic link libraries. Potete anche scaricare dalla memoria una DLL durante 
l'esecuzione quando non ne avete bisogno. Se quel programma e' l'unico che usa quella libreria, questa 
sara' scaricata immediatamente dalla memoria. Ma se la DLL e' ancora usata da qualche altro 
programma, essa rimarra' in memoria finche' l'ultimo programma che se ne serve non la scarica. 
Comunque, il linker ha un lavoro piu' difficile da compiere quando esegue gli address fixups (correzione 
degli indirizzi, NdT) per il file eseguibile finale. Poiche' non puo' "estrarre" le funzioni e inserirle nel file 
eseguibile finale, in qualche modo vi deve inserire abbastanza informazioni riguardanti la DLL e funzioni 
che gli permettano di localizzare e caricare la corretta DLL durante l'esecuzione. E' qui che entra in gioco 
la import library. Una import library contiene le informazioni riguardanti la DLL che rappresenta. Il linker 
puo' estrarre le informazioni di cui ha bisogno dalle import libraries e inserirle nel file eseguibile. Quando 
il loader di Windows carica il programma in memoria, vede che quest'ultimo si linka ad una DLL, e percio'
cerca questa DLL e la carica nell'address space del processo ed esegue gli address fixups per le chiamate 
alle funzioni nella DLL. Potete scegliere di caricare voi stessi la DLL senza basarvi sul loader di Windows. 
Questo metodo ha i suoi pro e i suoi contro: 

 Non necessita di una import library, percio' potete caricare ed usare qualsiasi DLL anche se non 
c'e' una relativa import library. Comunque, dovete sempre conoscere le funzioni al suo interno, 
quanti parametri richiedono e cosi' via. 

 Quando incaricate il loader di caricare la DLL per il vostro programma, se il loader non riesce a 
trovare la DLL vi riportera' "Una file .DLL necessario al programma, xxxxx.dll e' mancante" e 
puf! Il vostro programma non potra' mai partire anche se quella DLL non e' essenziale per le 
sue operazioni. Se caricate la DLL manualmente, invece, quando la DLL non viene trovata e non
e' essenziale alle operazioni, il vostro programma puo' semplicemente comunicare la cosa 
all'utente e continuare con l'esecuzione. 

 Potete chiamare funzioni *non documentate* che non sono incluse nelle import libraries. 
Ammesso ovviamente che abbiate alcune informazioni riguardo a tali funzioni. 

 Se usate LoadLibrary, dovete chiamare GetProcAddress per ciascuna funzione che volete 
chiamare. GetProcAddress trova l'entrypoint address di una funzione in una particolare DLL. 
Percio' il vostro codice potrebbe essere un po' piu' ingombrante e lento, ma non di molto. 

Essay
Visti i vantaggi/svantaggi della chiamata a LoadLibrary, entriamo ora nei dettagli di come creare una DLL.
Il codice che segue e' lo scheletro di una DLL. 

;---------------------------------------------------------------------------------------------
-----------
;                                         DLLSkeleton.asm
;---------------------------------------------------------------------------------------------
-----------
include windows.inc
includelib user32.lib
includelib kernel32.lib

.data

.code
DllEntry proc hInstDLL:HINSTANCE, reason:DWORD, reserved:DWORD
        mov  eax,TRUE
        ret
DllEntry Endp
;---------------------------------------------------------------------------------------------
-----------
;                                         Questa e' una funzione inutile
; Non esegue nessuna azione. L'ho messa qui per mostrarvi dove potete inserire le funzioni
; in una DLL.
;---------------------------------------------------------------------------------------------
-----------
TestFunction proc
    ret



TestFunction endp

End DllEntry

;---------------------------------------------------------------------------------------------
-----------
;                                          DLLSkeleton.def
;---------------------------------------------------------------------------------------------
-----------
LIBRARY   DLLSkeleton
EXPORTS   TestFunction

Il programma precedente e' lo scheletro della DLL. Ogni DLL deve avere una funzione di entrypoint. 
Windows chiamera' la funzione di entrypoint ogni volta che: 

 La DLL e' caricata per la prima volta 

 La DLL e' scaricata dalla memoria 

 Un thread viene creato nello stesso processo 

 Un thread viene distrutto stesso processo 
DllEntry proc hInstDLL:HINSTANCE, reason:DWORD, reserved:DWORD
        mov  eax,TRUE
        ret
DllEntry Endp

Potete chiamare la funzione di entrypoint in qualsiasi modo, basta che poi inserite un relativo END 
<Nome della funzione di Entrypoint>. Questa funzione prende tre parametri, dei quali solo i primi due 
sono importanti. 

hInstDLL e' il module handle della DLL. Non e' lo stesso dell'instance handle del processo. 

reason puo' essere uno dei seguenti quattro valori: 

 DLL_PROCESS_ATTACH La DLL riceve questo valore quando e' "iniettata" per la prima volta 
nell'address space del processo. Potete sfruttare questa opportunita' per inzializzare quello che 
vi serve. 

 DLL_PROCESS_DETACH La DLL riceve questo valore quando viene scaricata dall'address space 
del processo. Potete usare questa opportunita' per eseguire un po' di "pulizia" come la 
deallocazione della memoria e cosi' via. 

 DLL_THREAD_ATTACH La DLL riceve questo valore quando il processo crea un nuovo thread. 

 DLL_THREAD_DETACH La DLL riceve questo valore quando un thread nel processo viene 
distrutto. 

Voi restituite TRUE in eax se volete che la DLL continui nell'esecuzione. Se restituite FALSE, la DLL non 
verra' caricata. Ad esempio, se il vostro codice di inizializzazione deve allocare della memoria e non riesce
a farlo con successo, la funzione di entrypoint dovrebbe restituire FALSE per indicare che la DLL non puo' 
essere eseguita. Potete inserire le vostre funzioni nella DLL prima o dopo la funzione di entrypoint. Ma se 
volete che esse siano "chiamabili" dagli altri programmi, dovete metterne i nomi nella lista di export nel 
file di definizione di modulo (module definition file, .def). Una DLL ha bisogno di un file di definizione di 
modulo nella sua fase di sviluppo. Vi daremo un'occhiata adesso. 

LIBRARY   DLLSkeleton
EXPORTS   TestFunction

Normalmente dovete inserire la prima linea. Lo statement LIBRARY definisce il nome del modulo interno 
della DLL. Dovreste farlo combaciare con il nome del file della DLL. Lo statement EXPORTS dice al linker 
quali funzioni della DLL vengono esportate, cioe' saranno "chiamabili" da altri programmi. Nell'esempio, 
vogliamo che altri programmi possano chiamare la funzione TestFunction, percio' ne inseriamo il nome 
dopo lo statement EXPORTS. Un altro cambiamento e' lo switch del linker. Dovete usare lo switch /'DLL 
e /DEF:<il vostro file def> tra gli switches del linker, come in questo modo: 

link /DLL /SUBSYSTEM:WINDOWS /DEF:DLLSkeleton.def /LIBPATH:c:\masm\lib DLLSkeleton.obj



Gli switches dell'assembler sono gli stessi, cioe' /c /coff /Cp. Percio' dopo che avete linkato l'object file, 
otterrete .dll e .lib. Il file .lib e' la import library che potete usare per linkare altri programmi che usano le
funzioni nella DLL. Adesso vi mostrero' come usare LoadLibrary per caricare una DLL. 

;---------------------------------------------------------------------------------------------
-----------
;                                               UseDLL.asm
;---------------------------------------------------------------------------------------------
-----------
include windows.inc
includelib kernel32.lib
includelib user32.lib

.data
LibName db "DLLSkeleton.dll",0
FunctionName db "TestHello",0
DllNotFound db "Cannot load library",0
AppName db "Load Library",0
FunctionNotFound db "TestHello function not found",0

.data?
hLib dd ?                                         ; l'handle della libreria (DLL)
TestHelloAddr dd ?                        ; l'indirizzo della funzione TestHello

.code
start:
        invoke LoadLibrary,addr LibName
;---------------------------------------------------------------------------------------------
-----------
; Chiamate LoadLibrary con il nome della DLL desiderata. Se la chiamata ha
; successo, restituira' l'handle della libreria (DLL). Se no, restituira' NULL.
; Potete passare l'handle della libreria a GetProcAddress o a qualsiasi funzione
; che richiede l'handle di una libreria come parametro.
;---------------------------------------------------------------------------------------------
-----------
        .if eax==NULL
                invoke MessageBox,NULL,addr DllNotFound,addr AppName,MB_OK
        .else
                mov hLib,eax
                invoke GetProcAddress,hLib,addr FunctionName
;---------------------------------------------------------------------------------------------
-----------
; Quando ottenete l'handle della libreria, lo passate a GetProcAddress con l'indirizzo
; del nome della funzione che volete chiamare in quella DLL. Questa restituisce
; l'indirizzo della funzione, se ha successo. Altrimenti, restituisce NULL.
; Gli indirizzi delle funzioni non cambiano a meno che voi non scarichiate e ricarichiate
; la libreria. Pertanto, potete inserirli in variabili globali per uso futuro.
;---------------------------------------------------------------------------------------------
-----------
                .if eax==NULL
                        invoke MessageBox,NULL,addr FunctionNotFound,addr AppName,MB_OK
                .else
                        mov TestHelloAddr,eax
                        call [TestHelloAddr]
;---------------------------------------------------------------------------------------------
-----------
; Successivamente, potete chiamare la funzione con una semplice call con la variabile
; contenente l'indirizzo della funzione come operando.
;---------------------------------------------------------------------------------------------
-----------
                .endif
                invoke FreeLibrary,hLib
;---------------------------------------------------------------------------------------------
-----------
; Quando non avete piu' bisogno della libreria, scaricatela con FreeLibrary.
;---------------------------------------------------------------------------------------------
-----------
        .endif
        invoke ExitProcess,NULL
end start

Percio' potete osservare che l'utilizzo di LoadLibrary e' un po' piu' impegnativo ma e' anche piu' flessibile.

Introduzione
Impareremo cosa sono i controlli comuni (Common Controls) e come usarli. Questo tutorial sara' 
semplicemente una breve introduzione ad essi. Notate che questo esempio usa windows.inc dal package 
di MASM32. Ricordo che MASM32 potete scaricarlo da qui 

Tools
Esempio 

Essay
Windows 95 contiene diversi miglioramenti nell'interfaccia utente rispetto a Windows 3.1x. Essi rendono 
la GUI piu' ricca. Molti di questi erano gia' largamente usati prima che Windows 95 arrivasse nei negozi, 
come la status bar, le toolbars etc. I programmatori dovevano programmarli da soli. Ora la Microsoft li ha
inclusi in Windows 9x e NT. Li tratteremo qui. Questi sono i nuovi controlli: 

 Toolbar 

 Tooltip 

 Status bar 

 Property sheet 

 Property page 

 Tree view 

 List view 

 Animation 

 Drag list 

 Header 

 Hot-key 

 Image list 

 Progress bar 

 Right edit 

 Tab 

 Trackbar 

 Up-down 

Visto che ce ne sono diversi, caricarli tutti contemporaneamente in memoria e registrarli sarebbe uno 
spreco di risorse. Tutti questi controlli, ad eccezione del rich edit control, sono contenuti in comctl32.dll 
che le applicazioni possono caricare quando vogliono usare i controlli. Il rich edit control risiede nella sua 
propria dll, richedXX.dll, poiche' e' molto ingombrante e complicato. Potete caricare comctl32.dll 
includendo una call a InitCommonControls nel vostro programma. InitCommonControls e' una funzione
in comctl32.dll, cosi', riferendosi ad essa in un punto qualsiasi nel vostro codice, farete in modo che il PE 
loader carichi comctl32.dll quando il vostro programma parte. Non dovete eseguirla, semplicemente 
includetela nel vostro codice da qualche parte. Questa funzione non fa NIENTE! La sua unica 
istruzione e' "ret". Il suo unico scopo e' includere una referenza a comctl32.dll nella import section 
cosicche' il PE loader la carichi quando il progamma viene eseguito. Il vero cavallo di battaglia e' la 
funzione di entrypoint della DLL che registra tutte le classi quando la dll e' caricata. i controlli comuni 
vengono creati basandosi su quelle classi proprio come gli altri controlli di child window come edit, listbox
etc. Il rich edit e' tutto un altro discorso. Se volete usarlo, dovete chiamare LoadLibrary per caricarlo 
esplicitamente e poi chiamare FreeLibrary per scaricarlo. Ora sappiamo come crearli. Potete usare un 
editor di risorse per incorporarli nei dialog boxes o potete crearli da soli. Praticamente tutti i controlli 
comuni vengono creati chiamando CreateWindowEx o CreateWindow, a cui viene passato il nome della 
classe del controllo. Alcuni controlli comuni hanno funzioni di creazione specifiche, che comunque sono 
solo dei wrappers (un po' come un file batch che chiama un exe con dei parametri, NdT) attorno a 
CreateWindowEx per rendere piu' semplice la creazione di tali controlli. Le funzioni di creazione esistenti 
sono elencate qui sotto: 

 CreateToolbarEx 

 CreateStatusWindow 

 CreatePropertySheetPage 

 PropertySheet 

 ImageList_Create 

Per creare controlli comuni, dovete conoscere i nomi delle loro class. Sono elencati qui: 

Nome Class Common Control 

ToolbarWindow32 Toolbar 

tooltips_class32 Tooltip 

msctls_statusbar32 Status bar 

SysTreeView32 Tree view 

SysListView32 List view 

SysAnimate32 Animation 

SysHeader32 Header 

msctls_hotkey32 Hot-key 

msctls_progress32 Progress bar 

RICHEDIT Rich edit 

msctls_updown32 Up-down 

SysTabControl32 Tab 



I controlli property sheets e property pages, e image list hanno le loro specifiche funzioni di creazione. Il 
controllo drag list e' una listbox modificata quindi non ha la propria classe. I precedenti nomi di class 
sono verificati controllando lo script di risorse generato dall'editor di risorse del Visual C++. Essi 
differiscono dai nomi delle class elencati nella win32 api reference della Borland e in Programming 
Windows 95 di Charles Petzold. La lista che ho inserito qui e' quella accurata. Questi controlli comuni 
possono usare stili generici di finestre come WS_CHILD etc. Inoltre hanno stili specifici come TVS_XXXXX 
per il tree view control, LVS_xxxx per il list view control, etc. La Win32 api reference e' la vostra migliore 
amica per quanto riguarda questo. Ora che sappiamo come creare i controlli comuni, possiamo spostarci 
sui metodi di comunicazione tra i controlli comuni e i loro genitori. A differenza dei controlli di child 
window, i controlli comuni non comunicano con i genitori via WM_COMMAND. Al contrario, mandano 
messaggi WM_NOTIFY alla parent window quando qualche evento interessante avviene per i controlli 
comuni. Il genitore puo' controllare i figli mandandogli messaggi. Ci sono anche molti nuovi messaggi per
questi nuovi controlli. Dovreste consultare la vostra win32 api reference per maggiori dettagli. 
Esaminiamo i controlli progress bar e status bar nell'esempio seguente. 

CODICE CAMPIONE 

.386

.model flat,stdcall
include windows.inc
include user32.inc
include kernel32.inc
include comctl32.inc
includelib comctl32.lib
includelib user32.lib
includelib kernel32.lib

WinMain PROTO :DWORD,:DWORD,:DWORD,:SDWORD

.const
IDC_PROGRESS equ 1            ; control IDs
IDC_STATUS equ 2
IDC_TIMER  equ 3

.data
ClassName  db "CommonControlWinClass",0
AppName    db "Common Control Demo",0
ProgressClass  db "msctls_progress32",0        ; il class name della progress bar
Message  db "Finished!",0
TimerID  dd 0

.data?
hInstance  HINSTANCE ?
hwndProgress dd ?
hwndStatus dd ?
CurrentStep dd ?
.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
    invoke ExitProcess,eax
    invoke InitCommonControls

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:SDWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND

    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInstance
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_APPWORKSPACE
    mov   wc.lpszMenuName,NULL



    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,0
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
WS_OVERLAPPED+WS_CAPTION+WS_SYSMENU+WS_MINIMIZEBOX+WS_MAXIMIZEBOX+WS_VISIBLE,CW_USEDEFAULT,\
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    .while TRUE
         invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .endw
    mov eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    mov   eax,uMsg
    .if eax==WM_CREATE
         invoke CreateWindowEx,NULL,ADDR ProgressClass,NULL,\
            WS_CHILD+WS_VISIBLE,100,\
            200,300,20,hWnd,IDC_PROGRESS,\
            hInstance,NULL
        mov hwndProgress,eax
        mov eax,1000                ; il lParam del messaggio PBM_SETRANGE contiene il raggio
        mov CurrentStep,eax
        shl eax,16                    ; l'high range e' nella high word
        invoke SendMessage,hwndProgress,PBM_SETRANGE,0,eax
        invoke SendMessage,hwndProgress,PBM_SETSTEP,10,0
        invoke CreateStatusWindow,WS_CHILD+WS_VISIBLE,NULL,hWnd,IDC_STATUS
        mov hwndStatus,eax
        invoke SetTimer,hWnd,IDC_TIMER,100,NULL        ; crea un timer
        mov TimerID,eax
    .elseif eax==WM_DESTROY
        invoke PostQuitMessage,NULL
        .if TimerID!=0
            invoke KillTimer,hWnd,TimerID
        .endif
    .elseif eax==WM_TIMER        ; quando un avviene un timer event
        invoke SendMessage,hwndProgress,PBM_STEPIT,0,0    ; aumenta il progresso nella 
progress bar
        sub CurrentStep,10
        .if CurrentStep==0
            invoke KillTimer,hWnd,TimerID
            mov TimerID,0
            invoke SendMessage,hwndStatus,SB_SETTEXT,0,addr Message
            invoke MessageBox,hWnd,addr Message,addr AppName,MB_OK+MB_ICONINFORMATION
            invoke SendMessage,hwndStatus,SB_SETTEXT,0,0
            invoke SendMessage,hwndProgress,PBM_SETPOS,0,0
        .endif
    .else
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
WndProc endp
end start

Analisi



invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
          invoke ExitProcess,eax
          invoke InitCommonControls

Ho deliberatamente messo InitCommonControls dopo ExitProcess per dimostrare che 
InitCommonControls e' li' solo per dare un riferimento a comctl32.dll nella import section. Come potete 
vedere, i controlli comuni funzionano anche se InitCommonControls non viene eseguito. 

.if eax==WM_CREATE
               invoke CreateWindowEx,NULL,ADDR ProgressClass,NULL,\
                  WS_CHILD+WS_VISIBLE,100,\
                  200,300,20,hWnd,IDC_PROGRESS,\
                  hInstance,NULL
              mov hwndProgress,eax

Qui e' dove creiamo il controllo comune. Notate che questa call a CreateWindowEx contiene come hWnd 
quello della parent window. Essa specifica anche una control ID per identificare questo controllo. 
Comunque, visto che abbiamo il window handle del controllo, questa ID non e' usata. Tutti i controlli di 
child window devono avere lo stile WS_CHILD. 

mov eax,1000
              mov CurrentStep,eax
              shl eax,16
              invoke SendMessage,hwndProgress,PBM_SETRANGE,0,eax
              invoke SendMessage,hwndProgress,PBM_SETSTEP,10,0

Dopo che la progress bar e' stata creata, ne settiamo il raggio. Il raggio predefinito e' da 0 a 100. Se non 
vi soddisfa, potete specificare il vostro raggio con il messaggio PBM_SETRANGE. Il lParam di questo 
messaggio contiene il raggio, quello massimo nella high word e quello minimo nella low word. Potete 
specificare quanto tempo richiede un singolo passo usando il messaggio PBM_SETSTEP. L'esempio lo 
setta a 10 che significa che quando mandate un messaggio PBM_STEPIT alla progress bar, l'indicatore di 
progresso aumentera' di 10. Potete anche configurare il vostro proprio livello di indicatore mandando 
messaggi PBM_SETPOS. Questo messaggio vi da' maggiore controllo sulla progress bar. 

invoke CreateStatusWindow,WS_CHILD+WS_VISIBLE,NULL,hWnd,IDC_STATUS
              mov hwndStatus,eax
              invoke SetTimer,hWnd,IDC_TIMER,100,NULL        ; crea un timer
              mov TimerID,eax

Successivamente, creiamo una status bar chiamando CreateStatusWindow. Questa call e' semplice da 
capire percio' non la commentero'. Dopo che la status window e' creata, creiamo un timer. In questo 
esempio, aggiorneremo la progress bar ad un intervallo regolare di 100 ms, percio' dobbiamo creare un 
controllo timer. Qui sotto c'e' il prototipo di funzione per SetTimer. 

SetTimer PROTO hWnd:DWORD, TimerID:DWORD, TimeInterval:DWORD, lpTimerProc:DWORD

hWnd : handle della parent window 

TimerID : un identificatore di timer (nonzero). Potete creare il vostro identificatore personale. 

TimerInterval : l'intervallo in millisecondi che deve passare prima che il timer chiami la timer procedure
o mandi un messaggio WM_TIMER 

lpTimerProc : l'indirizzo della timer function che sara' chiamata quando l'intervallo del timer e' passato. 
Se invece questo parametro e' NULL, il timer mandera' un messaggio WM_TIMER alla parent window. 

Se questa call riesce, la funzione restituisce il TimerID. Se fallisce, restituisce 0. E' per questo che il timer
identifer deve essere diverso da zero. 

.elseif eax==WM_TIMER
              invoke SendMessage,hwndProgress,PBM_STEPIT,0,0
              sub CurrentStep,10
              .if CurrentStep==0
                  invoke KillTimer,hWnd,TimerID



                  mov TimerID,0
                  invoke SendMessage,hwndStatus,SB_SETTEXT,0,addr Message
                  invoke MessageBox,hWnd,addr Message,addr AppName,MB_OK+MB_ICONINFORMATION
                  invoke SendMessage,hwndStatus,SB_SETTEXT,0,0
                  invoke SendMessage,hwndProgress,PBM_SETPOS,0,0
              .endif

Quando l'intervallo di tempo specificato termina, il timer manda un messaggio WM_TIMER. Metterete il 
codice che deve essere eseguito qui. In questo esempio, aggiorniamo la progress bar e quindi 
controlliamo se il limite massimo e' stato raggiunto. Se si, eliminiamo il timer e quindi settiamo il testo 
nella status window con il messaggio SB_SETTEXT. Una message box viene mostrata e quando l'utente 
clicka OK, cancelliamo il testo nella status bar e la progress bar. 

Introduzione
In questo tutorial, impareremo come usare un tree view control. Inoltre, impareremo anche come fare il 
drag and drop sotto un tree view control e come usare una lista di immagini con esso. 

Tools
Esempio 

Essay
Un tree view control è un genere speciale di finestra che rappresenta gli oggetti nell'ordine gerarchico. Un
esempio di esso è il pannello di sinistra di Windows Explorer. Potete usare questo controllo per mostrare i
rapporti fra gli oggetti. Potete creare un tree view control chiamando CreateWindowEx, passando " 
SysTreeView32 " come nome della classe o incorporandolo in una finestra di dialogo. Non dimenticatevi di
mettere la chiamata a InitCommonControls nel vostro codice. Ci sono parecchi stili specifici per il tree 
view control. Questi tre sono quelli principalmente usati. 

TVS_HASBUTTONS: Visualizza i tasti più (+) e meno (-) vicino alle voci principali. L'utente clicca sui 
tasti per espandere o sprofondare la lista di voci figlie di una voce principale. Per includere i tasti con le 
voci alla radice di un albero, TVS_LINESATROOT deve essere specificato. 

TVS_HASLINES: Usa righe per mostrare la gerarchia delle voci. 

TVS_LINESATROOT: Usa righe per collegare le voci alla radice dell'albero. Questo valore è ignorato se 
TVS_HASLINES non è specificato. 

Il tree view control, come altri common controls, comunica con la finestra generatrice tramite messaggi. 
La finestra generatrice può trasmettere i vari messaggi ad esso ed il tree view control può trasmettere i 
messaggi " di notifica " alla relativa finestra generatrice. A questo riguardo, il tree view control non è 
differente da qualsiasi finestra. Quando qualcosa che lo interessa avviene, trasmette un messaggio 
WM_NOTIFY alla finestra generatrice accompagnato da informazioni. 

WM_NOTIFY 

wParam: Control ID, questo valore non è garantito che sia unico così noi non lo usiamo. Invece, usiamo 
hwndFrom o IDFrom membro della struttura NMHDR puntata da lParam 

lParam: Puntatore alla struttura NMHDR. Alcuni comandi possono passare un puntatore a una struttura 
più grande ma essa deve avere una struttura NMHDR come relativo primo membro. Quindi, quando avete
lParam, potete essere sicuri che punta almeno a una struttura NMHDR. 

(Date uno sguardo all'API Reference per maggiori informazioni :D NdT) 

Esaminiamo la struttura NMHDR. 

NMHDR struct    DWORD
          hwndFrom    DWORD ?
          idFrom      DWORD ?
          code        DWORD ?
      NMHDR ends

hwndFrom è l' handle del controllo che trasmette il messaggio WM_NOTIFY. 

idFrom è l'ID del controllo che trasmette questo messaggio WM_NOTIFY. 

code è il messaggio attuale che il controllo desidera trasmettere alla finestra generatrice. 

Le notifiche di un tree view sono quelle con TVN_ all'inizio del nome. I messaggi del tree view sono quelli 
con TVM_, come TVM_CREATEDRAGIMAGE. Il tree view control trasmette TVN_xxxx nel membro code di 
NMHDR. La finestra generatrice può trasmettere TVM_xxxx per gestirlo. 

Aggiunta di voci ad un tree view control

Dopo che avete creato un tree view control, potete aggiungere delle voci ad esso. Potete fare questo 
inviando TVM_INSERTITEM ad esso. 

TVM_INSERTITEM 

wParam = 0; 

lParam = puntatore a una TV_INSERTSTRUCT; 

Avete bisogno di conoscere una certa terminologia a questo punto circa il rapporto fra le voci nel tree 
view control. Una voce può essere principale, figlia, o entrambe allo stesso tempo. Una voce principale è 
una voce che ha un certo numero di sottovoci connesse ad essa. Allo stesso tempo, la voce principale può
essere figlia di una certa altra voce. Una voce senza un genitore è chiamata voce di radice. Ci possono 
essere molte voci di radice in un tree view control. Ora esamineremo la struttura TV_INSERTSTRUCT 



TV_INSERTSTRUCT STRUCT DWORD
        hParent       DWORD ?
        hInsertAfter  DWORD ?
                            ITEMTYPE <>
      TV_INSERTSTRUCT ENDS

hParent: Handle della voce principale. Se questo membro è il valore TVI_ROOT o NULL, la voce è 
inserita alla radice del tree-view control. 

hInsertAfter: Handle della voce dopo che una nuova voce è stata inserita o uno dei seguenti valori: 

 TVI_FIRST ==> Inserisce la voce all'inizio della lista. 

 TVI_LAST ==> Inserisce la voce alla fine della lista. 

 TVI_SORT ==> Inserisce la voce nella lista in ordine alfabetico. 

ITEMTYPE UNION
              itemex TVITEMEX <>
              item TVITEM <>
      ITEMTYPE ENDS

Useremo soltanto TVITEM. 

TV_ITEM STRUCT DWORD
        imask             DWORD      ?
        hItem             DWORD      ?
        state             DWORD      ?
        stateMask         DWORD      ?
        pszText           DWORD      ?
        cchTextMax        DWORD      ?
        iImage            DWORD      ?
        iSelectedImage    DWORD      ?
        cChildren         DWORD      ?
        lParam            DWORD      ?
      TV_ITEM ENDS

Questa struttura è usata per trasmettere o ricevere informazioni circa una voce del tree view, secondo i 
messaggi. Per esempio, con TVM_INSERTITEM, è usata per specificare gli attributi della voce da 
inserire nel tree view control. Con TVM_GETITEM, sarà riempita di informazioni sulla voce selezionata 
nel tree view. 

imask è usato per specificare quale(i) membro(i) della struttura TV_ITEM è(sono) valido(i). Per 
esempio, se il valore di imask è TVIF_TEXT , significa che soltanto il membro pszText è valido. Potete 
unire insieme parecchi flags. 

hItem è l' handle di una voce del tree view. Ogni voce ha una propria handle, come l' handle di una 
finestra. Se desiderate fare qualcosa con una voce, dovete selezionarla tramite la relativa handle. 

pszText è il puntatore ad una stringa null-terminata che è l'etichetta di una voce del tree view. 

cchTextMax è usato soltanto quando desiderate richiamare l'etichetta di una voce del tree view. Poichè 
fornirete il puntatore a un buffer in pszText, Window deve conoscere la dimensione del buffer fornito. 
Dovete specificare la dimensione in questo membro. 

iImage ed iSelectedImage si riferiscono all'indice in una lista di immagini che contiene le immagini da 
mostrare quando la voce non è selezionata e quando è selezionata. Se ricordate il pannello di sinistra di 
Windows Explorer, le immagini delle cartelle sono specificate da questi due membri. Per inserire una voce
nel tree view control, dovete riempire almeno hParent, hInsertAfter e dovreste riempire anche imask e 
pszText. 

Aggiunta di immagini al tree view control



Se desiderate mettere un'immagine alla sinistra dell'etichetta di una voce del tree view, dovete creare 
una lista di immagine ed associarla al tree view control. Potete creare una lista di immagine chiamando 
ImageList_Create. 

ImageList_Create PROTO 
          cx:DWORD, 
          cy:DWORD, 
          flags:DWORD,
          cInitial:DWORD,
          cGrow:DWORD

Questa funzione se riuscita restituisce l' handle ad una lista vuota di immagini. cx: larghezza di ogni 
immagine in questa lista di immagini, in pixel. 

cy: altezza di ogni immagine in questa lista di immagini, in pixel. Ogni immagine in una lista di immagini 
deve essere uguale l'un l'altra nel formato. Se specificate una bitmap più grande, Window userà cx e cy 
per * tagliarla * in parecchie parti. Così dovreste preparare la vostra immagine come striscie di immagini 
con le dimensioni identiche. 

flags: specifica i tipi di immagini in questa lista ,se sono a colori o monocromatiche e la loro intensità di 
colore. Consultate la vostra documentazione win32 api per più particolari. 

cInitial: il numero di immagini che questa lista inizialmente conterrà. Window userà questa info per 
allocare la memoria per le immagini. 

cGrow: la quantità di immagini che la lista può incorporare quando il sistema deve ridimensionarla per 
inserire nuove immagini. Questo parametro rappresenta il numero di nuove immagini che la lista 
ridimensionata può contenere. 

Una lista di immagine non è una finestra! È soltanto un deposito di immagini ad uso di altre finestre. 
Dopo che una lista di immagine è stata creata, potete aggiungere le immagini ad essa chiamando 
ImageList_Add 

ImageList_Add PROTO himl:DWORD, hbmImage:DWORD, hbmMask:DWORD

Questa funzione restituisce -1 se infruttuosa. 

himl: handle della lista di immagine a cui desiderate aggiungere l' immagine. È il valore restituito da una 
chiamata riuscita a ImageList_Create 

hbmImage: handle della bitmap da aggiungere alla lista di immagini. Solitamente memorizzate la 
bitmap nelle risorse e la caricate con la chiamata di LoadBitmap. Si noti che non dovete specificare il 
numero di immagini contenute in questa bitmap perché questa informazione è determinata dai parametri 
cx e cy passati da ImageList_Create. 

hbmMask: handle della bitmask. Se nessuna maschera è usata con la lista di immagini, questo 
parametro è ignorato. 

Normalmente, aggiungeremo soltanto due immagini alla lista per usarle con il tree view control: una 
usata quando la voce del tree view non è selezionata e l'altra quando è selezionata. Quando la lista di 
immagini è pronta, la associate con il tree view control trasmettendo TVM_SETIMAGELIST al tree view 
control. 

TVM_SETIMAGELIST 

wParam = tipo di lista di immagini da settare. Ci sono due scelte: 

 TVSIL_NORMAL Setta la lista di immagini normali, che contiene le immagini selezionata e non 
selezionata per una voce del tree-view. 

 TVSIL_STATE Setta la lista di immagini condizionate, che contiene le immagini per le voci del 
tree-view che sono in una condizione prestabilita dall'utente. 



lParam = handle della lista di immagini 

Richiamare le informazioni circa una voce del tree view

Potete richiamare le informazioni di una voce del tree view trasmettendo il messaggio TVM_GETITEM. 

TVM_GETITEM 

wParam = 0 

lParam = puntatore alla struttura TV_ITEM da riempire con le informazioni 

Prima che trasmettiate questo messaggio, dovete riempire il membro imask con i flag(s) che specificano 
quale membro(i) di TV_ITEM desiderate che Window riempia. E più importante, dovete riempire hItem 
con l'handle della voce di cui desiderate ottenere le informazioni. E questo crea un problema: Come 
potete conoscere l'handle della voce di cui desiderate richiamare l'Info? Dovrete memorizzare tutte 
l'handle del tree view? La risposta è abbastanza semplice: non dovete. Potete trasmettere il messaggio 
TVM_GETNEXTITEM al tree view control per ottenere l'handle della voce del tree view che ha gli 
attributi che avete specificato. Per esempio, potete chiedere l'handle della prima voce figlia, della voce 
alla radice, della voce selezionata e così via. 

TVM_GETNEXTITEM 

wParam = flag 

lParam = handle per una voce del tree view (necessaria soltanto per alcuni valori di flag) 

Il valore in wParam è molto importante. Presento tutti i flags qui sotto: 

 TVGN_CARET Ritorna la voce attualmente selezionata. 

 TVGN_CHILD Ritorna la prima voce figlia della voce specificata dal parametro hItem 

 TVGN_DROPHILITE Ritorna la voce che è l'obiettivo di un'operazione di drag-and-drop. 

 TVGN_FIRSTVISIBLE Ritorna la prima voce visibile. 

 TVGN_NEXT Ritorna la successiva voce figlia. 

 TVGN_NEXTVISIBLE Ritorna la successiva voce visibile seguente a quella specificata. La voce 
specificata deve essere visibile. Usate il messaggio TVM_GETITEMRECT per determinare se una 
voce è visibile. 

 TVGN_PARENT Ritorna la voce genitore della voce specificata. 

 TVGN_PREVIOUS Ritorna la precedente voce figlia. 

 TVGN_PREVIOUSVISIBLE Ritorna la prima voce visibile che precede quella specificata. La voce
specificata deve essere visibile. Usate il messaggio TVM_GETITEMRECT per determinare se una 
voce è visibile. 

 TVGN_ROOT Ritorna la voce radice del tree-view control. 

Si può notare che, potete richiamare l'handle della voce del tree view a cui siete interessati tramite 
questo messaggio. SendMessage restituisce l'handle della voce del tree view se riuscita. Potete allora 
inserire l'handle restituita nel membro hItem di TV_ITEM da usare con il messaggio TVM_GETITEM. 

Operazione di Drag and Drop nel tree view control

Questa parte è il motivo per cui ho deciso di scrivere questo tutorial. Quando ho provato a seguire 
l'esempio del riferimento delle api win32 ( win32.hlp di InPrise), sono stato molto frustrato perché le 
informazioni vitali sono difettose. Dalle prove e dagli errori, infine ho capito come effettuare il drag & 
drop in un tree view control e non auguro a nessuno di percorrere la stessa strada che ho fatto io. Sotto è
implementato in punti il drag & drop in un tree view control. 



1) Quando l'utente prova a trascinare una voce, il tree view control trasmette la notifica 
TVN_BEGINDRAG alla finestra generatrice. Potete usare questa occasione per creare un'immagine per il
drag che è l'immagine che sarà usata per rappresentare la voce mentre è trascinata. Potete trasmettere 
TVM_CREATEDRAGIMAGE al tree view control per dirgli di creare un'immagine di drag per l'immagine 
che attualmente è usata dalla voce che sarà trascinata. Il tree view control creerà una lista di immagini 
con solo un' immagine per il drag e restituirà l'handle della lista di immagini a voi. 

2) Dopo che l'immagine di drag è stata creata, specificate l' hotspot dell'immagine di drag chiamando 
ImageList_BeginDrag. 

ImageList_BeginDrag PROTO 
             himlTrack:DWORD,
             iTrack:DWORD,
             dxHotspot:DWORD,
             dyHotspot:DWORD

himlTrack è l'handle della lista di immagini che contiene l'immagine di drag. 

iTrack è l'indice nella lista di immagine che specifica l'immagine di drag 

dxHotspot specifica la distanza relativa dell' hotspot nel piano orizzontale dell'immagine di drag che sarà
usata al posto del cursore del mouse, in modo da specificare quale parte dell'immagine è l' hotspot. 

dyHotspot specifica la distanza relativa dell' hotspot nel piano verticale. Normalmente, iTrack deve 
essere 0 se specificate al tree view control di creare l'immagine di drag per voi e dxHotspot , dyHotspot 
possono essere 0 se desiderate che l'angolo superiore di sinistra dell'immagine di drag sia l'hotspot. 

3) Quando l'immagine di drag è pronta per essere visualizzata, chiamiamo 'ImageList_DragEnter per 
visualizzare l'immagine di drag nella finestra. 

ImageList_DragEnter PROTO hwndLock:DWORD, x:DWORD, y:DWORD

hwndLock è l'handle della finestra che possiede l'immagine di drag. L'immagine di drag non potrà 
spostarsi nelle parti esterne alla finestra. 

x e y sono x-e y-coordinate del posto in cui l'immagine di drag dovrebbe inizialmente essere visualizzata.
Si noti che questi valori sono relativi all'angolo superiore sinistro della finestra, non alla zona del client. 

4) Ora che l'immagine di drag è visualizzata nella finestra, dovrete supportare il funzionamento del drag 
nel tree view control. Tuttavia, c'è un piccolo problema. Dobbiamo controllare il percorso di drag con 
WM_MOUSEMOVE e la posizione di drop con i messaggi WM_LBUTTONUP. Tuttavia, se l'immagine di 
drag è sopra altre finestre figlie, la finestra generatrice non riceverà mai alcun messaggio del mouse. La 
soluzione è di bloccare l'input del mouse con SetCapture usando la chiamata, i messaggi del mouse 
saranno diretti verso la finestra specificata senza riguardo a dove il cursore del mouse si trova. 

5) Tramite la ricezione di WM_MOUSEMOVE , aggiornerete il percorso di drag con la chiamata di 
ImageList_DragMove. Questa funzione sposta l'immagine che si sta trascinando durante l'operazione di 
drag-and-drop. Ancora, se volete, potete evidenziare la voce sottostante l'immagine di drag trasmettendo
TVM_HITTEST per controllare se l'immagine di drag è sopra qualche voce. Se è così, potete trasmettere
TVM_SELECTITEM con il flag TVGN_DROPHILITE per evidenziare la voce. Si noti che prima di inviare il 
messaggio TVM_SELECTITEM , dovete in primo luogo nascondere l'immagine di drag altrimenti la 
vostra immagine di drag lascerà delle tracce indesiderate. Potete nascondere l'immagine di drag 
chiamando ImageList_DragShowNolock e, dopo che l'operazione di evidenziamento è finita, chiamerete 
ImageList_DragShowNolock ancora per mostrare l'immagine di drag. 

6) Quando l'utente libera il tasto di sinistra del mouse, dovete fare parecchie cose. Se avete evidenziato 
una voce, dovete rilasciarla trasmettendo TVM_SELECTITEM con ancora il flag TVGN_DROPHILITE, 
ma questo volta, lParam DEVE essere zero. Se non rilasciate la voce, otterrete un effetto indesiderato: 
quando selezionerete una certa altra voce, quella voce sarà racchiusa in un rettangolo ma sarà ancora 
evidenziata l'ultima voce selezionata. Successivamente, dovete chiamare ImageList_DragLeave 
seguito da ImageList_EndDrag. Dovete liberare il mouse chiamando ReleaseCapture. Se avete creato
una lista di immagini, dovete distruggerla chiamando ImageList_Destroy. Dopo questo, potete 



procedere con cosa il vostro programma desiderate che faccia quando l'operazione di drag & drop è 
completata. 

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
include \masm32\include\comctl32.inc
include \masm32\include\gdi32.inc
includelib \masm32\lib\gdi32.lib
includelib \masm32\lib\comctl32.lib
includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib

WinMain PROTO :DWORD,:DWORD,:DWORD,:DWORD
.const
IDB_TREE equ 4006                ; ID of the bitmap resource
.data
ClassName  db "TreeViewWinClass",0
AppName    db "Tree View Demo",0
TreeViewClass  db "SysTreeView32",0
Parent  db "Parent Item",0
Child1  db "child1",0
Child2  db "child2",0
DragMode  dd FALSE                ; a flag to determine if we are in drag mode

.data?
hInstance  HINSTANCE ?
hwndTreeView dd ?            ; handle of the tree view control
hParent  dd ?                        ; handle of the root tree view item
hImageList dd ?                    ; handle of the image list used in the tree view control
hDragImageList  dd ?        ; handle of the image list used to store the drag image

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
    invoke ExitProcess,eax
    invoke InitCommonControls

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInst
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_APPWORKSPACE
    mov   wc.lpszMenuName,NULL
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,eax
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,
                          ADDR ClassName,ADDR AppName, 
WS_OVERLAPPED+WS_CAPTION+WS_SYSMENU+WS_MINIMIZEBOX+WS_MAXIMIZEBOX+WS_VISIBLE,CW_USEDEFAULT,
           CW_USEDEFAULT,200,400,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    .while TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg



        invoke DispatchMessage, ADDR msg
    .endw
    mov eax,msg.wParam
    ret
WinMain endp

WndProc proc uses edi hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    LOCAL tvinsert:TV_INSERTSTRUCT
    LOCAL hBitmap:DWORD
    LOCAL tvhit:TV_HITTESTINFO
    .if uMsg==WM_CREATE
        invoke CreateWindowEx,NULL,ADDR TreeViewClass,NULL,\
            WS_CHILD+WS_VISIBLE+TVS_HASLINES+TVS_HASBUTTONS+TVS_LINESATROOT,0,\
            0,200,400,hWnd,NULL,\
            hInstance,NULL            ; Create the tree view control
        mov hwndTreeView,eax
        invoke ImageList_Create,16,16,ILC_COLOR16,2,10    ; create the associated image list
        mov hImageList,eax
        invoke LoadBitmap,hInstance,IDB_TREE        ; load the bitmap from the resource
        mov hBitmap,eax
        invoke ImageList_Add,hImageList,hBitmap,NULL    ; Add the bitmap into the image list
        invoke DeleteObject,hBitmap    ; always delete the bitmap resource
        invoke SendMessage,hwndTreeView,TVM_SETIMAGELIST,0,hImageList
        mov tvinsert.hParent,NULL
        mov tvinsert.hInsertAfter,TVI_ROOT
        mov tvinsert.item.imask,TVIF_TEXT+TVIF_IMAGE+TVIF_SELECTEDIMAGE
        mov tvinsert.item.pszText,offset Parent
        mov tvinsert.item.iImage,0
        mov tvinsert.item.iSelectedImage,1
        invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert
        mov hParent,eax
        mov tvinsert.hParent,eax
        mov tvinsert.hInsertAfter,TVI_LAST
        mov tvinsert.item.pszText,offset Child1
        invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert
        mov tvinsert.item.pszText,offset Child2
        invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert
    .elseif uMsg==WM_MOUSEMOVE
        .if DragMode==TRUE
            mov eax,lParam
            and eax,0ffffh
            mov ecx,lParam
            shr ecx,16
            mov tvhit.pt.x,eax
            mov tvhit.pt.y,ecx
            invoke ImageList_DragMove,eax,ecx
            invoke ImageList_DragShowNolock,FALSE
            invoke SendMessage,hwndTreeView,TVM_HITTEST,NULL,addr tvhit
            .if eax!=NULL
                invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_DROPHILITE,eax
            .endif
            invoke ImageList_DragShowNolock,TRUE
        .endif
    .elseif uMsg==WM_LBUTTONUP
        .if DragMode==TRUE
            invoke ImageList_DragLeave,hwndTreeView
            invoke ImageList_EndDrag
            invoke ImageList_Destroy,hDragImageList
            invoke SendMessage,hwndTreeView,TVM_GETNEXTITEM,TVGN_DROPHILITE,0
            invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_CARET,eax
            invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_DROPHILITE,0
            invoke ReleaseCapture
            mov DragMode,FALSE
        .endif
    .elseif uMsg==WM_NOTIFY
        mov edi,lParam
        assume edi:ptr NM_TREEVIEW
        .if [edi].hdr.code==TVN_BEGINDRAG
            invoke SendMessage,hwndTreeView,TVM_CREATEDRAGIMAGE,0,[edi].itemNew.hItem
            mov hDragImageList,eax
            invoke ImageList_BeginDrag,hDragImageList,0,0,0
            invoke ImageList_DragEnter,hwndTreeView,[edi].ptDrag.x,[edi].ptDrag.y
            invoke SetCapture,hWnd
            mov DragMode,TRUE



        .endif
        assume edi:nothing
    .elseif uMsg==WM_DESTROY
        invoke PostQuitMessage,NULL
    .else
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
WndProc endp
end start

Analisi
Nella risposta a WM_CREATE , create il tree view control 

invoke CreateWindowEx,NULL,ADDR TreeViewClass,NULL,\
            WS_CHILD+WS_VISIBLE+TVS_HASLINES+TVS_HASBUTTONS+TVS_LINESATROOT,0,\
            0,200,400,hWnd,NULL,\
            hInstance,NULL

Notare gli stili. TVS_xxxx sono gli stili di specifica del tree view. 

invoke ImageList_Create,16,16,ILC_COLOR16,2,10 
mov hImageList,eax 
invoke LoadBitmap,hInstance,IDB_TREE
mov hBitmap,eax 
invoke ImageList_Add,hImageList,hBitmap,NULL 
invoke DeleteObject,hBitmap 
invoke SendMessage,hwndTreeView,TVM_SETIMAGELIST,0,hImageList

Successivamente, create una lista vuota di immagini che accetterà immagini nel formato 16x16 pixel, 
colore a 16 bit ed inizialmente, conterrà 2 immagini ma potrà essere espansa a 10 se sarà necessario. 
Quindi caricate la bitmap dalle risorse e aggiungetela alla lista di immagini appena creata. Dopo questo , 
cancellate l'handle della bitmap poiché non sarà più usata. Quando la lista di immagini è pronta, 
associatela con il tree view control trasmettendo TVM_SETIMAGELIST al controllo. 

mov tvinsert.hParent,NULL 
mov tvinsert.hInsertAfter,TVI_ROOT 
mov tvinsert.u.item.imask,TVIF_TEXT+TVIF_IMAGE+TVIF_SELECTEDIMAGE 
mov tvinsert.u.item.pszText,offset Parent 
mov tvinsert.u.item.iImage,0 
mov tvinsert.u.item.iSelectedImage,1 
invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert

Inserite le voci nel tree view control, cominciando dalla voce alla radice. Poiché sarà la voce alla radice, il 
membro hParent è NULL e hInsertAfter è TVI_ROOT. Il membro imask specifica che i membri pszText, 
iImage e iSelectedImage della struttura TV_ITEM sono validi. Riempite questi tre membri di valori adatti. 
pszText contiene l'etichetta della voce alla radice, iImage è l'indice dell'immagine nella lista di immagini 
che sarà visualizzata alla sinistra della voce non selezionata, ed iSelectedImage è l'indice dell'immagine 
nella lista di immagini che sarà visualizzata quando la voce sarà selezionata. Quando tutti i membri sono 
riempiti, trasmettete il messaggio TVM_INSERTITEM al tree view control per aggiungere la voce alla 
radice. 

mov hParent,eax 
mov tvinsert.hParent,eax 
mov tvinsert.hInsertAfter,TVI_LAST 
mov tvinsert.u.item.pszText,offset Child1 
invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert 
mov tvinsert.u.item.pszText,offset Child2 
invoke SendMessage,hwndTreeView,TVM_INSERTITEM,0,addr tvinsert



Dopo che la voce alla radice è stata aggiunta, potete attaccare le voci figlie, ad essa. hParent ora è 
riempito con l'handle della voce del genitore. Ed useremo le stesse immagini nella lista di immagini in 
modo da non cambiare i membri iSelectedImage e iImage. 

.elseif uMsg==WM_NOTIFY
        mov edi,lParam
        assume edi:ptr NM_TREEVIEW
        .if [edi].hdr.code==TVN_BEGINDRAG
            invoke SendMessage,hwndTreeView,TVM_CREATEDRAGIMAGE,0,[edi].itemNew.hItem
            mov hDragImageList,eax
            invoke ImageList_BeginDrag,hDragImageList,0,0,0
            invoke ImageList_DragEnter,hwndTreeView,[edi].ptDrag.x,[edi].ptDrag.y
            invoke SetCapture,hWnd
            mov DragMode,TRUE
        .endif
        assume edi:nothing

Ora quando l'utente prova a trascinare una voce, il tree view control trasmette il messaggio WM_NOTIFY 
con TVN_BEGINDRAG come codice. lParam è il puntatore ad una struttura NM_TREEVIEW che contiene 
parecchie informazioni di cui abbiamo bisogno. Mettete il relativo valore in edi ed usate edi come 
puntatore alla struttura NM_TREEVIEW. assume edi:ptr NM_TREEVIEW è un modo di dire a MASM di 
trattare edi come puntatore alla struttura NM_TREEVIEW. Quindi crate un'immagine di drag trasmettendo
TVM_CREATEDRAGIMAGE al tree view control. Esso restituisce l'handle alla lista di immagini 
recentemente creata con l'immagine di drag all'interno. Chiamate ImageList_BeginDrag per regolare 
l'hotspot nell'immagine di drag. Quindi azionate l'operazione di drag chiamando ImageList_DragEnter. 
Questa funzione visualizza l'immagine di drag alla posizione specificata nella finestra specificata. Usiamo 
la struttura ptDrag che è un membro della struttura NM_TREEVIEW come il punto in cui l'immagine di 
drag dovrebbe essere inizialmente mostrata. Dopo ciò, bloccate l'input del mouse e regolate i flags per 
indicare che inizia la modalità di drag. 

.elseif uMsg==WM_MOUSEMOVE
        .if DragMode==TRUE
            mov eax,lParam
            and eax,0ffffh
            mov ecx,lParam
            shr ecx,16
            mov tvhit.pt.x,eax
            mov tvhit.pt.y,ecx
            invoke ImageList_DragMove,eax,ecx
            invoke ImageList_DragShowNolock,FALSE
            invoke SendMessage,hwndTreeView,TVM_HITTEST,NULL,addr tvhit
            .if eax!=NULL
                invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_DROPHILITE,eax
            .endif
            invoke ImageList_DragShowNolock,TRUE
        .endif

Ora ci concentreremo su WM_MOUSEMOVE. Quando l'utente trascina l'immagine di drag avanti, la nostra 
finestra generatrice riceve i messaggi WM_MOUSEMOVE. In risposta a questi messaggi, aggiornate la 
posizione dell'immagine di drag con ImageList_DragMove. Dopo ciò, controllate se l'immagine di drag è 
sopra una certa voce. Fate questo trasmettendo il messaggio TVM_HITTEST al tree view control con un 
punto da controllare. Se l'immagine di drag è sopra una certa voce, evidenziate quella voce trasmettendo
il messaggio TVM_SELECTITEM con il flag TVGN_DROPHILITE al tree view control. Durante l'operazione di
evidenziamento, nascondete l'immagine di drag in modo che non lasci delle macchie sgradevoli sul tree 
view control. 

.elseif uMsg==WM_LBUTTONUP
        .if DragMode==TRUE
            invoke ImageList_DragLeave,hwndTreeView
            invoke ImageList_EndDrag
            invoke ImageList_Destroy,hDragImageList
            invoke SendMessage,hwndTreeView,TVM_GETNEXTITEM,TVGN_DROPHILITE,0
            invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_CARET,eax
            invoke SendMessage,hwndTreeView,TVM_SELECTITEM,TVGN_DROPHILITE,0
            invoke ReleaseCapture
            mov DragMode,FALSE
        .endif



Quando l'utente libera il tasto di sinistra del mouse, l'operazione di drag è alla fine. Lasciate la modalità di
drag chiamando ImageList_DragLeave, seguita da ImageList_EndDrag e da ImageList_Destroy. Ricreate 
le voci del tree view in modo corretto, controllate anche l'ultima voce evidenziata e la selezionate. Dovete
poi deselezionarla altrimenti le altre voci non verranno evidenziate quando selezionate. Ed infine, liberate 
il bloccaggio del mouse. 

Introduzione
In questo tutorial, impareremo circa il subclassing di finestra, che cosa è e come usarlo a nostro 
vantaggio. 

Tools
Esempio 

Essay
Se programmate sotto Window da un certo tempo, troverete alcuni casi dove una finestra ha quasi gli 
attributi di cui avete bisogno nel vostro programma ma non tutti. Avete incontrato una situazione dove 
desideravate un certo genere speciale di casella di modifica che potesse filtrare fuori un certo testo 
indesiderato? La cosa diretta da fare è di scrivere il codice della vostra finestra. Ma è un lavoro duro e che
richiede realmente del tempo. Il subclassing di una finestra ci può aiutare. In una nutshell, il subclassing 
di finestre permette di " prendere il controllo " della finestra subclassata. Avrete il controllo assoluto su di
essa. Facciamo un esempio per capire meglio. Supponete di avere bisogno di una casella di testo che 
accetti soltanto i numeri esadecimali. Se usate una semplice casella di modifica, voi non sapete cosa e 
quando l' utente digita qualcosa oltre ai numeri hex nella vostra casella di testo, ie se l'utente digita " 
zb+q * " nella vostra casella di testo, non potete fare qualche cosa con essa tranne che il rifiuto della 
stringa di testo intera. Ciò non è professionale. Essenzialmente, avete bisogno della capacità di 
esaminare ogni carattere che l'utente ha digitato nella casella di testo nel momento in cui l'ha digitato. 
Esamineremo ora come fare questo. Quando l'utente digita qualcosa in una casella di testo, Window 
trasmette il messaggio WM_CHAR alla procedura della finestra casella di modifica. Questa procedura 
risiede all'interno di Window in modo da non poterla modificare. Ma possiamo ridirezionare il flusso 
dei messaggi alla nostra procedura per la finestra. In modo che la nostra procedura per la finestra 
ottenga per prima tutti i messaggi che Window trasmette alla casella di modifica. Se la nostra procedura 
per la finestra sceglie di intervenire sul messaggio, può farlo così. Ma se non desidera maneggiare il 
messaggio, può passarlo alla procedura originale di Window. In questo modo, la nostra procedura per la 
finestra si inserisce fra Window e la casella di modifica. Guardare il flusso qui sotto: 

Prima del Subclassing 

Windows ==> procedura per la finestra casella di modifica 

Dopo il Subclassing 

Windows ==> nostra procedura per la finestra -----> procedura di Window per la casella di modifica 

Ora incentriamo la nostra attenzione su come fare il subclass a una finestra. Si noti che il subclassing non
è limitato ai controls, può essere usato con tutte le finestre. Pensiamo a come Window sa dove la 
procedura per la finestra casella di modifica risiede. Un' ipotesi?.....lpfnWndProc della struttura 
WNDCLASSEX. Se possiamo sostituire questo membro con l'indirizzo della nostra procedura per la 
finestra, Window trasmetterà preferibilmente i messaggi alla nostro proc per la finestra. Possiamo fare 
questo chiamando SetWindowLong. 

SetWindowLong PROTO hWnd:DWORD, nIndex:DWORD, dwNewLong:DWORD

hWnd = handle della finestra per cambiare il valore nella struttura WNDCLASSEX 

nIndex == valore da cambiare. 

GWL_EXSTYLE Setta un nuovo stile esteso per la finestra. 

GWL_STYLE Setta un nuovo stile per la finestra. 

GWL_WNDPROC Setta un nuovo indirizzo per la procedura della finestra. 

GWL_HINSTANCE Setta una nuova handle per l' istanza dell' applicazione. 

GWL_ID Setta un nuovo identificatore per la finestra. 

GWL_USERDATA Setta il valore a 32-bit associato alla finestra. Ogni finestra ha un corrispondente valore 
a 32-bit progettato ad uso dell'applicazione che ha creato la finestra. 

dwNewLong = il valore sostitutivo. 

Così il nostro lavoro è facile: Codifichiamo una proc per la finestra che maneggierà i messaggi per la 
casella di modifica ed allora chiameremo SetWindowLong con il flag GWL_WNDPROC , passaremo 
l'indirizzo della nostra proc per la finestra nel terzo parametro. Se la funzione riesce, il valore di ritorno è 
il precedente numero dell'intero a 32-bit specificato, nel nostro caso, l'indirizzo della procedura originale 
per la finestra. Dobbiamo memorizzare questo valore per usarlo all'interno della nostra procedura. 
Ricordartevi che ci saranno alcuni messaggi che non desideriamo maneggiare, li passeremo alla 
procedura originale per la finestra. Possiamo fare questo chiamando la funzione CallWindowProc. 

CallWindowProc PROTO 
               lpPrevWndFunc:DWORD,
               hWnd:DWORD,
               Msg:DWORD,
               wParam:DWORD,
               lParam:DWORD

lpPrevWndFunc = l'indirizzo della procedura originale per la finestra. 

I quattro parametri restanti sono quelli passati alla nostra procedura per la finestra. Li passeremo alla 
CallWindowProc. 

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
include \masm32\include\comctl32.inc
includelib \masm32\lib\comctl32.lib



includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib

WinMain PROTO :DWORD,:DWORD,:DWORD,:DWORD
EditWndProc PROTO :DWORD,:DWORD,:DWORD,:DWORD

.data
ClassName  db "SubclassWinClass",0
AppName    db "Subclassing Demo",0
EditClass  db "EDIT",0
Message  db "You pressed Enter in the text box!",0

.data?
hInstance  HINSTANCE ?
hwndEdit dd ?
OldWndProc dd ?

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInst
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_APPWORKSPACE
    mov   wc.lpszMenuName,NULL
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,eax
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
 WS_OVERLAPPED+WS_CAPTION+WS_SYSMENU+WS_MINIMIZEBOX+WS_MAXIMIZEBOX+WS_VISIBLE,CW_USEDEFAULT,\
           CW_USEDEFAULT,350,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    .while TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .endw
    mov eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    .if uMsg==WM_CREATE
        invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR EditClass,NULL,\
            WS_CHILD+WS_VISIBLE+WS_BORDER,20,\
            20,300,25,hWnd,NULL,\
            hInstance,NULL
        mov hwndEdit,eax
        invoke SetFocus,eax
        ;-----------------------------------------
        ; Subclass it!
        ;-----------------------------------------
        invoke SetWindowLong,hwndEdit,GWL_WNDPROC,addr EditWndProc
        mov OldWndProc,eax
    .elseif uMsg==WM_DESTROY
        invoke PostQuitMessage,NULL



    .else
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
WndProc endp

EditWndProc PROC hEdit:DWORD,uMsg:DWORD,wParam:DWORD,lParam:DWORD
    .if uMsg==WM_CHAR
        mov eax,wParam
        .if (al>="0" && al<="9") || (al>="A" && al<="F") || (al>="a" && al<="f") || 
al==VK_BACK
            .if al>="a" && al<="f"
                sub al,20h
            .endif
            invoke CallWindowProc,OldWndProc,hEdit,uMsg,eax,lParam
            ret
        .endif
    .elseif uMsg==WM_KEYDOWN
        mov eax,wParam
        .if al==VK_RETURN
            invoke MessageBox,hEdit,addr Message,addr AppName,MB_OK+MB_ICONINFORMATION
            invoke SetFocus,hEdit
        .else
            invoke CallWindowProc,OldWndProc,hEdit,uMsg,wParam,lParam
            ret
        .endif
    .else
        invoke CallWindowProc,OldWndProc,hEdit,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
EditWndProc endp
end start

Analisi
invoke SetWindowLong,hwndEdit,GWL_WNDPROC,addr EditWndProc 
mov OldWndProc,eax

Dopo che la casella di modifica è stata creata, noi la subclassiamo chiamando SetWindowLong, 
sostituiamo l'indirizzo della procedura originale per la finestra con la nostra procedura per la finestra. Si 
noti che memorizziamo l'indirizzo della procedura originale per la finestra per usarlo con CallWindowProc. 
Notare EditWndProc è una procedura ordinaria per la finestra. 

.if uMsg==WM_CHAR
              mov eax,wParam
              .if (al>="0" && al<="9") || (al>="A" && al<="F") || (al>="a" && al<="f") || 
al==VK_BACK
                  .if al>="a" && al<="f"
                      sub al,20h
                  .endif
                  invoke CallWindowProc,OldWndProc,hEdit,uMsg,eax,lParam
                  ret
              .endif

All'interno di EditWndProc, filtriamo i messaggi WM_CHAR. Se il carattere è fra 0-9 o a-f, lo accettiamo 
passando il messaggio alla procedura originale per la finestra. Se è un carattere minuscolo, lo 
convertiamo in maiuscolo aggiungendogli 20h. Notate che, se il carattere non è fra quelli che prevediamo,
lo scartiamo. Non lo passiamo alla proc originale per la finestra. Così quando l'utente digita qualcosa 
tranne 0-9 o a-f, il carattere non compare nella casella di modifica. 

.elseif uMsg==WM_KEYDOWN
        mov eax,wParam



        .if al==VK_RETURN
            invoke MessageBox,hEdit,addr Message,addr AppName,MB_OK+MB_ICONINFORMATION
            invoke SetFocus,hEdit
        .else
            invoke CallWindowProc,OldWndProc,hEdit,uMsg,wParam,lParam
            ret
        .end

Desidero dimostrare la potenza del subclassing ulteriormente intrappolando il tasto Enter. EditWndProc 
controlla se il messaggio WM_KEYDOWN è VK_RETURN (il tasto Enter). Se è così, visualizza una finestra 
di messaggio che dice che " avete premuto il tasto Enter nella casella di testo!". Se non è il tasto Enter, 
passa il messaggio alla procedura originale per la finestra. Potete usare il subclassing di una finestra per 
prendere il controllo su altre finestre. È una delle tecniche più potenti che dovreste avere nel vostro 
arsenale. 

Tutorial 21: Pipe

Introduzione
In questo tutorial, esploreremo il pipe, che cos'è e per che cosa possiamo usarlo. Per renderlo più 
interessante, inserirò nella tecnica come cambiare il colore del testo e del background di una casella di 
modifica. 

Tools
Esempio 

Essay
Un pipe, traducendo banalmente dall'inglese, è un "tubo". In informatica identifica uno strumento che 
consente di scambiare dati fra due processi differenti, o all'interno dello stesso processo. È come un 
walkie-talkie: date all'altra parte un set e potrà usarlo per comunicare con voi. 

Tipi di pipe

Ci sono due tipi di pipe: pipe anonimi e con nome (named). Il pipe anonimo è, chiaramente, anonimo: 
cioè è possibile usarlo senza conoscere il relativo nome (per esempio se è stato ereditato da un processo 
padre). Un named pipe è l'opposto: è necessario conoscere il relativo nome prima che di poterlo usare. I 
pipe si possono anche distinguere in unidirezionale o bidirezionale. In un pipe unidirezionale i dati 
possono entrare soltanto in un senso: da un' estremità all' altra; in un pipe bidirezionale, invece, i dati 
possono essere scambiati fra entrambe le estremità. Un pipe anonimo è sempre unidirezionale mentre un
named pipe può essere unidirezionale o bidirezionale. I named pipe sono utilizzati solitamente in un 
ambiente di rete in cui un server può collegarsi a parecchi client. 

In questo tutorial, esamineremo in certi dettagli il pipe anonimo. Lo scopo principale del pipe anonimo è 
di essere usato come comunicazione tra un processo padre ed i processi figli o tra processi figli. 

Utilità dei pipe anonimi

Il pipe anonimo è molto utile quando si sta realizzando un'applicazione a console. Tuttavia, 
un'applicazione a console è un programma completamente 32-bit. Può usare tutte le funzione GUI, come 
gli altri programmi GUI. 

Un'applicazione a console ha tre handle che può utilizzare per i relativi ingressi e uscite. Sono chiamati 
handle standard (stdin per l'input, stdout per l'output e stderr per l'output di "scorta", solitamente gli 
errori). Un'applicazione a console può richiamare questi tre handle standard chiamando la funzione 
GetStdHandle, specificando l' handle che si desidera ottenere. Un'applicazione GUI non ha 
necessariamente una console attaccata: chiamando GetStdHandle, questa fallirà. Se il programma 
necessita di una console DOS è sufficiente chiamare AllocConsole per crearne una nuova. Tuttavia non 
dimenticatevi di chiamare FreeConsole quando avete finito con la console. 

Il pipe anonimo è utilizzato più frequentemente per riorientare l'input e/o l'uscita di un'applicazione figlia 
a console. Il processo genitore può essere una console o un'applicazione GUI ma il figlio deve essere 
un'applicazione a console affinchè questo funzioni. Com'è (ormai) noto, un'applicazione a console utilizza 
gli handle standard per i relativi ingressi ed uscite. Se desideriamo riorientare l'input e/o l'uscita di 
un'applicazione a console, possiamo sostituire l' handle con l' handle di un'estremità del pipe. 
L'applicazione non saprà che sta utilizzando l'handle di un'estremità del pipe, la userà come handle 
standard. Ciò è simile al concetto di polimorfismo, parlando a oggetti. E non è nemmeno necessario 
modificare il processo figlio in nessun modo. 

Un' altra cosa che dovreste conoscere circa un'applicazione a console è da dove ottiene questi handle 
standard. Quando un'applicazione a console è creata il processo genitore ha due scelte: può creare una 
nuova console per il figlio o può lasciare che il figlio erediti la propria console. Affinchè il secondo metodo 
funzioni il processo genitore deve essere un'applicazione a console o, se è un'applicazione GUI, deve 
prima chiamare AllocConsole per allocare una sezione comandi. 

Un po' di codice

Cominciamo il lavoro: per creare un pipe anonimo bisogna chiamare CreatePipe, che ha il seguente 
prototipo: 

CreatePipe proto pReadHandle:DWORD,
           pWriteHandle:DWORD,
           pPipeAttributes:DWORD,
           nBufferSize:DWORD

 pReadHandle è un puntatore ad una variabile dword che riceverà l'handle dell'estremità di 
lettura del pipe. 

 pWriteHandle è un puntatore ad una variabile dword che riceverà l'handle dell'estremità di 
scrittura del pipe. 

 pPipeAttributes puntatore ad una struttura SECURITY_ATTRIBUTES che determina se gli handle
restituiti per la lettura & scrittura sono ereditate dai processi figli. 

 nBufferSize è la dimensione suggerita del buffer che il pipe riserverà per l'uso. Potete usare 
NULL per dire alla funzione di usare il formato di default. 

Se la chiamata riesce, il valore di ritorno è diverso da zero. Se fallisse, il valore di ritorno è zero. Dopo 
che la chiamata è riuscita otterrete gli handle: uno dell'estremità di lettura del pipe e l'altro dell'estremità
di scrittura. 

Redirigere gli stream

Notate che il mio metodo differisce da quello del riferimento win32 api della Borland. Il metodo nel 
riferimento win32 api presuppone che il processo genitore sia un'applicazione a console ed il figlio può 
così ereditare gli handle standard. Per la maggior parte delle volte dovremo riorientare l'uscita 
dell'applicazione a console ad una di tipo GUI. 

1) Creare un pipe anonimo con CreatePipe. Non dimenticate di settare il membro bInheritable della 
struttura SECURITY_ATTRIBUTES su TRUE in modo che le handles siano ereditabili. 



2) Ora dobbiamo preparare i parametri che passeremo a CreateProcess poiché la useremo per caricare 
l'applicazione a console figlia. Una struttura importante è la struttura STARTUPINFO. Questa struttura 
determina la condizione iniziale della finestra principale del processo figlio quando compare per la prima 
volta. Questa struttura è vitale al nostro scopo: potete nascondere la finestra principale e passare un 
handle del pipe al processo a console figlio con essa. Di seguito ci sono i membri che dovete riempire: 

 cb: la dimensione della struttura STARTUPINFO 

 dwFlags: i flag che determinano quali membri della struttura sono validi e anche la condizione di 
visibilità/nascosta della finestra principale. Per il nostro scopo, dovrete usare una combinazione 
di STARTF_USESHOWWINDOW e di STARTF_USESTDHANDLES 

 hStdOutput e hStdError: gli handle che desiderate che il processo figlio utilizzi come handles 
standard di output/error. Per il nostro scopo, passeremo l'handle di scrittura del pipe per l'uscita
e l'errore standard al figlio. Così quando il figlio produce qualcosa di output/error standard, 
realmente passa l'Info tramite il pipe al processo genitore. 

 wShowWindow governa la condizione di visibilità/nascosta della finestra principale. Per il nostro 
scopo, non desideriamo mostrare la finestra della console del figlio così mettiamo SW_HIDE in 
questo membro. 

3) Chiamare CreateProcess per caricare l'applicazione figlia. Dopo la riuscita di CreateProcess il figlio è 
ancora disattivo: è caricato nella memoria ma non è in esecuzione immediatamente. 

4) Chiudere l'handle di scrittura del pipe: poiché il processo genitore non ha bisogno di usare l'handle di 
scrittura del pipe ed il pipe non funzionerà se ci saranno più di un' estremità di scrittura, DOBBIAMO 
chiuderla prima della lettura dei dati dal pipe. Tuttavia non chiudete l'handle di scrittura prima della 
chiamata a CreateProcess, altrimenti il vostro pipe sarà rotto. Dovreste chiuderla subito dopo che 
CreateProcess è ritornata e prima di leggere i dati dall'estremità di lettura del pipe. 

5) Ora potete leggere i dati dall' estremità di lettura del pipe con ReadFile. Usando ReadFile il processo 
figlio inizia ad essere eseguito, e quando scriverà qualcosa nell'handle standard per l'uscita (che è 
attualmente l'handle dell'estremità di scrittura del pipe), i dati saranno trasmessi tramite il pipe 
all'estremità di lettura. Dovete chiamare ripetutamente ReadFile fino a che non restituisca 0 quando le 
informazioni non saranno più dati da leggere. Potete fare qualsiasi cosa con i dati letti dal pipe. Nel nostro
esempio, li ho messi in una text box. 

6) Chiudere l' handle di lettura del pipe. 

Esempio

.386
    .model flat,stdcall
    option casemap:none
    include \masm32\include\windows.inc
    include \masm32\include\user32.inc
    include \masm32\include\kernel32.inc
    include \masm32\include\gdi32.inc
    includelib \masm32\lib\gdi32.lib
    includelib \masm32\lib\user32.lib
    includelib \masm32\lib\kernel32.lib

    WinMain PROTO :DWORD,:DWORD,:DWORD,:DWORD

    .const
    IDR_MAINMENU equ 101         ; ID del menu principale
    IDM_ASSEMBLE equ 40001

    .data
    ClassName            db "PipeWinClass",0
    AppName              db "One-way Pipe Example",0 EditClass db "EDIT",0
    CreatePipeError     db "Error during pipe creation",0
    CreateProcessError     db "Error during process creation",0
    CommandLine     db "ml /c /coff /Cp test.asm",0

    .data?
    hInstance HINSTANCE ?
    hwndEdit dd ?



    .code
    start:
        invoke GetModuleHandle, NULL
        mov hInstance,eax
        invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
        invoke ExitProcess,eax

    WinMain proc hInst:DWORD,hPrevInst:DWORD,CmdLine:DWORD,CmdShow:DWORD
        LOCAL wc:WNDCLASSEX
        LOCAL msg:MSG
        LOCAL hwnd:HWND
        mov wc.cbSize,SIZEOF WNDCLASSEX
        mov wc.style, CS_HREDRAW or CS_VREDRAW mov wc.lpfnWndProc, OFFSET WndProc
        mov wc.cbClsExtra,NULL
        mov wc.cbWndExtra,NULL
        push hInst
        pop wc.hInstance
        mov wc.hbrBackground,COLOR_APPWORKSPACE
        mov wc.lpszMenuName,IDR_MAINMENU
        mov wc.lpszClassName,OFFSET ClassName
        invoke LoadIcon,NULL,IDI_APPLICATION
        mov wc.hIcon,eax
        mov wc.hIconSm,eax
        invoke LoadCursor,NULL,IDC_ARROW
        mov wc.hCursor,eax
        invoke RegisterClassEx, addr wc
        invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName, 
WS_OVERLAPPEDWINDOW+WS_VISIBLE,CW_USEDEFAULT, CW_USEDEFAULT,400,200,NULL,NULL, hInst,NULL
        mov hwnd,eax
        .while TRUE
            invoke GetMessage, ADDR msg,NULL,0,0
            .BREAK .IF (!eax)
            invoke TranslateMessage, ADDR msg
            invoke DispatchMessage, ADDR msg
        .endw
        mov eax,msg.wParam
        ret
    WinMain endp

    WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
        LOCAL rect:RECT
        LOCAL hRead:DWORD
        LOCAL hWrite:DWORD
        LOCAL startupinfo:STARTUPINFO
        LOCAL pinfo:PROCESS_INFORMATION
        LOCAL buffer[1024]:byte
        LOCAL bytesRead:DWORD
        LOCAL hdc:DWORD
        LOCAL sat:SECURITY_ATTRIBUTES
        .if uMsg==WM_CREATE
            invoke CreateWindowEx,NULL,addr EditClass, NULL, WS_CHILD+ WS_VISIBLE+ 
ES_MULTILINE+ ES_AUTOHSCROLL+ ES_AUTOVSCROLL, 0, 0, 0, 0, hWnd, NULL, hInstance, NULL
            mov hwndEdit,eax
        .elseif uMsg==WM_CTLCOLOREDIT
            invoke SetTextColor,wParam,Yellow
            invoke SetBkColor,wParam,Black
           invoke GetStockObject,BLACK_BRUSH
            ret
        .elseif uMsg==WM_SIZE
            mov edx,lParam
            mov ecx,edx
            shr ecx,16
            and edx,0ffffh
            invoke MoveWindow,hwndEdit,0,0,edx,ecx,TRUE
        .elseif uMsg==WM_COMMAND
           .if lParam==0
                mov eax,wParam
                .if ax==IDM_ASSEMBLE
                    mov sat.niLength,sizeof SECURITY_ATTRIBUTES
                    mov sat.lpSecurityDescriptor,NULL
                    mov sat.bInheritHandle,TRUE
                    invoke CreatePipe,addr hRead,addr hWrite,addr sat,NULL
                    .if eax==NULL



                        invoke MessageBox, hWnd, addr CreatePipeError, addr AppName, 
MB_ICONERROR+ MB_OK
                    .else
                        mov startupinfo.cb,sizeof STARTUPINFO
                        invoke GetStartupInfo,addr startupinfo
                        mov eax, hWrite
                        mov startupinfo.hStdOutput,eax
                        mov startupinfo.hStdError,eax
                        mov startupinfo.dwFlags, STARTF_USESHOWWINDOW+ STARTF_USESTDHANDLES
                        mov startupinfo.wShowWindow,SW_HIDE
                        invoke CreateProcess, NULL, addr CommandLine, NULL, NULL, TRUE, NULL, 
NULL, NULL, addr startupinfo, addr pinfo
                        .if eax==NULL
                            invoke MessageBox,hWnd,addr CreateProcessError,addr         
AppName,MB_ICONERROR+MB_OK
                        .else
                            invoke CloseHandle,hWrite
                            .while TRUE
                                invoke RtlZeroMemory,addr buffer,1024
                                invoke ReadFile,hRead,addr buffer,1023,addr bytesRead,NULL
                                .if eax==NULL
                                    .break
                                .endif
                                invoke SendMessage,hwndEdit,EM_SETSEL,-1,0
                                invoke SendMessage,hwndEdit,EM_REPLACESEL,FALSE,addr buffer
                            .endw
                        .endif
                        invoke CloseHandle,hRead
                    .endif
                .endif
            .endif
        .elseif uMsg==WM_DESTROY
            invoke PostQuitMessage,NULL
        .else
            invoke DefWindowProc,hWnd,uMsg,wParam,lParam ret
        .endif
        xor eax,eax
        ret
    WndProc endp
    end start

Analisi
L'esempio chiamerà ml.exe per assemblare un file chiamato test.asm e per riorientare l'uscita di ml.exe 
nell'area client della text box. Quando il programma è caricato registra il codice per la classe finestra e 
crea la finestra principale come di consueto. La prima cosa che fa durante la creazione della finestra 
principale è di creare una casella di testo che sarà usato per visualizzare l'uscita di ml.exe. Ora la parte 
interessante: cambieremo il colore di background e del testo della casella di modifica. Quando la casella 
di modifica sta per disegnare la sua zona client, trasmette il messaggio WM_CTLCOLOREDIT al relativo 
genitore. wParam contiene l'handle del device context che la casella di modifica userà per scrivere la 
relativa zona client. Possiamo usare questa occasione per modificare le caratteristiche dell'HDC. 

.elseif uMsg==WM_CTLCOLOREDIT
            invoke SetTextColor,wParam,Yellow
            invoke SetTextColor,wParam,Black
            invoke GetStockObject,BLACK_BRUSH
            ret

SetTextColor cambia il colore del testo in giallo. SetTextColor cambia il colore di background del testo a 
nero. Ed infine, otteniamo l' handle del brush nero che restituiamo a Window. Con il messaggio 
WM_CTLCOLOREDIT dovete restituire un handle ad un brush che Windows userà per disegnare il 
background della casella di testo. Nel nostro esempio desidero che il background sia nero quindi 
restituisco l'handle del brush nero a Windows. Ora, quando l'utente seleziona Assemble dal menu, viene 
creato un pipe anonimo. 



.if ax==IDM_ASSEMBLE
                mov sat.niLength,sizeof SECURITY_ATTRIBUTES
                mov sat.lpSecurityDescriptor,NULL
                mov sat.bInheritHandle,TRUE

Prima di chiamare CreatePipe dobbiamo riempire la struttura SECURITY_ATTRIBUTES: Si noti che 
possiamo usare NULL nel membro lpSecurityDescriptor se non ci preoccupiamo della sicurezza. Ed il 
membro bInheritHandle deve essere TRUE in modo che le handles del pipe siano ereditabili dal processo 
figlio. 

invoke CreatePipe,addr hRead,addr hWrite,addr sat,NULL

Dopo questo chiamiamo CreatePipe che, se riuscita, riempirà le variabili hWrite e hRead rispettivamente 
con gli handle dell'estremità di lettura e scrittura del pipe. 

mov startupinfo.cb,sizeof STARTUPINFO
                    invoke GetStartupInfo,addr startupinfo
                    mov eax, hWrite
                    mov startupinfo.hStdOutput,eax
                    mov startupinfo.hStdError,eax
                    mov startupinfo.dwFlags, STARTF_USESHOWWINDOW+ STARTF_USESTDHANDLES
                    mov startupinfo.wShowWindow,SW_HIDE

Successivamente dobbiamo riempire la struttura STARTUPINFO. Chiamiamo GetStartupInfo per riempire 
la struttura STARTUPINFO con i valori di default del processo genitore. DOVETE riempire la struttura 
STARTUPINFO con questa chiamata se desiderate che il vostro codice possa lavorare sia sotto win9x che 
NT. Dopo che la chiamata a GetStartupInfo è ritornata potete modificare i membri importanti della 
struttura. Copiamo l'handle dell'estremità di scrittura del pipe in hStdOutput ed in hStdError poiché 
desideriamo che il processo fglio le usi anziché gli handle standard di default per output/error. 
Desideriamo anche nascondere la finestra della sezione comandi del processo figlio, quindi mettiamo il 
valore SW_HIDE nel membro wShowWidow. Ed infine dobbiamo indicare che i membri hStdOutput, 
hStdError e wShowWindow sono validi e deve essere fatto specificando i flags 
STARTF_USESHOWWINDOW e STARTF_USESTDHANDLES nel membro dwFlags. 

invoke CreateProcess, NULL, addr CommandLine, NULL, NULL, TRUE, NULL, NULL, NULL, addr 
startupinfo, addr pinfo

Ora creiamo il processo figlio con la chiamata di CreateProcess. Si noti che il parametro bInheritHandles 
deve essere regolato su TRUE affinchè l' handle del pipe funzioni. 

invoke CloseHandle,hWrite

Dopo aver creato con successo il processo figlio, dobbiamo chiudere l'estremità di scrittura del pipe. 
Ricordatevi che abbiamo passato l'handle di scrittura al processo figlio tramite la struttura STARTUPINFO:
se non chiudiamo l'handle di scrittura dalla nostra estremità, ci saranno due estremità di scrittura, e il 
pipe non funzionerà. Dobbiamo chiudere l'handle di scrittura dopo CreateProcess ma prima di leggere i 
dati dall'estremità del pipe. 

.while TRUE 
        invoke RtlZeroMemory,addr buffer,1024 
        invoke ReadFile,hRead,addr buffer,1023,addr bytesRead,NULL 
          .if eax==NULL 
          .break 
          .endif 
        invoke SendMessage,hwndEdit,EM_SETSEL,-1,0 
        invoke SendMessage,hwndEdit,EM_REPLACESEL,FALSE,addr buffer 
.endw

Ora siamo pronti a leggere i dati dall'uscita standard del processo figlio. Rimarremo in un ciclo infinito 
fino a che non ci saranno più dati coerenti da leggere dal pipe. Chiamiamo RtlZeroMemory per riempire il 
buffer di zeri quindi chiamiamo ReadFile, passando l' handle di lettura del pipe al posto dell'handle di un 
file. Si noti che leggiamo soltanto un massimo di 1023 byte poiché vogliamo che i dati siano una stringa 
ASCIIZ che possiamo passare a una casella di testo. 



Quando ReadFile ritorna con i dati nel buffer, riempiamo la casella di modifica con i dati. Tuttavia c'è un 
piccolo problema. Se usiamo SetWindowText per immettere i dati nella casella di modifica, i nuovi dati 
saranno scritti sopra i dati attuali! Desideriamo che i dati siano aggiunti alla conclusione dei dati esistenti.

Per raggiungere lo scopo prima spostiamo il cursore verso la fine del testo nella casella di modifica 
trasmettendo il messaggio EM_SETSEL con wParam == -1, poi aggiungiamo i dati a quel punto con il 
messaggio EM_REPLACESEL. 

invoke CloseHandle,hRead

Quando ReadFile restituisce NULL, usciamo dal ciclo e chiudiamo l'handle di lettura. 

Tutorial 22: Superclassing

Introduzione
In questo tutorial, impareremo circa il superclassing, che cos' è e che cosa fa. Imparerete anche come 
fornire il percorso del tasto Tab tra i controlli nella vostra finestra. 

Tools
Esempio 

Essay
Nella vostra carriera di programmazione, certamente incontrerete una situazione dove avrete bisogno di 
parecchi controlli con un compartamento * un po' * differente. Per esempio, potreste avere bisogno di 10
casella di modifica che accettino soltanto numeri. Ci sono parecchi modi per realizzare quest' obiettivo: 

 Creare il vostro codice per la classe ed usarlo al posto dei controlli; 

 Creare queste casella di modifica ed allora subclassarle tutte; 

 Superclassare le caselle di modifica. 

Il primo metodo è troppo noioso. Dovete effettuare voi ogni funzionalità della casella di modifica. 
Un'operazione da non prendere sottogamba. Il secondo metodo è migliore di quello precedente ma 
ancora di troppo lavoro. È giusto se subclassate soltanto alcuni controlli ma sarebbe un incubo 
subclassare una dozzina circa di controlli. Il superclassing è la tecnica che dovreste usare per questa 
situazione. Il superclassing è il metodo che usate per * prendere il controllo * di una categoria particolare
di finestre. * Prendere il controllo *, significa che potete modificare le proprietà di una categoria di 
finestre per soddisfare il vostro scopo e quindi creare un' insieme di controlli. I punti del superclassing 
sono descritto qui sotto: 

 chiamate GetClassInfoEx per ottenere le informazioni sulla categoria di finestre che desiderate 
superclassare. GetClassInfoEx richiede un puntatore ad una struttura WNDCLASSEX che sarà 
riempita di informazioni se la chiamata ritorna con successo. 

 Modificate i membri di WNDCLASSEX che volete. Comunque, ci sono due membri che DOVETE 
modificare: 

1) hInstance Dovete mettere l' handle dell' istanza del vostro programma in questo membro. 

2) lpszClassName Dovete fornire un puntatore ad un nuovo nome della classe. Non è necessario che 
modifichiate il membro lpfnWndProc ma la maggior parte delle volte, dovrete farlo. Ricordatevi di salvare 
il valore originale di lpfnWndProc se desiderate chiamarlo con CallWindowProc. 

 Registrate le modifiche alla struttura WNDCLASSEX. Avrete una nuova classe di finestre che avrà 
parecchie caratteristiche della vecchia classe di finestre. 

 Create le finestre dalla nuova classe. 

Il Superclassing è migliore del subclassing se desiderate creare molti controlli con le stesse 
caratteristiche. 

Esempio

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib

WM_SUPERCLASS equ WM_USER+5
WinMain PROTO :DWORD,:DWORD,:DWORD,:DWORD
EditWndProc PROTO :DWORD,:DWORD,:DWORD,:DWORD

.data
ClassName  db "SuperclassWinClass",0
AppName    db "Superclassing Demo",0
EditClass  db "EDIT",0
OurClass db "SUPEREDITCLASS",0
Message  db "You pressed the Enter key in the text box!",0

.data?
hInstance dd ?
hwndEdit dd 6 dup(?)
OldWndProc dd ?

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND

    mov wc.cbSize,SIZEOF WNDCLASSEX
    mov wc.style, CS_HREDRAW or CS_VREDRAW
    mov wc.lpfnWndProc, OFFSET WndProc
    mov wc.cbClsExtra,NULL
    mov wc.cbWndExtra,NULL
    push hInst
    pop wc.hInstance
    mov wc.hbrBackground,COLOR_APPWORKSPACE
    mov wc.lpszMenuName,NULL
    mov wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov wc.hIcon,eax
    mov wc.hIconSm,eax
    invoke LoadCursor,NULL,IDC_ARROW
    mov wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE+WS_EX_CONTROLPARENT,ADDR ClassName,ADDR AppName,
        
WS_OVERLAPPED+WS_CAPTION+WS_SYSMENU+WS_MINIMIZEBOX+WS_MAXIMIZEBOX+WS_VISIBLE,CW_USEDEFAULT,



           CW_USEDEFAULT,350,220,NULL,NULL,
           hInst,NULL
    mov hwnd,eax

    .while TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .endw
     mov eax,msg.wParam
    ret
WinMain endp

WndProc proc uses ebx edi hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    LOCAL wc:WNDCLASSEX
    .if uMsg==WM_CREATE
        mov wc.cbSize,sizeof WNDCLASSEX
        invoke GetClassInfoEx,NULL,addr EditClass,addr wc
        push wc.lpfnWndProc
        pop OldWndProc
        mov wc.lpfnWndProc, OFFSET EditWndProc
        push hInstance
        pop wc.hInstance
        mov wc.lpszClassName,OFFSET OurClass
        invoke RegisterClassEx, addr wc
        xor ebx,ebx
        mov edi,20
        .while ebx<6
            invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR OurClass,NULL,
                 WS_CHILD+WS_VISIBLE+WS_BORDER,20,
                 edi,300,25,hWnd,ebx,
                 hInstance,NULL
            mov dword ptr [hwndEdit+4*ebx],eax
            add edi,25
            inc ebx
        .endw
        invoke SetFocus,hwndEdit
    .elseif uMsg==WM_DESTROY
        invoke PostQuitMessage,NULL
    .else
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
WndProc endp

EditWndProc PROC hEdit:DWORD,uMsg:DWORD,wParam:DWORD,lParam:DWORD
    .if uMsg==WM_CHAR
        mov eax,wParam
        .if (al>="0" && al<="9") || (al>="A" && al<="F") || (al>="a" && al<="f") || 
al==VK_BACK
            .if al>="a" && al<="f"
               sub al,20h
            .endif
            invoke CallWindowProc,OldWndProc,hEdit,uMsg,eax,lParam
            ret
        .endif
    .elseif uMsg==WM_KEYDOWN
        mov eax,wParam
        .if al==VK_RETURN
            invoke MessageBox,hEdit,addr Message,addr AppName,MB_OK+MB_ICONINFORMATION
            invoke SetFocus,hEdit
        .elseif al==VK_TAB
            invoke GetKeyState,VK_SHIFT
            test eax,80000000
            .if ZERO?
                invoke GetWindow,hEdit,GW_HWNDNEXT
                .if eax==NULL
                    invoke GetWindow,hEdit,GW_HWNDFIRST
                .endif
            .else
                invoke GetWindow,hEdit,GW_HWNDPREV



                .if eax==NULL
                    invoke GetWindow,hEdit,GW_HWNDLAST
                .endif
            .endif
            invoke SetFocus,eax
            xor eax,eax
            ret
        .else
            invoke CallWindowProc,OldWndProc,hEdit,uMsg,wParam,lParam
            ret
        .endif
    .else
        invoke CallWindowProc,OldWndProc,hEdit,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
EditWndProc endp
end start

Analisi
Il programma creerà una finestra semplice con 6 caselle di modifica " alterate " nella relativa zona client. 
Le caselle di modifica accetteranno soltanto le cifre hex. In realtà, ho modificato l'esempio del subclassing
per fare il superclassing. Il programma comincia normalmente e la parte interessante è quando la finestra
principale viene creata: 

.if uMsg==WM_CREATE
         mov wc.cbSize,sizeof WNDCLASSEX
        invoke GetClassInfoEx,NULL,addr EditClass,addr wc

Dobbiamo in primo luogo riempire la struttura WNDCLASSEX con i dati della classe che desideriamo 
superclassare, in questo caso, essa è la classe EDIT. Ricordatevi che dovete settare il membro cbSize 
della struttura WNDCLASSEX prima di chiamare GetClassInfoEx altrimenti la struttura WNDCLASSEX non 
sarà riempita correttamente. Dopo che GetClassInfoEx ritorna, wc è riempita con tutte le informazioni per
creare una nuova classe di finestre. 

push wc.lpfnWndProc
        pop OldWndProc
        mov wc.lpfnWndProc, OFFSET EditWndProc
        push hInstance
        pop wc.hInstance
        mov wc.lpszClassName,OFFSET OurClass

Adesso dobbiamo modificare alcuni membri di wc. Il primo è il puntatore alla procedura per la finestra. 
Successivamente dobbiamo sostituire la nostra procedura per la finestra al posto di quella originale, 
dobbiamo salvarne il valore precedente in una variabile così che potremo chiamarlo con CallWindowProc. 
Queta tecnica è identica al subclassing eccetto che voi modificate la struttura WNDCLASSEX direttamente
senza dover chiamare SetWindowLong. I seguenti due membri devono essere cambiati altrimenti non 
sarete abilitati a registrare la vostra nuova classe di finestre, hInstance e lpsClassName. Dovete 
rimpiazzare il valore originale di hInstance con l' hInstance del vostro programma. E dovete scegliere un 
nuovo nome per la nuova classe. 

invoke RegisterClassEx, addr wc

Quando tutto è pronto, registrate la nuova classe. Otterrete una nuova classe con alcune caratteristiche 
della vecchia classe. 

xor ebx,ebx
        mov edi,20
        .while ebx<6
            invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR OurClass,NULL,
                 WS_CHILD+WS_VISIBLE+WS_BORDER,20,



                 edi,300,25,hWnd,ebx,
                 hInstance,NULL
            mov dword ptr [hwndEdit+4*ebx],eax
            add edi,25
            inc ebx
        .endw
        invoke SetFocus,hwndEdit

Ora che abbiamo registrato la classe, possiamo creare le finestre basate su essa. Nel precedente snippet, 
uso ebx come contatore del numero di finestre create. edi è usato come coordinata y dell' angolo 
superiore sinistro della finestra. Quando una finestra viene creata, la relativa handle viene memorizzata 
in un array di dwords. Quando tutte le finestre sono state create, regoliamo il fuoco per l'input sulla 
prima finestra. A questo punto, avete ottenuto 6 casella di modifica che accettano soltanto le cifre hex. 
La proc sostitutiva delle finestre gestisce il filtro. Realmente, è identica alla proc delle finestre 
nell'esempio del subclassing. Come potete vedere, non dovete fare il lavoro supplementare del 
subclassing. 

Inserisco uno snippet di codice per maneggiare il percorso del tab per rendere questo esempio più 
sugoso. Normalmente, se mettete i controlli su una finestra di dialogo, la finestre di dialogo gestisce i 
tasti di percorso per voi in modo che voi possono usare il tab per andare al successivo controllo o lo shift-
tab per andare di nuovo al controllo precedente. Viceversa, tale caratteristica non è disponibile se 
mettete i vostri controlli su una finestra semplice. Dovete subclassarli in modo da poter gestire il tasto 
Tab voi stessi. Nel nostro esempio, non abbiamo bisogno di subclassare i controlli uno per uno perché 
sono già superclassati, in modo che possiamo fornire un " gestore centrale della navigazione " per loro. 

.elseif al==VK_TAB
            invoke GetKeyState,VK_SHIFT
            test eax,80000000
            .if ZERO?
                invoke GetWindow,hEdit,GW_HWNDNEXT
                .if eax==NULL
                    invoke GetWindow,hEdit,GW_HWNDFIRST
                .endif
            .else
                invoke GetWindow,hEdit,GW_HWNDPREV
                .if eax==NULL
                    invoke GetWindow,hEdit,GW_HWNDLAST
                .endif
            .endif
            invoke SetFocus,eax
            xor eax,eax
            ret

Il suddetto snippet di codice proviene dalla procedura EditWndClass. Controlla se l'utente preme il tasto 
Tab, in caso affermativo, esso chiama GetKeyState per controllare se il tasto SHIFT è premuto. 
GetKeyState restituisce un valore in eax che determina se il tasto specificato è premuto oppure no. Se il 
tasto è premuto, il bit più alto di eax è settato. Altrimente, il bit più alto è azzerato. Così verifichiamo il 
valore di ritorno con 80000000h. Se il bit più alto è settato, significa che l'utente sta premendo shift+tab 
e dobbiamo gestirlo separatamente. Se l'utente preme solo il tasto Tab, noi chiamiamo GetWindow per 
richiamare l' handle del controllo seguente. Usiamo il flag GW_HWNDNEXT per dire a GetWindow di 
ottenere l' handle della finestra che è la seguente in linea all' hEdit corrente. Se questa funzione 
restituisce NULL, lo interpretiamo come se non ci siano più handle da ottenere così l' hEdit corrente è l' 
ultimo controllo in linea. Noi " salteremo in circolo " al primo controllo chiamando GetWindow con il flag 
GW_HWNDFIRST. Similmente al caso del Tab, shift-tab funzionerà giustamente al contrario. 

Tutorial 23: Tray Icon

Introduzione
In questo tutorial, impareremo come mettere le icone nel system tray e come creare/usare un popup 
menù. 

Tools
Esempio 

Essay
Il system tray è la regione rettangolare nella taskbar dove parecchie icone risiedono. Normalmente, 
vedrete almeno un orologio digitale. Potete anche mettere delle icone nel system tray. Sotto sono 
riportate le operazione che dovete effettuare per mettere un'icona nel system tray: 

1) Riempire una struttura NOTIFYCONDATA che ha i seguenti membri: 

 cbSize La dimensione di questa struttura. 

 hwnd Handle della finestra che riceverà le notifiche quando un evento del mouse si presenterà 
sopra l'icona nel tray. 

 uID Una costante che è usata come ID dell'icona. Siete voi a decidere questo valore. Nel caso 
abbiate più di un' icona nel tray, potrete controllare da quale icona nel tray la notifica del mouse
proviene. 

 uFlags Specifica quali membri di questa struttura sono validi: 

■ NIF_ICON Il membro hIcon è valido. 

■ NIF_MESSAGE Il membro uCallbackMessage è valido. 

■ NIF_TIP Il membro szTip è valido. 

 uCallbackMessage Il messaggio da noi dichiarato che Window trasmetterà alla finestra 
specificata nel membro hwnd quando un' evento del mouse si presenterà sopra l'icona nel tray. 
Dichiarate voi questo messaggio. 

 hIcon L' handle dell'icona che desiderate mettere nel system tray 

 szTip Un array di 64-byte che contiene la stringa di testo che sarà usata come tooltip quando il 
mouse si fermerà sopra l'icona nel tray. 

2) Chiamare Shell_NotifyIcon che è definita in shell32.inc. Questa funzione ha il seguente prototipo: 

Shell_NotifyIcon PROTO dwMessage:DWORD ,pnid:DWORD

dwMessage è il tipo di messaggio che trasmettete alla shell: 

- NIM_ADD Aggiunge un'icona alla status area. 

- NIM_DELETE Cancella un'icona dalla status area. 

- NIM_MODIFY Modifica un'icona nella status area. 

pnid è il puntatore ad una struttura NOTIFYICONDATA riempita di valori adeguati. 

Se desiderate aggiungere un'icona al tray, usate il messaggio NIM_ADD, se desiderate rimuovere l'icona, 
usate NIM_DELETE. 

Questo è tutto ciò che si deve conoscere. Ma la maggior parte delle volte, non sarete soddisfatti di 
mettere appena un'icona nel system tray. Vorrete poter rispondere agli eventi del mouse sopra l'icona nel
tray. Potete fare questo elaborando il messaggio che avete specificato nel membro uCallbackMessage 
della struttura NOTIFYICONDATA. Questo messaggio ha i seguenti valori in wParam ed in lParam (grazie 
mille a s__d per l'Info): 

 wParam contiene l'ID dell'icona. E' lo stesso valore che mettete nel membro uID della struttura 
NOTIFYICONDATA. 

 lParam la low word contiene il messaggio del mouse. Per esempio, se l'utente clicca il tasto destro
del mouse sull'icona, lParam conterrà WM_RBUTTONDOWN. 

La maggior parte delle icone nel tray, tuttavia, visualizzano un popup menù quando l' utente clicca il 
tasto destro su di esse. Possiamo implementare questa funzione creando un popup menù e quindi 
chiamando TrackPopupMenu per visualizzarlo. I punti sono descritti qui sotto: 

1. Creare un popup menù chiamando CreatePopupMenu. Questa funzione crea un menù vuoto. 
Restituisce l' handle del menù in eax se riuscita. 

2. Aggiungere le voci di menù ad esso con AppendMenu, InsertMenu o InsertMenuItem. 
3. Se desiderate visualizzare il popup menù dove si trova il cursore del mouse, chiamate 

GetCursorPos per ottenere le coordinate del cursore nello schermo e quindi chiamate 
TrackPopupMenu per visualizzare il menù. Quando l'utente seleziona una voce del menù, 
Window trasmette il messaggio WM_COMMAND alla vostra procedura per la finestra come una 
selezione normale da un menù. 

Nota: Attenti a due comportamenti anomali quando usate un popup menù con un'icona nel tray: 

1. Quando il popup menù è visualizzato, se cliccate dovunque fuori dal menù, il popup menù non 
sparirà immediatamente come dovrebbe. Questo comportamento accade perché la finestra che 
riceve le notifiche dal popup menù DEVE essere la finestra dalla priorità più alta. La chiamata di 
SetForegroundWindow correggerà il problema. 

2. Dopo aver chiamato SetForegroundWindow, verificherete che in un primo momento il popup 
menù sarà visualizzato, funzionà bene ma nei momenti successivi, il popup menù si mostrerà e 
si chiuderà immediatamente. Questo comportamento è " intenzionale ", cito da MSDN. Il 
passaggio di esecuzione al programma che è il proprietario dell'icona nel tray nell' immediato 
futuro è necessario per risolvere il problema. Potete forzare questo passaggio di esecuzione 
postando un messaggio alla finestra del programma. Usate PostMessage, non SendMessage! 

Esempio



.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
include \masm32\include\shell32.inc
includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\shell32.lib

WM_SHELLNOTIFY equ WM_USER+5
IDI_TRAY equ 0
IDM_RESTORE equ 1000
IDM_EXIT equ 1010
WinMain PROTO :DWORD,:DWORD,:DWORD,:DWORD

.data
ClassName  db "TrayIconWinClass",0
AppName    db "TrayIcon Demo",0
RestoreString db "&Restore",0
ExitString   db "E&xit Program",0

.data?
hInstance dd ?
note NOTIFYICONDATA <>
hPopupMenu dd ?

.code
start:
    invoke GetModuleHandle, NULL
    mov    hInstance,eax
    invoke WinMain, hInstance,NULL,NULL, SW_SHOWDEFAULT
    invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD
    LOCAL wc:WNDCLASSEX
    LOCAL msg:MSG
    LOCAL hwnd:HWND
    mov   wc.cbSize,SIZEOF WNDCLASSEX
    mov   wc.style, CS_HREDRAW or CS_VREDRAW or CS_DBLCLKS
    mov   wc.lpfnWndProc, OFFSET WndProc
    mov   wc.cbClsExtra,NULL
    mov   wc.cbWndExtra,NULL
    push  hInst
    pop   wc.hInstance
    mov   wc.hbrBackground,COLOR_APPWORKSPACE
    mov   wc.lpszMenuName,NULL
    mov   wc.lpszClassName,OFFSET ClassName
    invoke LoadIcon,NULL,IDI_APPLICATION
    mov   wc.hIcon,eax
    mov   wc.hIconSm,eax
    invoke LoadCursor,NULL,IDC_ARROW
    mov   wc.hCursor,eax
    invoke RegisterClassEx, addr wc
    invoke CreateWindowEx,WS_EX_CLIENTEDGE,ADDR ClassName,ADDR AppName,\
WS_OVERLAPPED+WS_CAPTION+WS_SYSMENU+WS_MINIMIZEBOX+WS_MAXIMIZEBOX+WS_VISIBLE,CW_USEDEFAULT,\
           CW_USEDEFAULT,350,200,NULL,NULL,\
           hInst,NULL
    mov   hwnd,eax
    .while TRUE
        invoke GetMessage, ADDR msg,NULL,0,0
        .BREAK .IF (!eax)
        invoke TranslateMessage, ADDR msg
        invoke DispatchMessage, ADDR msg
    .endw
    mov eax,msg.wParam
    ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
    LOCAL pt:POINT
    .if uMsg==WM_CREATE
        invoke CreatePopupMenu



        mov hPopupMenu,eax
        invoke AppendMenu,hPopupMenu,MF_STRING,IDM_RESTORE,addr RestoreString
        invoke AppendMenu,hPopupMenu,MF_STRING,IDM_EXIT,addr ExitString
    .elseif uMsg==WM_DESTROY
        invoke DestroyMenu,hPopupMenu
        invoke PostQuitMessage,NULL
    .elseif uMsg==WM_SIZE
        .if wParam==SIZE_MINIMIZED
            mov note.cbSize,sizeof NOTIFYICONDATA
            push hWnd
            pop note.hwnd
            mov note.uID,IDI_TRAY
            mov note.uFlags,NIF_ICON+NIF_MESSAGE+NIF_TIP
            mov note.uCallbackMessage,WM_SHELLNOTIFY
            invoke LoadIcon,NULL,IDI_WINLOGO
            mov note.hIcon,eax
            invoke lstrcpy,addr note.szTip,addr AppName
            invoke ShowWindow,hWnd,SW_HIDE
            invoke Shell_NotifyIcon,NIM_ADD,addr note
        .endif
    .elseif uMsg==WM_COMMAND
        .if lParam==0
            invoke Shell_NotifyIcon,NIM_DELETE,addr note
            mov eax,wParam
            .if ax==IDM_RESTORE
                invoke ShowWindow,hWnd,SW_RESTORE
            .else
                invoke DestroyWindow,hWnd
            .endif
        .endif
    .elseif uMsg==WM_SHELLNOTIFY
        .if wParam==IDI_TRAY
            .if lParam==WM_RBUTTONDOWN
                invoke GetCursorPos,addr pt
                invoke SetForegroundWindow,hWnd
                invoke TrackPopupMenu,hPopupMenu,TPM_RIGHTALIGN,pt.x,pt.y,NULL,hWnd,NULL
                invoke PostMessage,hWnd,WM_NULL,0,0
            .elseif lParam==WM_LBUTTONDBLCLK
                invoke SendMessage,hWnd,WM_COMMAND,IDM_RESTORE,0
            .endif
        .endif
    .else
        invoke DefWindowProc,hWnd,uMsg,wParam,lParam
        ret
    .endif
    xor eax,eax
    ret
WndProc endp

end start

Analisi
Il programma visualizzerà una finestra semplice. Quando premerete il tasto di minimizzazione, essa si 
nasconderà e metterà un'icona nel system tray. Quando farete un doppio-click sull'icona, la finestra si 
rimostrerà e rimuoverà l'icona dal system tray. Quando farete un click con il pulsante destro sull' icona, 
un popup menù sarà visualizzato. Potrete scegliere se rivisualizzare la finestra o uscire da esso. 

.if uMsg==WM_CREATE
        invoke CreatePopupMenu
        mov hPopupMenu,eax
        invoke AppendMenu,hPopupMenu,MF_STRING,IDM_RESTORE,addr RestoreString
        invoke AppendMenu,hPopupMenu,MF_STRING,IDM_EXIT,addr ExitString

Quando la finestra principale è stata creata, il programma creerà un popup menù e collegherà due voci di
menù. AppendMenu ha la seguente sintassi: 



AppendMenu PROTO hMenu:DWORD, uFlags:DWORD, uIDNewItem:DWORD, lpNewItem:DWORD

 hMenu è l' handle del menù a cui desiderate collegare la voce di menù 

 uFlags dice a Window se la voce di menù che deve essere collegata sarà una bitmap o una 
stringa o una voce owner-draw, abilitata, oscurata o inattiva ecc. Potete ottenere la lista 
completa dal riferimento win32 api. Nel nostro esempio, usiamo MF_STRING che significa che la
voce di menù è una stringa. 

 uIDNewItem è l'ID della voce di menù. Questo è un valore dichiarato dal programmatore che è 
usato per rappresentare la voce di menù. 

 lpNewItem specifica il contenuto della voce di menù, in base a cosa specificate nel membro 
uFlags. Poiché specifichiamo MF_STRING nel membro uFlags, lpNewItem deve contenere il 
puntatore alla stringa da visualizzare nel popup menù. 

Dopo che il popup menù è stato creato, la finestra principale aspetta pazientemente che l'utente prema il 
tasto minimizza. Quando una finestra è minimizzata, riceve il messaggio WM_SIZE con il valore 
SIZE_MINIMIZED in wParam. 

.elseif uMsg==WM_SIZE
        .if wParam==SIZE_MINIMIZED
            mov note.cbSize,sizeof NOTIFYICONDATA
            push hWnd
            pop note.hwnd
            mov note.uID,IDI_TRAY
            mov note.uFlags,NIF_ICON+NIF_MESSAGE+NIF_TIP
            mov note.uCallbackMessage,WM_SHELLNOTIFY
            invoke LoadIcon,NULL,IDI_WINLOGO
            mov note.hIcon,eax
            invoke lstrcpy,addr note.szTip,addr AppName
            invoke ShowWindow,hWnd,SW_HIDE
            invoke Shell_NotifyIcon,NIM_ADD,addr note
        .endif

Usiamo questa occasione per riempire la struttura NOTIFYICONDATA. IDI_TRAY è una costante definita 
all'inizio del codice sorgente. Potete settarla con qualunque valore gradite. Non è importante perché 
avete soltanto un' icona nel tray. Ma se metteste parecchie icone nel system tray, avreste bisogno di un 
ID unico per ogni icona del tray. Impostiamo tutti i flags nel membro uFlags perché specifichiamo 
un'icona (NIF_ICON), specifichiamo un nostro messaggio (NIF_MESSAGE) e specifichiamo il testo del 
tooltip (NIF_TIP). WM_SHELLNOTIFY è giusto un nostro messaggio definito come WM_USER+5. Il valore 
reale non è importante a condizione che sia unico. Nell' esempio uso l'icona winlogo per il tray ma potete 
usare qualsiasi icona del vostro programma. Carichiamo essa dalle risorse con LoadIcon e mettiamo l' 
handle restituita nel membro hIcon. Infine, riempiamo szTip con il testo che desideriamo che la shell 
visualizzi quando il mouse è sopra l'icona. Nascondiamo la finestra principale per dare l'illusione di una " 
minimizzazione-a-icona-nel tray ". Successivamente chiameremo Shell_NotifyIcon con il messaggio 
NIM_ADD per aggiungere l'icona al system tray. 

Ora la nostra finestra principale è nascosta e l'icona è nel system tray. Se passate il mouse sopra ad 
essa, vedrete un tooltip che visualizza il testo che abbiamo messo nel membro szTip. Successivamente, 
se fate un doppio-click sull' icona, la finestra principale riapparirà e l'icona nel tray sparirà. 

.elseif uMsg==WM_SHELLNOTIFY
        .if wParam==IDI_TRAY
            .if lParam==WM_RBUTTONDOWN
                invoke GetCursorPos,addr pt
                invoke SetForegroundWindow,hWnd
                invoke TrackPopupMenu,hPopupMenu,TPM_RIGHTALIGN,pt.x,pt.y,NULL,hWnd,NULL
                invoke PostMessage,hWnd,WM_NULL,0,0
            .elseif lParam==WM_LBUTTONDBLCLK
                invoke SendMessage,hWnd,WM_COMMAND,IDM_RESTORE,0
            .endif
        .endif

Quando un evento del mouse si presenta sopra l'icona nel tray, la vostra finestra riceve il messaggio 
WM_SHELLNOTIFY che è il messaggio che avete specificato nel membro uCallbackMessage. Ricordatevi 
che alla ricezione di questo messaggio, wParam contiene l'ID dell'icona nel tray e lParam contiene il 
messaggio reale del mouse. Nel codice qui sopra, controlliamo in primo luogo se il messaggio che 
proviene dall'icona che è nel tray ci interessa. Se è così, controlliamo il messaggio reale del mouse. 



Poiché siamo soltanto interessati al click destro e al doppio-click-sinistro del mouse, elaboriamo soltanto i
messaggi WM_RBUTTONDOWN e WM_LBUTTONDBLCLK. Se il messaggio del mouse è 
WM_RBUTTONDOWN, chiamiamo GetCursorPos per ottenere le coordinate correnti del cursore del mouse 
nello schermo. Quando la funzione ritorna, la struttura POINT contiene le coordinate del cursore del 
mouse nello schermo. Per coordinate nello schermo, intendo le coordinate dell' intero schermo senza 
considerare nessun contorno di finestra. Per esempio, se la risoluzione dello schermo è 640*480, l' 
angolo di destra-più basso dello schermo è x==639 e y==479. Se desiderate convertire le coordinate 
nello schermo in coordinate della finestra, usate la funzione ScreenToClient. Tuttavia, per il nostro scopo, 
desideriamo visualizzare il popup menù all' attuale posizione del cursore del mouse con la chiamata di 
TrackPopupMenu che richiede le coordinate nello schermo, noi possiamo usare le coordinate restituite 
direttamente da GetCursorPos. TrackPopupMenu ha la seguente sintassi: 

TrackPopupMenu PROTO hMenu:DWORD, uFlags:DWORD,  x:DWORD,  y:DWORD, nReserved:DWORD, 
hWnd:DWORD, prcRect:DWORD

 hMenu è l' handle del popup menù da visualizzare; 

 uFlagsspecifica le opzioni della funzione. Come dove posizionare il menù in relazione alle 
coordinate specificate successivamente e quale tasto del mouse sarà utilizzato per richiamare il 
menù. Nel nostro esempio, usiamo TPM_RIGHTALIGN per posizionare il popup menù alla sinistra
delle coordinate. 

 x e y specificano la posizione del menù nelle coordinate dello schermo. 

 nReserved deve essere NULL 

 hWnd è l' handle della finestra che riceverà i messaggi dal menù. 

 prcRect è il rettangolo nello schermo in cui è possibile cliccare senza far sparire il menù. 
Normalmente noi mettiamo NULL qui così quando l'utente clicca dovunque fuori dal popup 
menù, il menù sparisce. 

Quando l' utente fa' un doppio-click sull'icona nel tray, noi trasmettiamo il messaggio WM_COMMAND alla
nostra finestra specificando IDM_RESTORE per emulare la seleziona della voce di menù Restore da parte 
dell' utente e quindi rivisualizzare la finestra principale e rimuovere l'icona dal system tray. Per poter 
ricevere il messaggio di doppio click, la finestra principale deve avere lo stile CS_DBLCLKS. 

invoke Shell_NotifyIcon,NIM_DELETE,addr note
            mov eax,wParam
            .if ax==IDM_RESTORE
                invoke ShowWindow,hWnd,SW_RESTORE
            .else
                invoke DestroyWindow,hWnd
            .endif

Quando l'utente seleziona la voce di menù Restore, rimuoviamo l'icona dal tray chiamando ancora 
Shell_NotifyIcon, questo volta però specifichiamo NIM_DELETE come messaggio. Quindi, ristabiliamo la 
finestra principale alla relativa condizione originale. Se l'utente seleziona la voce di menù Exit, 
rimuoviamo ancora l'icona dal tray e distruggiamo la finestra principale chiamando DestroyWindow. 

Introduction
Ristrutturato il 2/01/2007 

Un bel giorno decisi di imparare a programmare in assembler. Girando sulla rete inceppai nei FAVOLOSI 
tutorial di Iczelion sulla programmazione a 32 bit in assembler, non quei pallosi programmi stile dos ma 
finalmente dialog, finestre ecc. 

Pultoppo, andando avanti, notai che la traduzione in italiano arrivava fino al 23 su un totale di 35. Tra i 
non tradotti c'era anche qualcosa di veramente interessante per il reversing (tipo le DebugApi). 

Poi mi sono detto: L'inglese lo conosco percui facciamo un favore alla nazione e traduciamo questi cavolo 
di tutorial. Questo fu l'inizio… 

Tools
Allegato (codice sorgente + eseguibile già compilato) 

RadAsm (IDE perfetto per programmare in asm) 

Essay
Teoria: 

I Windows Hooks possono essere considerati una delle caratteristiche più potenti di windows. Con essi, si 
può "agganciare" eventi che accadono in un nostro processo o in altri. Con l' "Hooking" dici a windows di 
una funzione, chiamata filter function (funzione di filtro) o "hook procedure", che verrà chiamata ogni 
volta che l'evento a cui siamo interessati accade. 

Ci sono due tipi di Hook: LOCAL HOOK e REMOTE HOOK. 

1. LOCAL HOOK agganciano un evento che avviene in un nostro processo 
2. REMOTE HOOK agganciano un evento che avviene in altri processi. Ce ne sono due tipi: 

Thread-Specific, i quali agganciano eventi che avvengono in uno specifico thread. In poche parole, li 
userai quando vuoi osservare un evento in uno specifico thread, in un processo specifico. System-wide, 
che agganciano tutti gli eventi che avvengono in qualsiasi thread di qualsiasi processo nel sistema 

Quando installi degli hooks, ricordati che pesano sulla performance del sistema. I System-Wide sono i più
instabili. Quando gli eventi scelti passeranno per la nostra filter function, in nostro computer può 
rallentare di molto. Perciò se decidi di usare i System Hook, dovresti usarli con giudizio e sganciarli 
appena non ti servono più. 

Inoltre, c'è una grande possibilità di crash quando entri in conflitto con altri processi e se ci sono errori 
nella filter function. In questo caso con esso possono terminare anche altri processi. RICORDA: IL 
POTERE VA USATO CON RESPONSABILITA' 

Prima di usarli con efficienza, dobbiamo capire bene come funzionano gli hook. Quando crei un hook, 
windows crea una struttura di dati in memoria contenente informazioni dell'hook, e la aggiunge a una 
linked-list (lista collegata) dove sono presenti tutti gli hook. Il nuovo hook sarà aggiunto davanti agli altri.
Quando l'evento scelto accade, se installi un LOCAL HOOK, sarà chiamata la filter-function del processo e 
così via. Ma, se c'è un REMOTE HOOK, il sistema deve sistemare il codice per l'hook nell' address space 
degli altri processi. E il sistema lo può fare solo se la funzione (filter-function) risiede in una DLL, perciò, 
se vuoi usare un REMOTE HOOK, la tua hook procedure deve stare in una DLL. Ci sono solo due 
eccezioni: JOURNAL RECORD e JOURNAL PLAYBACK HOOKS. La hook procedure di questi due hooks deve 
risiedere nel thread che li installa. I motivi sono che tutti e due gli hook trattano con intercettazioni di 
basso livello degli eventi dell'input dell'hardware. Gli eventi di input devono essere registrati/riavviati 
nell'ordine in cui appaiono. Se il codice è in una DLL, gli eventi di input potrebbero espandersi in altri 
thread ed è impossibile conoscerne l'ordine. La soluzione sarebbe di mettere la procedura di hook di 
questi due in un singolo thread, ovvero il thread che li installa. 

Ci sono 14 tipi di hooks: 

1. WH_CALLWNDPROC, chiamato quando è chiamato un SendMessage 
2. WH_CALLWNDPROCRET, chiamato quando un SendMessage ritorna 
3. WH_GETMESSAGE, chiamato quando è chiamato un GetMessage o un PeekMessage 
4. WH_KEYBOARD, chiamato quando GetMessage o PeekMessage restituiscono WM_KEYUP o 

WM_KEYDOWN alla coda di messaggi 
5. WH_MOUSE, quando GetMessage o PeekMessage danno un qualsiasi messaggio del mouse 
6. WH_HARDARE, quando GetMessage o PeekMessage danno un altro messaggio hardware che 

non sia della tastiera o del mouse 
7. WH_MSGFILTER, chiamato quando in una dialogbox, un menù o una scroll-bar stanno 

processando un messaggio. Questo hook è locale. È specifico per gli oggetti che hanno un 
message-loop interno 

8. WH_SYSMSGFILTER, come sopra, solo che è system-wide 
9. WH_JOURNALRECORD, chiamato quando windows restituisce un messaggio dalla coda degli 

hardware input 
10. WH_JOURNALPLAYBACK, chiamato quando un evento è richiesto dalla coda di sistema degli 

hardware input 
11. WH_SHELL, chiamato quando succede qualcosa di interessante riguardo la shell come quando 

la task-bar ha bisogno di ridisegnare i suoi pulsanti. 
12. WH_CBT usato specialmente per i computer-based training 
13. WH_FOREGROUNDIDLE, usato internamente a windows. Poca utilità per applicazioni classiche
14. WH_DEBUG, usato per debugare la hooking procedure. 

Adesso che conosciamo un po' di teoria, possiamo osservare come si installano/disinstallano gli hooks. 

Per installare un hook, si usa SetWindowsHookEx: 

SetWindowsHookEx PROTO HookType:DWORD, pHookProc:DWORD, hInstance:DWORD, 
ThreadID:DWORD 



 Hook Type è uno dei valori sopra (WH_*) 

 pHookProc è l'indirizzo della hook procedure che verrà chiamata per processare il messaggio per
l'hook specificato. Se è un REMOTE HOOK deve stare in una DLL, altrimenti deve stare nel tuo 
processo 

 hInstance è l'Instance-handle della DLL che contiene la hook procedure. Se è un hook locale, va 
impostata su NULL 

 ThreadID è l'ID del thread dove vuoi che l'hook si agganci per spiare. Questo parametro è quello
che determina se un hook è local o remote. Se il parametro è NULL, Windows vedrà l'hook come
system-wide remote hook che coinvolge tutti i thread del sistema. Se specifichi un ID di un tuo 
thread in un tuo processo, l'hook sarà local; Se l'ID è di un thread di un altro processo, sarà un 
thread-specific remote hook. Ci sono due eccezioni a questa regola: WH_JOURNALRECORD e 
WH_JOURNALPLAYBACK sono sempre local system-wide hooks che non richiedono di stare in 
una DLL e WH_SYSMSGFILTER è sempre un system-wide remote hook. Quest'ultimo è identico 
WH_MSGFILTER con ThreadID==0. 

Se la chiamata ha successo, eax conterrà l'handle dell' hook. Altrimnti eax sarà NULL. Devi salvare 
l'handle per "sganciare" l'hook più tardi. 

Puoi sganciare un hook chiamando UnhookWindowsHookEx, che prende solo un parametro, l'handle 
dell'hook che vuoi sganciare. Se la chiamata riesce, eax sarà !=0 altrimenti sarà NULL. Adesso che sai 
come agganciare/sganciare gli hook, possiamo esaminare la hook procedure. La hook procedure verrà 
chiamata ogni volta che si presenterà l'evento con coi è stata associata. Per esempio, se installi un 
WH_MOUSE, quando si presenta un evento da mouse, la hook procedure verrà chiamata. 
Indipendentemente dal tipo di hook, la hook procedure ha sempre la seguente forma: 

<nome> proto nCode:DWORD, wParam:DWORD, lParam:DWORD 

 nCode specifica il codice dell'hook 

 wParam & lParam contengono informazioni extra nell'evento 

 <nome> può essere sostituito con qualsiasi altro nome. 

Il significato di nCode, lParam, wParam dipende dal tipo di hook installato. Anche il valore di ritorno varia,
per esempio: 

WH_CALLWNDPROC 

-nCode può essere solo HC_ACTION, che ci dice che c'è un messaggio mandato a una finestra 

-wParam contiene il messaggio che è stato mandato 

-lParam punta a una struttura CWPSTRUCT 

-valore di ritorno non è usato, è zero 

WH_MOUSE 

-nCode può essere HC_ACTION o HC_NOREMOVE 

-wParam contiene il messaggio del mouse 



-lParam punta a una struttura MOUSEHOOKSTRUCT 

-valore di ritorno zero se il messaggio deve essere processato, 1 se il messaggio sarà scartato 

Per avere ulteriori informazioni sul significato dei parametri e dei valori di ritorno, cercare nella vostra 
guida API. 

Ricorda che gli hook sono legati a una linked-list, dove, l'hook più recente è in testa. Quando arriva 
l'evento, windows chiamerà il primo hook della catena. La responsabilità di chiamare l'hook seguente è 
della tua hook procedure. Puoi scegliere di non chiamare l'hook seguente, ma sarebbe meglio conoscere 
quello che stai facendo. La maggior parte delle volte è bene fare pratica chiamando la hook-procedure 
seguente, in modo che gli altri hook possano avere il proprio turno. Per chiamare l'hook successivo si usa 
CallNexHookEx che ha il seguente prototipo: 

CallNextHookEx proto hHook:DWORD, nCode:DWORD, wParam:DWORD, lParam:DWORD 

 nHook è l'handle del nostro hook. Questa funzione usa questo handle per sfogliarsi la linked-list e
per cercarsi la hook procedure da chiamare 

 nCode, wParam e lParam possono essere riempiti con quelli restituiti da Windows 

RICORDATE: la hook procedure deve risiedere in una DLL che sarà mappata negli altri processi. Quando 
windows mappa la DLL nei processi, non mapperà la sezione dati. In poche parole, tutti i processi si 
condividono una singola copia di codice (hook procedure) ma hanno ognuno la propria DLL nella sezione 
dati. può essere una brutta sorpresa per gli incauti. Potresti pensare che quando metti un valore in una 
variabile di una DLL, quel valore sarà condiviso tra tutti i processi che la importano nel loro process-
space. Ma non è vero. Nelle situazioni normali, questo comportamento è desiderabile in quanto dà 
l'illusione che ogni processo abbia la propria DLL. Ma non quando ci sono di mezzo gli hook. Noi vogliamo 
che la DLL sia UGUALE per tutti i processi, inclusi i dati. SOLUZIONE: bisogna "marcare" la sezione come 
"condivisa". Puoi farlo specificando gli attributi di sezione della fase di link. Per il masm si fà: 

/SECTION:<section name>, S 

Il nome dei dati inizializzati è .data e quelli dei dati non-inizializzati è .bss. Per esempio, se vuoi creare 
una DLL che contenga una hook procedure e vuoi che la sezione di dati non-inizializzati sia condivisa, 
devi linkare con la seguente riga: 

link /section:.bss,S /DLL /SUBSYSTEM:WINDOWS .......... 

"S" ci dice che la sezione è condivisa 

;------------------------------------ Programma principale----------------------------

.386

.model flat,stdcall    (se volete risparmiarvi tutto questo, usate il mio personale include
file allegato Death_Reaver.inc)

option casemap:none

 



include \masm32\include\windows.inc

include \masm32\include\user32.inc

include \masm32\include\kernel32.inc

 

include mousehook.inc

includelib mousehook.lib

 

includelib \masm32\lib\user32.lib

includelib \masm32\lib\kernel32.lib

 

wsprintfA proto C :DWORD,:DWORD,:VARARG

wsprintf TEXTEQU <wsprintfA>

 

.const

IDD_MAINDLG                   equ 101

IDC_CLASSNAME              equ 1000

IDC_HANDLE                     equ 1001

IDC_WNDPROC                 equ 1002

IDC_HOOK                         equ 1004

IDC_EXIT                           equ 1005

 

WM_MOUSEHOOK             equ WM_USER+6

 

DlgFunc PROTO :DWORD,:DWORD,:DWORD,:DWORD

 

.data

HookFlag dd FALSE

HookText db "&Hook",0

UnhookText db "&Unhook",0

template db "%lx",0

 

.data?

hInstance dd ?

hHook dd ?

 

.code

start:



    invoke GetModuleHandle,NULL

    mov hInstance,eax

    invoke DialogBoxParam,hInstance,IDD_MAINDLG,NULL,addr DlgFunc,NULL

    invoke ExitProcess,NULL

 

DlgFunc proc hDlg:DWORD,uMsg:DWORD,wParam:DWORD,lParam:DWORD

 

    LOCAL hLib:DWORD

    LOCAL buffer[128]:byte

    LOCAL buffer1[128]:byte

    LOCAL rect:RECT

    .if uMsg==WM_CLOSE

        .if HookFlag==TRUE

            invoke UninstallHook

        .endif

        invoke EndDialog,hDlg,NULL

    .elseif uMsg==WM_INITDIALOG

        invoke GetWindowRect,hDlg,addr rect

        invoke SetWindowPos, hDlg, HWND_TOPMOST, rect.left, rect.top, rect.right, rect.bottom,

                      SWP_SHOWWINDOW

    .elseif uMsg==WM_MOUSEHOOK

        invoke GetDlgItemText,hDlg,IDC_HANDLE,addr buffer1,128

        invoke wsprintf,addr buffer,addr template,wParam

        invoke lstrcmpi,addr buffer,addr buffer1

        .if eax!=0

            invoke SetDlgItemText,hDlg,IDC_HANDLE,addr buffer

        .endif

        invoke GetDlgItemText,hDlg,IDC_CLASSNAME,addr buffer1,128

        invoke GetClassName,wParam,addr buffer,128

        invoke lstrcmpi,addr buffer,addr buffer1

        .if eax!=0

            invoke SetDlgItemText,hDlg,IDC_CLASSNAME,addr buffer

        .endif

        invoke GetDlgItemText,hDlg,IDC_WNDPROC,addr buffer1,128

        invoke GetClassLong,wParam,GCL_WNDPROC

        invoke wsprintf,addr buffer,addr template,eax

        invoke lstrcmpi,addr buffer,addr buffer1



        .if eax!=0

            invoke SetDlgItemText,hDlg,IDC_WNDPROC,addr buffer

        .endif

    .elseif uMsg==WM_COMMAND

        .if lParam!=0

            mov eax,wParam

            mov edx,eax

            shr edx,16

            .if dx==BN_CLICKED

                .if ax==IDC_EXIT

                    invoke SendMessage,hDlg,WM_CLOSE,0,0

                .else

                    .if HookFlag==FALSE

                        invoke InstallHook,hDlg

                        .if eax!=NULL

                            mov HookFlag,TRUE

                            invoke SetDlgItemText,hDlg,IDC_HOOK,addr UnhookText

                        .endif

                    .else

                        invoke UninstallHook

                        invoke SetDlgItemText,hDlg,IDC_HOOK,addr HookText

                        mov HookFlag,FALSE

                        invoke SetDlgItemText,hDlg,IDC_CLASSNAME,NULL

                        invoke SetDlgItemText,hDlg,IDC_HANDLE,NULL

                        invoke SetDlgItemText,hDlg,IDC_WNDPROC,NULL

                    .endif

                .endif

            .endif

        .endif

    .else

        mov eax,FALSE

        ret

    .endif

    mov eax,TRUE

    ret

 

DlgFunc endp



end start

 

 ;--------------------------- Codice della DLL-------------------------

 

.386

.model flat,stdcall

option casemap:none

 

include \masm32\include\windows.inc

include \masm32\include\kernel32.inc

includelib \masm32\lib\kernel32.lib

include \masm32\include\user32.inc

includelib \masm32\lib\user32.lib

 

.const

WM_MOUSEHOOK equ WM_USER+6

 

.data

hInstance dd 0

 

.data?

hHook dd ?

hWnd dd ?

 

.code

DllEntry proc hInst:HINSTANCE, reason:DWORD, reserved1:DWORD

 

    .if reason==DLL_PROCESS_ATTACH

        push hInst

        pop hInstance

    .endif

    mov  eax,TRUE

    ret

 

DllEntry Endp

 

MouseProc proc nCode:DWORD,wParam:DWORD,lParam:DWORD



 

    invoke CallNextHookEx,hHook,nCode,wParam,lParam

    mov edx,lParam

    assume edx:PTR MOUSEHOOKSTRUCT

    invoke WindowFromPoint,[edx].pt.x,[edx].pt.y

    invoke PostMessage,hWnd,WM_MOUSEHOOK,eax,0

    assume edx:nothing

    xor eax,eax

    ret

 

MouseProc endp

 

InstallHook proc hwnd:DWORD

 

    push hwnd

    pop hWnd

    invoke SetWindowsHookEx,WH_MOUSE,addr MouseProc,hInstance,NULL

    mov hHook,eax

    ret

 

InstallHook endp

 

UninstallHook proc

 

    invoke UnhookWindowsHookEx,hHook

    ret

 

UninstallHook endp

 

End DllEntry

 

 ;------------------------ MakeFile per la DLL--------------------------

 

NAME=mousehook

$(NAME).dll: $(NAME).obj

        Link /SECTION:.bss,S  /DLL /DEF:$(NAME).def /SUBSYSTEM:WINDOWS /LIBPATH:c:\masm\lib 
             $(NAME).obj

$(NAME).obj: $(NAME).asm



        ml /c /coff /Cp $(NAME).asm

Analisi: 

L'esempio mostrerà una dialog-box con tre edit-control che andranno riempiti con il nome della classe, 
l'handle della finestra e l'indirizzo della windows-procedure associata alla finestra su cui è posato il 
cursore del mouse. Ci sono due pulsanti: "Hook" e "Exit". Quando si preme il pulsante "Hook" il 
programma aggancia l'input del mouse e il testo del pulsante cambia in "UnHook" (sgancia). Quando si 
muove il cursore sopra una finestra, le informazioni sulla finestra saranno mostrate sulla finestra 
principale del nostro programma. Quando si preme il pulsante "UnHook", il programma rimuove l'hook. Il 
programma principale usa una dialog-box come sua finestra. inoltre si definisce un messaggio 
personalizzato, WM_MOUSEHOOK che sarà usato come tramite tra il programma e la hook-DLL. Quando il
programma riceve tale messaggio, wParam contiene l'handle della finestra dove è il cursore. 
Naturalmente usare wParam è stata una decisione arbitraria. Ho deciso di mettere l'handle in wParam per
semplicità. Puoi scegliere il tuo metodo di comunicazione tra il programma e la hook-DLL a piacere. 

.if HookFlag==FALSE
              invoke InstallHook,hDlg
              .if eax!=NULL
                  mov HookFlag,TRUE
                  invoke SetDlgItemText,hDlg,IDC_HOOK,addr UnhookText
              .endif

Il programma definisce un flag,HookFlag, per monitorare lo stato dell'hook. È FALSE se l'hook non è 
installato e è TRUE se l'hook è installato. Quando un utente preme il pulsante "Hook" il programma 
controlla se l'hook è già stato installato. Se non lo è, chiama la funzione InstallHook della DLL per 
installarlo. Nota che passiamo l'handle della dialogbox principale come parametro della funzione cosichè 
la hook-DLL possa inviare i WM_MOUSEMOVE alla giusta finestra, cioè la nostra. Quando il programma è 
caricato, anche la hook-DLL è caricata. Le DLL sono caricate in memoria immediatamente dopo il 
programma principale. La funzione di entry-point della DLL è chiamata prima della prima istruzione del 
programma principale. Perciò quando il programma esegue la DLL, essa è già inizializzata. Noi abbiamo 
messo il seguente codice nella funzione di entry della hook-DLL: 

.if reason==DLL_PROCESS_ATTACH
        push hInst
        pop hInstance
    .endif

Il codice salva la sua l'instance handle in una variabile globale chiamata hInstance per usarla nella 
funzione InstallHook. Finchè la funzione di entry della DLL è chiamata prima delle altre funzioni, 
hInstance è sempre valida. Abbiamo messo hInstance nella sezione .data? in modo che sia utilizzabile da 
tutti i processi. Quando il cursore del mouse punta su una finestra, la hook-DLL è mappata nel processo. 
Immagina che ci sia già una DLL che occupa l'indirizzo di mappatura della hook-DLL, in questo caso la 
hook-DLL potrebbe essere rimappata in un altro indirizzo. Il valore di hInstance sarà aggiornato con il 
nuovo indirizzo. Quando l'utente preme il pulsante UnHook e successivamente Hook, SetWindowsHookEx 
sarà chiamato di nuovo. Però questa volta userà il nuovo indirizzo come instance handle che sarà 
sbagliato poiché nell'esempio l'indirizzo di mappatura della hook-DLL non è stato cambiato. L'hook 
diventerà un hook locale dove puoi agganciare solo gli eventi del mouse della tua finestra. Situazione 
poco desiderabile. 

InstallHook proc hwnd:DWORD
    push hwnd
    pop hWnd
    invoke SetWindowsHookEx,WH_MOUSE,addr MouseProc,hInstance,NULL
    mov hHook,eax
    ret
 InstallHook endp

La funzione InstallHook è molto semplice. Salva l'handle della finestra, passato come suo parametro, in 
una variabile globale chiamata hWnd per uso futuro. Quindi chiama SetWindowsHookEx per installare un 
mouse-hook. Il valore di ritorno di SetWindowsHookEx viene messo in un altra variabile globale chiamata
hHook per poterla poi usare con UnhookWindowsHookEx. Dopo che SetWindowsHookEx è chiamata, il 
mouse-hook è funzionante. Quando si verificherà un evento da mouse nel sistema, è chiamata MouseProc
(la tua procedura di hook). 



MouseProc proc nCode:DWORD,wParam:DWORD,lParam:DWORD
    invoke CallNextHookEx,hHook,nCode,wParam,lParam
    mov edx,lParam
    assume edx:PTR MOUSEHOOKSTRUCT
    invoke WindowFromPoint,[edx].pt.x,[edx].pt.y
    invoke PostMessage,hWnd,WM_MOUSEHOOK,eax,0
    assume edx:nothing
    xor eax,eax
    ret
 MouseProc endp

La prima cosa che fa è chiamare CallNextHookEx per dare agli altri hook la possibilità di processare 
l'evento del mouse. Dopodichè chiama la funzione WindowFromPoint per prendere l'handle della finestra 
che è a specifiche coordinate. Notare che usiamo una struttura POINT nella struttura 
MOUSEHOOKSTRUCT (puntata da lParam) come coordinate attuali del mouse. Poi, mandiamo l'handle al 
programma principale con PostMessage con il messaggio WM_MOUSEHOOK. Una cosa da ricordare: Non 
dovresti usare SendMessage dentro la hook-procedure, essa infatti può causare il blocco del messaggio. 
PostMessage è raccomandato. La struttura MOUSEHOOKSTRUCT è definita sotto: 

MOUSEHOOKSTRUCT STRUCT DWORD
  pt            POINT <>
  hwnd          DWORD      ?
  wHitTestCode  DWORD      ?
  dwExtraInfo   DWORD      ?
 MOUSEHOOKSTRUCT ENDS

 pt è la struttura POINT con le coordinate del cursore del mouse. 

 hwnd è l'handle della finestra che riceverà il messaggio. Di solito è la la finestra dove è il cursore,
ma non sempre. Set una finestra chiama SetCapture l'input del mouse non sarà portato a quella
finestra. Per questo motivo, non uso il membro hwnd ma invece chiamo WindowFromPoint. 

 wHitTestCode specifica il valore di hit-test. Il valore di hit-test dà più informazioni sulla posizione 
del mouse. Specifica su quale parte della finestra è il cursore del mouse. Per la lista completa, 
cerca informazioni su win32.hlp sotto WM_NCHITTEST 

 dwExtraInfo contiene le informazione extra associate col messaggio. Normalmente questo valore 
è impostato chiamando mouse_event dato dalla chiamata a GetMessageExtraInfo. 

Quando la finestra principale riceve WM_MOUSEHOOK, usa l'handle in wParam per prendere informazioni 
dalla finestra. 

.elseif uMsg==WM_MOUSEHOOK
        invoke GetDlgItemText,hDlg,IDC_HANDLE,addr buffer1,128
        invoke wsprintf,addr buffer,addr template,wParam
        invoke lstrcmpi,addr buffer,addr buffer1
        .if eax!=0
            invoke SetDlgItemText,hDlg,IDC_HANDLE,addr buffer
        .endif
        invoke GetDlgItemText,hDlg,IDC_CLASSNAME,addr buffer1,128
        invoke GetClassName,wParam,addr buffer,128
        invoke lstrcmpi,addr buffer,addr buffer1
        .if eax!=0
            invoke SetDlgItemText,hDlg,IDC_CLASSNAME,addr buffer
        .endif
        invoke GetDlgItemText,hDlg,IDC_WNDPROC,addr buffer1,128
        invoke GetClassLong,wParam,GCL_WNDPROC
        invoke wsprintf,addr buffer,addr template,eax
        invoke lstrcmpi,addr buffer,addr buffer1
        .if eax!=0
            invoke SetDlgItemText,hDlg,IDC_WNDPROC,addr buffer
        .endif

Pen non commettere errori, controlliamo che il testo che è nelle editbox sia uguale a quello da immettere.
Se è uguale, passiamo avanti. Prendiamo il nome della classe chiamando GetClassName, l'indirizzo della 
windows-procedure chiamando GetClassLong con GCL_WNDPROC e quindi le formattiamo e le mettiamo 
nei propri edit controls. 

invoke UninstallHook
     invoke SetDlgItemText,hDlg,IDC_HOOK,addr HookText



     mov HookFlag,FALSE
     invoke SetDlgItemText,hDlg,IDC_CLASSNAME,NULL
     invoke SetDlgItemText,hDlg,IDC_HANDLE,NULL
     invoke SetDlgItemText,hDlg,IDC_WNDPROC,NULL

Quando l'utente preme il tasto "UnHook", il programma chiama la funzione UninstallHook dalla hook-DLL.
UninstallHook chiama solo UnHookWindowsHookEx. Dopodichè cambia il testo del pulsante di nuovo in 
"Hook", HookFlag in FALSE e pulisce il contenuto delle edit-controls 

Lo switch del linker nel makefile: 

Link /SECTION:.bss,S /DLL /DEF:$(NAME).def /SUBSYSTEM:WINDOWS 

Specifica la sezione .bss come sezione condivisa per fare in modo che tutti i processi condividano la 
stessa sezione non-inizializzata della hook-DLL. Senza questo switch, la tua hook-DLL non funzionerebbe.

Introduction
Continua la traduzione dei tutorial di Iczelion… 

Tools
Allegato Masm v8 

Essay
Teoria 

Le bitmap possono essere considerate come foto nel computer. Ci sono molti tipi di foto usati dal 
computer ma Windows supporta di standard solo le Windows Bitmap Graphic Files (.bmp).Le bitmap 
trattate in questo tutorial saranno proprio windows bitmap graphic files. La via più facile per usare una 
bitmap è di usarla come risorsa. Ci sono due metodi per farlo. 

1)La si può inserire nel resource script(.rc) così: 

#define IDB_MYBITMAP 100
IDB_MYBITMAP BITMAP "c:\immagini\esempio.bmp"

Questo metodo utilizza una costante per definire una bitmap. La prima riga crea una costante chiamata 
IDB_MYBITMAP che ha il valore 100. Useremo questo nome per "passare" la bitmap al programma. La 
riga successiva dichiara una risorsa di tipo BITMAP. Dice al resource-compiler dove trovare il file bmp. 

2)L'altro metodo usa un nome per rappresentare la bitmap 

MyBitMap BITMAP "c:\immagini\esempio.bmp"

Con questo metodo passi la bitmap al programma tramite la stringa "MyBitMap" invece di un valore. 
Entrambi i metodi vanno bene finché sai che metodo stai usando. Adesso che abbiamo messo la bitmap 
nel file di risorse possiamo andare avanti mostrandola nella client area della nostra finestra. 

1) Chiamiamo LoadBitmap per prendere l'handle della bitmap. LoadBitmap ha il seguente prototipo: 

LoadBitmap proto hInstance:HINSTANCE,lpBitmapName:LPSTR

Questa funzione restituisce l'handle della bitmap. hInstance è l'instance-handel del tuo programma. 
lpBitmapName è un puntatore alla stringa che contiene il nome della bitmap(nel caso del secondo 
metodo). Se stai usando una costante per definire la bitmap, puoi metterne il valore qui. (nell'esempio 
sotto sarà 100). 

Un piccolo esempio in ordine: 

Primo metodo: 

.386

.model flat, stdcall

...............

.const
IDB_MYBITMAP    equ 100
...............
.data?
hInstance  dd ?
..............
.code
.............
   invoke GetModuleHandle,NULL
   mov hInstance,eax
............
   invoke LoadBitmap,hInstance,IDB_MYBITMAP
...........

Secondo Metodo: 

.386

.model flat, stdcall

................



.data
BitmapName  db "MyBitMap",0
...............
.data?
hInstance  dd ?
..............
.code
.............
   invoke GetModuleHandle,NULL
   mov hInstance,eax
............
   invoke LoadBitmap,hInstance,addr BitmapName
...........

2) Ottenere l'handle del Device Context (DC). Lo si può ottenere chiamando BeginPaint in risposta a 
WM_PAINT o chiamando GetDC dovunque. 

3) Creare un device context in memoria che abbia gli stessi attributi del device context appena ottenuto. 
L'idea è di creare una specie di superficie invisibile dove possiamo disegnare la nostra bitmap. Quando 
abbiamo finito questa operazione, basta copiare il contenuto della superficie invisibile nell'attuale device 
context con una funzione. Questo è un esempio della tecnica a "doppio buffer" usata per mostrare 
velocemente le immagini sullo schermo. Si puo creare questa superfice invisibile chiamando 
CreateCompatibleDC. 

CreateConpatibleDC proto hdc:HDC

Se la chiamata riesce, restituisce in eax l'handle del memory device context. hdc è l'handle del device 
context del quale vuoi che sia compatibile. 

4) Adesso che abbiamo la nostra superficie invisibile, possiamo disegnare inserendo la bitmap dentro al 
mDC(memory device context). Questo si può fare chiamando SelectObject con l'handle del mDC come 
primo parametro e l'handle della bitmap come secondo. SelectObject ha il seguente prototipo: 

SelectObject proto hdc:HDC, hGdiObject:DWORD

5) Ora la bitmap è disegnate nel mDC. Quello che rimane da fare è copiarlo nell'attuale device context, 
ossia il vero device context. Ci sono tante funzioni che ci permettono di farlo come BitBlt e StretchBlt. 
BitBlt copia semplicemente il contenuto di un DC a un altro e per questo è veloce, mentre, StretchBlt può
estendere o comprimere la bitmap per riempire l'area di output. Qui useremo BitBlt: 

BitBlt proto hdcDest:DWORD, nxDest:DWORD, nyDest:DWORD, 
nWidth:DWORD, nHeight:DWORD, hdcSrc:DWORD, nxSrc:DWORD,
nySrc:DWORD, dwROP:DWORD



hdcDest è l'handle del DC di destinazione 

nxDest,nyDest sono le coordinate dell'angolo superiore sinistro della zna di destinazione 

nWidth,nHeight sono la larghezza e l'altezza della zona di destinazione 

hdcSource è l'handle del DC da copiare 

nxSrc,nySrc sono le coordinate delangolo superiore sinistro da cui iniziare a copiare 

dwROP è il codice di "rasterizzazione" (acronimo ROP) che decide come combinare i dati della bitmap con 
il colore già esistente nella zona di destinazione per 

ottenere il risultato finale. La maggior parte delle volte si vuole sovrascrivere il colore esistente con uno 
nuovo. 

6) Quando hai finito con la bitmap, cancellala con una chaimata a DeleteObject. 

Finito! Per ricapitolare: hai bisogno di mettere la bitmap nel resource script. Quindi la carichi con 
LoadBitmap. Otterrai l'handle della bitmap. Dopo prendi l'handle del DC dell'area dove vuoi disegnare. 
Quindi crei un device context in memoria compatibile con il device context precedentemente ottenuto. 
Metti la bitmap nel mDC ottenuto e copi il contenuto del mDC nel DC reale. 

Esempio: 

.386

.model flat,stdcall

option casemap:none

include \masm32\include\windows.inc

include \masm32\include\user32.inc

include \masm32\include\kernel32.inc

include \masm32\include\gdi32.inc

includelib \masm32\lib\user32.lib

includelib \masm32\lib\kernel32.lib

includelib \masm32\lib\gdi32.lib

 

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD

IDB_MAIN   equ 1

 

.data

ClassName db "SimpleWin32ASMBitmapClass",0

AppName  db "Win32ASM Simple Bitmap Example",0

 



.data?

hInstance HINSTANCE ?

CommandLine LPSTR ?

hBitmap dd ?

 

.code

start:

 invoke GetModuleHandle, NULL

 mov    hInstance,eax

 invoke GetCommandLine

 mov    CommandLine,eax

 invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT

 invoke ExitProcess,eax

 

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD

 LOCAL wc:WNDCLASSEX

 LOCAL msg:MSG

 LOCAL hwnd:HWND

 mov   wc.cbSize,SIZEOF WNDCLASSEX

 mov   wc.style, CS_HREDRAW or CS_VREDRAW

 mov   wc.lpfnWndProc, OFFSET WndProc

 mov   wc.cbClsExtra,NULL

 mov   wc.cbWndExtra,NULL

 push  hInstance

 pop   wc.hInstance

 mov   wc.hbrBackground,COLOR_WINDOW+1

 mov   wc.lpszMenuName,NULL

 mov   wc.lpszClassName,OFFSET ClassName

 invoke LoadIcon,NULL,IDI_APPLICATION

 mov   wc.hIcon,eax

 mov   wc.hIconSm,eax

 invoke LoadCursor,NULL,IDC_ARROW

 mov   wc.hCursor,eax

 invoke RegisterClassEx, addr wc

 INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\

           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\

           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\



           hInst,NULL

 mov   hwnd,eax

 invoke ShowWindow, hwnd,SW_SHOWNORMAL

 invoke UpdateWindow, hwnd

 .while TRUE

  invoke GetMessage, ADDR msg,NULL,0,0

  .break .if (!eax)

  invoke TranslateMessage, ADDR msg

  invoke DispatchMessage, ADDR msg

 .endw

 mov     eax,msg.wParam

 ret

WinMain endp

 

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM

   LOCAL ps:PAINTSTRUCT

   LOCAL hdc:HDC

   LOCAL hMemDC:HDC

   LOCAL rect:RECT

   .if uMsg==WM_CREATE

      invoke LoadBitmap,hInstance,IDB_MAIN

      mov hBitmap,eax

   .elseif uMsg==WM_PAINT

      invoke BeginPaint,hWnd,addr ps

      mov    hdc,eax

      invoke CreateCompatibleDC,hdc

      mov    hMemDC,eax

      invoke SelectObject,hMemDC,hBitmap

      invoke GetClientRect,hWnd,addr rect

      invoke BitBlt,hdc,0,0,rect.right,rect.bottom,hMemDC,0,0,SRCCOPY

      invoke DeleteDC,hMemDC

      invoke EndPaint,hWnd,addr ps

 .elseif uMsg==WM_DESTROY

  invoke DeleteObject,hBitmap

  invoke PostQuitMessage,NULL

 .ELSE

  invoke DefWindowProc,hWnd,uMsg,wParam,lParam



  ret

 .ENDIF

 xor eax,eax

 ret

WndProc endp

end start
 
 ;---------------------------------------------------------------------
 ;                            Il resource script
 ;---------------------------------------------------------------------
 #define IDB_MAIN 1
 
 IDB_MAIN BITMAP "tweety78.bmp"

Analisi: Non c'è molto da analizzare in questo tutorial ;) 

#define IDB_MAIN 1
IDB_MAIN BITMAP "tweety78.bmp"

Definiamo una costante chiamata IDB_MAIN con valore 1.Usiamo questo valore come ID della bitmap. Il 
file incluso nelle risorse è "tweety78.bmp" che stà nella stessa cartella del resource script. 

.if uMsg==WM_CREATE
     invoke LoadBitmap,hInstance,IDB_MAIN
     mov hBitmap,eax

In risposta a WM_CREATE chiamiamo LoadBitmap per caricare la bitmap dalle risorse, passandogli l'ID 
della bitmap come secondo parametro. Otterremo l'handle della bitmap. Adesso che la bitmap è caricata, 
possiamo disegnarla nella client-area della nostra finestra. 

.elseif uMsg==WM_PAINT
     invoke BeginPaint,hWnd,addr ps
     mov    hdc,eax
     invoke CreateCompatibleDC,hdc
     mov    hMemDC,eax
     invoke SelectObject,hMemDC,hBitmap
     invoke GetClientRect,hWnd,addr rect
     invoke BitBlt,hdc,0,0,rect.right,rect.bottom,hMemDC,0,0,SRCCOPY
     invoke DeleteDC,hMemDC
     invoke EndPaint,hWnd,addr ps

Abbiamo scelto di disegnare la bitmap in risposta a WM_PAINT. Prima chiamiamo BeginPaint per ottenere
l'handle del DC. Quindi creiamo un DC compatibile in memoria con CreateCompatibleDC. Dopo mettiamo 
la bitmap nel DC in memoria con SelectObject. Determiniamo le dimensioni della client area con 
GetClientRect. 

Possiamo ora mostrare la bitmap nella client-area chiamando BitBlt che copia la bitmap dal mDC al DC 
reale. Quando abbiamo finito di disegnare, non abbiamo più bisogno del mDC e quindi lo cancelliamo con 
DeleteDC. Terminiamo la sezione di disegno con EndPaint. 



.elseif uMsg==WM_DESTROY
 invoke DeleteObject,hBitmap
 invoke PostQuitMessage,NULL

Quando non avremo più bisogno della bitmap, la cancelliamo con DeleteObject. 

Introduzione
Ora che sappiamo come usare una bitmap, siamo in grado di fare qualcosa di più avanzato con essa. 

Tools
Esempio 

Essay
Una splash-screen è una finestra che non ha la barra del titolo, il system menu, bordi che mostra soltanto
una bitmap per un pò e poi sparisce automaticamente. Di solito è usata durante l'avvio di un programma 
per mostrare il logo di tale programma o distrarre l'attenzione dell'utente mentre il programma ha lunghe
fasi di inizializzazione. In questo tutorial implementeremo una splash-screen. Il primo passo è di 
includere la bitmap nel file di risorse. Se ci pensi però, è una grande perdita di memoria caricare la 
bitmap che useremo solo una volta e tenerla in memoria fino alla chiusura del programma. Una soluzione
ottimale sarebbe di creare una "DLL di risorse" che contenga la bitmap e che abbia il solo compito di 
mostrare lo splash-screen. In questo modo puoi caricare la DLL quando vuoi mostrare la splash-screen e 
scaricarla quando non serve più. Perciò avremo due moduli: il programma principale e la "splash-DLL". 
Metteremo la bitmap nelle risorse della DLL. 

Lo schema generale che seguiremo sarà questo: 

1. Mettere la bitmap nella DLL come risorsa 
2. Il programma principale chiamerà LoadLibrary per caricare la DLL. 
3. La funzione di entry-point della DLL sarà chiamata. Creerà un timer settando il tempo in cui la 

splash-screen resterà visibile. Dopo registreremo e creeremo una finestra senza titolo e bordi e 
infine metteremo la bitmap nella client-area. 

4. Quando il tempo specificato sopra sarà passato, la splash-screen sarà rimossa dallo schermo e il 
controllo tornerà al programma principale. 

5. Il programma principale chiamerà FreeLibrary per scaricare la DLL dalla memoria e quindi farà i 
compiti che deve. 

Esamineremo i punti in dettaglio. 

Caricamento/Scaricamento DLL

Puoi caricare dinamicamente una DLL con LoadLibrary che ha la seguente sintassi: 

LoadLibrary PROTO lpDLLName:DWORD

Prende solo un parametro: l'indirizzo al quale è contenuto il nome della DLL da caricare in memoria. Se la
chiamata ha successo, ritorna con l'handle della DLL altrimenti ritorna con NULL. Per scaricare la DLL si 
chiama FreeLibrary 

FreeLibrary proto hLib:DWORD

Prende solo un parametro: l'handle della DLL da scaricare. Normalmente l'handle lo prendi da 
LoadLibrary. 

Come usare un Timer

Per prima cosa devi crearlo con SetTimer 

SetTimer proto hWnd:DWORD,TimerID:DWORD, uElapse:DWORD, lpTimerFunc:DWORD

hWnd è l'handle della finestra che riceverà i messaggi di notifica. Questo parametro può essere NULL per
specificare che non c'è nessuna finestra associata al timer; 

TimerID è un valore definito dell'utente che è usata come ID del timer; 

uElapse è l‘intervallo (ogni quando manda i messaggi di notifica, in millisecondi); 

lpTimerFunc è l'indirizzo di una funzione che processerà il messaggio di notifica del timer. Se è settato a
NULL i messaggi del timer saranno mandati alla finestra specificata da hWnd (WM_TIMER); 

SetTimer, se ha successo, restituisce l'ID di un timer. Altrimenti restituisce NULL. Quindi è meglio non 
associare a una TimerID zero 

Puoi creare un timer in due modi: 

 Se hai una finestra e vuoi che i messaggio di notifica del timer vadano alla tua finestra devi 
passare tutti i parametri a SetTimer e inoltre lpTimerFunc deve essere NULL. 

 Se non hai una finestra o non vuoi processare i messaggi del timer nella WndProc, devi passare 
NULL al valore di hWnd. Devi inoltre specificare l'indirizzo della timer-function che processerà i 
messaggi. Useremo il primo approccio nell'esempio. 

Quando il periodo di intervallo scade, un WM_TIMER è mandato alla finestra associata al timer. Per 
esempio se specifichi uElapse a 1000, la tua finestra riceverà un WM_TIMER ogni secondo. 

Quando non hai più bisogno di un timer, va distrutto con KillTimer 

KillTimer proto hWnd:DWORD,TimerID:DWORD

Esempio

;---------------------------------------------------------------------------------------------
-----------
;                         Il programma principale
;---------------------------------------------------------------------------------------------
-----------
.386
.model flat,stdcall
option casemap:none



include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib

WinMain proto :DWORD,:DWORD,:DWORD,:DWORD

.data
ClassName db "SplashDemoWinClass",0
AppName  db "Splash Screen Example",0
Libname db "splash.dll",0

.data?
hInstance HINSTANCE ?
CommandLine LPSTR ?
.code
start:
 invoke LoadLibrary,addr Libname
 .if eax!=NULL
    invoke FreeLibrary,eax
 .endif
 invoke GetModuleHandle, NULL
 mov    hInstance,eax
 invoke GetCommandLine
 mov    CommandLine,eax
 invoke WinMain, hInstance,NULL,CommandLine, SW_SHOWDEFAULT
 invoke ExitProcess,eax

WinMain proc hInst:HINSTANCE,hPrevInst:HINSTANCE,CmdLine:LPSTR,CmdShow:DWORD
 LOCAL wc:WNDCLASSEX
 LOCAL msg:MSG
 LOCAL hwnd:HWND
 mov   wc.cbSize,SIZEOF WNDCLASSEX
 mov   wc.style, CS_HREDRAW or CS_VREDRAW
 mov   wc.lpfnWndProc, OFFSET WndProc
 mov   wc.cbClsExtra,NULL
 mov   wc.cbWndExtra,NULL
 push  hInstance
 pop   wc.hInstance
 mov   wc.hbrBackground,COLOR_WINDOW+1
 mov   wc.lpszMenuName,NULL
 mov   wc.lpszClassName,OFFSET ClassName
 invoke LoadIcon,NULL,IDI_APPLICATION
 mov   wc.hIcon,eax
 mov   wc.hIconSm,eax
 invoke LoadCursor,NULL,IDC_ARROW
 mov   wc.hCursor,eax
 invoke RegisterClassEx, addr wc
 INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,
           WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,
           hInst,NULL
 mov   hwnd,eax
 invoke ShowWindow, hwnd,SW_SHOWNORMAL
 invoke UpdateWindow, hwnd
 .while TRUE
  invoke GetMessage, ADDR msg,NULL,0,0
  .break .if (!eax)
  invoke TranslateMessage, ADDR msg
  invoke DispatchMessage, ADDR msg
 .endw
 mov     eax,msg.wParam
 ret
WinMain endp

WndProc proc hWnd:HWND, uMsg:UINT, wParam:WPARAM, lParam:LPARAM
 .IF uMsg==WM_DESTROY
  invoke PostQuitMessage,NULL
 .ELSE
  invoke DefWindowProc,hWnd,uMsg,wParam,lParam
  ret
 .ENDIF
 xor eax,eax



 ret
WndProc endp
end start

;---------------------------------------------------------------------------------------------
-----------
;                            La Bitmap DLL
;---------------------------------------------------------------------------------------------
-----------
.386
.model flat, stdcall
include \masm32\include\windows.inc
include \masm32\include\user32.inc
include \masm32\include\kernel32.inc
include \masm32\include\gdi32.inc
includelib \masm32\lib\user32.lib
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\gdi32.lib
.data
BitmapName db "MySplashBMP",0
ClassName db "SplashWndClass",0
hBitMap dd 0
TimerID dd 0

.data
hInstance dd ?

.code

DllEntry proc hInst:DWORD, reason:DWORD, reserved1:DWORD
   .if reason==DLL_PROCESS_ATTACH  ; quando il processo è caricato
      push hInst
      pop hInstance
      call ShowBitMap
   .endif
   mov eax,TRUE
   ret
DllEntry Endp
ShowBitMap proc
        LOCAL wc:WNDCLASSEX
        LOCAL msg:MSG
        LOCAL hwnd:HWND
        mov   wc.cbSize,SIZEOF WNDCLASSEX
        mov   wc.style, CS_HREDRAW or CS_VREDRAW
        mov   wc.lpfnWndProc, OFFSET WndProc
        mov   wc.cbClsExtra,NULL
        mov   wc.cbWndExtra,NULL
        push  hInstance
        pop   wc.hInstance
        mov   wc.hbrBackground,COLOR_WINDOW+1
        mov   wc.lpszMenuName,NULL
        mov   wc.lpszClassName,OFFSET ClassName
        invoke LoadIcon,NULL,IDI_APPLICATION
        mov   wc.hIcon,eax
        mov   wc.hIconSm,0
        invoke LoadCursor,NULL,IDC_ARROW
        mov   wc.hCursor,eax
        invoke RegisterClassEx, addr wc
        INVOKE CreateWindowEx,NULL,ADDR ClassName,NULL,
           WS_POPUP,CW_USEDEFAULT,
           CW_USEDEFAULT,250,250,NULL,NULL,
           hInstance,NULL
        mov   hwnd,eax
        INVOKE ShowWindow, hwnd,SW_SHOWNORMAL
        .WHILE TRUE
                INVOKE GetMessage, ADDR msg,NULL,0,0
                .BREAK .IF (!eax)
                INVOKE TranslateMessage, ADDR msg
                INVOKE DispatchMessage, ADDR msg
        .ENDW
        mov     eax,msg.wParam
        ret
ShowBitMap endp
WndProc proc hWnd:DWORD,uMsg:DWORD,wParam:DWORD,lParam:DWORD



        LOCAL ps:PAINTSTRUCT
        LOCAL hdc:HDC
        LOCAL hMemoryDC:HDC
        LOCAL hOldBmp:DWORD
        LOCAL bitmap:BITMAP
        LOCAL DlgHeight:DWORD
        LOCAL DlgWidth:DWORD
        LOCAL DlgRect:RECT
        LOCAL DesktopRect:RECT

        .if uMsg==WM_DESTROY
                .if hBitMap!=0
                        invoke DeleteObject,hBitMap
                .endif
                invoke PostQuitMessage,NULL
        .elseif uMsg==WM_CREATE
                invoke GetWindowRect,hWnd,addr DlgRect
                invoke GetDesktopWindow
                mov ecx,eax
                invoke GetWindowRect,ecx,addr DesktopRect
                push  0
                mov  eax,DlgRect.bottom
                sub  eax,DlgRect.top
                mov  DlgHeight,eax
                push eax
                mov  eax,DlgRect.right
                sub  eax,DlgRect.left
                mov  DlgWidth,eax
                push eax
                mov  eax,DesktopRect.bottom
                sub  eax,DlgHeight
                shr  eax,1
                push eax
                mov  eax,DesktopRect.right
                sub  eax,DlgWidth
                shr  eax,1
                push eax
                push hWnd
                call MoveWindow
                invoke LoadBitmap,hInstance,addr BitmapName
                mov hBitMap,eax
                invoke SetTimer,hWnd,1,2000,NULL
                mov TimerID,eax
        .elseif uMsg==WM_TIMER
                invoke SendMessage,hWnd,WM_LBUTTONDOWN,NULL,NULL
                invoke KillTimer,hWnd,TimerID
        .elseif uMsg==WM_PAINT
                invoke BeginPaint,hWnd,addr ps
                mov hdc,eax
                invoke CreateCompatibleDC,hdc
                mov hMemoryDC,eax
                invoke SelectObject,eax,hBitMap
                mov hOldBmp,eax
                invoke GetObject,hBitMap,sizeof BITMAP,addr bitmap
                invoke StretchBlt,hdc,0,0,250,250,
                       hMemoryDC,0,0,bitmap.bmWidth,bitmap.bmHeight,SRCCOPY
                invoke SelectObject,hMemoryDC,hOldBmp
                invoke DeleteDC,hMemoryDC
                invoke EndPaint,hWnd,addr ps
        .elseif uMsg==WM_LBUTTONDOWN
                invoke DestroyWindow,hWnd
        .else
                invoke DefWindowProc,hWnd,uMsg,wParam,lParam
                ret
        .endif
        xor eax,eax
        ret
WndProc endp

End DllEntry



Analisi
Esamineremo il codice del programma per primo. 

invoke LoadLibrary,addr Libname
 .if eax!=NULL
    invoke FreeLibrary,eax
 .endif

Chiamiamo LoadLibrary per caricare la DLL chiamata "splash.dll". Dopodichè la scarichiamo con 
FreeLibrary. LoadLibrary ritornerà solo quando la DLL ha finito la sua inizializzazione. E' tutto quello che 
fà il programma. Il bello è nella DLL. 

.if reason==DLL_PROCESS_ATTACH  ; quando il processo è caricato
      push hInst
      pop hInstance
      call ShowBitMap

quando la DLL è caricata, Windows chiama la sua funzione di entry-point con il flag 
DLL_PROCESS_ATTACH. Usiamo questa opportunità per mostrare la splash-screen. 

Per prima cosa, mettiamo al sicuro l'handle della DLL per uso futuro. Quindi chiamiamo una funzione 
(ShowBitMap) la quale fa il vero lavoro. ShowBitMap registra una window-class(classe di finestra), crea 
una finestra ed entra in message-loop come al solito. La parte interessante è nella chiamata a 
CreateWindowsEx: 

INVOKE CreateWindowEx,NULL,ADDR ClassName,NULL,
           WS_POPUP,CW_USEDEFAULT,
           CW_USEDEFAULT,250,250,NULL,NULL,
           hInstance,NULL

Notare che lo stile finestra è solo WS_POPUP che creerà una finestra senza bordi e senza titolo. Inoltre 
limitiamo l'altezza e la larghezza della finestra a 250x250 pixel. 

Quando la finestra è creata, in WM_CREATE moviamo la finestra al centro dello schermo con il seguente 
codice: 

invoke GetWindowRect,hWnd,addr DlgRect
                invoke GetDesktopWindow
                mov ecx,eax
                invoke GetWindowRect,ecx,addr DesktopRect
                push  0
                mov  eax,DlgRect.bottom
                sub  eax,DlgRect.top
                mov  DlgHeight,eax
                push eax
                mov  eax,DlgRect.right
                sub  eax,DlgRect.left
                mov  DlgWidth,eax
                push eax
                mov  eax,DesktopRect.bottom
                sub  eax,DlgHeight
                shr  eax,1
                push eax
                mov  eax,DesktopRect.right
                sub  eax,DlgWidth
                shr  eax,1
                push eax
                push hWnd
                call MoveWindow

Restituisce le dimenzioni del desktop e la finestra quindi calcola le giuste coordinate dall'angolo superiore 
sinistro per mettersi al centro. 



invoke LoadBitmap,hInstance,addr BitmapName
                mov hBitMap,eax
                invoke SetTimer,hWnd,1,2000,NULL
                mov TimerID,eax

Poi carica la bitmap dal file di risorse con LoadBitmap e crea un timer con ID == 1 e l'intervallo di 2 
secondi (2000 mmsec). Il timer manderà WM_TIMER alla finestra ogni 2 secondi. 

.elseif uMsg==WM_PAINT
                invoke BeginPaint,hWnd,addr ps
                mov hdc,eax
                invoke CreateCompatibleDC,hdc
                mov hMemoryDC,eax
                invoke SelectObject,eax,hBitMap
                mov hOldBmp,eax
                invoke GetObject,hBitMap,sizeof BITMAP,addr bitmap
                invoke StretchBlt,hdc,0,0,250,250,
                       hMemoryDC,0,0,bitmap.bmWidth,bitmap.bmHeight,SRCCOPY
                invoke SelectObject,hMemoryDC,hOldBmp
                invoke DeleteDC,hMemoryDC
                invoke EndPaint,hWnd,addr ps

Quando la finestra riceve un WM_PAINT, crea un DC, un mDC (device context in memoria), ci inserisce la
bitmap, ne ottiene la grandezza con GetObject e infine mette la bitmap nella finestra con StretchBlt che 
funziona come BitBlt ma può allargare o stringere la bitmap per raggiungere la dimensioni volute. In 
questo caso vogliamo che la bitmap riempia la finestra percui usiamo StretchBlt invece che BitBlt. Dopo 
questa operazione cancelliamo il DC. 

.elseif uMsg==WM_LBUTTONDOWN
                invoke DestroyWindow,hWnd

Sarebbe frustrante per l'utente aspettare la scomparsa della splash-screen. Possiamo risolvere questo 
problema con una scelta: quando clicka sulla splash-screen, questa sparisce. E' l'unico motivo per il quale
processiamo WM_LBUTTONDOWN nella DLL. Quando riceve tale messaggio, la finestra viene distrutta 
chiamando DestroyWindow. 

.elseif uMsg==WM_TIMER
                invoke SendMessage,hWnd,WM_LBUTTONDOWN,NULL,NULL
                invoke KillTimer,hWnd,TimerID

Se l'utente sceglie di aspettare, la splash-screen sparirà quando il timer avrà raggiunto l'intervallo 
specificato (2 secondi). Possiamo farlo processando il messaggio WM_TIMER. Quando si riceviamo tale 
messaggio chiudiamo la finestra mandando un WM_LBUTTONDOWN alla finestra (con SendMessage). In 
questo modo ci risparmiamo di scrivere due volte il codice. Non abbiamo più bisogno del timer, percui lo 
distruggiamo con KillTimer. Quando la finestra è chiusa, la DLL ridarà il controllo al programma 
principale. 

Tutorial 27: Controllo Tooltip

Introduzione
Impareremo circa il Controllo Tooltip. Che cos'è, come crearlo e come usarlo. 

Tools
Esempio 

Essay
Un tooltip è una piccola finestra rettangolare che è mostrata quando il puntatore del mouse sta su 
un'area specifica. Una finestra tooltip contiene del testo che il programmatore vuole che sia mostrato. In 
questo caso, un tooltip ha lo stesso ruolo della status-bar ma sparisce quando l'utente clicka o muove il 
mouse fuori dall'area designata. Probabilmente sei abituato a vedere i tooltip associati ai pulsanti di una 
toolbar. Quei tooltip sono create automaticamente con la toolbar. Se vuoi i tooltips per altri 
controlli/finestre hai bisogno di crearli da te. Adesso che sai che è un tooltip, vediamo come crearlo e 
usarlo. I principali passi sono: 

1. Creare un controllo tooltip con CreateWindowsEx 
2. Definire una regione che tooltip controllerà circa i movimenti del puntatore del mouse 
3. Registrare e passare la regione al tooltip 
4. Controllare i messaggi del mouse sul tooltip. (questo passo può servire farlo subito, dipende dal 

metodo usato per prendere i messaggi) 

Esamineremo ogni passo in seguito. 

Creazione del Tooltip

Il tooltip è un controllo comune. Per questo devi chiamare InitCommonControls da qualche parte nel tuo 
codice cosicché MASM linkerà il tuo programma con comctl32.dll. I controlli tooltip si creano con 
CreateWindowEx. La situazione sarà come questa: 

.data
    TooltipClassName db "Tooltips_class32",0
    .code
    .....
    invoke InitCommonControls
    invoke CreateWindowEx, NULL, addr TooltipClassName, NULL, TIS_ALWAYSTIP, CW_USEDEFAULT, 
CW_USEDEFAULT, CW_USEDEFAULT, CW_USEDEFAULT, NULL, NULL, hInstance, NULL

Notare lo stile TIS_ALWAYSTIP. Questo stile specifica che il tooltip sarà mostrato quando il puntatore 
del mouse sarà sull'area designata, fregandosene (rende meglio il concetto. NTD) dello stato attuale della
finestra che contiene tale area. Se uso questo flag, quando metti il puntatore del mouse sull'area 
registrata(al tooltip), il tooltip apparirà, anche se la finestra puntata dal mouse è inattiva. Non c'è bisogno
di includere i stili WS_POPUP e WS_EX_TOOLWINDOW dentro CreateWindowEx perchè la windows-
procedure del tooltip li aggiunge automaticamente. Inoltre non serve specificare le coordinate e la 
grandezza del tooltip: il tooltip aggiusterà tali valori automaticamente per mostrare interamente il testo 
contenuto in esso e perciò li riempiamo con CW_USEDEFAULT in tutti e 4 i parametri. 

Specificare il "Tool"

Il tooltip è creato ma non è mostrato immediatamente. Noi vogliamo che il tooltip sia mostrato quando il 
puntatore del mouse punta in una qualche area. Adesso è arrivato il momento di specificare tale area. 
Questa area è chiamata "tool". Un tool è un'area rettangolare della client area di una finestra che il tooltip
controllerà per verificare se è presente il cursore del mouse. Se il cursore punta al tool, la finestra di 
tooltip apparirà. L'area rettangolare può coprire l'intera client-area o solo una parte. Per questo motivo, 
possiamo dividere i tool in due tipi: uno che è implementato come una finestra e l'altro è implementato 
come un'area rettangolare della client-area di una finestra. Entrambi trovano applicazioni. Il tool che 
copre l'intera area di una finestra è perlopiù usato con controlli come button, editbox e così via. In questo
caso non devi specificare le coordinate del tool, poiché questo coprirà l'intera client-area. Il tool che si 
basa su una zona rettangolare sulla client-area è utile quando vuoi dividere la client-area in tante 
"regioni" senza usare child-windows (finestre, controlli NTD). Con questo tipo di tool devi specificare le 
coordinate dell'angolo superiore sinistro e dimensioni del tool (altezza e larghezza). Il tool viene 
specificato con la struttura TOOLINFO che ha la seguente definizione: 

TOOLINFO STRUCT

  cbSize           DWORD      ?

  uFlags           DWORD      ?

  hWnd             DWORD      ?

  uId              DWORD      ?

  rect             RECT      <>

  hInst            DWORD      ?

  lpszText         DWORD      ?

  lParam           LPARAM     ?

TOOLINFO ENDS

Nome 
Campo 

Spiegazione 

cbSize La grandezza della struttura TOOLINFO. DEVI riempire questo membro. Windows non 
segnalerà errori se questo membro non è riempito a dovere ma riceverà strani, indicibili risultati

uFlag I flag che specificano le caratteristiche del tool. Questo valore può essere una combinazione dei 
valori sottostanti: 

 TTF_IDISHWND "ID is hWnd" (l'ID è l'hWnd, NTD). Se viene impostato, segnifica che 
vuoi che il tool copra tutta la client-area(primo tipo). Se usi questo flag, DEVI riempire il
membro uID della struttura con l'handle della finestra che vuoi utilizzare. Se non lo usi, 



significa che il tool coprirà una zona della client-area(secondo tipo). In questo caso devi 
riempire il membro rect con la deimenzione della zona. 

 TTF_CENTERTIP normalmente il tooltip apparirà a destra e sotto il cursore del mouse. 
Se usi questo flag, il tooltip apparirà direttamente sotto il tool e sarà centrato secondo 
la posizione del mouse. 

 TTF_RTLREADING questo flag è inutile se il tuo programma non è designato 
specificatamente persistemi arabici o ebrei. Questo flag mostra il testo del tooltip 
leggendo il testo da destra a sinstra. Sugli altri sistemi non funziona 

 TTF_SUBCLASS se usi questo flag, dici al tooltip di subclassare la finestra su cui è il 
tool cosicchè possa intercettare i messaggi del mouse che sono mandati alla finestra. E' 
un flag molto utile. Se non lo sui dovrai fare più lavoro per mandare i messaggi del 
mouse al tooltip. 

hWnd 

L'Handle della finestra che contiene il tool. Se però setti il flag TTF_IDISHWND questo campo è 
ignorato poiché Windows userà il valore nel membro uId come handle della finestra. Devi 
riempire questo campo se: 

 Non usi il flag TTF_IDISHWND (cioè usi un tool che non occupa tutta la client-area) 

 Metti il valore LPSTR_TEXTCALLBACK nel membro lpszText. Questo valore dice al 
controllo tooltip che quando deve mostrare il tooltip deve chiedere alla finestra che 
contiene il tool il testo da mostrare. E' una specie di aggiornamento dinamico del testo. 
Quindi, se vuoi cambiare il testo del tooltip in modo dinamico devi mettere il valore 
LPSTR_TEXTCALLBACK nel membro lpszText. Il tooltip invierà il messaggio di notifica 
TTN_NEEDTEXT alla finestra identificata dall'handle presente in hWnd 

uId 

Questo campo può avere due significati, dipende se il membro uFlags contiene il flag 
TTF_IDISHWND 

 Application-defined tool ID (l'ID del tool è definita dall'applicazione, NDT) se il flag 
TTF_IDISHWND non è usato. Poichè stai usando un tool che copre solo una parte della
client-area è logico che ci possono essere più tool in una client-area. Il controllo tooltip 
ha bisogno di distinguerli. In questo caso il valore in hWnd non è abbastanza poiché ci 
sono più tool sulla client-area. Perciò sono necessarie le application-defined ID. Le ID 
possono essere qualunque valore che però deve essere unico. 

 L'handle della finestra dove tutta la client-area è usata come tool se è usato il flag 
TTF_IDISHWND. Potresti chiederti perchè questo campo è usato per contenere 
l'handle della finestra invece del membro hWnd. La risposta è: il membro hWnd 
potrebbe già essere usato se il membro lpszText contiene LPSTR_TEXTCALLBACK e la
finestra che è responsabile dell'approvvigionamento del testo del tooltip non è la stessa 
che contiene il tool (puoi creare un programma che abbia una finestra che faccia tutti e 
due i compiti ma sarebbe troppo restrittivo. In questo caso la Microsoft ci dà più 
liberta...). (e quando mai ce l'ha data! NTD) 

rect 

Una struttura RECT che contiene le dimensioni del tool. Questa struttura definisce il rettangolo 
relativo all'angolo superiore sinistro della client-area della finestra indicata dal membro hWnd. 
In poche parole, devi riempire questa struttura se vuoi che il tool copra solo una parte della 
client-area. Il tooltip ignorerà questo membro se è specificato il flag TTF_IDISHWND (cioè se 
vuoi che il tool copra tutta la client-area). 

hInst 

L' hInstance-handle del luogo contenente la stringa che sarà usata come testo del tooltip se il 
valore in lpszText specifica l'ID della stringa. Può confondere: leggi prima la spiegazione di 
lpszText e capirai a che serve questo campo. Il tooltip ignora questo campo se lpszText non 
contiene l'ID di una stringa. 

lpszText Questo campo può avere molti valori: 

 Se specifichi il valore LPSTR_TEXTCALLBACK, il tooltip invierà TTN_NEEDTEXT come
messaggio di notifica alla finestra indicata dal valore in hWnd per avere il testo da 
mostrare nel tooltip. Questo è il metodo più dinamico per cambiare testo al tooltip. Puoi 
cambiare il testo ogni volta che il tooltip è mostrato. 

 Se specifichi un' ID di una stringa (intesa come risorsa, NTD) in questo campo, quando 
il tooltip avrà bisogno di mostrare il testo, cercherà la stringa nella string-table (è un 
tipo di risorsa NTD) presente nell'instance indicata dal membro hInst. Il tooltip 
identificherà una ID di una stringa controllando la word alta di questo membro . Questo 
perchè l'ID di una stringa è un valore a 16 bit (una word) e quindi la word alta sarà 
zero. Questo metodo è utile se in futuro vorrai tradurre il tuo programma. Essendo 



presente nel resource script, la string-table può essere modificata senza toccare il 
codice. Devi solo cambiare le stringe e il testo del tooltip cambierà senza rischio di bug. 

 Se il valore di questo campo non è LPSTR_TEXTCALLBACK e la word alta è zero, il 
tooltip interpreterà questo valore come un puntatore a una stringa che sarà usata come 
testo del tooltip. Questo è il metodo più facile ma è poco flessibile. 

Per ricapitolare, devi riempire la struttura TOOLINFO prima di mandarla al tooltip. Questa struttura 
descrive le caratteristiche del tool. 

Associare il tool al Controllo Tooltip

Dopo aver riempito la struttura TOOLINFO, devi inviarla al tooltip. Un tooltip può servire molti tool perciò 
di solito non è necessario creare più di un tooltip per finestra. Per associare un tool con un controllo 
tooltip, devi inviare il messaggio TTM_ADDTOOL al tooltip. wParam non è usato e lParam deve 
contenere l'indirizzo della struttura TOOLINFO che vuoi associare. 

;-------------------------------------------

;Piccolo esempio

;-------------------------------------------

.data?

ti TOOLINFO <>

.......

.code

.......

<Riempire la struttura TOOLINFO>

.......

invoke SendMessage, hwndTooltip, TTM_ADDTOOL, NULL, addr ti

;-------------------------------------------------------------------------

SendMessage restituirà TRUE se l'associamento è riuscito, altrimenti FALSE. Puoi dis-associare il tool 
inviando il messaggio TTM_DELTOOL al tooltip. 

Controllare i messaggi del mouse sul Tooltip

Quando il passo precedente è stato completato, il tooltip conosce quale area deve monitorare per 
controllare se avvengono messaggi dal mouse e quale testo deve visualizzare una volta apparso. L'unica 
cosa che manca è il "grilletto" per farlo partire. Facciamo un esempio: L'area specificata dal tool è sulla 
client-area dell'altra finestra. Come può il tooltip intercettare i messaggi del mouse destinati a quella 
finestra? Ha bisogno di fare un pò di ordine in modo che possa misurare la quantità di tempo in cui il 
cursore del mouse punta sul tool e, quando passa questa quantità di tempo, il tooltip apparirà. Ci sono 
due metodi per riuscire nell'impresa, uno che richiede la collaborazione della finestra che contiene il tool e
l'altra senza la collaborazione da parte della finestra. 

 La finestra che contiene il tool deve rilasciare messaggi del mouse al tooltip mandando il 
messaggio TTM_RELAYEVENT al controllo. il valore di lParam in questo messaggio deve 
contenere l'indirizzo di una struttura MSG che specifica il messaggio da rilasciare al tooltip. Il 
tooltip processa solo i seguenti messaggi: 

(per capire meglio, vi consiglio di guardare l'esempio, NDT) 

- WM_LBUTTONDOWN - WM_MOSEMNOVE - WM_LBUTTONUP - WM_RBUTTONDOWN - 
WM_MBUTTONDOWN - WM_RBUTTONUP - WM_MBUTTONUP 



tutti gli altri sono ignorati. Per questo nella window-procedure della finestra che contiene il tool, ci deve 
essere un "switch" che fa qualcosa del genere: 

WndProc proc hWnd:DWORD, uMsg:DWORD, wParam:DWORD, lParam:DWORD
.......
    if uMsg==WM_CREATE
        .............
    elseif uMsg==WM_LBUTTONDOWN || uMsg==WM_MOUSEMOVE || uMsg==WM_LBUTTONUP || 
uMsg==WM_RBUTTONDOWN || uMsg==WM_MBUTTONDOWN || uMsg==WM_RBUTTONUP || uMsg==WM_MBUTTONUP
        invoke SendMessage, hwndTooltip, TTM_RELAYEVENT, NULL, addr msg
        ..........

 Puoi specificare il flag TTF_SUBCLASS nel membro uFlag della struttura TOOLINFO. Questo 
flag dice al tooltip di subclassare la finestra che contiene il tool in modo da intercettare i 
messaggi del mouse senza la collaborazione della finestra. Questo metodo è più semplice da 
usare perchè non richiede altro coding che specificare il flag TTF_SUCLASS e il tooltip gestirà 
tutte le intercezioni dei messaggi da solo. 

Ce l'abbiamo fatta. A questo punto il tooltip è pienamente funzionante. Ci sono un sacco di messaggi utili 
relativi al tooltip che dovresti conoscere. 

 TIM_ACTIVATE. Se vuoi abilitare/disabilitare il tooltip in modo dinamico, questo è il messaggio 
che fa per te. Se wParam è TRUE, il tooltip è abilitato. Se è FALSE, il tooltip è disabilitato. Il 
tooltip, appena creato, è già attivato e quindi non serve attivarlo. 

 TIM_GETTOOLINFO e TIM_SETTOOLINFO. Se vuoi ottenere/cambiare i valori nella struttura 
TOOLINFO dopo che è stata assegnata al tooltip, usa questi messaggi. Hai bisogno di specificare
il tool a cui vuoi cambiare i membri uID e hWnd. Se vuoi solo cambiare il valore del membro 
rect usa il messaggio TIM_NEWTOOLRECT. Se vuoi cambiare solo il testo usa 
TIM_UPDATETIPTEXT. 

 TIM_SETDELAYTIME. Con questo messaggio, puoi specificare il tempo di sparizione del tooltip 
(Ovvero per quanto tempo rimane visibile NDT) 

Esempio
L'esempio seguente è una semplice dialog box con due pulsanti. La client-area è divisa in quattro parti: in
alto a sinistra, in altro a destra, in basso a sinistra e in basso a destra. Ogni area è considerata come un 
tool con il suo testo di tooltip. I due pulsanti hanno i loro testi per il tooltip. 

.386
    .model flat,stdcall
    option casemap:none
    include \masm32\include\windows.inc
    include \masm32\include\kernel32.inc
    include \masm32\include\user32.inc
    include \masm32\include\comctl32.inc
    includelib \masm32\lib\comctl32.lib
    includelib \masm32\lib\user32.lib
    includelib \masm32\lib\kernel32.lib
    DlgProc proto :DWORD,:DWORD,:DWORD,:DWORD
    EnumChild proto :DWORD,:DWORD
    SetDlgToolArea proto :DWORD,:DWORD,:DWORD,:DWORD,:DWORD
    .const
    IDD_MAINDIALOG equ 101
    .data
    ToolTipsClassName db "Tooltips_class32",0
    MainDialogText1 db "This is the upper left area of the dialog",0
    MainDialogText2 db "This is the upper right area of the dialog",0
    MainDialogText3 db "This is the lower left area of the dialog",0
    MainDialogText4 db "This is the lower right area of the dialog",0
    .data?
    hwndTool dd ?
    hInstance dd ?
    .code
    start:



        invoke GetModuleHandle,NULL
        mov hInstance,eax
        invoke DialogBoxParam,hInstance,IDD_MAINDIALOG,NULL,addr DlgProc,NULL
        invoke ExitProcess,eax

    DlgProc proc hDlg:DWORD,uMsg:DWORD,wParam:DWORD,lParam:DWORD
        LOCAL ti:TOOLINFO
        LOCAL id:DWORD
        LOCAL rect:RECT
        .if uMsg==WM_INITDIALOG
            invoke InitCommonControls
            invoke CreateWindowEx,NULL,ADDR ToolTipsClassName,NULL,\
                TTS_ALWAYSTIP,CW_USEDEFAULT,\
                CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\
                hInstance,NULL
            mov hwndTool,eax
            mov id,0
            mov ti.cbSize,sizeof TOOLINFO
            mov ti.uFlags,TTF_SUBCLASS
            push hDlg
            pop ti.hWnd
            invoke GetWindowRect,hDlg,addr rect
            invoke SetDlgToolArea,hDlg,addr ti,addr MainDialogText1,id,addr rect
            inc id
            invoke SetDlgToolArea,hDlg,addr ti,addr MainDialogText2,id,addr rect
            inc id
            invoke SetDlgToolArea,hDlg,addr ti,addr MainDialogText3,id,addr rect
            inc id
            invoke SetDlgToolArea,hDlg,addr ti,addr MainDialogText4,id,addr rect
            invoke EnumChildWindows,hDlg,addr EnumChild,addr ti
        .elseif uMsg==WM_CLOSE
            invoke EndDialog,hDlg,NULL
        .else
            mov eax,FALSE
            ret
        .endif
        mov eax,TRUE
        ret
    DlgProc endp

    EnumChild proc uses edi hwndChild:DWORD,lParam:DWORD
        LOCAL buffer[256]:BYTE
        mov edi,lParam
        assume edi:ptr TOOLINFO
        push hwndChild
        pop [edi].uId
        or [edi].uFlags,TTF_IDISHWND
        invoke GetWindowText,hwndChild,addr buffer,255
        lea eax,buffer
        mov [edi].lpszText,eax
        invoke SendMessage,hwndTool,TTM_ADDTOOL,NULL,edi
        assume edi:nothing
        ret
    EnumChild endp

    SetDlgToolArea proc uses edi esi hDlg:DWORD,lpti:DWORD,lpText:DWORD,id:DWORD,lprect:DWORD
        mov edi,lpti
        mov esi,lprect
        assume esi:ptr RECT
        assume edi:ptr TOOLINFO
        .if id==0
            mov [edi].rect.left,0
            mov [edi].rect.top,0
            mov eax,[esi].right
            sub eax,[esi].left
            shr eax,1
            mov [edi].rect.right,eax
            mov eax,[esi].bottom
            sub eax,[esi].top
            shr eax,1
            mov [edi].rect.bottom,eax
        .elseif id==1
            mov eax,[esi].right
            sub eax,[esi].left



            shr eax,1
            inc eax
            mov [edi].rect.left,eax
            mov [edi].rect.top,0
            mov eax,[esi].right
            sub eax,[esi].left
            mov [edi].rect.right,eax
            mov eax,[esi].bottom
            sub eax,[esi].top
            mov [edi].rect.bottom,eax
        .elseif id==2
            mov [edi].rect.left,0
            mov eax,[esi].bottom
            sub eax,[esi].top
            shr eax,1
            inc eax
            mov [edi].rect.top,eax
            mov eax,[esi].right
            sub eax,[esi].left
            shr eax,1
            mov [edi].rect.right,eax
            mov eax,[esi].bottom
            sub eax,[esi].top
            mov [edi].rect.bottom,eax
        .else
            mov eax,[esi].right
            sub eax,[esi].left
            shr eax,1
            inc eax
            mov [edi].rect.left,eax
            mov eax,[esi].bottom
            sub eax,[esi].top
            shr eax,1
            inc eax
            mov [edi].rect.top,eax
            mov eax,[esi].right
            sub eax,[esi].left
            mov [edi].rect.right,eax
            mov eax,[esi].bottom
            sub eax,[esi].top
            mov [edi].rect.bottom,eax
        .endif
        push lpText
        pop [edi].lpszText
        invoke SendMessage,hwndTool,TTM_ADDTOOL,NULL,lpti
        assume edi:nothing
        assume esi:nothing
        ret
    SetDlgToolArea endp
    end start

Analisi
Dopo che la finestra principale è stata creata, creiamo il tooltip con CreateWindowEx. 

invoke InitCommonControls
    invoke CreateWindowEx,NULL,ADDR ToolTipsClassName,NULL,\
           TTS_ALWAYSTIP,CW_USEDEFAULT,\
           CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\
           hInstance,NULL
    mov hwndTool,eax

Dopodichè, andiamo avanti definendo 4 tool, uno per ogni angolo della dialog-box. 

mov id,0        ; usato come ID del tool
    mov ti.cbSize,sizeof TOOLINFO
    mov ti.uFlags,TTF_SUBCLASS    ; dice al tooltip di subclassare la dialog-box



    push hDlg
    pop ti.hWnd    ; handle della finestra che contiene il tool
    invoke GetWindowRect,hDlg,addr rect    ; otteniamo le dimenzioni della client-area
    invoke SetDlgToolArea,hDlg,addr ti,addr MainDialogText1,id,addr rect

Inizializziamo i membri della struttura TOOLINFO. Notare che se vogliamo dividere la client-area in 4 
parti, abbiamo bisogno della dimensioni della client-area. Per questo chiamiamo GetWindowRect. Non 
vogliamo rilasciare i messaggi del mouse al tooltip a mano, perciò specifichiamo il flag TIF_SUBCLASS. 
SetDlgToolArea è una funzione che calcola "l'area rettangolare" di ogni tool e registra il tool al tooltip. 
Non andrò nei dettagli riguardo il calcolo, basti sapere che divide la client-area in 4 parti uguali. Quindi 
invia il messaggio TTM_ADDTOOL al tooltip passandogli l'indirizzo della struttura TOOLINFO in lParam. 

invoke SendMessage,hwndTool,TTM_ADDTOOL,NULL,lpti

Dopo che tutti i 4 tool sono registrati, possiamo andare avanti con i pulsanti sulla dialog-box. Possiamo 
gestire ogni pulsante tramite l'ID ma sarebbe noioso. Invece, useremo l'API EnumChildWindows per 
"numerare" tutti i controlli sulla dialog-box e registrarli al tooltip. EnumChildWindows ha la seguente 
sintassi: 

EnumChildWindows proto hWnd:DWORD, lpEnumFunc:DWORD, lParam:DWORD

hWnd è l'handle della parent-window. lpEnumFunc è l'indirizzo della funzione EnumChildProc che sarà 
chiamata per ogni controllo numerato. lParam è il valore application-defined che sarà passato alla 
funzione EnumChildProc. 

EnumChildProc ha la seguente sintassi: 

EnumChildProc proto hwndChild:DWORD, lParam:DWORD

hwndChild è l'handle del controllo numerato da EnumChildWindows. lParam è lo stesso passato a 
EnumChildWindows. 

Nel nostro esempio chiamiamo EnumChildWindows così: 

invoke EnumChildWindows,hDlg,addr EnumChild,addr ti

Passiamo l'indirizzo della struttura TOOLINFO in lParam perchè vogliamo registrare ogni controllo al 
tooltip nella funzione EnumChild. Se non usiamo questo metodo, dobbiamo dichiarare ti come variabile 
globale e potrebbero presentarsi bugs. Quando chiamiamo EnumChildWindows, Windows numererà i 
control presenti nella dialog-box e chiamerà la funzione EnumChildProc una volta per ogni controllo 
numerato. Nel nostro caso ci sono due controlli e quindi la EnumChildProc verrà chiamata 2 volte. 
(nell'esempio l'EnumChildProc è EnumChild NTD) 

La funzione EnumChild riempie i membri importanti della struttura TOOLINFO e quindi registra il tool al 
tooltip. 

EnumChild proc uses edi hwndChild:DWORD,lParam:DWORD
    LOCAL buffer[256]:BYTE
    mov edi,lParam
    assume edi:ptr TOOLINFO
    push hwndChild
    pop [edi].uId    ; usiamo l’intera client-area come tool
    or [edi].uFlags,TTF_IDISHWND
    invoke GetWindowText,hwndChild,addr buffer,255
    lea eax,buffer    ; usiamo il tsto della finestra come testo del tooltip
    mov [edi].lpszText,eax
    invoke SendMessage,hwndTool,TTM_ADDTOOL,NULL,edi
    assume edi:nothing
    ret
EnumChild endp



Notare che in questo caso usiamo un tipo diverso di tool: quello che copre l'intera client-area. Percui 
abbiamo bisogno di riempire il membro uID con l'handle della finestra che contiene il tool. Inoltre va 
settato il flag TTF_IDISHWND nel membro uFlags. 

Introduzione
In questo tutorial impareremo cosa Win32 offre per i programmatori riguardo gli strumenti di debugging. 
Dopo aver letto questo tutorial avrai imparato come debuggare un processo. 

Tools
Esempio 

Essay
Win32 possiede molte API che permettono ai programmatori alcune potenzialità di un debugger. Sono 
chiamate Win32 Debug-API o "primitives". Con esse puoi: 

 Caricare un programma o attaccarti a un programma in esecuzione; 

 Ottenere informazioni di basso livello sul programma che stai debuggando, come per esempio le 
process ID, indirizzo dell'entry-point, imagebase etc.; 

 Essere avvisato dagli eventi relativi al debugger come succede quando un processo/thread inizia/
esce, le DLL sono caricate/scaricate etc.; 

 Modificare il processo/thread che è stato debuggato. 

In poche parole, puoi scrivere un semplice debugger con queste API. Siccome l'argomento è vasto, è 
stato, di conseguenza, diviso in parti più maneggevoli: Il tutorial inizia con la prima parte. Qui verranno 
spiegati i concetti di base e il framework generale per usare le Debug-API. 

Gli step da seguire per usare le debug-API sono: 

1) Creare un processo o "attaccare" il tuo programma a un processo in esecuzione. Questo è il 
primo passo per usare le Debug-API. Finchè il tuo programma lavora da debugger ti serve un programma
da debuggare. Il programma che sta per essere debuggato è chiamato "Debuggee". Puoi acquisire il 
Debuggee in due modi: 

 Puoi creare il processo del debuggee da te con CreateProcess. Per poter creare un 
processo da debuggare devi specificare il flag DEBUG_PROCESS. Questo flag fa in modo che 
Windows sappia che vogliamo debuggare il processo. Windows manderà delle notificazioni al tuo
programma riguardo importanti eventi di debugging(debug-events) che si presenteranno nel 

debuggee. Il processo debuggee verrà immediatamente sospeso finchè il tuo programma non è 
pronto. Se il debuggee crea a sua volta processi-figli, Windows notificherà dei debug-events 
provenienti anche da tutti i processi figli. Di solito non si desidera tale comportamento. Lo puoi 
disabilitare specificando il flag DEBUG_ONLY_THIS_PROCESS combinato con DEBUG_PROCESS 
(in masm, si scrive 'DEBUG_ONLY_THIS_PROCESS or DEBUG_PROCESS' NTD). 

 Puoi "attaccare" il tuo programma a un processo già presente con DebugActiveProcess.

2) Aspettare i debug-events. Dopo che il tuo programma ha acquisito il debuggee, il thread primario 
del debuggee è sospeso e lo rimarrà finchè il tuo programma chiama WaitForDebugEvent. Questa 
funzione lavora come gli altri WaitFor..... cioè blocca il thread chiamato finchè i eventi attesi si 
verificheranno. In questo caso, aspetta i debug-events mandati da Windows. Vediamo il suo prototipo: 

WaitForDebugEvent proto lpDebugEvent:DWORD, dwMilliseconds:DWORD

lpDebugEvent è l'indirizzo di una struttura DEBUG_EVENT che sarà riempita con informazioni sul debug-
event che si è presentato nel debuggee; 

dwMilliseconds è il periodo di tempo in millisecondi per il quale questa funzione attenderà la venuta di 
debug-events. Se il periodo di tempo scade e non ci sono stati debug-events, WaitForDebugEvent 
ritornerà al caller (ovvero al pezzo di codice che l'ha chiamata NTD). Però se specifichi INFINITE 
(costante NTD) in questo argomento, la funzione non ritornerà al caller finchè non si verificherà un 
debug-event. Adesso esaminiamo la struttura DEBUG_EVENT in dettaglio: 

DEBUG_EVENT STRUCT
dwDebugEventCode dd ?
dwProcessId dd ?
dwThreadId dd ?
u DEBUGSTRUCT <>
DEBUG_EVENT ENDS

dwDebugEventCode contiene il valore che specifica il tipo di debug-event arrivato. Brevemente, qui ci 
possono essere tanti tipi di eventi, e perciò il tuo programma deve controllare il valore di questo campo in
modo che sappia quale tipo di evento è arrivato e agire di conseguenza. I valori possibili sono: 

Valore Significato 

CREATE_PROCESS_DEBUG_EVENT 

Quando un processo è creato. Questo evento sarà mandato quando
il debuggee è stato appena creato (ma non in eseguzione)o quando
il tuo programma si attacca a un processo in eseguzione con 
DebugActiveProcess. Questo è il primo evento che il tuo 
programma riceverà. 

EXIT_PROCESS_DEBUG_EVENT Quando un processo viene chiuso 

CREATE_THEAD_DEBUG_EVENT 

Quando un nuovo thread è creato nel processo del debuggee o 
quando il tuo programma si attacca a un processo in eseguzione. 
Notare che non si ricevono questo tipo di eventi quando il thread 
principale è creato 

EXIT_THREAD_DEBUG_EVENT 

Quando un thread del debuggee viene chiuso. Il tuo programma 
però non riceverà questo debug-event per il thread principale. In 
poche parole, puoi vedere il thread principale del debuggee come 
l'equivalente del processo del debuggee stesso. Quindi quando il 
tuo programma vede CREATE_PROCESS_DEBUG_EVENT è in realtà 
CREATE_THREAD_DEBUG_EVENT per il thread principale 

LOAD_DLL_DEBUG_EVENT 
Quando il debuggee carica una DLL., Il programma riceverà questo 
evento solo quando il PE Loader ha risolto i link alle DLL (scrive la 
IAT NTD) oppure quando il debuggee chiama LoadLibrary 

UNLOAD_DLL_DEBUG_EVENT Quando una DLL è scaricata dal processo del debuggee 

EXCEPTION_DEBUG_EVENT Quando viene generata un'eccezione nel processo del debuggee. 
IMPORTANTE:Questo evento arriverà appena prima che il debuggee
inizia a esegure la sua prima istruzione. L'eccezione è attualmente 
un debug break (int3). Quando vuoi "riesumare" il debuggee 



chiama ContinueDebugEvent con il flag DBG_CONTINUE. Non usare
il flag DBG_EXCEPTION_NOT_HANDLED altrimenti il debuggee si 
rifiuterà di eseguire su WinNT (e quindi anche su WinXP e Win2k 
NTD). In Win98 invece va bene. 

OUTPUT_DEBUG_STRING_EVENT 
Questo evento è generato quando il debuggee chiama la funzione 
DebugOutPutString per mandare una stringa di messaggio al tuo 
programma 

RIP_EVENT 
Quando arrivano degli errori di debug di sistema (System 
Debugging errors NTD) 

dwProcessId e dwThreadId sono le ID del processo e del thread nel quale arrivano gli eventi di debug.
Puoi usare questi valori come identificatori del processo/thread a cui sei interessato. Ricorda che se usi 
CreateProcess per caricare il debuggee, puoi anche prendere la process ID e la thread ID del debuggee 
nella struttura PROCESS_INFO. Puoi usare questi valori per distinguere i debug-event del debuggee e 
quelli dei child-process (nel caso tu non abbia specificato il flag DEBUG_ONLY_THIS_PROCESS) u è un
unione che contiene informazioni sul debug-event. Può essere una delle strutture sottostanti, dipende dal 
valore di dwDebugEventCode. 

valore in "dwDebugEventCode" Interpretazione di "u" 

CREATE_PROCESS_DEBUG_EVE
NT

Una 
strutturaCREATE_PROCESS_DEBUG_INFOchiamataCreateProcess
Info 

EXIT_PROCESS_DEBUG_EVENT Una struttura EXIT_PROCESS_DEBUG_INFO chiamata ExitProcess

CREATE_THEAD_DEBUG_EVENT
Una struttura CREATE_THREAD_DEBUG_INFO chiamata 
CreateThread 

EXIT_THREAD_DEBUG_EVENT Una struttura EXIT_THREAD_DEBUG_EVENT chiamata ExitThread 

LOAD_DLL_DEBUG_EVENT Una struttura LOAD_DLL_DEBUG_INFO chiamata LoadDll 

UNLOAD_DLL_DEBUG_EVENT Una struttura UNLOAD_DLL_DEBUG_INFO chiamata UnloadDll 

EXCEPTION_DEBUG_EVENT Una struttura EXCEPTION_DEBUG_INFO chiamata Exception 

OUTPUT_DEBUG_STRING_EVEN
T 

Una struttura OUTPUT_DEBUG_STRING_INFO chiamata 
DebugString 

RIP_EVENT Una struttura RIP_INFO chiamata RipInfo 

Non spiegherò nei dettagli tutte queste strutture in questo tutorial, solo la struttura 
CREATE_PROCESS_DEBUG_INFO sarà spiegata. Immaginiamo che il nostro programma chiamasse 
WaitForDebugEvent e che ritorni. La prima cosa che dovremo fare sarà esaminare il valore in 
dwDebugEventCode per vedere che tipo di debug-event è accaduto nel processo del debuggee. Per 
esempio se il valore in dwDebugEventCode è CREATE_PROCESS_DEBUG_EVENT puoi interpretare il 
membro "u" come CreateProcessInfo e accedervi tramite u.CreateProcessInfo. 

3)Fai qualunque cosa il tuo programma voglia fare in risposta ai debug-event.Quando 
WaitForDebugEvent ritorna, significa che un debug-event è appena accaduto nel processo del debugggee 
oppure il timeout è scaduto. Il tuo programma deve esaminare il valore in dwDebugEventCode in modo 
da reagire al debug-event in modo appropiato. In questo senso è come quando si processano i messaggi 
di Windows: tu scegli di gestirne alcuni e di tralasciarne altri. 

4)Fai continuare l'esecuzione del debuggee.Quando arriva un debug-event, Windows sopende il 
processo del debuggee. Quando hai finito di gestire gli eventi, devi "spingere" il debuggee in modo da 
farlo ripartire. Puoi farlo chiamando la funzione ContinueDebugEvent. 

ContinueDebugEvent proto dwProcessId:DWORD, dwThreadId:DWORD, dwContinueStatus:DWORD

Questa funzione "riesuma"(fa ripartire, NTD) il thread che è stato precedentemente sospeso a causa della
comparsa di un debug-event. dwProcessId e dwThreadId sono rispettivamente l'ID del processo e del 
thread che sarà riesumato. Di solito prendi questi valori dai membri dwProcessId e dwThreadId della 
struttura DEBUG_EVENT. dwContinueStatus specifica come continuare il thread che ha generato il 
debug-event. Ci sono due possibili valori: DBG_CONTINUE e DBG_EXCEPTION_NOT_HANDLED.Questi 



due valori fanno la stessa per tutti i debug-event: continuano il thread. L'eccezione è 
EXCEPTION_DEBUG_EVENT. Se il thread riporta un'eccezione, significa che si è avuta un'eccezione del 
thread del debuggee. Se specifichi DBG_CONTINUE, il thread ignorerà il proprio gestore di eccezioni (vedi
il corso NewBie N°13 NTD) e continuerà l'eseguzione. Per questo, il tuo programma deve esaminare e 
risolvere le eccezioni prima di riesumare il thread con DBG_CONTINUE oppure l'eccezione arriverà ancora
e ancora...Se specifichi DBG_EXCEPTION_NOT_HANDLED, il tuo programma sta dicendo a Windows che è
incapace di gestire l'eccezione. Windows dovrà usare il gestore di eccezioni di default del debuggee per 
poter gestire l'eccezione. In conclusione, se il debug-event si riferisce a un'eccezione avvenuta nel 
processo del debuggee, dovresti chiamare ContinueDebugEvent con il flag DBG_CONTINUE se il tuo 
programma ha già risolto l'eccezione. Altrimenti, il tuo programma deve chiamare ContinueDebugEvent 
con il flag DBG_EXCEPTION_NOT_HANDLED. Dovresti sempre usare DBG_CONTINUE tranne in un caso:il 
primo EXCEPTION_DEBUG_EVENT che ha il valore EXCEPTION_BREAKPOINT nel membro ExceptionCode. 
Quando il debuggee sta per eseguire la sua primissima istruzione, il tuo programma riceverà un 
eccezione (che è un tipo di debug-event NTD) che precisamente è un debug-break (INT 3). Se rispondi 
chiamando ContinueDebugEvent con il flag DBG_EXCEPTION_NOT_HANDLED, Windows NT (e quindi 
anche XP e 2000) si rifiuterà di eseguire il debuggee (perchè non ne tiene conto). Devi sempre usare il 
flag DBG_CONTINUE per dire a Windows che vuoi che il thread vada avanti. 

5)Continuare questo ciclo all'infinito finchè il processo del debuggee esce.Il tuo programma deve
rimanere in un loop infinito come un message-loop finchè il debuggee non chiude. Il loop dovrebbe 
essere tipo questo: 

.while TRUE
invoke WaitForDebugEvent, addr DebugEvent, INFINITE
.break .if DebugEvent.dwDebugEventCode==EXIT_PROCESS_DEBUG_EVENT
<Gestione dei debug-event>
invoke ContinueDebugEvent, DebugEvent.dwProcessId, DebugEvent.dwThreadId, 
DBG_EXCEPTION_NOT_HANDLED
.endw

Ecco la sorpresa: Una volta che sei partito a debuggare un programma, non puoi staccarti finchè il 
programma non esce. Facciamo un breve riepilogo: 

1. Creare un processo o "attaccare" il tuo programma a un processo in esecuzione; 
2. Aspettare i debug-events; 
3. Fai qualunque cosa il tuo programma voglia fare in risposta ai debug-event; 
4. Fai continuare l'esecuzione del debuggee; 
5. Continuare questo ciclo all'infinito finchè il processo del debuggee esce. 

Esempio
Questo esempio debugga un programma in Win32 e mostra importanti informazioni come l'handle del 
processo, ID, imagebase etc. 

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\kernel32.inc
include \masm32\include\comdlg32.inc
include \masm32\include\user32.inc
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\comdlg32.lib
includelib \masm32\lib\user32.lib
.data
AppName db "Win32 Debug Example no.1",0
ofn OPENFILENAME <>
FilterString db "Executable Files",0,"*.exe",0
              db "All Files",0,"*.*",0,0
ExitProc db "The debuggee exits",0
NewThread db "A new thread is created",0
EndThread db "A thread is destroyed",0



ProcessInfo db "File Handle: %lx ",0dh,0Ah
             db "Process Handle: %lx",0Dh,0Ah
             db "Thread Handle: %lx",0Dh,0Ah
             db "Image Base: %lx",0Dh,0Ah
             db "Start Address: %lx",0
.data?
buffer db 512 dup(?)
startinfo STARTUPINFO <>
pi PROCESS_INFORMATION <>
DBEvent DEBUG_EVENT <>
.code
start:
mov ofn.lStructSize,sizeof ofn
mov ofn.lpstrFilter, offset FilterString
mov ofn.lpstrFile, offset buffer
mov ofn.nMaxFile,512
mov ofn.Flags, OFN_FILEMUSTEXIST or OFN_PATHMUSTEXIST or OFN_LONGNAMES or OFN_EXPLORER or 
OFN_HIDEREADONLY
invoke GetOpenFileName, ADDR ofn
.if eax==TRUE
invoke GetStartupInfo,addr startinfo
invoke CreateProcess, addr buffer, NULL, NULL, NULL, FALSE, DEBUG_PROCESS+ 
DEBUG_ONLY_THIS_PROCESS, NULL, NULL, addr startinfo, addr pi
.while TRUE
    invoke WaitForDebugEvent, addr DBEvent, INFINITE
    .if DBEvent.dwDebugEventCode==EXIT_PROCESS_DEBUG_EVENT
       invoke MessageBox, 0, addr ExitProc, addr AppName, MB_OK+MB_ICONINFORMATION
       .break
   .elseif DBEvent.dwDebugEventCode==CREATE_PROCESS_DEBUG_EVENT
       invoke wsprintf, addr buffer, addr ProcessInfo, DBEvent.u.CreateProcessInfo.hFile, 
DBEvent.u.CreateProcessInfo.hProcess, DBEvent.u.CreateProcessInfo.hThread, 
DBEvent.u.CreateProcessInfo.lpBaseOfImage, DBEvent.u.CreateProcessInfo.lpStartAddress
       invoke MessageBox,0, addr buffer, addr AppName, MB_OK+MB_ICONINFORMATION    
   .elseif DBEvent.dwDebugEventCode==EXCEPTION_DEBUG_EVENT
       .if DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_BREAKPOINT
           invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, DBG_CONTINUE
          .continue
       .endif
   .elseif DBEvent.dwDebugEventCode==CREATE_THREAD_DEBUG_EVENT
       invoke MessageBox,0, addr NewThread, addr AppName, MB_OK+MB_ICONINFORMATION
   .elseif DBEvent.dwDebugEventCode==EXIT_THREAD_DEBUG_EVENT
       invoke MessageBox,0, addr EndThread, addr AppName, MB_OK+MB_ICONINFORMATION
   .endif
   invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, 
DBG_EXCEPTION_NOT_HANDLED
.endw
invoke CloseHandle,pi.hProcess
invoke CloseHandle,pi.hThread
.endif
invoke ExitProcess, 0
end start

Analisi
Il programma riempie la struttura OPENFILENAME e chiama GetOpenFileName per far scegliere all'utente 
il programma da debuggare. 

invoke GetStartupInfo,addr startinfo
invoke CreateProcess, addr buffer, NULL, NULL, NULL, FALSE, DEBUG_PROCESS+ 
DEBUG_ONLY_THIS_PROCESS, NULL, NULL, addr startinfo, addr pi

Quando l'utente ne sceglie uno, il programma chiama CreateProcess per caricare il programma. Poi 
chiama GetStartupInfo per riempire la struttura STARTUPINFO con i valori di default. Notare il fatto che 
stiamo usando DEBUG_PROCESS combinato con DEBUG_ONLY_THIS_PROCESS in modo da 
debuggare solo il nostro programma e non i processi figli di quest'ultimo. 



.while TRUE
invoke WaitForDebugEvent, addr DBEvent, INFINITE

Quando il debuggee è caricato, entriamo nel debug-loop infinito, dove viene chiamato 
WaitForDebugEvent. WaitForDebugEvent non ritornerà finchè non si verifica un debug-event nel 
debuggee, questo perchè abbiamo specificato il flag INFINITE come suo secondo parametro. Quando si 
verifica un debug-event, WaitForDebugEvent ritorna e DBEvent (cioè la nostra struttura NTD) è riempita 
con le informazioni riguardanti il debug-event. 

.if DBEvent.dwDebugEventCode==EXIT_PROCESS_DEBUG_EVENT
invoke MessageBox, 0, addr ExitProc, addr AppName, MB_OK+MB_ICONINFORMATION
.break

Prima controlliamo il valore di dwDebugEventCode: Se è EXIT_PROCESS_DEBUG_EVENT mostriamo il 
messaggio che dice "il debuggee è uscito" e quindi usciamo dal debug-loop. 

.elseif DBEvent.dwDebugEventCode==CREATE_PROCESS_DEBUG_EVENT
invoke wsprintf, addr buffer, addr ProcessInfo, DBEvent.u.CreateProcessInfo.hFile, 
DBEvent.u.CreateProcessInfo.hProcess, DBEvent.u.CreateProcessInfo.hThread, 
DBEvent.u.CreateProcessInfo.lpBaseOfImage, DBEvent.u.CreateProcessInfo.lpStartAddress
invoke MessageBox,0, addr buffer, addr AppName, MB_OK+MB_ICONINFORMATION

Se il valore di dwDebugEventCode è CREATE_PROCESS_DEBUG_EVENT, mostriamo un pò di 
informazioni sul debuggee, tutto su una message-box. Otteniamo queste informazioni da 
u.CreateProcessInfo. CreatProcessInfo è una struttura del tipo CREATE_PROCESS_DEBUG_INFO. Puoi 
prelevare altre informazioni su questa struttura dalla API Reference. 

.elseif DBEvent.dwDebugEventCode==EXCEPTION_DEBUG_EVENT

.if DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_BREAKPOINT
invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, DBG_CONTINUE
.continue
.endif

Se il valore di dwDebugEventCode è EXCEPTION_DEBUG_EVENT dobbiamo capire che tipo di 
eccezione è. E' una lunga serie di strutture nidificate (vedi tutorial N°13 per newbie! NTD) ma puoi 
ottenere il tipo d'eccezione dal membro ExceptionCode. Sei il valore di ExceptionCode è 
EXCEPTION_BREAKPOINT e arriva per la prima volta (oppure siamo sicuri che il debuggee non abbia 
nel codice degli int 3h) possiamo dire senza sbagliare che il debuggee sta per eseguire la sua prima 
istruzione. Abbiamo finito con il processing (ovvero di controllare i debug-event NTD), dobbiamo 
chiamare ContinueDebugEvent con il flag DBG_CONTINUE per permettere al debuggee di ripartire. 
Quindi aspettiamo il prossimo debug-event. 

.elseif DBEvent.dwDebugEventCode==CREATE_THREAD_DEBUG_EVENT
invoke MessageBox,0, addr NewThread, addr AppName, MB_OK+MB_ICONINFORMATION
.elseif DBEvent.dwDebugEventCode==EXIT_THREAD_DEBUG_EVENT
invoke MessageBox,0, addr EndThread, addr AppName, MB_OK+MB_ICONINFORMATION
.endif

Se il valore di dwDebugEventCode è CREATE_THREAD_DEBUG_EVENT o 
EXIT_THREAD_DEBUG_EVENT, mostriamo il testo che lo dice. 

invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, DBG_EXCEPTION_NOT_HANDLED
.endw

Eccetto per il caso di EXCEPTION_DEBUG_EVENT sopra, chiamiamo ContinueDebugEvent con il flag 
DBG_EXCEPTION_NOT_HANDLED. 

invoke CloseHandle,pi.hProcess
invoke CloseHandle,pi.hThread

Quando il debuggee esce, siamo fuori dal debug-loop e dobbiamo chiudere entrambi i processi e gli 
handle dei thread del debuggee. Chiuderli non significa chiudere il processo/thread. Significa solo che non
vogliamo usare più questi handle per riferirci al processo/thread. 

Introduzione
Continuiamo a proposito delle Win32 Debug API. In questo tutorial, impareremo come modificare il 
debuggee ( il debuggee indica il processo da debuggare NdT ). 

Tools
Esempio 

Essay
Nel precedente tutorial, abbiamo scoperto come caricare il processo da dubuggare ( debugge ;) NdT ) e 
gestire gli eventi che avvengono nel processo stesso. Per divenire utile, il nostro programma deve essere 
in grado di modificare il processo debuggato. Per tale scopo ci sono diverse API: 

 ReadProcessMemory Questa funzione ci permette di leggere la memoria del processo 
specificato. Il prototipo della funzione è il seguente: 

ReadProcessMemory proto hProcess:DWORD, lpBaseAddress:DWORD, lpBuffer:DWORD, nSize:DWORD, 
lpNumberOfBytesRead:DWORD

hProcess è l'handle del processo che si vuole leggere. 

lpBaseAddress specifica l'indirizzo alla quale vuoi far iniziare la lettura nel processo. Per esempio, se io 
voglio iniziare a leggere 4 byte dal processo debuggato all'indirizzo 401000h, il valore di questo 
parametro deve essere 401000h. 

lpBuffer è l'indirizzo del buffer che riceve i byte letti dal processo. 

nSize è il numero di byte che vuoi leggere. 

lpNumberOfBytesRead è l'indirizzo delle variabili di dimensione di una dword che riceve il numero di 
byte effettivamente letti. Se per voi questo parametro non è importante, potete utilizzare NULL. 

 WriteProcessMemory è la controparte di ReadProcessMemory. Vi permette di scrivere la 
memoria del processo. I parametri sono esattamente gli stessi di ReadProcessMemory. 

Le prossime due api necessitano di un ulteriore background di conoscenze. Al di sotto di un OS 
multitasking come Windows, contemporaneamente possono esserci diversi programmi in esecuzione. 
Windows impartisce una certa quantità di tempo ad ogni thread ( processo NdT ). Quando questa 
quantità di tempo si esaurisce, Windows blocca il thread presente e passa al prossimo thread che ha 
priorità maggiore. Poco prima di passare ad un'altro thread, Windows memorizza i valori del thread 
presente in un registro in modo tale che quando arriva il momento di ripristinare il thread, Windows è in 

grado di ripristinare l'ultimo "ambiente" di quel thread. I valori salvati nei registri sono chiamati 
"contesti". Torniamo al nostro argomento. Quando si verifica un'evento di debug, Windows sospende il 
processo da debuggare, e il suo contesto viene salvato. Poichè esso è sospeso, possiamo essere certi che 
i valori nel contesto rimarranno invariati. Siamo in grado di ottenere i valori nel contesto con 
GetThreadContext e possiamo cambiarli con SetThreadContext. Queste due API sono molto potenti. 
Con loro, hai a portata di mano il potere VxD-like sul processo da debuggare: è possibile modificare i 
valori salvati nel registro poco prima che venga ripristinato il debuggee, i valori nel contesto saranno 
scritti all'interno dei registri. Ogni cambiamento, grazie al contesto si riflette sul debuggee. Pensate: è 
anche possibile modificare il valore del registro EIP e deviare il flusso di esecuzione in qualsiasi posto 
preferisci! Sempre se non sarà in grado di farlo in condizioni normali. 

GetThreadContext proto hThread:DWORD, lpContext:DWORD

hThread è l'handle del thread di cui vuoi ottenere il contesto. 

lpContext è l'indirizzo della struttura CONTEXT che sarà riempito quando la funzione tornerà con 
successo. 

SetThreadContext ha esattamente gli stessi parametri. Guardiamo com'è una struttura CONTEXT: 

CONTEXT STRUCT

ContextFlags dd ?

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_DEBUG_REGISTERS

----------------------------------------------------------------------------------------------
-----------

iDr0 dd ?
iDr1 dd ?
iDr2 dd ?
iDr3 dd ?
iDr6 dd ?
iDr7 dd ?

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_FLOATING_POINT

----------------------------------------------------------------------------------------------
-----------

FloatSave FLOATING_SAVE_AREA <>

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_SEGMENTS

----------------------------------------------------------------------------------------------
-----------
regGs dd ?
regFs dd ?
regEs dd ?
regDs dd ?

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_INTEGER

----------------------------------------------------------------------------------------------



-----------

regEdi dd ?
regEsi dd ?
regEbx dd ?
regEdx dd ?
regEcx dd ?
regEax dd ?

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_CONTROL

----------------------------------------------------------------------------------------------
-----------

regEbp dd ?
regEip dd ?
regCs dd ?
regFlag dd ?
regEsp dd ?
regSs dd ? 

----------------------------------------------------------------------------------------------
-----------

;Questa sezione viene restituita se ContextFlags contiene il valore CONTEXT_EXTENDED_REGISTERS

----------------------------------------------------------------------------------------------
-----------

ExtendedRegisters db MAXIMUM_SUPPORTED_EXTENSION dup(?) CONTEXT ENDS

Come puoi vedere, i membri di questa struttura sono la simulazioni dei registri reali del processore. Prima
che tu possa usare questa struttura, hai bisogno di specificare che gruppo di registri vuoi leggere/scrivere
nel membro ContextFlags. Per esempio, se tu vuoi leggere/scrivere tutti i registri, puoi specificare in 
ContextFlags CONTEXT_FULL. Se tu vuoi leggere/scrivere solo regEbp, regEip, regCs, regFlag, regEsp 
or regSs, devi specificare in ContextFlags CONTEXT_CONTROL. 

Devi ricordare una cosa quando usi la struttura CONTEXT: quest'ultima deve essere allineata al termine 
della dword altrimenti si ottengono strani risultati sotto NT. Devi mettere "align dword" appena sopra la 
riga di dichiarazione, come questo: 

align dword
MyContext CONTEXT <>

Esempio
Il primo esempio dimostra l'uso di DebugActiveProcess. Per primo, abbiamo bisogno di avviare il target
chiamato win.exe che entra in un loop infinito appena prima che la finestra venga mostrata sullo 
schermo. Quando l'esempio è avviato, potrai "attaccarti" allo stesso win.exe e modificare il suo codice 
non appena esce dal loop infinito e mostra la propria finestra. 

.386

.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\kernel32.inc
include \masm32\include\comdlg32.inc
include \masm32\include\user32.inc
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\comdlg32.lib



includelib \masm32\lib\user32.lib

.data
AppName db "Win32 Debug Example no.2",0
ClassName db "SimpleWinClass",0
SearchFail db "Cannot find the target process",0
TargetPatched db "Target patched!",0
buffer dw 9090h

.data?
DBEvent DEBUG_EVENT <>
ProcessId dd ?
ThreadId dd ?
align dword
context CONTEXT <>

.code
start:
invoke FindWindow, addr ClassName, NULL
.if eax!=NULL
    invoke GetWindowThreadProcessId, eax, addr ProcessId
    mov ThreadId, eax
    invoke DebugActiveProcess, ProcessId
    .while TRUE
       invoke WaitForDebugEvent, addr DBEvent, INFINITE
       .break .if DBEvent.dwDebugEventCode==EXIT_PROCESS_DEBUG_EVENT
       .if DBEvent.dwDebugEventCode==CREATE_PROCESS_DEBUG_EVENT
          mov context.ContextFlags, CONTEXT_CONTROL
          invoke GetThreadContext,DBEvent.u.CreateProcessInfo.hThread, addr context           
          invoke WriteProcessMemory, DBEvent.u.CreateProcessInfo.hProcess, context.regEip
,addr buffer, 2, NULL
          invoke MessageBox, 0, addr TargetPatched, addr AppName, MB_OK+MB_ICONINFORMATION
       .elseif DBEvent.dwDebugEventCode==EXCEPTION_DEBUG_EVENT
          .if DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_BREAKPOINT
             invoke ContinueDebugEvent, DBEvent.dwProcessId,DBEvent.dwThreadId, DBG_CONTINUE
             .continue
          .endif
       .endif
       invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, 
DBG_EXCEPTION_NOT_HANDLED
   .endw
.else
    invoke MessageBox, 0, addr SearchFail, addr AppName,MB_OK+MB_ICONERROR .endif
invoke ExitProcess, 0
end start

;---------------------------------------------------------------------------------------------
------------
; Parte del codice di win.asm, il nostro processo da debuggare. Si tratta
; infatti della finestra dell'esempio del tutorial 2 nella quale è inserito
; un loop infinito prima che entri nel loop del messaggio.
;---------------------------------------------------------------------------------------------
------------

......
mov wc.hIconSm,eax
invoke LoadCursor,NULL,IDC_ARROW
mov wc.hCursor,eax
invoke RegisterClassEx, addr wc
INVOKE CreateWindowEx,NULL,ADDR ClassName,ADDR AppName,\ WS_OVERLAPPEDWINDOW,CW_USEDEFAULT,\ 
CW_USEDEFAULT,CW_USEDEFAULT,CW_USEDEFAULT,NULL,NULL,\ hInst,NULL
mov hwnd,eax
jmp $ <---- Qui il nostro loop infinito. Si compone di EB FE
invoke ShowWindow, hwnd,SW_SHOWNORMAL
invoke UpdateWindow, hwnd
.while TRUE
   invoke GetMessage, ADDR msg,NULL,0,0
   .break .if (!eax)
   invoke TranslateMessage, ADDR msg
   invoke DispatchMessage, ADDR msg
.endw
mov eax,msg.wParam
ret
WinMain endp



Analisi
invoke FindWindow, addr ClassName, NULL

Il nostro programma deve attaccarsi al debuggee con DebugActiveProcesss che richiede appunto il suo
ID. Possiamo ottenere l'ID del processo chiamando GetWindowsThreadProcessId che a sua volta 
come parametro ha bisogno dell'handle della finestra. Quindi per primo dobbiamo ottenere l'handle della 
finestra. Con FindWindow, possiamo specificare il nome della classe della finestra di cui abbiamo 
bisogno. Essa ritorna l'handle della finestra creata dalla classe della finestra stessa. Se essa ritorna NULL,
non è presentenessuna finestra della classe. 

.if eax!=NULL
    invoke GetWindowThreadProcessId, eax, addr ProcessId
    mov ThreadId, eax
    invoke DebugActiveProcess, ProcessId

Dopo che abbiamo ottenuto l'ID del processo, possiamo chiamare DebugActiveProcess. Poi entriamo 
nel loop di debug in attesa di eventi dello stesso. 

.if DBEvent.dwDebugEventCode==CREATE_PROCESS_DEBUG_EVENT
          mov context.ContextFlags, CONTEXT_CONTROL
          invoke GetThreadContext,DBEvent.u.CreateProcessInfo.hThread, addr context

Quando otteniamo CREATE_PROCESS_DEBUG_INFO, significa che il debuggee è sospeso, quindi 
pronto per poter effettuare un'intervento "chirurgico" sul processo. In questo esempio, sovrascriveremo 
l'istruzione del loop infinito del debuggee (0EBh 0FEh) con i NOP ( 90h 90h ). Prima abbiamo ottenuto 
l'indirizzo dell'istruzione. Poichè il debuggee è già nel loop, possiamo "attaccare" il nostro programma a 
lui, eip punterà già all'istruzione. Tutto quello che c'è da fare è ottenere il valore di EIP. Usiamo 
GetThreadContext per il raggiungimento di tale obiettivo. Abbiamo impostato il membro ContextFlags 
con CONTEXT_CONTROL in modo tale da dire a GetThreadContext che vogliamo riempire i membri 
"control" del registro della struttura CONTEXT. 

invoke WriteProcessMemory, DBEvent.u.CreateProcessInfo.hProcess, context.regEip ,addr buffer, 
2, NULL

Adesso che abbiamo ottenuto il valore di EIP, possiamo chiamare WriteProcessMemory per 
sovrascrivere l'istruzione " jmp $ " con i NOP, quindi effettivamente aiuta il debuggee ad uscire dal loop 
infinito. Dopo che è stato visualizzato il messaggio per l'utente, chiamiamo ContinueDebugEvent per 
ripristinare il debuggee. Poichè l'istruzione " jmp $ " è sovrascritta dai NOP, il debuggee sarà in grado di 
continuare con la visualizzazione della finestra e entrare nel loop del messaggio. La dimostrazione è che 
vedremo la finestra sullo schermo. 

L'altro esempio usa un'approcio differente per interrompere il loop infinito del debuggee. 

.......

.......

.if DBEvent.dwDebugEventCode==CREATE_PROCESS_DEBUG_EVENT
   mov context.ContextFlags, CONTEXT_CONTROL
   invoke GetThreadContext,DBEvent.u.CreateProcessInfo.hThread, addr context
   add context.regEip,2
   invoke SetThreadContext,DBEvent.u.CreateProcessInfo.hThread, addr context
   invoke MessageBox, 0, addr LoopSkipped, addr AppName, MB_OK+MB_ICONINFORMATION
.......
.......

Viene chiamato ancora GetThreadContext per ottenere il valore corrente dell'EIP ma invece di 
sovrascrivere l'istruzione " jmp $ "incrementiano il valore di regEip a 2 per poter "saltare" l'istruzione. Il 
risultato è che quando il debuggee riacquista il controllo, riprende l'esecuzione dall'istruzione successiva a
quella relativa al " jmp $ ". 



Adesso puoi vedere la potenza di Get/SetThreadContext. Puoi anche modificare i registri delle altre 
immagini come pure i loro valori e di coseguenza si rifletteranno al debuggee. Nel processo da debuggare
è possibile inserire anche istruzioni int 3 per mettere breakpoint. 

In questo tutorial, continuiamo l'esplorazione delle API Win 32 di Debug. In particolare, impareremo 
come "tracciare" il debuggee. 

Tools
Esempio 

Essay
Se avete usato un debugger, dovreste avere familiarità con il tracing [ NdT ovvero l'operazione di 
"tracciamento" del debuggee ]. Quando si "traccia" un programma, il programma si blocca dopo 
l'esecuzione di ogni istruzione, dando la possibilità di esaminare i valori dei/della registri/memoria. Il 
tracing è chiamato: Single-Stepping. La modalità Single-Step [ NdT tradotto letteralmente significa 
"singolo passo", ovvero quello che effettua il processo di debug con il debugge ] è fornita dalla CPU 
stessa. L'ottavo bit del flag register è chiamato trap flag. Se questo flag ( bit ) è stato impostato, la CPU 
viene eseguita in modalità single-step. La CPU genererà un eccezione di debug dopo ogni istruzione. 
Dopo che l'eccezione di debug è stata generata, il trap flag viene cancellato automaticamente. Nel 
debuggee noi possiamo fare anche un single-step usando le API Win32 di Debug. I passi sono i seguenti: 

1. Chiamare GetThreadContext, specificando CONTEXT_CONTROL in ContextFlag, per ottenere 
il valore del flag register; 

2. Impostare il bit trap nel membro regFlag della struttura CONTEXT; 
3. Chiamare SetThreadContext; 
4. Come al solito attendere gli eventi di debug. Il debuggee sarà eseguito nella modalità single-step.

Dopo che ogni istruzione viene eseguita, avremo un EXCEPTION_DEBUG_EVENT con il valore
EXCEPTION_SINGLE_STEP in u.Exception.pExceptionRecord.ExceptionCode; 

5. Se c'è bisogno di tracciare la prossima istruzione, bisogna impostare di nuovo il bit trap. 

Esempio
.386
.model flat,stdcall
option casemap:none
include \masm32\include\windows.inc
include \masm32\include\kernel32.inc
include \masm32\include\comdlg32.inc
include \masm32\include\user32.inc
includelib \masm32\lib\kernel32.lib
includelib \masm32\lib\comdlg32.lib
includelib \masm32\lib\user32.lib

.data
AppName db "Win32 Debug Example no.4",0
ofn OPENFILENAME <>
FilterString db "Executable Files",0,"*.exe",0
             db "All Files",0,"*.*",0,0
ExitProc db "The debuggee exits",0Dh,0Ah
         db "Total Instructions executed : %lu",0
TotalInstruction dd 0

.data?
buffer db 512 dup(?)
startinfo STARTUPINFO <>
pi PROCESS_INFORMATION <>
DBEvent DEBUG_EVENT <>

http://quequero.org/uicwiki/images/Tut30.zip


align dword
context CONTEXT <>

.code
start:
mov ofn.lStructSize,SIZEOF ofn
mov ofn.lpstrFilter, OFFSET FilterString
mov ofn.lpstrFile, OFFSET buffer
mov ofn.nMaxFile,512
mov ofn.Flags, OFN_FILEMUSTEXIST or OFN_PATHMUSTEXIST or OFN_LONGNAMES or OFN_EXPLORER or 
OFN_HIDEREADONLY
invoke GetOpenFileName, ADDR ofn
.if eax==TRUE
    invoke GetStartupInfo,addr startinfo
    invoke CreateProcess, addr buffer, NULL, NULL, NULL, FALSE, DEBUG_PROCESS+ 
DEBUG_ONLY_THIS_PROCESS, NULL, NULL, addr startinfo, addr pi
    .while TRUE
       invoke WaitForDebugEvent, addr DBEvent, INFINITE
       .if DBEvent.dwDebugEventCode==EXIT_PROCESS_DEBUG_EVENT
          invoke wsprintf, addr buffer, addr ExitProc, TotalInstruction
          invoke MessageBox, 0, addr buffer, addr AppName, MB_OK+MB_ICONINFORMATION
          .break
       .elseif DBEvent.dwDebugEventCode==EXCEPTION_DEBUG_EVENT           .if 
DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_BREAKPOINT
             mov context.ContextFlags, CONTEXT_CONTROL
             invoke GetThreadContext, pi.hThread, addr context
             or context.regFlag,100h
             invoke SetThreadContext,pi.hThread, addr context
             invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, DBG_CONTINUE
             .continue
          .elseif DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_SINGLE_STEP
             inc TotalInstruction
             invoke GetThreadContext,pi.hThread,addr context or context.regFlag,100h
             invoke SetThreadContext,pi.hThread, addr context
             invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId,DBG_CONTINUE
             .continue
          .endif
       .endif
       invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, 
DBG_EXCEPTION_NOT_HANDLED
    .endw
.endif
invoke CloseHandle,pi.hProcess
invoke CloseHandle,pi.hThread
invoke ExitProcess, 0
end start

Analisi
Il programma mostra la dialog box per aprire il file. Quando l'utente sceglie un file eseguibile, 
quest'ultimo viene eseguito in modalità single-step, contando il numero di istruzioni eseguite fino a 
quando il programma non esce. 

.elseif DBEvent.dwDebugEventCode==EXCEPTION_DEBUG_EVENT .if 
DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_BREAKPOINT

Cogliamo questa occasione per impostare il debuggee in modalità single-step. Ricorda che Windows invia 
un EXCEPTION_BREAKPOINT non appena che la prima istruzione del debuggee venga eseguita. 

mov context.ContextFlags, CONTEXT_CONTROL
invoke GetThreadContext, pi.hThread, addr context

Chiamiamo GetThreadContext per riempire la struttura CONTEXT con i valori contenuti nei registri del 
debuggee. Più precisamente, abbiamo bisogno del valore corrente nel flag register. 

or context.regFlag,100h



Settiamo il bit trap ( 8° bit ) nell'image del registro flag. 

invoke SetThreadContext,pi.hThread, addr context
invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId, DBG_CONTINUE
.continue

Poi chiamiamo SetThreadContext per sovrascrivere il valore nella struttura CONTEXT con un nuovo 
valore o più, e chiamiamo ContinueDebugEvent con il flag DBG_CONTINUE per riprendere il 
debuggee. 

.elseif DBEvent.u.Exception.pExceptionRecord.ExceptionCode==EXCEPTION_SINGLE_STEP
  inc TotalInstruction

Quando viene eseguita un istruzione nel debugge, riceviamo un EXCEPTION_DEBUG_EVENT. 
Dobbiamo così esaminare il valore di u.Exception.pExceptionRecord.ExceptionCode. Se il valore è 
EXCEPTION_SINGLE_STEP, allora viene generato questo evento di debug a causa della modalità 
single-step. In questo caso, noi siamo in grado si incrementare la variabile TotalInstruction di uno 
perchè sappiamo per certo che è stata eseguita un istruzione nel debuggee. 

invoke GetThreadContext,pi.hThread,addr context or context.regFlag,100h
invoke SetThreadContext,pi.hThread, addr context
invoke ContinueDebugEvent, DBEvent.dwProcessId, DBEvent.dwThreadId,DBG_CONTINUE
.continue

Poichè il trap flag viene cancellato dopo che viene generata un'eccezione di debug, dobbiamo impostare 
di nuovo il trap flag se vogliamo continuare in modalità single-step. 

Attenzione: Non usate l'esempio in questo tutorial con programmi grandi: il tracing è LENTO. Potrebbe 
essere necessario attendere una decina di minuti prima di poter chiudere il debuggee. 

